
Changelog for v1.6
    

See the upgrade guide to upgrade from Phoenix 1.5.x.
Phoenix v1.6 requires Elixir v1.9+.
1.6.9 (2022-05-16)
Bug Fixes
	[phx.gen.release] Fix generated .dockerignore comment

1.6.8 (2022-05-06)
Bug Fixes
	[phx.gen.release] Fix Ecto check failing to find ecto in certain cases

1.6.7 (2022-04-14)
Enhancements
	[Endpoint] Add Endpoint init telemetry event
	[Endpoint] Prioritize user :http configuration for ranch  to fix inet_backend failing to be respected
	[Logger] Support log_module in router metadata
	[phx.gen.release] Don't handle assets in Docker when directory doesn't exist
	[phx.gen.release] Skip generating migration files when ecto_sql is not installed

JavaScript Client Enhancements
	Switch to .mjs files for ESM for better compatibility across build tools

JavaScript Client Bug Fixes
	Fix LongPoll callbacks in JS client causing errors on connection close

1.6.6 (2022-01-04)
Bug Fixes
	[Endpoint] Fix check_origin: :conn failing to match scheme

1.6.5 (2021-12-16)
Enhancements
	[Endpoint] Support check_origin: :conn to enforce origin on the connection's host, port, and scheme

Bug Fixes
	Fix LiveView upload testing errors caused by Phoenix.ChannelTest

1.6.4 (2021-12-08)
Bug Fixes
	Fix incorrect phx.gen.release output

1.6.3 (2021-12-07)
Enhancements
	Add new phx.gen.release task for release and docker based deployments
	Add fullsweep_after option to the websocket transport
	Add :force_watchers option to Phoenix.Endpoint for running watchers even when web server is not started

Bug Fixes
	Fix Endpoint log: false failing to disable logging

JavaScript Client Bug Fixes
	Do not attempt to reconnect automatically if client gracefully closes connection

1.6.2 (2021-10-08)
Bug Fixes
	[phx.new] Fix external flag to esbuild using incorrect syntax

1.6.1 (2021-10-08)
Enhancements
	[phx.new] Add external flag to esbuild for fonts and image path loading
	[phx.gen.auth] No longer set argon2 as the default hash algorithm for phx.gen.auth in favor of bcrypt for performance reasons on smaller hardware

Bug Fixes
	Fix race conditions logging debug duplicate channel joins when no duplicate existed

JavaScript Client Bug Fixes
	Export commonjs modules for backwards compatibility

1.6.0 (2021-09-24) 🚀
Enhancements
	[ConnTest] Add path_params/2 for retrieving router path parameters out of dynamically returned URLs.

JavaScript Client Bug Fixes
	Fix LongPoll transport undefined readyState check

1.6.0-rc.1 (2021-09-22)
Enhancements
	[mix phx.gen.auth] Validate bcrypt passwords are no longer than 72 bytes
	re-enable phx.routes task to support back to back invocations, such as for aliased mix route tasks
	[mix phx.gen.html] Remove comma after for={@changeset} on form.html.heex

JavaScript Client Bug Fixes
	Fix messages for duplicate topic being dispatched to old channels

1.6.0-rc.0 (2021-08-26)
Enhancements
	[CodeReloader] Code reloading can now pick up changes to .beam files if they were compiled in a separate OS process than the Phoenix server
	[Controller] Do not create compile-time dependency for action_fallback
	[Endpoint] Allow custom error response from socket handler
	[Endpoint] Do not require a pubsub server in the socket (only inside channels)
	[mix phx.digest.clean] Add --all flag to mix phx.digest.clean
	[mix phx.gen.auth] Add mix phx.gen.auth generator
	[mix phx.gen.context] Support enum types and the redact option when declaring fields
	[mix phx.gen.notifier] A new generator to build notifiers that by default deliver emails
	[mix phx.new] Update mix phx.new to require Elixir v1.12 and use the new config/runtime.exs
	[mix phx.new] Set plug_init_mode: :runtime in generated config/test.exs
	[mix phx.new] Add description to Ecto telemetry metrics
	[mix phx.new] Use Ecto.Adapters.SQL.Sandbox.start_owner!/2 in generators - this approach provides proper shutdown semantics for apps using LiveView and Presence
	[mix phx.new] Add --install and --no-install options to phx.new
	[mix phx.new] Add --database sqlite3 option to phx.new
	[mix phx.new] Remove usage of Sass
	[mix phx.new] New applications now depend on Swoosh to deliver emails
	[mix phx.new] No longer generate a socket file by default, instead one can run mix phx.gen.socket
	[mix phx.new] No longer generates a home page using LiveView, instead one can run mix phx.gen.live
	[mix phx.new] LiveView is now included by default. Passing --no-live will comment out lines in app.js and Endpoint
	[mix phx.server] Add --open flag
	[Router] Do not add compile time deps in pipe_through
	[View] Extracted Phoenix.View into its own project to facilitate reuse

JavaScript Client Enhancements
	Add new replaceTransport function to socket with extended onError API to allow simplified LongPoll fallback
	Fire each event in a separate task for the LongPoll transport to fix ordering
	Optimize presence syncing

Bug fixes
	[Controller] Return normalized paths in current_path/1 and current_path/2
	[mix phx.gen.live] Fix a bug where tests with utc_datetime and boolean fields did not pass out of the box

JavaScript Client Bug fixes
	Bind to beforeunload instead of unload to solve Firefox connection issues
	Fix presence onJoin including current metadata in new presence

Deprecations
	[mix compile.phoenix] Adding the :phoenix compiler to your mix.exs (compilers: [:phoenix] ++ Mix.compilers()) is no longer required from Phoenix v1.6 forward if you are running on Elixir v1.11. Remove it from your mix.exs and you should gain faster compilation times too
	[Endpoint] Phoenix now requires Cowboy v2.7+

Breaking changes
	[View] @view_module and @view_template are no longer set. Use Phoenix.Controller.view_module/1 and Phoenix.Controller.view_template/1 respectively, or pass explicit assigns from Phoenix.View.render.

v1.5
The CHANGELOG for v1.5 releases can be found in the v1.5 branch.



  

    
Overview
    

Phoenix is a web development framework written in Elixir which implements the server-side Model View Controller (MVC) pattern. Many of its components and concepts will seem familiar to those of us with experience in other web frameworks like Ruby on Rails or Python's Django.
Phoenix provides the best of both worlds - high developer productivity and high application performance. It also has some interesting new twists like channels for implementing realtime features and pre-compiled templates for blazing speed.
If you are already familiar with Elixir, great! If not, there are a number of places to learn. The Elixir guides and the Elixir learning resources page are two great places to start.
The guides that you are currently looking at provide an overview of all parts that make Phoenix. Here is a rundown of what they provide:
	Introduction - the guides you are currently reading. They will cover how to get your first application up and running

	Guides - in-depth guides covering the main components in Phoenix and Phoenix applications

	Authentication - in-depth guide covering how to use mix phx.gen.auth

	Real-time components - in-depth guides covering Phoenix's built-in real-time components

	Testing - in-depth guides about testing

	Deployment - in-depth guides about deployment

	How-to's - a collection of articles on how to achieve certain things with Phoenix


If you would prefer to read these guides as an EPUB, click here!
Note, these guides are not a step-by-step introduction to Phoenix. If you want a more structured approach to learning the framework, we have a large community and many books, courses, and screencasts available. See our community page for a complete list.
Let's get Phoenix installed.



  

    
Installation
    

In order to build a Phoenix application, we will need a few dependencies installed in our Operating System:
	the Erlang VM and the Elixir programming language
	a database - Phoenix recommends PostgreSQL, but you can pick others or not use a database at all
	and other optional packages.

Please take a look at this list and make sure to install anything necessary for your system. Having dependencies installed in advance can prevent frustrating problems later on.
Elixir 1.12 or later
Phoenix is written in Elixir, and our application code will also be written in Elixir. We won't get far in a Phoenix app without it! The Elixir site maintains a great Installation Page to help.
If we have just installed Elixir for the first time, we will need to install the Hex package manager as well. Hex is necessary to get a Phoenix app running (by installing dependencies) and to install any extra dependencies we might need along the way.
Here's the command to install Hex (If you have Hex already installed, it will upgrade Hex to the latest version):
$ mix local.hex

Erlang 22 or later
Elixir code compiles to Erlang byte code to run on the Erlang virtual machine. Without Erlang, Elixir code has no virtual machine to run on, so we need to install Erlang as well.
When we install Elixir using instructions from the Elixir Installation Page,  we will usually get Erlang too. If Erlang was not installed along with Elixir, please see the Erlang Instructions section of the Elixir Installation Page for instructions.
Phoenix
To check that we are on Elixir 1.12 and Erlang 22 or later, run:
elixir -v
Erlang/OTP 22 [erts-10.7] [source] [64-bit] [smp:8:8] [async-threads:10] [hipe] [kernel-poll:false] [dtrace]

Elixir 1.12.1

Once we have Elixir and Erlang, we are ready to install the Phoenix application generator:
$ mix archive.install hex phx_new

The phx.new generator is now available to generate new applications in the next guide, called Up and Running. The flags mentioned below are command line options to the generator; see all available options by calling mix help phx.new.
PostgreSQL
PostgreSQL is a relational database server. Phoenix configures applications to use it by default, but we can switch to MySQL, MSSQL, or SQLite3 by passing the --database flag when creating a new application.
In order to talk to databases, Phoenix applications use another Elixir package, called Ecto. If you don't plan to use databases in your application, you can pass the --no-ecto flag.
However, if you are just getting started with Phoenix, we recommend you to install PostgreSQL and make sure it is running. The PostgreSQL wiki has installation guides for a number of different systems.
inotify-tools (for Linux users)
Phoenix provides a very handy feature called Live Reloading. As you change your views or your assets, it automatically reloads the page in the browser. In order for this functionality to work, you need a filesystem watcher.
macOS and Windows users already have a filesystem watcher, but Linux users must install inotify-tools. Please consult the inotify-tools wiki for distribution-specific installation instructions.
Summary
At the end of this section, you must have installed Elixir, Hex, Phoenix, and PostgreSQL. Now that we have everything installed, let's create our first Phoenix application and get up and running.



  

    
Up and Running
    

Let's get a Phoenix application up and running as quickly as possible.
Before we begin, please take a minute to read the Installation Guide. By installing any necessary dependencies beforehand, we'll be able to get our application up and running smoothly.
We can run mix phx.new from any directory in order to bootstrap our Phoenix application. Phoenix will accept either an absolute or relative path for the directory of our new project. Assuming that the name of our application is hello, let's run the following command:
$ mix phx.new hello

A note about Ecto: Ecto allows our Phoenix application to communicate with a data store, such as PostgreSQL, MySQL, and others. If our application will not require this component we can skip this dependency by passing the --no-ecto flag to mix phx.new.

To learn more about mix phx.new you can read the Mix Tasks Guide.

mix phx.new hello
* creating hello/config/config.exs
* creating hello/config/dev.exs
* creating hello/config/prod.exs
...

Fetch and install dependencies? [Yn]

Phoenix generates the directory structure and all the files we will need for our application.
Phoenix promotes the usage of git as version control software: among the generated files we find a .gitignore. We can git init our repository, and immediately add and commit all that hasn't been marked ignored.

When it's done, it will ask us if we want it to install our dependencies for us. Let's say yes to that.
Fetch and install dependencies? [Yn] Y
* running mix deps.get
* running mix deps.compile

We are almost there! The following steps are missing:

    $ cd hello

Then configure your database in config/dev.exs and run:

    $ mix ecto.create

Start your Phoenix app with:

    $ mix phx.server

You can also run your app inside IEx (Interactive Elixir) as:

    $ iex -S mix phx.server

Once our dependencies are installed, the task will prompt us to change into our project directory and start our application.
Phoenix assumes that our PostgreSQL database will have a postgres user account with the correct permissions and a password of "postgres". If that isn't the case, please see the Mix Tasks Guide to learn more about the mix ecto.create task.
Ok, let's give it a try. First, we'll cd into the hello/ directory we've just created:
$ cd hello

Now we'll create our database:
$ mix ecto.create
Compiling 13 files (.ex)
Generated hello app
The database for Hello.Repo has been created

In case the database could not be created, see the guides for the mix ecto.create for general troubleshooting.
Note: if this is the first time you are running this command, Phoenix may also ask to install Rebar. Go ahead with the installation as Rebar is used to build Erlang packages.

And finally, we'll start the Phoenix server:
$ mix phx.server
[info] Running HelloWeb.Endpoint with cowboy 2.9.0 at 127.0.0.1:4000 (http)
[info] Access HelloWeb.Endpoint at http://localhost:4000
[watch] build finished, watching for changes...
...

If we choose not to have Phoenix install our dependencies when we generate a new application, the mix phx.new task will prompt us to take the necessary steps when we do want to install them.
Fetch and install dependencies? [Yn] n

We are almost there! The following steps are missing:

    $ cd hello
    $ mix deps.get

Then configure your database in config/dev.exs and run:

    $ mix ecto.create

Start your Phoenix app with:

    $ mix phx.server

You can also run your app inside IEx (Interactive Elixir) as:

    $ iex -S mix phx.server

By default, Phoenix accepts requests on port 4000. If we point our favorite web browser at http://localhost:4000, we should see the Phoenix Framework welcome page.
[image: Phoenix Welcome Page]
If your screen looks like the image above, congratulations! You now have a working Phoenix application. In case you can't see the page above, try accessing it via http://127.0.0.1:4000 and later make sure your OS has defined "localhost" as "127.0.0.1".
To stop it, we hit ctrl-c twice.
Now you are ready to explore the world provided by Phoenix! See our community page for books, screencasts, courses, and more.
Alternatively, you can continue reading these guides to have a quick introduction into all the parts that make your Phoenix application. If that's the case, you can read the guides in any order or start with our guide that explains the Phoenix directory structure.



  

    
Community
    

The Elixir and Phoenix communities are friendly and welcoming. All questions and comments are valuable, so please come join the discussion!
There are a number of places to connect with community members at all experience levels.
	We're on Libera IRC in the #elixir channel.

	Request an invitation and join the #phoenix channel on Slack.

	Feel free to join and check out the #phoenix channel on Discord.

	Read about bug reports or open an issue in the Phoenix issue tracker.

	Ask or answer questions about Phoenix on Elixir Forum or Stack Overflow.

	To discuss new features in the framework, email the phoenix-core mailing list.

	Follow the Phoenix Framework on Twitter.

	The source for these guides is on GitHub. To help improve the guides, please report an issue or open a pull request.


Books
	Programming Phoenix LiveView - Interactive Elixir Web Programming Without Writing Any JavaScript - 2021 (by Bruce Tate and Sophie DeBenedetto)

	Real-Time Phoenix - Build Highly Scalable Systems with Channels (by Stephen Bussey - 2020)

	Programming Phoenix 1.4 (by Bruce Tate and Phoenix core team members Chris McCord and José Valim - 2019)

	Phoenix in Action (by Geoffrey Lessel - 2019)

	Phoenix Inside Out - Book Series (by Shankar Dhanasekaran - 2017)

	Functional Web Development with Elixir, OTP, and Phoenix Rethink the Modern Web App (by Lance Halvorsen - 2017)


Screencasts/Courses
	Phoenix LiveView Free Course (by The Pragmatic Studio - 2020)

	Groxio LiveView: Self Study Program (by Bruce Tate - 2020)

	Alchemist Camp: Learn Elixir and Phoenix by building (2018-2022)

	The Complete Elixir and Phoenix Bootcamp Master Functional Programming Techniques with Elixir and Phoenix while Learning to Build Compelling Web Applications (by Stephen Grider - 2017)

	Discover Elixir & Phoenix (by Tristan Edwards - 2017)

Directory structure
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

When we use mix phx.new to generate a new Phoenix application, it builds a top-level directory structure like this:
├── _build
├── assets
├── config
├── deps
├── lib
│   ├── hello
│   ├── hello.ex
│   ├── hello_web
│   └── hello_web.ex
├── priv
└── test

We will go over those directories one by one:
	_build - a directory created by the mix command line tool that ships as part of Elixir that holds all compilation artifacts. As we have seen in "Up and Running", mix is the main interface to your application. We use Mix to compile our code, create databases, run our server, and more. This directory must not be checked into version control and it can be removed at any time. Removing it will force Mix to rebuild your application from scratch.

	assets - a directory that keeps everything related to source front-end assets, such as JavaScript and CSS, and automatically managed by the esbuild tool.

	config - a directory that holds your project configuration. The config/config.exs file is the entry point for your configuration. At the end of the config/config.exs, it imports environment specific configuration, which can be found in config/dev.exs, config/test.exs, and config/prod.exs. Finally, config/runtime.exs is executed and it is the best place to read secrets and other dynamic configuration.

	deps - a directory with all of our Mix dependencies. You can find all dependencies listed in the mix.exs file, inside the defp deps do function definition. This directory must not be checked into version control and it can be removed at any time. Removing it will force Mix to download all deps from scratch.

	lib - a directory that holds your application source code. This directory is broken into two subdirectories, lib/hello and lib/hello_web. The lib/hello directory will be responsible to host all of your business logic and business domain. It typically interacts directly with the database - it is the "Model" in Model-View-Controller (MVC) architecture. lib/hello_web is responsible for exposing your business domain to the world, in this case, through a web application. It holds both the View and Controller from MVC. We will discuss the contents of these directories with more detail in the next sections.

	priv - a directory that keeps all resources that are necessary in production but are not directly part of your source code. You typically keep database scripts, translation files, and more in here. Static and generated assets, sourced from the assets directory, are also served from here by default.

	test - a directory with all of our application tests. It often mirrors the same structure found in lib.


The lib/hello directory
The lib/hello directory hosts all of your business domain. Since our project does not have any business logic yet, the directory is mostly empty. You will only find three files:
lib/hello
├── application.ex
├── mailer.ex
└── repo.ex

The lib/hello/application.ex file defines an Elixir application named Hello.Application. That's because at the end of the day Phoenix applications are simply Elixir applications. The Hello.Application module defines which services are part of our application:
children = [
  # Start the Ecto repository
  Hello.Repo,
  # Start the Telemetry supervisor
  HelloWeb.Telemetry,
  # Start the PubSub system
  {Phoenix.PubSub, name: Hello.PubSub},
  # Start the Endpoint (http/https)
  HelloWeb.Endpoint
  # Start a worker by calling: Hello.Worker.start_link(arg)
  # {Hello.Worker, arg}
]
If it is your first time with Phoenix, you don't need to worry about the details right now. For now, suffice it to say our application starts a database repository, a PubSub system for sharing messages across processes and nodes, and the application endpoint, which effectively serves HTTP requests. These services are started in the order they are defined and, whenever shutting down your application, they are stopped in the reverse order.
You can learn more about applications in Elixir's official docs for Application.
The lib/hello/mailer.ex file holds the Hello.Mailer module, which defines the main interface to deliver e-mails:
defmodule Hello.Mailer do
  use Swoosh.Mailer, otp_app: :hello
end
In the same lib/hello directory, we will find a lib/hello/repo.ex. It defines a Hello.Repo module which is our main interface to the database. If you are using Postgres (the default database), you will see something like this:
defmodule Hello.Repo do
  use Ecto.Repo,
    otp_app: :hello,
    adapter: Ecto.Adapters.Postgres
end
And that's it for now. As you work on your project, we will add files and modules to this directory.
The lib/hello_web directory
The lib/hello_web directory holds the web-related parts of our application. It looks like this when expanded:
lib/hello_web
├── controllers
│   └── page_controller.ex
├── templates
│   ├── layout
│   │   ├── app.html.heex
│   │   ├── live.html.heex
│   │   └── root.html.heex
│   └── page
│       └── index.html.heex
├── views
│   ├── error_helpers.ex
│   ├── error_view.ex
│   ├── layout_view.ex
│   └── page_view.ex
├── endpoint.ex
├── gettext.ex
├── router.ex
└── telemetry.ex

All of the files which are currently in the controllers, templates, and views directories are there to create the "Welcome to Phoenix!" page we saw in the "Up and running" guide.
By looking at templates and views directories, we can see Phoenix provides features for handling layouts and error pages out of the box.
Besides the directories mentioned, lib/hello_web has four files at its root. lib/hello_web/endpoint.ex is the entry-point for HTTP requests. Once the browser accesses http://localhost:4000, the endpoint starts processing the data, eventually leading to the router, which is defined in lib/hello_web/router.ex. The router defines the rules to dispatch requests to "controllers", which then uses "views" and "templates" to render HTML pages back to clients. We explore these layers in length in other guides, starting with the "Request life-cycle" guide coming next.
Through Telemetry, Phoenix is able to collect metrics and send monitoring events of your application. The lib/hello_web/telemetry.ex file defines the supervisor responsible for managing the telemetry processes. You can find more information on this topic in the Telemetry guide.
Finally, there is a lib/hello_web/gettext.ex file which provides internationalization through Gettext. If you are not worried about internationalization, you can safely skip this file and its contents.
The assets directory
The assets directory contains source files related to front-end assets, such as JavaScript and CSS. Since Phoenix v1.6, we use esbuild to compile assets, which is managed by the esbuild Elixir package. The integration with esbuild is baked into your app. The relevant config can be found in your config/config.exs file.
Your other static assets are placed in the priv/static folder, where priv/static/assets is kept for generated assets. Everything in priv/static is served by the Plug.Static plug configured in lib/hello_web/endpoint.ex.  When running in dev mode (MIX_ENV=dev), Phoenix watches for any changes you make in the assets directory, and then takes care of updating your front end application in your browser as you work.
Note that when you first create your Phoenix app using mix phx.new it is possible to specify options that will affect the presence and layout of the assets directory.  In fact, Phoenix apps can bring their own front end tools or not have a front-end at all (handy if you're writing an API for example).  For more information you can run mix help phx.new or see the documentation in Mix tasks.
If the default esbuild integration does not cover your needs, for example because you want to use another build tool, you can switch to a custom assets build.
As for CSS, Phoenix ships with a handful of custom styles as well as the Milligram CSS Framework, providing a minimal setup for projects. You may move to any CSS framework of your choice. Additional references can be found in the asset management guide.

Request life-cycle
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

The goal of this guide is to talk about Phoenix's request life-cycle. This guide will take a practical approach where we will learn by doing: we will add two new pages to our Phoenix project and comment on how the pieces fit together along the way.
Let's get on with our first new Phoenix page!
Adding a new page
When your browser accesses http://localhost:4000/, it sends a HTTP request to whatever service is running on that address, in this case our Phoenix application. The HTTP request is made of a verb and a path. For example, the following browser requests translate into:
	Browser address bar	Verb	Path
	http://localhost:4000/	GET	/
	http://localhost:4000/hello	GET	/hello
	http://localhost:4000/hello/world	GET	/hello/world

There are other HTTP verbs. For example, submitting a form typically uses the POST verb.
Web applications typically handle requests by mapping each verb/path pair into a specific part of your application. This matching in Phoenix is done by the router. For example, we may map "/articles" to a portion of our application that shows all articles. Therefore, to add a new page, our first task is to add a new route.
A new route
The router maps unique HTTP verb/path pairs to controller/action pairs which will handle them. Controllers in Phoenix are simply Elixir modules. Actions are functions that are defined within these controllers.
Phoenix generates a router file for us in new applications at lib/hello_web/router.ex. This is where we will be working for this section.
The route for our "Welcome to Phoenix!" page from the previous Up And Running Guide looks like this.
get "/", PageController, :index
Let's digest what this route is telling us. Visiting http://localhost:4000/ issues an HTTP GET request to the root path. All requests like this will be handled by the index/2 function in the HelloWeb.PageController module defined in lib/hello_web/controllers/page_controller.ex.
The page we are going to build will say "Hello World, from Phoenix!" when we point our browser to http://localhost:4000/hello.
The first thing we need to do is to create the page route for a new page. Let's open up lib/hello_web/router.ex in a text editor. For a brand new application, it looks like this:
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end

  pipeline :api do
    plug :accepts, ["json"]
  end

  scope "/", HelloWeb do
    pipe_through :browser

    get "/", PageController, :index
  end

  # Other scopes may use custom stacks.
  # scope "/api", HelloWeb do
  #   pipe_through :api
  # end

  # ...
end
For now, we'll ignore the pipelines and the use of scope here and just focus on adding a route. We will discuss those in the Routing guide.
Let's add a new route to the router that maps a GET request for /hello to the index action of a soon-to-be-created HelloWeb.HelloController inside the scope "/" do block of the router:
scope "/", HelloWeb do
  pipe_through :browser

  get "/", PageController, :index
  get "/hello", HelloController, :index
end
A new controller
Controllers are Elixir modules, and actions are Elixir functions defined in them. The purpose of actions is to gather the data and perform the tasks needed for rendering. Our route specifies that we need a HelloWeb.HelloController module with an index/2 function.
To make the index action happen, let's create a new lib/hello_web/controllers/hello_controller.ex file, and make it look like the following:
defmodule HelloWeb.HelloController do
  use HelloWeb, :controller

  def index(conn, _params) do
    render(conn, "index.html")
  end
end
We'll save a discussion of use HelloWeb, :controller for the Controllers guide. For now, let's focus on the index action.
All controller actions take two arguments. The first is conn, a struct which holds a ton of data about the request. The second is params, which are the request parameters. Here, we are not using params, and we avoid compiler warnings by prefixing it with _.
The core of this action is render(conn, "index.html"). It tells Phoenix to render "index.html". The modules responsible for rendering are called views. By default, Phoenix views are named after the controller, so Phoenix is expecting a HelloWeb.HelloView to exist and handle "index.html" for us.
Note: Using an atom as the template name also works render(conn, :index). In these cases, the template will be chosen based off the Accept headers, e.g. "index.html" or "index.json".

A new view
Phoenix views act as the presentation layer. For example, we expect the output of rendering "index.html" to be a complete HTML page. To make our lives easier, we often use templates for creating those HTML pages.
Let's create a new view. Create lib/hello_web/views/hello_view.ex and make it look like this:
defmodule HelloWeb.HelloView do
  use HelloWeb, :view
end
Now in order to add templates to this view, we need to add files to the lib/hello_web/templates/hello directory. Note the controller name (HelloController), the view name (HelloView), and the template directory (hello) all follow the same naming convention and are named after each other.
A template file has the following structure: NAME.FORMAT.TEMPLATING_LANGUAGE. In our case, we will create an index.html.heex file at lib/hello_web/templates/hello/index.html.heex. ".heex" stands for "HTML+EEx". EEx is a library for embedding Elixir that ships as part of Elixir itself. "HTML+EEx" is a Phoenix extension of EEx that is HTML aware, with support for HTML validation, components, and automatic escaping of values. The latter protects you from security vulnerabilities like Cross-Site-Scripting with no extra work on your part.
Create lib/hello_web/templates/hello/index.html.heex and make it look like this:
<section class="phx-hero">
  <h2>Hello World, from Phoenix!</h2>
</section>
Now that we've got the route, controller, view, and template, we should be able to point our browsers at http://localhost:4000/hello and see our greeting from Phoenix! (In case you stopped the server along the way, the task to restart it is mix phx.server.)
[image: Phoenix Greets Us]
There are a couple of interesting things to notice about what we just did. We didn't need to stop and restart the server while we made these changes. Yes, Phoenix has hot code reloading! Also, even though our index.html.heex file consists of only a single section tag, the page we get is a full HTML document. Our index template is rendered into the application layout: lib/hello_web/templates/layout/app.html.heex. If you open it, you'll see a line that looks like this:
<%= @inner_content %>
Which injects our template into the layout before the HTML is sent off to the browser.
A note on hot code reloading: Some editors with their automatic linters may prevent hot code reloading from working. If it's not working for you, please see the discussion in this issue.

From endpoint to views
As we built our first page, we could start to understand how the request life-cycle is put together. Now let's take a more holistic look at it.
All HTTP requests start in our application endpoint. You can find it as a module named HelloWeb.Endpoint in lib/hello_web/endpoint.ex. Once you open up the endpoint file, you will see that, similar to the router, the endpoint has many calls to plug. Plug is a library and a specification for stitching web applications together. It is an essential part of how Phoenix handles requests and we will discuss it in detail in the Plug guide coming next.
For now, it suffices to say that each plug defines a slice of request processing. In the endpoint you will find a skeleton roughly like this:
defmodule HelloWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :hello

  plug Plug.Static, ...
  plug Plug.RequestId
  plug Plug.Telemetry, ...
  plug Plug.Parsers, ...
  plug Plug.MethodOverride
  plug Plug.Head
  plug Plug.Session, ...
  plug HelloWeb.Router
end
Each of these plugs have a specific responsibility that we will learn later. The last plug is precisely the HelloWeb.Router module. This allows the endpoint to delegate all further request processing to the router. As we now know, its main responsibility is to map verb/path pairs to controllers. The controller then tells a view to render a template.
At this moment, you may be thinking this can be a lot of steps to simply render a page. However, as our application grows in complexity, we will see that each layer serves a distinct purpose:
	endpoint (Phoenix.Endpoint) - the endpoint contains the common and initial path that all requests go through. If you want something to happen on all requests, it goes to the endpoint.

	router (Phoenix.Router) - the router is responsible for dispatching verb/path to controllers. The router also allows us to scope functionality. For example, some pages in your application may require user authentication, others may not.

	controller (Phoenix.Controller) - the job of the controller is to retrieve request information, talk to your business domain, and prepare data for the presentation layer.

	view  (Phoenix.View) - the view handles the structured data from the controller and converts it to a presentation to be shown to users.


Let's do a quick recap and how the last three components work together by adding another page.
Another new page
Let's add just a little complexity to our application. We're going to add a new page that will recognize a piece of the URL, label it as a "messenger" and pass it through the controller into the template so our messenger can say hello.
As we did last time, the first thing we'll do is create a new route.
Another new route
For this exercise, we're going to reuse HelloController created at the previous step and add a new show action. We'll add a line just below our last route, like this:
scope "/", HelloWeb do
  pipe_through :browser

  get "/", PageController, :index
  get "/hello", HelloController, :index
  get "/hello/:messenger", HelloController, :show
end
Notice that we use the :messenger syntax in the path. Phoenix will take whatever value that appears in that position in the URL and convert it into a parameter. For example, if we point the browser at: http://localhost:4000/hello/Frank, the value of "messenger" will be "Frank".
Another new Action
Requests to our new route will be handled by the HelloWeb.HelloController show action. We already have the controller at lib/hello_web/controllers/hello_controller.ex, so all we need to do is edit that controller and add a show action to it. This time, we'll need to extract the messenger from the parameters so that we can pass it (the messenger) to the template. To do that, we add this show function to the controller:
def show(conn, %{"messenger" => messenger}) do
  render(conn, "show.html", messenger: messenger)
end
Within the body of the show action, we also pass a third argument to the render function, a key-value pair where :messenger is the key, and the messenger variable is passed as the value.
If the body of the action needs access to the full map of parameters bound to the params variable, in addition to the bound messenger variable, we could define show/2 like this:
def show(conn, %{"messenger" => messenger} = params) do
  ...
end
It's good to remember that the keys of the params map will always be strings, and that the equals sign does not represent assignment, but is instead a pattern match assertion.
Another new template
For the last piece of this puzzle, we'll need a new template. Since it is for the show action of HelloController, it will go into the lib/hello_web/templates/hello directory and be called show.html.heex. It will look surprisingly like our index.html.heex template, except that we will need to display the name of our messenger.
To do that, we'll use the special EEx tags for executing Elixir expressions: <%= %>. Notice that the initial tag has an equals sign like this: <%= . That means that any Elixir code that goes between those tags will be executed, and the resulting value will replace the tag in the HTML output. If the equals sign were missing, the code would still be executed, but the value would not appear on the page.
And this is what the template should look like:
<section class="phx-hero">
  <h2>Hello World, from <%= @messenger %>!</h2>
</section>
Our messenger appears as @messenger. We call "assigns" the values passed from the controller to views. It is a special bit of metaprogrammed syntax which stands in for assigns.messenger. The result is much nicer on the eyes and much easier to work with in a template.
We're done. If you point your browser to http://localhost:4000/hello/Frank, you should see a page that looks like this:
[image: Frank Greets Us from Phoenix]
Play around a bit. Whatever you put after /hello/ will appear on the page as your messenger.

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Request life-cycle guide.

Plug lives at the heart of Phoenix's HTTP layer, and Phoenix puts Plug front and center. We interact with plugs at every step of the request life-cycle, and the core Phoenix components like endpoints, routers, and controllers are all just plugs internally. Let's jump in and find out just what makes Plug so special.
Plug is a specification for composable modules in between web applications. It is also an abstraction layer for connection adapters of different web servers. The basic idea of Plug is to unify the concept of a "connection" that we operate on. This differs from other HTTP middleware layers such as Rack, where the request and response are separated in the middleware stack.
At the simplest level, the Plug specification comes in two flavors: function plugs and module plugs.
Function plugs
In order to act as a plug, a function needs to:
	accept a connection struct (%Plug.Conn{}) as its first argument, and connection options as its second one;
	return a connection struct.

Any function that meets these two criteria will do. Here's an example.
def introspect(conn, _opts) do
  IO.puts """
  Verb: #{inspect(conn.method)}
  Host: #{inspect(conn.host)}
  Headers: #{inspect(conn.req_headers)}
  """

  conn
end
This function does the following:
	It receives a connection and options (that we do not use)
	It prints some connection information to the terminal
	It returns the connection

Pretty simple, right? Let's see this function in action by adding it to our endpoint in lib/hello_web/endpoint.ex. We can plug it anywhere, so let's do it by inserting plug :introspect right before we delegate the request to the router:
defmodule HelloWeb.Endpoint do
  ...

  plug :introspect
  plug HelloWeb.Router

  def introspect(conn, _opts) do
    IO.puts """
    Verb: #{inspect(conn.method)}
    Host: #{inspect(conn.host)}
    Headers: #{inspect(conn.req_headers)}
    """

    conn
  end
end
Function plugs are plugged by passing the function name as an atom. To try the plug out, go back to your browser and fetch http://localhost:4000. You should see something like this printed in your shell terminal:
Verb: "GET"
Host: "localhost"
Headers: [...]

Our plug simply prints information from the connection. Although our initial plug is very simple, you can do virtually anything you want inside of it. To learn about all fields available in the connection and all of the functionality associated to it, see the documentation for Plug.Conn.
Now let's look at the other plug variant, the module plugs.
Module plugs
Module plugs are another type of plug that let us define a connection transformation in a module. The module only needs to implement two functions:
	init/1 which initializes any arguments or options to be passed to call/2
	call/2 which carries out the connection transformation. call/2 is just a function plug that we saw earlier

To see this in action, let's write a module plug that puts the :locale key and value into the connection assign for downstream use in other plugs, controller actions, and our views. Put the contents below in a file named lib/hello_web/plugs/locale.ex:
defmodule HelloWeb.Plugs.Locale do
  import Plug.Conn

  @locales ["en", "fr", "de"]

  def init(default), do: default

  def call(%Plug.Conn{params: %{"locale" => loc}} = conn, _default) when loc in @locales do
    assign(conn, :locale, loc)
  end

  def call(conn, default) do
    assign(conn, :locale, default)
  end
end
To give it a try, let's add this module plug to our router, by appending plug HelloWeb.Plugs.Locale, "en"  to our :browser pipeline in lib/hello_web/router.ex:
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_flash
    plug :protect_from_forgery
    plug :put_secure_browser_headers
    plug HelloWeb.Plugs.Locale, "en"
  end
  ...
In the init/1 callback, we pass a default locale to use if none is present in the params. We also use pattern matching to define multiple call/2 function heads to validate the locale in the params, and fall back to "en" if there is no match. The assign/3 is a part of the Plug.Conn module and it's how we store values in the conn data structure.
To see the assign in action, go to the layout in lib/hello_web/templates/layout/app.html.heex and add the following close to the main container:
<main class="container">
  <p>Locale: <%= @locale %></p>
Go to http://localhost:4000/ and you should see the locale exhibited. Visit http://localhost:4000/?locale=fr and you should see the assign changed to "fr". Someone can use this information alongside Gettext to provide a fully internationalized web application.
That's all there is to Plug. Phoenix embraces the plug design of composable transformations all the way up and down the stack. Let's see some examples!
Where to plug
The endpoint, router, and controllers in Phoenix accept plugs.
Endpoint plugs
Endpoints organize all the plugs common to every request, and apply them before dispatching into the router with its custom pipelines. We added a plug to the endpoint like this:
defmodule HelloWeb.Endpoint do
  ...

  plug :introspect
  plug HelloWeb.Router
The default endpoint plugs do quite a lot of work. Here they are in order:
	Plug.Static - serves static assets. Since this plug comes before the logger, requests for static assets are not logged.

	Phoenix.LiveDashboard.RequestLogger - sets up the Request Logger for Phoenix LiveDashboard, this will allow you to have the option to either pass a query parameter to stream requests logs or to enable/disable a cookie that streams requests logs from your dashboard.

	Plug.RequestId - generates a unique request ID for each request.

	Plug.Telemetry - adds instrumentation points so Phoenix can log the request path, status code and request time by default.

	Plug.Parsers - parses the request body when a known parser is available. By default, this plug can handle URL-encoded, multipart and JSON content (with Jason). The request body is left untouched if the request content-type cannot be parsed.

	Plug.MethodOverride - converts the request method to PUT, PATCH or DELETE for POST requests with a valid _method parameter.

	Plug.Head - converts HEAD requests to GET requests and strips the response body.

	Plug.Session - a plug that sets up session management. Note that fetch_session/2 must still be explicitly called before using the session, as this plug just sets up how the session is fetched.


In the middle of the endpoint, there is also a conditional block:
  if code_reloading? do
    socket "/phoenix/live_reload/socket", Phoenix.LiveReloader.Socket
    plug Phoenix.LiveReloader
    plug Phoenix.CodeReloader
    plug Phoenix.Ecto.CheckRepoStatus, otp_app: :hello
  end
This block is only executed in development. It enables:
	live reloading - if you change a CSS file, they are updated in-browser without refreshing the page;
	code reloading - so we can see changes to our application without restarting the server;
	check repo status - which makes sure our database is up to date, raising a readable and actionable error otherwise.

Router plugs
In the router, we can declare plugs inside pipelines:
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
    plug HelloWeb.Plugs.Locale, "en"
  end

  scope "/", HelloWeb do
    pipe_through :browser

    get "/", PageController, :index
  end
Routes are defined inside scopes and scopes may pipe through multiple pipelines. Once a route matches, Phoenix invokes all plugs defined in all pipelines associated to that route. For example, accessing "/" will pipe through the :browser pipeline, consequently invoking all of its plugs.
As we will see in the routing guide, the pipelines themselves are plugs. There, we will also discuss all plugs in the :browser pipeline.
Controller plugs
Finally, controllers are plugs too, so we can do:
defmodule HelloWeb.PageController do
  use HelloWeb, :controller

  plug HelloWeb.Plugs.Locale, "en"
In particular, controller plugs provide a feature that allows us to execute plugs only within certain actions. For example, you can do:
defmodule HelloWeb.PageController do
  use HelloWeb, :controller

  plug HelloWeb.Plugs.Locale, "en" when action in [:index]
And the plug will only be executed for the index action.
Plugs as composition
By abiding by the plug contract, we turn an application request into a series of explicit transformations. It doesn't stop there. To really see how effective Plug's design is, let's imagine a scenario where we need to check a series of conditions and then either redirect or halt if a condition fails. Without plug, we would end up with something like this:
defmodule HelloWeb.MessageController do
  use HelloWeb, :controller

  def show(conn, params) do
    case Authenticator.find_user(conn) do
      {:ok, user} ->
        case find_message(params["id"]) do
          nil ->
            conn |> put_flash(:info, "That message wasn't found") |> redirect(to: "/")
          message ->
            if Authorizer.can_access?(user, message) do
              render(conn, :show, page: message)
            else
              conn |> put_flash(:info, "You can't access that page") |> redirect(to: "/")
            end
        end
      :error ->
        conn |> put_flash(:info, "You must be logged in") |> redirect(to: "/")
    end
  end
end
Notice how just a few steps of authentication and authorization require complicated nesting and duplication? Let's improve this with a couple of plugs.
defmodule HelloWeb.MessageController do
  use HelloWeb, :controller

  plug :authenticate
  plug :fetch_message
  plug :authorize_message

  def show(conn, params) do
    render(conn, :show, page: conn.assigns[:message])
  end

  defp authenticate(conn, _) do
    case Authenticator.find_user(conn) do
      {:ok, user} ->
        assign(conn, :user, user)
      :error ->
        conn |> put_flash(:info, "You must be logged in") |> redirect(to: "/") |> halt()
    end
  end

  defp fetch_message(conn, _) do
    case find_message(conn.params["id"]) do
      nil ->
        conn |> put_flash(:info, "That message wasn't found") |> redirect(to: "/") |> halt()
      message ->
        assign(conn, :message, message)
    end
  end

  defp authorize_message(conn, _) do
    if Authorizer.can_access?(conn.assigns[:user], conn.assigns[:message]) do
      conn
    else
      conn |> put_flash(:info, "You can't access that page") |> redirect(to: "/") |> halt()
    end
  end
end
To make this all work, we converted the nested blocks of code and used halt(conn) whenever we reached a failure path. The halt(conn) functionality is essential here: it tells Plug that the next plug should not be invoked.
At the end of the day, by replacing the nested blocks of code with a flattened series of plug transformations, we are able to achieve the same functionality in a much more composable, clear, and reusable way.
To learn more about plugs, see the documentation for the Plug project, which provides many built-in plugs and functionalities.

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Request life-cycle guide.

Routers are the main hubs of Phoenix applications. They match HTTP requests to controller actions, wire up real-time channel handlers, and define a series of pipeline transformations scoped to a set of routes.
The router file that Phoenix generates, lib/hello_web/router.ex, will look something like this one:
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end

  pipeline :api do
    plug :accepts, ["json"]
  end

  scope "/", HelloWeb do
    pipe_through :browser

    get "/", PageController, :index
  end

  # Other scopes may use custom stacks.
  # scope "/api", HelloWeb do
  #   pipe_through :api
  # end
  # ...
end
Both the router and controller module names will be prefixed with the name you gave your application suffixed with Web.
The first line of this module, use HelloWeb, :router, simply makes Phoenix router functions available in our particular router.
Scopes have their own section in this guide, so we won't spend time on the scope "/", HelloWeb do block here. The pipe_through :browser line will get a full treatment in the "Pipelines" section of this guide. For now, you only need to know that pipelines allow a set of plugs to be applied to different sets of routes.
Inside the scope block, however, we have our first actual route:
get "/", PageController, :index
get is a Phoenix macro that corresponds to the HTTP verb GET. Similar macros exist for other HTTP verbs, including POST, PUT, PATCH, DELETE, OPTIONS, CONNECT, TRACE, and HEAD.
Examining routes
Phoenix provides an excellent tool for investigating routes in an application: mix phx.routes.
Let's see how this works. Go to the root of a newly-generated Phoenix application and run mix phx.routes. You should see something like the following, generated with all routes you currently have:
$ mix phx.routes
page_path  GET  /  HelloWeb.PageController :index
...

The route above tells us that any HTTP GET request for the root of the application will be handled by the index action of the HelloWeb.PageController.
page_path is an example of what Phoenix calls a path helper, and we'll talk about those very soon.
Resources
The router supports other macros besides those for HTTP verbs like get, post, and put. The most important among them is resources. Let's add a resource to our lib/hello_web/router.ex file like this:
scope "/", HelloWeb do
  pipe_through :browser

  get "/", PageController, :index
  resources "/users", UserController
  ...
end
For now it doesn't matter that we don't actually have a HelloWeb.UserController.
Run mix phx.routes once again at the root of your project. You should see something like the following:
...
user_path  GET     /users           HelloWeb.UserController :index
user_path  GET     /users/:id/edit  HelloWeb.UserController :edit
user_path  GET     /users/new       HelloWeb.UserController :new
user_path  GET     /users/:id       HelloWeb.UserController :show
user_path  POST    /users           HelloWeb.UserController :create
user_path  PATCH   /users/:id       HelloWeb.UserController :update
           PUT     /users/:id       HelloWeb.UserController :update
user_path  DELETE  /users/:id       HelloWeb.UserController :delete
...

This is the standard matrix of HTTP verbs, paths, and controller actions. For a while, this was known as RESTful routes, but most consider this a misnomer nowadays. Let's look at them individually, in a slightly different order.
	A GET request to /users will invoke the index action to show all the users.
	A GET request to /users/:id/edit will invoke the edit action with an ID to retrieve an individual user from the data store and present the information in a form for editing.
	A GET request to /users/new will invoke the new action to present a form for creating a new user.
	A GET request to /users/:id will invoke the show action with an id to show an individual user identified by that ID.
	A POST request to /users will invoke the create action to save a new user to the data store.
	A PATCH request to /users/:id will invoke the update action with an ID to save the updated user to the data store.
	A PUT request to /users/:id will also invoke the update action with an ID to save the updated user to the data store.
	A DELETE request to /users/:id will invoke the delete action with an ID to remove the individual user from the data store.

If we don't need all these routes, we can be selective using the :only and :except options to filter specific actions.
Let's say we have a read-only posts resource. We could define it like this:
resources "/posts", PostController, only: [:index, :show]
Running mix phx.routes shows that we now only have the routes to the index and show actions defined.
post_path  GET     /posts      HelloWeb.PostController :index
post_path  GET     /posts/:id  HelloWeb.PostController :show
Similarly, if we have a comments resource, and we don't want to provide a route to delete one, we could define a route like this.
resources "/comments", CommentController, except: [:delete]
Running mix phx.routes now shows that we have all the routes except the DELETE request to the delete action.
comment_path  GET    /comments           HelloWeb.CommentController :index
comment_path  GET    /comments/:id/edit  HelloWeb.CommentController :edit
comment_path  GET    /comments/new       HelloWeb.CommentController :new
comment_path  GET    /comments/:id       HelloWeb.CommentController :show
comment_path  POST   /comments           HelloWeb.CommentController :create
comment_path  PATCH  /comments/:id       HelloWeb.CommentController :update
              PUT    /comments/:id       HelloWeb.CommentController :update

The Phoenix.Router.resources/4 macro describes additional options for customizing resource routes.
Path helpers
Path helpers are dynamically defined functions. They allow us to retrieve the path corresponding to a given controller-action pair. The name of each path helper is derived from the name of the controller used in the route definition. For our controller HelloWeb.PageController, page_path is the function that will return the path, which in this case is the root of our application.
Let's see it in action. Run iex -S mix at the root of the project. When we call the page_path function on our router helpers with the endpoint or connection and action as arguments, it returns the path to us.
iex> HelloWeb.Router.Helpers.page_path(HelloWeb.Endpoint, :index)
"/"
This is significant because we can use the page_path function in a template to link to the root of our application. We can then use this helper in our templates:
<%= link "Welcome Page!", to: Routes.page_path(@conn, :index) %>
Note that path helpers are dynamically defined on the Router.Helpers module for an individual application. For us, that is HelloWeb.Router.Helpers.
The reason we can use Routes.page_path instead of the full HelloWeb.Router.Helpers.page_path name is because HelloWeb.Router.Helpers is aliased as Routes by default in the view_helpers/0 block defined inside lib/hello_web.ex. This definition is made available to our templates through use HelloWeb, :view.
We can, of course, use HelloWeb.Router.Helpers.page_path(@conn, :index) instead, but the convention is to use the aliased version for conciseness. Note that the alias is only set automatically for use in views, controllers and templates - outside these, you need either the full name, or to alias it yourself inside the module definition with alias HelloWeb.Router.Helpers, as: Routes.
Using path helpers makes it easy to ensure our controllers, views, and templates link to pages our router can actually handle.
More on path helpers
When we ran mix phx.routes for our user resource, it listed the user_path as the path helper function for each output line. Here is what that translates to for each action:
iex> alias HelloWeb.Router.Helpers, as: Routes
iex> alias HelloWeb.Endpoint

iex> Routes.user_path(Endpoint, :index)
"/users"

iex> Routes.user_path(Endpoint, :show, 17)
"/users/17"

iex> Routes.user_path(Endpoint, :new)
"/users/new"

iex> Routes.user_path(Endpoint, :create)
"/users"

iex> Routes.user_path(Endpoint, :edit, 37)
"/users/37/edit"

iex> Routes.user_path(Endpoint, :update, 37)
"/users/37"

iex> Routes.user_path(Endpoint, :delete, 17)
"/users/17"
What about paths with query strings? By adding an optional fourth argument of key-value pairs, the path helpers will return those pairs in the query string.
iex> Routes.user_path(Endpoint, :show, 17, admin: true, active: false)
"/users/17?admin=true&active=false"
What if we need a full URL instead of a path? Just replace _path with _url:
iex> Routes.user_url(Endpoint, :index)
"http://localhost:4000/users"
The *_url functions will get the host, port, proxy port, and SSL information needed to construct the full URL from the configuration parameters set for each environment. We'll talk about configuration in more detail in its own guide. For now, you can take a look at config/dev.exs file in your own project to see those values.
Whenever possible, prefer to pass a conn (or @conn in your views) in place of an endpoint module.
Nested resources
It is also possible to nest resources in a Phoenix router. Let's say we also have a posts resource that has a many-to-one relationship with users. That is to say, a user can create many posts, and an individual post belongs to only one user. We can represent that by adding a nested route in lib/hello_web/router.ex like this:
resources "/users", UserController do
  resources "/posts", PostController
end
When we run mix phx.routes now, in addition to the routes we saw for users above, we get the following set of routes:
...
user_post_path  GET     /users/:user_id/posts           HelloWeb.PostController :index
user_post_path  GET     /users/:user_id/posts/:id/edit  HelloWeb.PostController :edit
user_post_path  GET     /users/:user_id/posts/new       HelloWeb.PostController :new
user_post_path  GET     /users/:user_id/posts/:id       HelloWeb.PostController :show
user_post_path  POST    /users/:user_id/posts           HelloWeb.PostController :create
user_post_path  PATCH   /users/:user_id/posts/:id       HelloWeb.PostController :update
                PUT     /users/:user_id/posts/:id       HelloWeb.PostController :update
user_post_path  DELETE  /users/:user_id/posts/:id       HelloWeb.PostController :delete
...
We see that each of these routes scopes the posts to a user ID. For the first one, we will invoke PostController's index action, but we will pass in a user_id. This implies that we would display all the posts for that individual user only. The same scoping applies for all these routes.
When calling path helper functions for nested routes, we will need to pass the IDs in the order they came in the route definition. For the following show route, 42 is the user_id, and 17 is the post_id. Let's remember to alias our HelloWeb.Endpoint before we begin.
iex> alias HelloWeb.Endpoint
iex> HelloWeb.Router.Helpers.user_post_path(Endpoint, :show, 42, 17)
"/users/42/posts/17"
Again, if we add a key-value pair to the end of the function call, it is added to the query string.
iex> HelloWeb.Router.Helpers.user_post_path(Endpoint, :index, 42, active: true)
"/users/42/posts?active=true"
If we hadn't aliased the Helpers module as we did before (remember it is only automatically aliased for views, templates and controllers), and since we are inside iex, we'll have to do it ourselves:
iex> alias HelloWeb.Router.Helpers, as: Routes
iex> alias HelloWeb.Endpoint
iex> Routes.user_post_path(Endpoint, :index, 42, active: true)
"/users/42/posts?active=true"
Scoped routes
Scopes are a way to group routes under a common path prefix and scoped set of plugs. We might want to do this for admin functionality, APIs, and especially for versioned APIs. Let's say we have user-generated reviews on a site, and that those reviews first need to be approved by an administrator. The semantics of these resources are quite different, and they might not share the same controller. Scopes enable us to segregate these routes.
The paths to the user-facing reviews would look like a standard resource.
/reviews
/reviews/1234
/reviews/1234/edit
...

The administration review paths can be prefixed with /admin.
/admin/reviews
/admin/reviews/1234
/admin/reviews/1234/edit
...

We accomplish this with a scoped route that sets a path option to /admin like this one. We can nest this scope inside another scope, but instead, let's set it by itself at the root, by adding to lib/hello_web/router.ex the following:
scope "/admin", HelloWeb.Admin do
  pipe_through :browser

  resources "/reviews", ReviewController
end
We define a new scope where all routes are prefixed with /admin and all controllers are under the HelloWeb.Admin namespace.
Running mix phx.routes again, in addition to the previous set of routes we get the following:
...
review_path  GET     /admin/reviews           HelloWeb.Admin.ReviewController :index
review_path  GET     /admin/reviews/:id/edit  HelloWeb.Admin.ReviewController :edit
review_path  GET     /admin/reviews/new       HelloWeb.Admin.ReviewController :new
review_path  GET     /admin/reviews/:id       HelloWeb.Admin.ReviewController :show
review_path  POST    /admin/reviews           HelloWeb.Admin.ReviewController :create
review_path  PATCH   /admin/reviews/:id       HelloWeb.Admin.ReviewController :update
             PUT     /admin/reviews/:id       HelloWeb.Admin.ReviewController :update
review_path  DELETE  /admin/reviews/:id       HelloWeb.Admin.ReviewController :delete
...

This looks good, but there is a problem here. Remember that we wanted both user-facing review routes /reviews and the admin ones /admin/reviews. If we now include the user-facing reviews in our router under the root scope like this:
scope "/", HelloWeb do
  pipe_through :browser

  ...
  resources "/reviews", ReviewController
end

scope "/admin", HelloWeb.Admin do
  pipe_through :browser

  resources "/reviews", ReviewController
end
and we run mix phx.routes, we get this output:
...
review_path  GET     /reviews                 HelloWeb.ReviewController :index
review_path  GET     /reviews/:id/edit        HelloWeb.ReviewController :edit
review_path  GET     /reviews/new             HelloWeb.ReviewController :new
review_path  GET     /reviews/:id             HelloWeb.ReviewController :show
review_path  POST    /reviews                 HelloWeb.ReviewController :create
review_path  PATCH   /reviews/:id             HelloWeb.ReviewController :update
             PUT     /reviews/:id             HelloWeb.ReviewController :update
review_path  DELETE  /reviews/:id             HelloWeb.ReviewController :delete
...
review_path  GET     /admin/reviews           HelloWeb.Admin.ReviewController :index
review_path  GET     /admin/reviews/:id/edit  HelloWeb.Admin.ReviewController :edit
review_path  GET     /admin/reviews/new       HelloWeb.Admin.ReviewController :new
review_path  GET     /admin/reviews/:id       HelloWeb.Admin.ReviewController :show
review_path  POST    /admin/reviews           HelloWeb.Admin.ReviewController :create
review_path  PATCH   /admin/reviews/:id       HelloWeb.Admin.ReviewController :update
             PUT     /admin/reviews/:id       HelloWeb.Admin.ReviewController :update
review_path  DELETE  /admin/reviews/:id       HelloWeb.Admin.ReviewController :delete

The actual routes we get all look right, except for the path helper review_path at the beginning of each line. We are getting the same helper for both the user facing review routes and the admin ones, which is not correct.
We can fix this problem by adding an as: :admin option to our admin scope:
scope "/admin", HelloWeb.Admin, as: :admin do
  pipe_through :browser

  resources "/reviews", ReviewController
end
mix phx.routes now shows us we have what we are looking for.
...
      review_path  GET     /reviews                        HelloWeb.ReviewController :index
      review_path  GET     /reviews/:id/edit               HelloWeb.ReviewController :edit
      review_path  GET     /reviews/new                    HelloWeb.ReviewController :new
      review_path  GET     /reviews/:id                    HelloWeb.ReviewController :show
      review_path  POST    /reviews                        HelloWeb.ReviewController :create
      review_path  PATCH   /reviews/:id                    HelloWeb.ReviewController :update
                   PUT     /reviews/:id                    HelloWeb.ReviewController :update
      review_path  DELETE  /reviews/:id                    HelloWeb.ReviewController :delete
...
admin_review_path  GET     /admin/reviews                  HelloWeb.Admin.ReviewController :index
admin_review_path  GET     /admin/reviews/:id/edit         HelloWeb.Admin.ReviewController :edit
admin_review_path  GET     /admin/reviews/new              HelloWeb.Admin.ReviewController :new
admin_review_path  GET     /admin/reviews/:id              HelloWeb.Admin.ReviewController :show
admin_review_path  POST    /admin/reviews                  HelloWeb.Admin.ReviewController :create
admin_review_path  PATCH   /admin/reviews/:id              HelloWeb.Admin.ReviewController :update
                   PUT     /admin/reviews/:id              HelloWeb.Admin.ReviewController :update
admin_review_path  DELETE  /admin/reviews/:id              HelloWeb.Admin.ReviewController :delete

The path helpers now return what we want them to as well. Run iex -S mix and give it a try yourself.
iex> HelloWeb.Router.Helpers.review_path(HelloWeb.Endpoint, :index)
"/reviews"

iex> HelloWeb.Router.Helpers.admin_review_path(HelloWeb.Endpoint, :show, 1234)
"/admin/reviews/1234"
What if we had a number of resources that were all handled by admins? We could put all of them inside the same scope like this:
scope "/admin", HelloWeb.Admin, as: :admin do
  pipe_through :browser

  resources "/images",  ImageController
  resources "/reviews", ReviewController
  resources "/users",   UserController
end
Here's what mix phx.routes tells us:
...
 admin_image_path  GET     /admin/images            HelloWeb.Admin.ImageController :index
 admin_image_path  GET     /admin/images/:id/edit   HelloWeb.Admin.ImageController :edit
 admin_image_path  GET     /admin/images/new        HelloWeb.Admin.ImageController :new
 admin_image_path  GET     /admin/images/:id        HelloWeb.Admin.ImageController :show
 admin_image_path  POST    /admin/images            HelloWeb.Admin.ImageController :create
 admin_image_path  PATCH   /admin/images/:id        HelloWeb.Admin.ImageController :update
                   PUT     /admin/images/:id        HelloWeb.Admin.ImageController :update
 admin_image_path  DELETE  /admin/images/:id        HelloWeb.Admin.ImageController :delete
admin_review_path  GET     /admin/reviews           HelloWeb.Admin.ReviewController :index
admin_review_path  GET     /admin/reviews/:id/edit  HelloWeb.Admin.ReviewController :edit
admin_review_path  GET     /admin/reviews/new       HelloWeb.Admin.ReviewController :new
admin_review_path  GET     /admin/reviews/:id       HelloWeb.Admin.ReviewController :show
admin_review_path  POST    /admin/reviews           HelloWeb.Admin.ReviewController :create
admin_review_path  PATCH   /admin/reviews/:id       HelloWeb.Admin.ReviewController :update
                   PUT     /admin/reviews/:id       HelloWeb.Admin.ReviewController :update
admin_review_path  DELETE  /admin/reviews/:id       HelloWeb.Admin.ReviewController :delete
  admin_user_path  GET     /admin/users             HelloWeb.Admin.UserController :index
  admin_user_path  GET     /admin/users/:id/edit    HelloWeb.Admin.UserController :edit
  admin_user_path  GET     /admin/users/new         HelloWeb.Admin.UserController :new
  admin_user_path  GET     /admin/users/:id         HelloWeb.Admin.UserController :show
  admin_user_path  POST    /admin/users             HelloWeb.Admin.UserController :create
  admin_user_path  PATCH   /admin/users/:id         HelloWeb.Admin.UserController :update
                   PUT     /admin/users/:id         HelloWeb.Admin.UserController :update
  admin_user_path  DELETE  /admin/users/:id         HelloWeb.Admin.UserController :delete

This is great, exactly what we want. Note how every route, path helper and controller is properly namespaced.
Scopes can also be arbitrarily nested, but you should do it carefully as nesting can sometimes make our code confusing and less clear. With that said, suppose that we had a versioned API with resources defined for images, reviews, and users. Then technically, we could set up routes for the versioned API like this:
scope "/api", HelloWeb.Api, as: :api do
  pipe_through :api

  scope "/v1", V1, as: :v1 do
    resources "/images",  ImageController
    resources "/reviews", ReviewController
    resources "/users",   UserController
  end
end
You can run mix phx.routes to see how these definitions will look like.
Interestingly, we can use multiple scopes with the same path as long as we are careful not to duplicate routes. The following router is perfectly fine with two scopes defined for the same path:
defmodule HelloWeb.Router do
  use Phoenix.Router
  ...
  scope "/", HelloWeb do
    pipe_through :browser

    resources "/users", UserController
  end

  scope "/", AnotherAppWeb do
    pipe_through :browser

    resources "/posts", PostController
  end
  ...
end
If we do duplicate a route — which means two routes having the same path and the same alias — we'll get this familiar warning:
warning: this clause cannot match because a previous clause at line 16 always matches

Pipelines
We have come quite a long way in this guide without talking about one of the first lines we saw in the router: pipe_through :browser. It's time to fix that.
Pipelines are a series of plugs that can be attached to specific scopes. If you are not familiar with plugs, we have an in-depth guide about them.
Routes are defined inside scopes and scopes may pipe through multiple pipelines. Once a route matches, Phoenix invokes all plugs defined in all pipelines associated to that route. For example, accessing / will pipe through the :browser pipeline, consequently invoking all of its plugs.
Phoenix defines two pipelines by default, :browser and :api, which can be used for a number of common tasks. In turn we can customize them as well as create new pipelines to meet our needs.
The :browser and :api pipelines
As their names suggest, the :browser pipeline prepares for routes which render requests for a browser, and the :api pipeline prepares for routes which produce data for an API.
The :browser pipeline has six plugs: The plug :accepts, ["html"] defines the accepted request format or formats. :fetch_session, which, naturally, fetches the session data and makes it available in the connection. :fetch_live_flash, which fetches any flash messages from LiveView and merges them with the controller flash messages. Then, the plug :put_root_layout will store the root layout for rendering purposes. Later :protect_from_forgery and :put_secure_browser_headers, protects form posts from cross-site forgery.
Currently, the :api pipeline only defines plug :accepts, ["json"].
The router invokes a pipeline on a route defined within a scope. Routes outside of a scope have no pipelines. Although the use of nested scopes is discouraged (see above the versioned API example), if we call pipe_through within a nested scope, the router will invoke all pipe_through's from parent scopes, followed by the nested one.
Those are a lot of words bunched up together. Let's take a look at some examples to untangle their meaning.
Here's another look at the router from a newly generated Phoenix application, this time with the /api scope uncommented back in and a route added.
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end

  pipeline :api do
    plug :accepts, ["json"]
  end

  scope "/", HelloWeb do
    pipe_through :browser

    get "/", PageController, :index
  end

  # Other scopes may use custom stacks.
  scope "/api", HelloWeb do
    pipe_through :api

    resources "/reviews", ReviewController
  end
  # ...
end
When the server accepts a request, the request will always first pass through the plugs in our endpoint, after which it will attempt to match on the path and HTTP verb.
Let's say that the request matches our first route: a GET to /. The router will first pipe that request through the :browser pipeline - which will fetch the session data, fetch the flash, and execute forgery protection - before it dispatches the request to PageController's index action.
Conversely, suppose the request matches any of the routes defined by the resources/2 macro. In that case, the router will pipe it through the :api pipeline — which currently only performs content negotiation — before it dispatches further to the correct action of the HelloWeb.ReviewController.
If no route matches, no pipeline is invoked and a 404 error is raised.
Let's stretch these ideas out a little bit more. What if we need to pipe requests through both :browser and one or more custom pipelines? We simply pipe_through a list of pipelines, and Phoenix will invoke them in order.
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end
  ...

  scope "/reviews" do
    pipe_through [:browser, :review_checks, :other_great_stuff]

    resources "/", HelloWeb.ReviewController
  end
end
Here's another example with two scopes that have different pipelines:
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end
  ...

  scope "/", HelloWeb do
    pipe_through :browser

    resources "/posts", PostController
  end

  scope "/reviews", HelloWeb do
    pipe_through [:browser, :review_checks]

    resources "/", ReviewController
  end
end
In general, the scoping rules for pipelines behave as you might expect. In this example, all routes will pipe through the :browser pipeline. However, only the reviews resources routes will  pipe through the :review_checks pipeline. Since we declared both pipes pipe_through [:browser, :review_checks] in a list of pipelines, Phoenix will pipe_through each of them as it invokes them in order.
Creating new pipelines
Phoenix allows us to create our own custom pipelines anywhere in the router. To do so, we call the pipeline/2 macro with these arguments: an atom for the name of our new pipeline and a block with all the plugs we want in it.
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end

  pipeline :review_checks do
    plug :ensure_authenticated_user
    plug :ensure_user_owns_review
  end

  scope "/reviews", HelloWeb do
    pipe_through [:browser, :review_checks]

    resources "/", ReviewController
  end
end
Note that pipelines themselves are plugs, so we can plug a pipeline inside another pipeline. For example, we could rewrite the review_checks pipeline above to automatically invoke browser, simplifying the downstream pipeline call:
  pipeline :review_checks do
    plug :browser
    plug :ensure_authenticated_user
    plug :ensure_user_owns_review
  end

  scope "/reviews", HelloWeb do
    pipe_through :review_checks

    resources "/", ReviewController
  end
Forward
The Phoenix.Router.forward/4 macro can be used to send all requests that start with a particular path to a particular plug. Let's say we have a part of our system that is responsible (it could even be a separate application or library) for running jobs in the background, it could have its own web interface for checking the status of the jobs. We can forward to this admin interface using:
defmodule HelloWeb.Router do
  use HelloWeb, :router

  ...

  scope "/", HelloWeb do
    ...
  end

  forward "/jobs", BackgroundJob.Plug
end
This means that all routes starting with /jobs will be sent to the HelloWeb.BackgroundJob.Plug module. Inside the plug, you can match on subroutes, such as /pending and /active that shows the status of certain jobs.
We can even mix the forward/4 macro with pipelines. If we wanted to ensure that the user was authenticated and was an administrator in order to see the jobs page, we could use the following in our router.
defmodule HelloWeb.Router do
  use HelloWeb, :router

  ...

  scope "/" do
    pipe_through [:authenticate_user, :ensure_admin]
    forward "/jobs", BackgroundJob.Plug
  end
end
This means the plugs in the authenticate_user and ensure_admin pipelines will be called before the BackgroundJob.Plug allowing them to send an appropriate response and halt the request accordingly.
The opts that are received in the init/1 callback of the Module Plug can be passed as a third argument. For example, maybe the background job lets you set the name of your application to be displayed on the page. This could be passed with:
forward "/jobs", BackgroundJob.Plug, name: "Hello Phoenix"
There is a fourth router_opts argument that can be passed. These options are outlined in the Phoenix.Router.scope/2 documentation.
BackgroundJob.Plug can be implemented as any Module Plug discussed in the Plug guide. Note though it is not advised to forward to another Phoenix endpoint. This is because plugs defined by your app and the forwarded endpoint would be invoked twice, which may lead to errors.
Summary
Routing is a big topic, and we have covered a lot of ground here. The important points to take away from this guide are:
	Routes which begin with an HTTP verb name expand to a single clause of the match function.
	Routes declared with resources expand to 8 clauses of the match function.
	Resources may restrict the number of match function clauses by using the only: or except: options.
	Any of these routes may be nested.
	Any of these routes may be scoped to a given path.
	Using the as: option in a scope can reduce duplication.
	Using the helper option for scoped routes eliminates unreachable paths.

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the request life-cycle guide.

Phoenix controllers act as intermediary modules. Their functions — called actions — are invoked from the router in response to HTTP requests. The actions, in turn, gather all the necessary data and perform all the necessary steps before invoking the view layer to render a template or returning a JSON response.
Phoenix controllers also build on the Plug package, and are themselves plugs. Controllers provide the functions to do almost anything we need to in an action. If we do find ourselves looking for something that Phoenix controllers don't provide, we might find what we're looking for in Plug itself. Please see the Plug guide or the Plug documentation for more information.
A newly generated Phoenix app will have a single controller named PageController, which can be found at lib/hello_web/controllers/page_controller.ex which looks like this:
defmodule HelloWeb.PageController do
  use HelloWeb, :controller

  def index(conn, _params) do
    render(conn, "index.html")
  end
end
The first line below the module definition invokes the __using__/1 macro of the HelloWeb module, which imports some useful modules.
PageController gives us the index action to display the Phoenix welcome page associated with the default route Phoenix defines in the router.
Actions
Controller actions are just functions. We can name them anything we like as long as they follow Elixir's naming rules. The only requirement we must fulfill is that the action name matches a route defined in the router.
For example, in lib/hello_web/router.ex we could change the action name in the default route that Phoenix gives us in a new app from index:
get "/", PageController, :index
to home:
get "/", PageController, :home
as long as we change the action name in PageController to home as well, the welcome page will load as before.
defmodule HelloWeb.PageController do
  ...

  def home(conn, _params) do
    render(conn, "index.html")
  end
end
While we can name our actions whatever we like, there are conventions for action names which we should follow whenever possible. We went over these in the routing guide, but we'll take another quick look here.
	index   - renders a list of all items of the given resource type
	show    - renders an individual item by ID
	new     - renders a form for creating a new item
	create  - receives parameters for one new item and saves it in a data store
	edit    - retrieves an individual item by ID and displays it in a form for editing
	update  - receives parameters for one edited item and saves the item to a data store
	delete  - receives an ID for an item to be deleted and deletes it from a data store

Each of these actions takes two parameters, which will be provided by Phoenix behind the scenes.
The first parameter is always conn, a struct which holds information about the request such as the host, path elements, port, query string, and much more. conn comes to Phoenix via Elixir's Plug middleware framework. More detailed information about conn can be found in the Plug.Conn documentation.
The second parameter is params. Not surprisingly, this is a map which holds any parameters passed along in the HTTP request. It is a good practice to pattern match against parameters in the function signature to provide data in a simple package we can pass on to rendering. We saw this in the request life-cycle guide when we added a messenger parameter to our show route in lib/hello_web/controllers/hello_controller.ex.
defmodule HelloWeb.HelloController do
  ...

  def show(conn, %{"messenger" => messenger}) do
    render(conn, "show.html", messenger: messenger)
  end
end
In some cases — often in index actions, for instance — we don't care about parameters because our behavior doesn't depend on them. In those cases, we don't use the incoming parameters, and simply prefix the variable name with an underscore, calling it _params. This will keep the compiler from complaining about the unused variable while still keeping the correct arity.
Rendering
Controllers can render content in several ways. The simplest is to render some plain text using the text/2 function which Phoenix provides.
For example, let's rewrite the show action from HelloController to return text instead. For that, we could do the following.
def show(conn, %{"messenger" => messenger}) do
  text(conn, "From messenger #{messenger}")
end
Now /hello/Frank in your browser should display From messenger Frank as plain text without any HTML.
A step beyond this is rendering pure JSON with the json/2 function. We need to pass it something that the Jason library can decode into JSON, such as a map. (Jason is one of Phoenix's dependencies.)
def show(conn, %{"messenger" => messenger}) do
  json(conn, %{id: messenger})
end
If we again visit /hello/Frank in the browser, we should see a block of JSON with the key id mapped to the string "Frank".
{"id": "Frank"}
The json/2 function is useful for writing APIs and there is also the html/2 function for rendering HTML, but most of the times we use Phoenix views to build our responses. For this, Phoenix includes the render/3 function. It is specially important for HTML responses, as Phoenix Views provide performance and security benefits.
Let's rollback our show action to what we originally wrote in the request life-cycle guide:
defmodule HelloWeb.HelloController do
  use HelloWeb, :controller

  def show(conn, %{"messenger" => messenger}) do
    render(conn, "show.html", messenger: messenger)
  end
end
In order for the render/3 function to work correctly, the controller and view must share the same root name (in this case Hello), and it also must have the same root name as the template directory (in this case hello) where the show.html.heex template lives. In other words, HelloController requires HelloView, and HelloView requires the existence of the lib/hello_web/templates/hello directory, which must contain the show.html.heex template.
render/3 will also pass the value which the show action received for messenger from the parameters as an assign.
If we need to pass values into the template when using render, that's easy. We can pass a keyword like we've seen with messenger: messenger, or we can use Plug.Conn.assign/3, which conveniently returns conn.
  def show(conn, %{"messenger" => messenger}) do
    conn
    |> Plug.Conn.assign(:messenger, messenger)
    |> render("show.html")
  end
Note: Using Phoenix.Controller imports Plug.Conn, so shortening the call to assign/3 works just fine.
Passing more than one value to our template is as simple as connecting assign/3 functions together:
  def show(conn, %{"messenger" => messenger}) do
    conn
    |> assign(:messenger, messenger)
    |> assign(:receiver, "Dweezil")
    |> render("show.html")
  end
Generally speaking, once all assigns are configured, we invoke the view layer. The view layer then renders show.html alongside the layout and a response is sent back to the browser.
Views and templates have their own guide, so we won't spend much time on them here. What we will look at is how to assign a different layout, or none at all, from inside a controller action.
Assigning layouts
Layouts are just a special subset of templates. They live in the templates/layout folder (lib/hello_web/templates/layout). Phoenix created three for us when we generated our app. The default root layout is called root.html.heex, and it is the layout into which all templates will be rendered by default.
Since layouts are really just templates, they need a view to render them. This one is LayoutView which is defined in lib/hello_web/views/layout_view.ex. Since Phoenix generated this view for us, we won't have to create a new one as long as we put the layouts we want to render inside the lib/hello_web/templates/layout directory.
Before we create a new layout, though, let's do the simplest possible thing and render a template with no layout at all.
The Phoenix.Controller module provides the put_root_layout/2 function for us to switch root layouts. This takes conn as its first argument and a string for the basename of the layout we want to render. It also accepts false to disable the layout altogether.
You can edit the index action of PageController in lib/hello_web/controllers/page_controller.ex to look like this.
def index(conn, _params) do
  conn
  |> put_root_layout(false)
  |> render("index.html")
end
After reloading http://localhost:4000/, we should see a very different page, one with no title, logo image, or CSS styling at all.
Now let's actually create another layout and render the index template into it. As an example, let's say we had a different layout for the admin section of our application which didn't have the logo image. To do this, let's copy the existing root.html.heex to a new file admin.html.heex in the same directory lib/hello_web/templates/layout. Then let's replace the lines in admin.html.heex that displays the logo with the word "Administration".
Remove these lines:
<a href="https://phoenixframework.org/" class="phx-logo">
  <img src={Routes.static_path(@conn, "/images/phoenix.png")} alt="Phoenix Framework Logo"/>
</a>
Replace them with:
<p>Administration</p>
Then, pass the basename of the new layout into put_root_layout/2 in our index action in lib/hello_web/controllers/page_controller.ex.
def index(conn, _params) do
  conn
  |> put_root_layout("admin.html")
  |> render("index.html")
end
When we load the page, we should be rendering the admin layout without a logo and with the word "Administration".
Overriding rendering formats
Rendering HTML through a template is fine, but what if we need to change the rendering format on the fly? Let's say that sometimes we need HTML, sometimes we need plain text, and sometimes we need JSON. Then what?
Phoenix allows us to change formats on the fly with the _format query string parameter. To make this happen, Phoenix requires an appropriately named view and an appropriately named template in the correct directory.
As an example, let's take PageController's index action from a newly generated app. Out of the box, this has the right view PageView, the right templates directory (lib/hello_web/templates/page), and the right template for rendering HTML (index.html.heex.)
def index(conn, _params) do
  render(conn, "index.html")
end
What it doesn't have is an alternative template for rendering text. Let's add one at lib/hello_web/templates/page/index.text.eex. Here is our example index.text.eex template.
OMG, this is actually some text.
There are just a few more things we need to do to make this work. We need to tell our router that it should accept the text format. We do that by adding text to the list of accepted formats in the :browser pipeline. Let's open up lib/hello_web/router.ex and change plug :accepts to include text as well as html like this.
defmodule HelloWeb.Router do
  use HelloWeb, :router

  pipeline :browser do
    plug :accepts, ["html", "text"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
  end
...
We also need to tell the controller to render a template with the same format as the one returned by Phoenix.Controller.get_format/1. We do that by substituting the name of the template "index.html" with the atom version :index.
def index(conn, _params) do
  render(conn, :index)
end
If we go to http://localhost:4000/?_format=text, we will see "OMG, this is actually some text.".
Sending responses directly
If none of the rendering options above quite fits our needs, we can compose our own using some of the functions that Plug gives us. Let's say we want to send a response with a status of "201" and no body whatsoever. We can do that with the Plug.Conn.send_resp/3 function.
Edit the index action of PageController in lib/hello_web/controllers/page_controller.ex to look like this:
def index(conn, _params) do
  send_resp(conn, 201, "")
end
Reloading http://localhost:4000 should show us a completely blank page. The network tab of our browser's developer tools should show a response status of "201" (Created). Some browsers (Safari) will download the response, as the content type is not set.
To be specific about the content type, we can use put_resp_content_type/2 in conjunction with send_resp/3.
def index(conn, _params) do
  conn
  |> put_resp_content_type("text/plain")
  |> send_resp(201, "")
end
Using Plug functions in this way, we can craft just the response we need.
Setting the Content Type
Analogous to the _format query string param, we can render any sort of format we want by modifying the HTTP Content-Type Header and providing the appropriate template.
If we wanted to render an XML version of our index action, we might implement the action like this in lib/hello_web/page_controller.ex.
def index(conn, _params) do
  conn
  |> put_resp_content_type("text/xml")
  |> render("index.xml", content: some_xml_content)
end
We would then need to provide an index.xml.eex template which created valid XML, and we would be done.
For a list of valid content mime-types, please see the MIME library.
Setting the HTTP Status
We can also set the HTTP status code of a response similarly to the way we set the content type. The Plug.Conn module, imported into all controllers, has a put_status/2 function to do this.
Plug.Conn.put_status/2 takes conn as the first parameter and as the second parameter either an integer or a "friendly name" used as an atom for the status code we want to set. The list of status code atom representations can be found in Plug.Conn.Status.code/1 documentation.
Let's change the status in our PageController index action.
def index(conn, _params) do
  conn
  |> put_status(202)
  |> render("index.html")
end
The status code we provide must be a valid number.
Redirection
Often, we need to redirect to a new URL in the middle of a request. A successful create action, for instance, will usually redirect to the show action for the resource we just created. Alternately, it could redirect to the index action to show all the things of that same type. There are plenty of other cases where redirection is useful as well.
Whatever the circumstance, Phoenix controllers provide the handy redirect/2 function to make redirection easy. Phoenix differentiates between redirecting to a path within the application and redirecting to a URL — either within our application or external to it.
In order to try out redirect/2, let's create a new route in lib/hello_web/router.ex.
defmodule HelloWeb.Router do
  ...

  scope "/", HelloWeb do
    ...
    get "/", PageController, :index
    get "/redirect_test", PageController, :redirect_test
    ...
  end
end
Then we'll change PageController's index action of our controller to do nothing but to redirect to our new route.
defmodule HelloWeb.PageController do
  use HelloWeb, :controller

  def index(conn, _params) do
    redirect(conn, to: "/redirect_test")
  end
end

Actually, we should make use of the path helpers, which are the preferred approach to link to any page within our application, as we learned about in the routing guide.
def index(conn, _params) do
  redirect(conn, to: Routes.page_path(conn, :redirect_test))
end
Finally, let's define in the same file the action we redirect to, which simply renders the index, but now under a new address:
def redirect_test(conn, _params) do
  render(conn, "index.html")
end
When we reload our welcome page, we see that we've been redirected to /redirect_test which shows the original welcome page. It works!
If we care to, we can open up our developer tools, click on the network tab, and visit our root route again. We see two main requests for this page - a get to / with a status of 302, and a get to /redirect_test with a status of 200.
Notice that the redirect function takes conn as well as a string representing a relative path within our application. For security reasons, the :to helper can only redirect to paths within your application. If you want to redirect to a fully-qualified path or an external URL, you should use :external instead:
def index(conn, _params) do
  redirect(conn, external: "https://elixir-lang.org/")
end
Flash messages
Sometimes we need to communicate with users during the course of an action. Maybe there was an error updating a schema, or maybe we just want to welcome them back to the application. For this, we have flash messages.
The Phoenix.Controller module provides the put_flash/3 and get_flash/2 functions to help us set and retrieve flash messages as a key-value pair. Let's set two flash messages in our HelloWeb.PageController to try this out.
To do this we modify the index action as follows:
defmodule HelloWeb.PageController do
  ...
  def index(conn, _params) do
    conn
    |> put_flash(:info, "Welcome to Phoenix, from flash info!")
    |> put_flash(:error, "Let's pretend we have an error.")
    |> render("index.html")
  end
end
In order to see our flash messages, we need to be able to retrieve them and display them in a template layout. One way to do the first part is with get_flash/2 which takes conn and the key we care about. It then returns the value for that key.
For our convenience, the application layout, lib/hello_web/templates/layout/app.html.heex, already has markup for displaying flash messages.
<p class="alert alert-info" role="alert"><%= get_flash(@conn, :info) %></p>
<p class="alert alert-danger" role="alert"><%= get_flash(@conn, :error) %></p>
When we reload the welcome page, our messages should appear just above "Welcome to Phoenix!"
The flash functionality is handy when mixed with redirects. Perhaps you want to redirect to a page with some extra information. If we reuse the redirect action from the previous section, we can do:
  def index(conn, _params) do
    conn
    |> put_flash(:info, "Welcome to Phoenix, from flash info!")
    |> put_flash(:error, "Let's pretend we have an error.")
    |> redirect(to: Routes.page_path(conn, :redirect_test))
  end
Now if you reload the welcome page, you will be redirected and the flash messages will be shown once more.
Besides put_flash/3 and get_flash/2, the Phoenix.Controller module has another useful function worth knowing about. clear_flash/1 takes only conn and removes any flash messages which might be stored in the session.
Phoenix does not enforce which keys are stored in the flash. As long as we are internally consistent, all will be well. :info and :error, however, are common and are handled by default in our templates.
Action fallback
Action fallback allows us to centralize error handling code in plugs, which are called when a controller action fails to return a %Plug.Conn{} struct. These plugs receive both the conn which was originally passed to the controller action along with the return value of the action.
Let's say we have a show action which uses with to fetch a blog post and then authorize the current user to view that blog post. In this example we might expect fetch_post/1 to return {:error, :not_found} if the post is not found and authorize_user/3 might return {:error, :unauthorized} if the user is unauthorized. We could use ErrorView which is generated by Phoenix for every new application to handle these error paths accordingly:
defmodule HelloWeb.MyController do
  use Phoenix.Controller

  def show(conn, %{"id" => id}, current_user) do
    with {:ok, post} <- fetch_post(id),
         :ok <- authorize_user(current_user, :view, post) do
      render(conn, "show.json", post: post)
    else
      {:error, :not_found} ->
        conn
        |> put_status(:not_found)
        |> put_view(HelloWeb.ErrorView)
        |> render(:"404")

      {:error, :unauthorized} ->
        conn
        |> put_status(403)
        |> put_view(HelloWeb.ErrorView)
        |> render(:"403")
    end
  end
end
Now imagine you may need to implement similar logic for every controller and action handled by your API. This would result in a lot of repetition.
Instead we can define a module plug which knows how to handle these error cases specifically. Since controllers are module plugs, let's define our plug as a controller:
defmodule HelloWeb.MyFallbackController do
  use Phoenix.Controller

  def call(conn, {:error, :not_found}) do
    conn
    |> put_status(:not_found)
    |> put_view(HelloWeb.ErrorView)
    |> render(:"404")
  end

  def call(conn, {:error, :unauthorized}) do
    conn
    |> put_status(403)
    |> put_view(HelloWeb.ErrorView)
    |> render(:"403")
  end
end
Then we can reference our new controller as the action_fallback and simply remove the else block from our with:
defmodule HelloWeb.MyController do
  use Phoenix.Controller

  action_fallback HelloWeb.MyFallbackController

  def show(conn, %{"id" => id}, current_user) do
    with {:ok, post} <- fetch_post(id),
         :ok <- authorize_user(current_user, :view, post) do
      render(conn, "show.json", post: post)
    end
  end
end
Whenever the with conditions do not match, HelloWeb.MyFallbackController will receive the original conn as well as the result of the action and respond accordingly.

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the request life-cycle guide.

The main job of a Phoenix view is to render the body of the response which gets sent back to browsers and to API clients. Most of the time, we use templates to build these responses, but we can also craft them by hand. We will learn how.
Rendering templates
Phoenix assumes a strong naming convention from controllers to views to the templates they render. PageController requires a PageView to render templates in the lib/hello_web/templates/page/ directory. While all of these can be customizable (see Phoenix.View and Phoenix.Template for more information), we recommend users stick with Phoenix' convention.
A newly generated Phoenix application has three view modules - ErrorView, LayoutView, and PageView -  which are all in the lib/hello_web/views/ directory.
Let's take a quick look at LayoutView.
defmodule HelloWeb.LayoutView do
  use HelloWeb, :view
end
That's simple enough. There's only one line, use HelloWeb, :view. This line calls the view/0 function defined in HelloWeb which sets up the basic imports and configuration for our views and templates.
All of the imports and aliases we make in our view will also be available in our templates. That's because templates are effectively compiled into functions inside their respective views. For example, if you define a function in your view, you will be able to invoke it directly from the template. Let's see this in practice.
Open up our application layout template, lib/hello_web/templates/layout/root.html.heex, and change this line,
<%= live_title_tag assigns[:page_title] || "Hello", suffix: " · Phoenix Framework" %>
to call a title/0 function, like this.
<title><%= title() %></title>
Now let's add a title/0 function to our LayoutView.
defmodule HelloWeb.LayoutView do
  use HelloWeb, :view

  def title() do
    "Awesome New Title!"
  end
end
When we reload our home page, we should see our new title. Since templates are compiled inside the view, we could invoke the view function simply as title(), otherwise we would have to type HelloWeb.LayoutView.title().
As you may recall, Elixir templates use .heex, which stands for  "HTML+EEx". EEx is an Elixir library that uses <%= expression %> to execute Elixir expressions and interpolate their results into the template. This is frequently used to display assigns we have set by way of the @ shortcut. In your controller, if you invoke:
  render(conn, "show.html", username: "joe")
Then you can access said username in the templates as <%= @username %>. In addition to displaying assigns and functions, we can use pretty much any Elixir expression. For example, in order to have conditionals:
<%= if some_condition? do %>
  <p>Some condition is true for user: <%= @username %></p>
<% else %>
  <p>Some condition is false for user: <%= @username %></p>
<% end %>
or even loops:
<table>
  <tr>
    <th>Number</th>
    <th>Power</th>
  </tr>
<%= for number <- 1..10 do %>
  <tr>
    <td><%= number %></td>
    <td><%= number * number %></td>
  </tr>
<% end %>
</table>
Did you notice the use of <%= %> versus <% %> above? All expressions that output something to the template must use the equals sign (=). If this is not included the code will still be executed but nothing will be inserted into the template.
HTML extensions
Besides allowing interpolation of Elixir expressions via <%= %>, .heex templates come with HTML-aware extensions. For example, let's see what happens if you try to interpolate a value with "<" or ">" in it, which would lead to HTML injection:
<%= "<b>Bold?</b>" %>
Once you render the template, you will see the literal <b> on the page. This means users cannot inject HTML content on the page. If you want to allow them to do so, you can call raw, but do so with extreme care:
<%= raw "<b>Bold?</b>" %>
Another super power of HEEx templates is validation of HTML and lean interpolation syntax of attributes. You can write:
<div title="My div" class={@class}>
  <p>Hello <%= @username %></p>
</div>
Notice how you could simply use key={value}. HEEx will automatically handle special values such as false to remove the attribute or a list of classes.
To interpolate a dynamic number of attributes in a keyword list or map, do:
<div title="My div" {@many_attributes}>
  <p>Hello <%= @username %></p>
</div>
Also, try removing the closing </div> or renaming it to </div-typo>. HEEx templates will let you know about your error.
HTML components
The last feature provided by HEEx is the idea of components. Components are pure functions that can be either local (same module) or remote (external module).
HEEx allows invoking those function components directly in the template using an HTML-like notation. For example, a remote function:
<MyApp.Weather.city name="Kraków"/>
A local function can be invoked with a leading dot:
<.city name="Kraków"/>
where the component could be defined as follows:
defmodule MyApp.Weather do
  use Phoenix.Component

  def city(assigns) do
    ~H"""
    The chosen city is: <%= @name %>.
    """
  end

  def country(assigns) do
    ~H"""
    The chosen country is: <%= @name %>.
    """
  end
end
In the example above, we used the ~H sigil syntax to embed HEEx templates directly into our modules. We have already invoked the city component and calling the country component wouldn't be different:
<div title="My div" {@many_attributes}>
  <p>Hello <%= @username %></p>
  <MyApp.Weather.country name="Brazil" />
</div>
You can learn more about components in Phoenix.Component.
Understanding template compilation
Phoenix templates are compiled into Elixir code, which make them extremely performant. Let's learn more about this.
When a template is compiled into a view, it is simply compiled as a render/2 function that expects two arguments: the template name and the assigns.
You can prove this by temporarily adding this function clause to your PageView module in lib/hello_web/views/page_view.ex.
defmodule HelloWeb.PageView do
  use HelloWeb, :view

  def render("index.html", assigns) do
    "rendering with assigns #{inspect Map.keys(assigns)}"
  end
end
Now if you fire up the server with mix phx.server and visit http://localhost:4000, you should see the following text below your layout header instead of the main template page:
rendering with assigns [:conn]

By defining our own clause in render/2, it takes higher priority than the template, but the template is still there, which you can verify by simply removing the newly added clause.
Pretty neat, right? At compile-time, Phoenix precompiles all *.html.heex templates and turns them into render/2 function clauses on their respective view modules. At runtime, all templates are already loaded in memory. There's no disk reads, complex file caching, or template engine computation involved.
Manually rendering templates
So far, Phoenix has taken care of putting everything in place and rendering views for us. However, we can also render views directly.
Let's create a new template to play around with, lib/hello_web/templates/page/test.html.heex:
This is the message: <%= @message %>
This doesn't correspond to any action in our controller, which is fine. We'll exercise it in an IEx session. At the root of our project, we can run iex -S mix, and then explicitly render our template. Let's give it a try by calling Phoenix.View.render/3 with the view name, the template name, and a set of assigns we might have wanted to pass and we got the rendered template as a string:
iex(1)> Phoenix.View.render(HelloWeb.PageView, "test.html", message: "Hello from IEx!")
%Phoenix.LiveView.Rendered{
  dynamic: #Function<1.71437968/1 in Hello16Web.PageView."test.html"/1>,
  fingerprint: 142353463236917710626026938006893093300,
  root: false,
  static: ["This is the message: ", ""]
}
The output we got above is not very helpful. That's the internal representation of how Phoenix keeps our rendered templates. Luckily, we can convert them into strings with render_to_string/3:
iex(2)> Phoenix.View.render_to_string(HelloWeb.PageView, "test.html", message: "Hello from IEx!")
"This is the message: Hello from IEx!"
That's much better! Let's test out the HTML escaping, just for fun:
iex(3)> Phoenix.View.render_to_string(HelloWeb.PageView, "test.html", message: "<script>badThings();</script>")
"This is the message: &lt;script&gt;badThings();&lt;/script&gt;"
Sharing views and templates
Now that we have acquainted ourselves with Phoenix.View.render/3, we are ready to share views and templates from inside other views and templates. We use render/3 to compose our templates and at the end Phoenix will convert them all into the proper representation to send to the browser.
For example, if you want to render the test.html template from inside our layout, you can invoke render/3 directly from the layout lib/hello_web/templates/layout/root.html.heex:
<%= Phoenix.View.render(HelloWeb.PageView, "test.html", message: "Hello from layout!") %>
If you visit the welcome page, you should see the message from the layout.
Since Phoenix.View is automatically imported into our templates, we could even skip the Phoenix.View module name and simply invoke render(...) directly:
<%= render(HelloWeb.PageView, "test.html", message: "Hello from layout!") %>
If you want to render a template within the same view, you can skip the view name, and simply call render("test.html", message: "Hello from sibling template!") instead. For example, open up lib/hello_web/templates/page/index.html.heex and add this at the top:
<%= render("test.html", message: "Hello from sibling template!") %>
Now if you visit the Welcome page, you see the template results also shown.
Layouts
Layouts are just templates. They have a view, just like other templates. In a newly generated app, this is lib/hello_web/views/layout_view.ex. You may be wondering how the string resulting from a rendered view ends up inside a layout. That's a great question! If we look at lib/hello_web/templates/layout/root.html.heex, just about at the end of the <body>, we will see this.
<%= @inner_content %>
In other words, the inner template is placed in the @inner_content assign.
Rendering JSON
The view's job is not only to render HTML templates. Views are about data presentation. Given a bag of data, the view's purpose is to present that in a meaningful way given some format, be it HTML, JSON, CSV, or others. Many web apps today return JSON to remote clients, and Phoenix views are great for JSON rendering.
Phoenix uses the Jason library to encode JSON, so all we need to do in our views is to format the data we would like to respond with as a list or a map, and Phoenix will do the rest.
While it is possible to respond with JSON back directly from the controller and skip the view, Phoenix views provide a much more structured approach for doing  so. Let's take our PageController, and see what it may look like when we respond with some static page maps as JSON, instead of HTML.
defmodule HelloWeb.PageController do
  use HelloWeb, :controller

  def show(conn, _params) do
    page = %{title: "foo"}

    render(conn, "show.json", page: page)
  end

  def index(conn, _params) do
    pages = [%{title: "foo"}, %{title: "bar"}]

    render(conn, "index.json", pages: pages)
  end
end
Here, we have our show/2 and index/2 actions returning static page data. Instead of passing in "show.html" to render/3 as the template name, we pass "show.json". This way, we can have views that are responsible for rendering HTML as well as JSON by pattern matching on different file types.
defmodule HelloWeb.PageView do
  use HelloWeb, :view

  def render("index.json", %{pages: pages}) do
    %{data: Enum.map(pages, fn page -> %{title: page.title} end)}
  end

  def render("show.json", %{page: page}) do
    %{data: %{title: page.title}}
  end
end
In the view we see our render/2 function pattern matching on "index.json", "show.json", and "page.json". The "index.json" and "show.json" are the ones requested directly from the controller. They also match on the assigns sent by the controller. Phoenix understands the .json extension and will take care of converting the data-structures we return into JSON.  "index.json" will respond like this:
{
  "data": [
    {
     "title": "foo"
    },
    {
     "title": "bar"
    },
 ]
}
And "show.json" like this:
{
  "data": {
    "title": "foo"
  }
}
However, there is some duplication between index.json and show.json, as both encode the same logic on how to render pages. We can address this by moving the page rendering to a separate function clause and using render_many/3 and render_one/3 to reuse it:
defmodule HelloWeb.PageView do
  use HelloWeb, :view

  def render("index.json", %{pages: pages}) do
    %{data: render_many(pages, HelloWeb.PageView, "page.json")}
  end

  def render("show.json", %{page: page}) do
    %{data: render_one(page, HelloWeb.PageView, "page.json")}
  end

  def render("page.json", %{page: page}) do
    %{title: page.title}
  end
end
The render_many/3 function takes the data we want to respond with (pages), a view, and a string to pattern match on the render/2 function defined on view. It will map over each item in pages and call PageView.render("page.json", %{page: page}). render_one/3 follows the same signature, ultimately using the render/2 matching page.json to specify what each page looks like.
It's useful to build our views like this so that they are composable. Imagine a situation where our Page has a has_many relationship (#NOTE: We haven't talked about has_many relationship yet#) with Author, and depending on the request, we may want to send back author data with the page. We can easily accomplish this with a new render/2:
defmodule HelloWeb.PageView do
  use HelloWeb, :view
  alias HelloWeb.AuthorView

  def render("page_with_authors.json", %{page: page}) do
    %{title: page.title,
      authors: render_many(page.authors, AuthorView, "author.json")}
  end

  def render("page.json", %{page: page}) do
    %{title: page.title}
  end
end
The name used in assigns is determined from the view. For example PageView will use %{page: page} and AuthorView will use %{author: author}. This can be overridden with the as option. Let's assume that the author view uses %{writer: writer} instead of %{author: author}:
def render("page_with_authors.json", %{page: page}) do
  %{title: page.title,
    authors: render_many(page.authors, AuthorView, "author.json", as: :writer)}
end
Error pages
Phoenix has a view called ErrorView which lives in lib/hello_web/views/error_view.ex. The purpose of ErrorView is to handle errors in a general way, from one centralized location.  Similar to the views we built in this guide, error views can return both HTML and JSON responses. See the Custom Error Pages How-To for more information.

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Most web applications today need some form of data validation and persistence. In the Elixir ecosystem, we have Ecto to enable this. Before we jump into building database-backed web features, we're going to focus on the finer details of Ecto to give a solid base to build our web features on top of. Let's get started!
Phoenix uses Ecto to provide builtin support to the following databases:
	PostgreSQL (via postgrex)
	MySQL (via myxql)
	MSSQL (via tds)
	ETS (via etso)
	SQLite3 (via ecto_sqlite3)

Newly generated Phoenix projects include Ecto with the PostgreSQL adapter by default. You can pass the --database option to change or --no-ecto flag to exclude this.
Ecto also provides support for other databases and it has many learning resources available. Please check out Ecto's README for general information.
This guide assumes that we have generated our new application with Ecto integration and that we will be using PostgreSQL. The introductory guides cover how to get your first application up and running. For instructions on switching to MySQL, please see the Using MySQL section.
Using the schema and migration generator
Once we have Ecto and PostgreSQL installed and configured, the easiest way to use Ecto is to generate an Ecto schema through the phx.gen.schema task. Ecto schemas are a way for us to specify how Elixir data types map to and from external sources, such as database tables. Let's generate a User schema with name, email, bio, and number_of_pets fields.
$ mix phx.gen.schema User users name:string email:string \
bio:string number_of_pets:integer

* creating ./lib/hello/user.ex
* creating priv/repo/migrations/20170523151118_create_users.exs

Remember to update your repository by running migrations:

   $ mix ecto.migrate

A couple of files were generated with this task. First, we have a user.ex file, containing our Ecto schema with our schema definition of the fields we passed to the task. Next, a migration file was generated inside priv/repo/migrations/ which will create our database table that our schema maps to.
With our files in place, let's follow the instructions and run our migration:
$ mix ecto.migrate
Compiling 1 file (.ex)
Generated hello app

[info]  == Running Hello.Repo.Migrations.CreateUsers.change/0 forward

[info]  create table users

[info]  == Migrated in 0.0s

Mix assumes that we are in the development environment unless we tell it otherwise with MIX_ENV=prod mix ecto.migrate.
If we log in to our database server, and connect to our hello_dev database, we should see our users table. Ecto assumes that we want an integer column called id as our primary key, so we should see a sequence generated for that as well.
$ psql -U postgres

Type "help" for help.

postgres=# \connect hello_dev
You are now connected to database "hello_dev" as user "postgres".
hello_dev=# \d
                List of relations
 Schema |       Name        |   Type   |  Owner
--------+-------------------+----------+----------
 public | schema_migrations | table    | postgres
 public | users             | table    | postgres
 public | users_id_seq      | sequence | postgres
(3 rows)
hello_dev=# \q

If we take a look at the migration generated by phx.gen.schema in priv/repo/migrations/, we'll see that it will add the columns we specified. It will also add timestamp columns for inserted_at and updated_at which come from the timestamps/1 function.
defmodule Hello.Repo.Migrations.CreateUsers do
  use Ecto.Migration

  def change do
    create table(:users) do
      add :name, :string
      add :email, :string
      add :bio, :string
      add :number_of_pets, :integer

      timestamps()
    end
  end
end
And here's what that translates to in the actual users table.
$ psql
hello_dev=# \d users
Table "public.users"
Column         |            Type             | Modifiers
---------------+-----------------------------+----------------------------------------------------
id             | bigint                      | not null default nextval('users_id_seq'::regclass)
name           | character varying(255)      |
email          | character varying(255)      |
bio            | character varying(255)      |
number_of_pets | integer                     |
inserted_at    | timestamp without time zone | not null
updated_at     | timestamp without time zone | not null
Indexes:
"users_pkey" PRIMARY KEY, btree (id)

Notice that we do get an id column as our primary key by default, even though it isn't listed as a field in our migration.
Repo configuration
Our Hello.Repo module is the foundation we need to work with databases in a Phoenix application. Phoenix generated it for us in lib/hello/repo.ex, and this is what it looks like.
defmodule Hello.Repo do
  use Ecto.Repo,
    otp_app: :hello,
    adapter: Ecto.Adapters.Postgres
end
It begins by defining the repository module. Then it configures our otp_app name, and the adapter – Postgres, in our case.
Our repo has three main tasks - to bring in all the common query functions from [Ecto.Repo], to set the otp_app name equal to our application name, and to configure our database adapter. We'll talk more about how to use Hello.Repo in a bit.
When phx.new generated our application, it included some basic repository configuration as well. Let's look at config/dev.exs.
...
# Configure your database
config :hello, Hello.Repo,
  username: "postgres",
  password: "postgres",
  hostname: "localhost",
  database: "hello_dev",
  show_sensitive_data_on_connection_error: true,
  pool_size: 10
...
We also have similar configuration in config/test.exs and config/runtime.exs (formerly config/prod.secret.exs) which can also be changed to match your actual credentials.
The schema
Ecto schemas are responsible for mapping Elixir values to external data sources, as well as mapping external data back into Elixir data structures. We can also define relationships to other schemas in our applications. For example, our User schema might have many posts, and each post would belong to a user. Ecto also handles data validation and type casting with changesets, which we'll discuss in a moment.
Here's the User schema that Phoenix generated for us.
defmodule Hello.User do
  use Ecto.Schema
  import Ecto.Changeset
  alias Hello.User

  schema "users" do
    field :bio, :string
    field :email, :string
    field :name, :string
    field :number_of_pets, :integer

    timestamps()
  end

  @doc false
  def changeset(user, attrs) do
    user
    |> cast(attrs, [:name, :email, :bio, :number_of_pets])
    |> validate_required([:name, :email, :bio, :number_of_pets])
  end
end
Ecto schemas at their core are simply Elixir structs. Our schema block is what tells Ecto how to cast our %User{} struct fields to and from the external users table. Often, the ability to simply cast data to and from the database isn't enough and extra data validation is required. This is where Ecto changesets come in. Let's dive in!
Changesets and validations
Changesets define a pipeline of transformations our data needs to undergo before it will be ready for our application to use. These transformations might include type-casting, user input validation, and filtering out any extraneous parameters. Often we'll use changesets to validate user input before writing it to the database. Ecto repositories are also changeset-aware, which allows them not only to refuse invalid data, but also perform the minimal database updates possible by inspecting the changeset to know which fields have changed.
Let's take a closer look at our default changeset function.
def changeset(user, attrs) do
  user
  |> cast(attrs, [:name, :email, :bio, :number_of_pets])
  |> validate_required([:name, :email, :bio, :number_of_pets])
end
Right now, we have two transformations in our pipeline. In the first call, we invoke Ecto.Changeset.cast/3, passing in our external parameters and marking which fields are required for validation.
cast/3 first takes a struct, then the parameters (the proposed updates), and then the final field is the list of columns to be updated. cast/3 also will only take fields that exist in the schema.
Next, Ecto.Changeset.validate_required/3 checks that this list of fields is present in the changeset that cast/3 returns. By default with the generator, all fields are required.
We can verify this functionality in IEx. Let's fire up our application inside IEx by running iex -S mix. In order to minimize typing and make this easier to read, let's alias our Hello.User struct.
$ iex -S mix

iex> alias Hello.User
Hello.User

Next, let's build a changeset from our schema with an empty User struct, and an empty map of parameters.
iex> changeset = User.changeset(%User{}, %{})
#Ecto.Changeset<
  action: nil,
  changes: %{},
  errors: [
    name: {"can't be blank", [validation: :required]},
    email: {"can't be blank", [validation: :required]},
    bio: {"can't be blank", [validation: :required]},
    number_of_pets: {"can't be blank", [validation: :required]}
  ],
  data: #Hello.User<>,
  valid?: false
>
Once we have a changeset, we can check if it is valid.
iex> changeset.valid?
false
Since this one is not valid, we can ask it what the errors are.
iex> changeset.errors
[
  name: {"can't be blank", [validation: :required]},
  email: {"can't be blank", [validation: :required]},
  bio: {"can't be blank", [validation: :required]},
  number_of_pets: {"can't be blank", [validation: :required]}
]
Now, let's make number_of_pets optional. In order to do this, we simply remove it from the list in the changeset/2 function, in Hello.User.
    |> validate_required([:name, :email, :bio])
Now casting the changeset should tell us that only name, email, and bio can't be blank. We can test that by running recompile() inside IEx and then rebuilding our changeset.
iex> recompile()
Compiling 1 file (.ex)
:ok

iex> changeset = User.changeset(%User{}, %{})
#Ecto.Changeset<
  action: nil,
  changes: %{},
  errors: [
    name: {"can't be blank", [validation: :required]},
    email: {"can't be blank", [validation: :required]},
    bio: {"can't be blank", [validation: :required]}
  ],
  data: #Hello.User<>,
  valid?: false
>

iex> changeset.errors
[
  name: {"can't be blank", [validation: :required]},
  email: {"can't be blank", [validation: :required]},
  bio: {"can't be blank", [validation: :required]}
]
What happens if we pass a key-value pair that is neither defined in the schema nor required?
Inside our existing IEx shell, let's create a params map with valid values plus an extra random_key: "random value".
iex> params = %{name: "Joe Example", email: "joe@example.com", bio: "An example to all", number_of_pets: 5, random_key: "random value"}
%{
  bio: "An example to all",
  email: "joe@example.com",
  name: "Joe Example",
  number_of_pets: 5,
  random_key: "random value"
}
Next, let's use our new params map to create another changeset.
iex> changeset = User.changeset(%User{}, params)
#Ecto.Changeset<
  action: nil,
  changes: %{
    bio: "An example to all",
    email: "joe@example.com",
    name: "Joe Example",
    number_of_pets: 5
  },
  errors: [],
  data: #Hello.User<>,
  valid?: true
> 
Our new changeset is valid.
iex> changeset.valid?
true
We can also check the changeset's changes - the map we get after all of the transformations are complete.
iex(9)> changeset.changes
%{bio: "An example to all", email: "joe@example.com", name: "Joe Example",
  number_of_pets: 5}
Notice that our random_key key and "random_value" value have been removed from the final changeset. Changesets allow us to cast external data, such as user input on a web form or data from a CSV file into valid data into our system. Invalid parameters will be stripped and bad data that is unable to be cast according to our schema will be highlighted in the changeset errors.
We can validate more than just whether a field is required or not. Let's take a look at some finer-grained validations.
What if we had a requirement that all biographies in our system must be at least two characters long? We can do this easily by adding another transformation to the pipeline in our changeset which validates the length of the bio field.
def changeset(user, attrs) do
  user
  |> cast(attrs, [:name, :email, :bio, :number_of_pets])
  |> validate_required([:name, :email, :bio, :number_of_pets])
  |> validate_length(:bio, min: 2)
end
Now, if we try to cast data containing a value of "A" for our user's bio, we should see the failed validation in the changeset's errors.
iex> recompile()

iex> changeset = User.changeset(%User{}, %{bio: "A"})

iex> changeset.errors[:bio]
{"should be at least %{count} character(s)",
 [count: 2, validation: :length, kind: :min, type: :string]}
If we also have a requirement for the maximum length that a bio can have, we can simply add another validation.
def changeset(user, attrs) do
  user
  |> cast(attrs, [:name, :email, :bio, :number_of_pets])
  |> validate_required([:name, :email, :bio, :number_of_pets])
  |> validate_length(:bio, min: 2)
  |> validate_length(:bio, max: 140)
end
Let's say we want to perform at least some rudimentary format validation on the email field. All we want to check for is the presence of the @. The Ecto.Changeset.validate_format/3 function is just what we need.
def changeset(user, attrs) do
  user
  |> cast(attrs, [:name, :email, :bio, :number_of_pets])
  |> validate_required([:name, :email, :bio, :number_of_pets])
  |> validate_length(:bio, min: 2)
  |> validate_length(:bio, max: 140)
  |> validate_format(:email, ~r/@/)
end
If we try to cast a user with an email of "example.com", we should see an error message like the following:
iex> recompile()

iex> changeset = User.changeset(%User{}, %{email: "example.com"})

iex> changeset.errors[:email]
{"has invalid format", [validation: :format]}
There are many more validations and transformations we can perform in a changeset. Please see the Ecto Changeset documentation for more information.
Data persistence
We've explored migrations and schemas, but we haven't yet persisted any of our schemas or changesets. We briefly looked at our repository module in lib/hello/repo.ex earlier, and now it's time to put it to use.
Ecto repositories are the interface into a storage system, be it a database like PostgreSQL or an external service like a RESTful API. The Repo module's purpose is to take care of the finer details of persistence and data querying for us. As the caller, we only care about fetching and persisting data. The Repo module takes care of the underlying database adapter communication, connection pooling, and error translation for database constraint violations.
Let's head back over to IEx with iex -S mix, and insert a couple of users into the database.
iex> alias Hello.{Repo, User}
[Hello.Repo, Hello.User]

iex> Repo.insert(%User{email: "user1@example.com"})
[debug] QUERY OK db=6.5ms queue=0.5ms idle=1358.3ms
INSERT INTO "users" ("email","inserted_at","updated_at") VALUES ($1,$2,$3) RETURNING "id" ["user1@example.com", ~N[2021-02-25 01:58:55], ~N[2021-02-25 01:58:55]]
{:ok,
 %Hello.User{
   __meta__: #Ecto.Schema.Metadata<:loaded, "users">,
   bio: nil,
   email: "user1@example.com",
   id: 1,
   inserted_at: ~N[2021-02-25 01:58:55],
   name: nil,
   number_of_pets: nil,
   updated_at: ~N[2021-02-25 01:58:55]
 }}

iex> Repo.insert(%User{email: "user2@example.com"})
[debug] QUERY OK db=1.3ms idle=1402.7ms
INSERT INTO "users" ("email","inserted_at","updated_at") VALUES ($1,$2,$3) RETURNING "id" ["user2@example.com", ~N[2021-02-25 02:03:28], ~N[2021-02-25 02:03:28]]
{:ok,
 %Hello.User{
   __meta__: #Ecto.Schema.Metadata<:loaded, "users">,
   bio: nil,
   email: "user2@example.com",
   id: 2,
   inserted_at: ~N[2021-02-25 02:03:28],
   name: nil,
   number_of_pets: nil,
   updated_at: ~N[2021-02-25 02:03:28]
 }}
We started by aliasing our User and Repo modules for easy access. Next, we called Repo.insert/2 with a User struct. Since we are in the dev environment, we can see the debug logs for the query our repository performed when inserting the underlying %User{} data. We received a two-element tuple back with {:ok, %User{}}, which lets us know the insertion was successful.
We could also insert a user by passing a changeset to Repo.insert/2. If the changeset is valid, the repository will use an optimized database query to insert the record, and return a two-element tuple back, as above. If the changeset is not valid, we receive a two-element tuple consisting of :error plus the invalid changeset.
With a couple of users inserted, let's fetch them back out of the repo.
iex> Repo.all(User)
[debug] QUERY OK source="users" db=5.8ms queue=1.4ms idle=1672.0ms
SELECT u0."id", u0."bio", u0."email", u0."name", u0."number_of_pets", u0."inserted_at", u0."updated_at" FROM "users" AS u0 []
[
  %Hello.User{
    __meta__: #Ecto.Schema.Metadata<:loaded, "users">,
    bio: nil,
    email: "user1@example.com",
    id: 1,
    inserted_at: ~N[2021-02-25 01:58:55],
    name: nil,
    number_of_pets: nil,
    updated_at: ~N[2021-02-25 01:58:55]
  },
  %Hello.User{
    __meta__: #Ecto.Schema.Metadata<:loaded, "users">,
    bio: nil,
    email: "user2@example.com",
    id: 2,
    inserted_at: ~N[2021-02-25 02:03:28],
    name: nil,
    number_of_pets: nil,
    updated_at: ~N[2021-02-25 02:03:28]
  }
]
That was easy! Repo.all/1 takes a data source, our User schema in this case, and translates that to an underlying SQL query against our database. After it fetches the data, the Repo then uses our Ecto schema to map the database values back into Elixir data structures according to our User schema. We're not just limited to basic querying – Ecto includes a full-fledged query DSL for advanced SQL generation. In addition to a natural Elixir DSL, Ecto's query engine gives us multiple great features, such as SQL injection protection and compile-time optimization of queries. Let's try it out.
iex> import Ecto.Query
Ecto.Query

iex> Repo.all(from u in User, select: u.email)
[debug] QUERY OK source="users" db=0.8ms queue=0.9ms idle=1634.0ms
SELECT u0."email" FROM "users" AS u0 []
["user1@example.com", "user2@example.com"]
First, we imported [Ecto.Query], which imports the from/2 macro of Ecto's Query DSL. Next, we built a query which selects all the email addresses in our users table. Let's try another example.
iex)> Repo.one(from u in User, where: ilike(u.email, "%1%"),
                               select: count(u.id))
[debug] QUERY OK source="users" db=1.6ms SELECT count(u0."id") FROM "users" AS u0 WHERE (u0."email" ILIKE '%1%') []
1
Now we're starting to get a taste of Ecto's rich querying capabilities. We used Repo.one/2 to fetch the count of all users with an email address containing 1, and received the expected count in return. This just scratches the surface of Ecto's query interface, and much more is supported such as sub-querying, interval queries, and advanced select statements. For example, let's build a query to fetch a map of all user id's to their email addresses.
iex> Repo.all(from u in User, select: %{u.id => u.email})
[debug] QUERY OK source="users" db=0.9ms
SELECT u0."id", u0."email" FROM "users" AS u0 []
[
  %{1 => "user1@example.com"},
  %{2 => "user2@example.com"}
]
That little query packed a big punch. It both fetched all user emails from the database and efficiently built a map of the results in one go. You should browse the Ecto.Query documentation to see the breadth of supported query features.
In addition to inserts, we can also perform updates and deletes with Repo.update/2 and Repo.delete/2 to update or delete a single schema. Ecto also supports bulk persistence with the Repo.insert_all/3, Repo.update_all/3, and Repo.delete_all/2 functions.
There is quite a bit more that Ecto can do and we've only barely scratched the surface. With a solid Ecto foundation in place, we're now ready to continue building our app and integrate the web-facing application with our backend persistence. Along the way, we'll expand our Ecto knowledge and learn how to properly isolate our web interface from the underlying details of our system. Please take a look at the Ecto documentation for the rest of the story.
In our contexts guide, we'll find out how to wrap up our Ecto access and business logic behind modules that group related functionality. We'll see how Phoenix helps us design maintainable applications, and we'll find out about other neat Ecto features along the way.
Using MySQL
Phoenix applications are configured to use PostgreSQL by default, but what if we want to use MySQL instead? In this guide, we'll walk through changing that default whether we are about to create a new application, or whether we have an existing one configured for PostgreSQL.
If we are about to create a new application, configuring our application to use MySQL is easy. We can simply pass the --database mysql flag to phx.new and everything will be configured correctly.
$ mix phx.new hello_phoenix --database mysql

This will set up all the correct dependencies and configuration for us automatically. Once we install those dependencies with mix deps.get, we'll be ready to begin working with Ecto in our application.
If we have an existing application, all we need to do is switch adapters and make some small configuration changes.
To switch adapters, we need to remove the Postgrex dependency and add a new one for Myxql instead.
Let's open up our mix.exs file and do that now.
defmodule HelloPhoenix.MixProject do
  use Mix.Project

  . . .
  # Specifies your project dependencies.
  #
  # Type `mix help deps` for examples and options.
  defp deps do
    [
      {:phoenix, "~> 1.4.0"},
      {:phoenix_ecto, "~> 4.4"},
      {:ecto_sql, "~> 3.6"},
      {:myxql, ">= 0.0.0"},
      ...
    ]
  end
end
Next, we need to configure our adapter to use the default MySQL credentials by updating config/dev.exs:
config :hello_phoenix, HelloPhoenix.Repo,
username: "root",
password: "",
database: "hello_phoenix_dev"
If we have an existing configuration block for our HelloPhoenix.Repo, we can simply change the values to match our new ones. You also need to configure the correct values in the config/test.exs and config/runtime.exs (formerly config/prod.secret.exs) files as well.
The last change is to open up lib/hello_phoenix/repo.ex and make sure to set the :adapter to Ecto.Adapters.MyXQL.
Now all we need to do is fetch our new dependency, and we'll be ready to go.
$ mix do deps.get, compile

With our new adapter installed and configured, we're ready to create our database.
$ mix ecto.create

The database for HelloPhoenix.Repo has been created.
We're also ready to run any migrations, or do anything else with Ecto that we may choose.
$ mix ecto.migrate
[info] == Running HelloPhoenix.Repo.Migrations.CreateUser.change/0 forward
[info] create table users
[info] == Migrated in 0.2s

Other options
While Phoenix uses the Ecto project to interact with the data access layer, there are many other data access options, some even built into the Erlang standard library. ETS – available in Ecto via etso – and DETS are key-value data stores built into OTP. OTP also provides a relational database called Mnesia with its own query language called QLC. Both Elixir and Erlang also have a number of libraries for working with a wide range of popular data stores.
The data world is your oyster, but we won't be covering these options in these guides.



  

  
    
    Contexts - Phoenix v1.6.9
    
    

    



  
  

    
Contexts
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Request life-cycle guide.

Requirement: This guide expects that you have gone through the Ecto guide.

So far, we've built pages, wired up controller actions through our routers, and learned how Ecto allows data to be validated and persisted. Now it's time to tie it all together by writing web-facing features that interact with our greater Elixir application.
When building a Phoenix project, we are first and foremost building an Elixir application. Phoenix's job is to provide a web interface into our Elixir application. Naturally, we compose our applications with modules and functions, but simply defining a module with a few functions isn't enough when designing an application. We need to consider the boundaries between modules and how to group functionality. In other words, it's vital to think about application design when writing code.
Thinking about design
Contexts are dedicated modules that expose and group related functionality. For example, anytime you call Elixir's standard library, be it Logger.info/1 or Stream.map/2, you are accessing different contexts. Internally, Elixir's logger is made of multiple modules, but we never interact with those modules directly. We call the Logger module the context, exactly because it exposes and groups all of the logging functionality.
By giving modules that expose and group related functionality the name contexts, we help developers identify these patterns and talk about them. At the end of the day, contexts are just modules, as are your controllers, views, etc.
In Phoenix, contexts often encapsulate data access and data validation. They often talk to a database or APIs. Overall, think of them as boundaries to decouple and isolate parts of your application. Let's use these ideas to build out our web application. Our goal is to build an ecommerce system where we can showcase products, allow users to add products to their cart, and complete their orders.
How to read this guide: Using the context generators is a great way for beginners and intermediate Elixir programmers alike to get up and running quickly while thoughtfully designing their applications. This guide focuses on those readers.

Adding a Catalog Context
An ecommerce platform has wide-reaching coupling across a codebase so it's important to think upfront about writing well-defined interfaces. With that in mind, our goal is to build a product catalog API that handles creating, updating, and deleting the products available in our system. We'll start off with the basic features of showcasing our products, and we will add shopping cart features later. We'll see how starting with a solid foundation with isolated boundaries allows us to grow our application naturally as we add functionality.
Phoenix includes the mix phx.gen.html, mix phx.gen.json, mix phx.gen.live, and mix phx.gen.context generators that apply the ideas of isolating functionality in our applications into contexts. These generators are a great way to hit the ground running while Phoenix nudges you in the right direction to grow your application. Let's put these tools to use for our new product catalog context.
In order to run the context generators, we need to come up with a module name that groups the related functionality that we're building. In the Ecto guide, we saw how we can use Changesets and Repos to validate and persist user schemas, but we didn't integrate this with our application at large. In fact, we didn't think about where a "user" in our application should live at all. Let's take a step back and think about the different parts of our system. We know that we'll have products to showcase on pages for sale, along with descriptions, pricing, etc. Along with selling products, we know we'll need to support carting, order checkout, and so on. While the products being purchased are related to the cart and checkout processes, showcasing a product and managing the exhibition of our products is distinctly different than tracking what a user has placed in their cart or how an order is placed. A Catalog context is a natural place for the management of our product details and the showcasing of those products we have for sale.
Naming things is hard. If you're stuck when trying to come up with a context name when the grouped functionality in your system isn't yet clear, you can simply use the plural form of the resource you're creating. For example, a Products context for managing products. As you grow your application and the parts of your system become clear, you can simply rename the context to a more refined one.

To jump-start our catalog context, we'll use mix phx.gen.html which creates a context module that wraps up Ecto access for creating, updating, and deleting products, along with web files like controllers and templates for the web interface into our context. Run the following command at your project root:
$ mix phx.gen.html Catalog Product products title:string \
description:string price:decimal views:integer

* creating lib/hello_web/controllers/product_controller.ex
* creating lib/hello_web/templates/product/edit.html.heex
* creating lib/hello_web/templates/product/form.html.heex
* creating lib/hello_web/templates/product/index.html.heex
* creating lib/hello_web/templates/product/new.html.heex
* creating lib/hello_web/templates/product/show.html.heex
* creating lib/hello_web/views/product_view.ex
* creating test/hello_web/controllers/product_controller_test.exs
* creating lib/hello/catalog/product.ex
* creating priv/repo/migrations/20210201185747_create_products.exs
* creating lib/hello/catalog.ex
* injecting lib/hello/catalog.ex
* creating test/hello/catalog_test.exs
* injecting test/hello/catalog_test.exs
* creating test/support/fixtures/catalog_fixtures.ex
* injecting test/support/fixtures/catalog_fixtures.ex

Add the resource to your browser scope in lib/hello_web/router.ex:

    resources "/products", ProductController


Remember to update your repository by running migrations:

    $ mix ecto.migrate

Note: we are starting with the basics for modeling an ecommerce system. In practice, modeling such systems yields more complex relationships such as product variants, optional pricing, multiple currencies, etc. We'll keep things simple in this guide, but the foundations will give you a solid starting point to building such a complete system.

Phoenix generated the web files as expected in lib/hello_web/. We can also see our context files were generated inside a lib/hello/catalog.ex file and our product schema in the directory of the same name. Note the difference between lib/hello and lib/hello_web. We have a Catalog module to serve as the public API for product catalog functionality, as well as an Catalog.Product struct, which is an Ecto schema for casting and validating product data. Phoenix also provided web and context tests for us, it also included test helpers for creating entities via the Hello.Catalog context, which we'll look at later. For now, let's follow the instructions and add the route according to the console instructions, in lib/hello_web/router.ex:
  scope "/", HelloWeb do
    pipe_through :browser

    get "/", PageController, :index
+   resources "/products", ProductController
  end
With the new route in place, Phoenix reminds us to update our repo by running mix ecto.migrate, but first we need to make a few tweaks to the generated migration in priv/repo/migrations/*_create_products.exs:
  def change do
    create table(:products) do
      add :title, :string
      add :description, :string
-     add :price, :decimal
+     add :price, :decimal, precision: 15, scale: 6, null: false
-     add :views, :integer
+     add :views, :integer, default: 0, null: false

      timestamps()
    end
We modified our price column to a specific precision of 15, scale of 6, along with a not-null constraint. This ensures we store currency with proper precision for any mathematical operations we may perform. Next, we added a default value and not-null constraint to our views count. With our changes in place, we're ready to migrate up our database. Let's do that now:
$ mix ecto.migrate
14:09:02.260 [info]  == Running 20210201185747 Hello.Repo.Migrations.CreateProducts.change/0 forward

14:09:02.262 [info]  create table products

14:09:02.273 [info]  == Migrated 20210201185747 in 0.0s

Before we jump into the generated code, let's start the server with mix phx.server and visit http://localhost:4000/products. Let's follow the "New Product" link and click the "Save" button without providing any input. We should be greeted with the following output:
Oops, something went wrong! Please check the errors below.
When we submit the form, we can see all the validation errors inline with the inputs. Nice! Out of the box, the context generator included the schema fields in our form template and we can see our default validations for required inputs are in effect. Let's enter some example product data and resubmit the form:
Product created successfully.

Title: Metaprogramming Elixir
Description: Write Less Code, Get More Done (and Have Fun!)
Price: 15.000000
Views: 0
If we follow the "Back" link, we get a list of all products, which should contain the one we just created. Likewise, we can update this record or delete it. Now that we've seen how it works in the browser, it's time to take a look at the generated code.
Starting With Generators
That little mix phx.gen.html command packed a surprising punch. We got a lot of functionality out-of-the-box for creating, updating, and deleting products in our catalog. This is far from a full-featured app, but remember, generators are first and foremost learning tools and a starting point for you to begin building real features. Code generation can't solve all your problems, but it will teach you the ins and outs of Phoenix and nudge you towards the proper mindset when designing your application.
Let's first check out the ProductController that was generated in lib/hello_web/controllers/product_controller.ex:
defmodule HelloWeb.ProductController do
  use HelloWeb, :controller

  alias Hello.Catalog
  alias Hello.Catalog.Product

  def index(conn, _params) do
    products = Catalog.list_products()
    render(conn, "index.html", products: products)
  end

  def new(conn, _params) do
    changeset = Catalog.change_product(%Product{})
    render(conn, "new.html", changeset: changeset)
  end

  def create(conn, %{"product" => product_params}) do
    case Catalog.create_product(product_params) do
      {:ok, product} ->
        conn
        |> put_flash(:info, "Product created successfully.")
        |> redirect(to: Routes.product_path(conn, :show, product))

      {:error, %Ecto.Changeset{} = changeset} ->
        render(conn, "new.html", changeset: changeset)
    end
  end

  def show(conn, %{"id" => id}) do
    product = Catalog.get_product!(id)
    render(conn, "show.html", product: product)
  end
  ...
end
We've seen how controllers work in our controller guide, so the code probably isn't too surprising. What is worth noticing is how our controller calls into the Catalog context. We can see that the index action fetches a list of products with Catalog.list_products/0, and how products are persisted in the create action with Catalog.create_product/1. We haven't yet looked at the catalog context, so we don't yet know how product fetching and creation is happening under the hood – but that's the point. Our Phoenix controller is the web interface into our greater application. It shouldn't be concerned with the details of how products are fetched from the database or persisted into storage. We only care about telling our application to perform some work for us. This is great because our business logic and storage details are decoupled from the web layer of our application. If we move to a full-text storage engine later for fetching products instead of a SQL query, our controller doesn't need to be changed. Likewise, we can reuse our context code from any other interface in our application, be it a channel, mix task, or long-running process importing CSV data.
In the case of our create action, when we successfully create a product, we use Phoenix.Controller.put_flash/3 to show a success message, and then we redirect to the router's product_path show page. Conversely, if Catalog.create_product/1 fails, we render our "new.html" template and pass along the Ecto changeset for the template to lift error messages from.
Next, let's dig deeper and check out our Catalog context in lib/hello/catalog.ex:
defmodule Hello.Catalog do
  @moduledoc """
  The Catalog context.
  """

  import Ecto.Query, warn: false
  alias Hello.Repo

  alias Hello.Catalog.Product

  @doc """
  Returns the list of products.

  ## Examples

      iex> list_products()
      [%Product{}, ...]

  """
  def list_products do
    Repo.all(Product)
  end
  ...
end
This module will be the public API for all product catalog functionality in our system. For example, in addition to product detail management, we may also handle product category classification and product variants for things like optional sizing, trims, etc. If we look at the list_products/0 function, we can see the private details of product fetching. And it's super simple. We have a call to Repo.all(Product). We saw how Ecto repo queries worked in the Ecto guide, so this call should look familiar. Our list_products function is a generalized function name specifying the intent of our code – namely to list products. The details of that intent where we use our Repo to fetch the products from our PostgreSQL database is hidden from our callers. This is a common theme we'll see re-iterated as we use the Phoenix generators. Phoenix will push us to think about where we have different responsibilities in our application, and then to wrap up those different areas behind well-named modules and functions that make the intent of our code clear, while encapsulating the details.
Now we know how data is fetched, but how are products persisted? Let's take a look at the Catalog.create_product/1 function:
  @doc """
  Creates a product.

  ## Examples

      iex> create_product(%{field: value})
      {:ok, %Product{}}

      iex> create_product(%{field: bad_value})
      {:error, %Ecto.Changeset{}}

  """
  def create_product(attrs \\ %{}) do
    %Product{}
    |> Product.changeset(attrs)
    |> Repo.insert()
  end
There's more documentation than code here, but a couple of things are important to highlight. First, we can see again that our Ecto Repo is used under the hood for database access. You probably also noticed the call to Product.changeset/2. We talked about changesets before, and now we see them in action in our context.
If we open up the Product schema in lib/hello/catalog/product.ex, it will look immediately familiar:
defmodule Hello.Catalog.Product do
  use Ecto.Schema
  import Ecto.Changeset

  schema "products" do
    field :description, :string
    field :price, :decimal
    field :title, :string
    field :views, :integer

    timestamps()
  end

  @doc false
  def changeset(product, attrs) do
    product
    |> cast(attrs, [:title, :description, :price, :views])
    |> validate_required([:title, :description, :price, :views])
  end
end
This is just what we saw before when we ran mix phx.gen.schema, except here we see a @doc false above our changeset/2 function. This tells us that while this function is publicly callable, it's not part of the public context API. Callers that build changesets do so via the context API. For example, Catalog.create_product/1 calls into our Product.changeset/2 to build the changeset from user input. Callers, such as our controller actions, do not access Product.changeset/2 directly. All interaction with our product changesets is done through the public Catalog context.
Adding Catalog functions
As we've seen, your context modules are dedicated modules that expose and group related functionality. Phoenix generates generic functions, such as list_products and update_product, but they only serve as a basis for you to grow your business logic and application from. Let's add one of the basic features of our catalog by tracking product page view count.
For any ecommerce system, the ability to track how many times a product page has been viewed is essential for marketing, suggestions, ranking, etc. While we could try to use the existing Catalog.update_product function, along the lines of Catalog.update_product(product, %{views: product.views + 1}), this would not only be prone to race conditions, but it would also require the caller to know too much about our Catalog system. To see why the race condition exists, let's walk through the possible execution of events:
Intuitively, you would assume the following events:
	User 1 loads the product page with count of 13
	User 1 saves the product page with count of 14
	User 2 loads the product page with count of 14
	User 2 saves the product page with count of 15

While in practice this would happen:
	User 1 loads the product page with count of 13
	User 2 loads the product page with count of 13
	User 1 saves the product page with count of 14
	User 2 saves the product page with count of 14

The race conditions would make this an unreliable way to update the existing table since multiple callers may be updating out of date view values. There's a better way.
Let's think of a function that describes what we want to accomplish. Here's how we would like to use it:
product = Catalog.inc_page_views(product)
That looks great. Our callers will have no confusion over what this function does, and we can wrap up the increment in an atomic operation to prevent race conditions.
Open up your catalog context (lib/hello/catalog.ex), and add this new function:
  def inc_page_views(%Product{} = product) do
    {1, [%Product{views: views}]} =
      from(p in Product, where: p.id == ^product.id, select: [:views])
      |> Repo.update_all(inc: [views: 1])

    put_in(product.views, views)
  end
We built a query for fetching the current product given its ID which we pass to Repo.update_all. Ecto's Repo.update_all allows us to perform batch updates against the database, and is perfect for atomically updating values, such as incrementing our views count. The result of the repo operation returns the number of updated records, along with the selected schema values specified by the select option. When we receive the new product views, we use put_in(product.views, views) to place the new view count within the product struct.
With our context function in place, let's make use of it in our product controller. Update your show action in lib/hello_web/controllers/product_controller.ex to call our new function:
  def show(conn, %{"id" => id}) do
    product =
      id
      |> Catalog.get_product!()
      |> Catalog.inc_page_views()

    render(conn, "show.html", product: product)
  end
We modified our show action to pipe our fetched product into Catalog.inc_page_views/1, which will return the updated product. Then we rendered our template just as before. Let's try it out. Refresh one of your product pages a few times and watch the view count increase.
We can also see our atomic update in action in the ecto debug logs:
[debug] QUERY OK source="products" db=0.5ms idle=834.5ms
UPDATE "products" AS p0 SET "views" = p0."views" + $1 WHERE (p0."id" = $2) RETURNING p0."views" [1, 1]
Good work!
As we've seen, designing with contexts gives you a solid foundation to grow your application from. Using discrete, well-defined APIs that expose the intent of your system allows you to write more maintainable applications with reusable code. Now that we know how to start extending our context API, lets explore handling relationships within a context.
In-context Relationships
Our basic catalog features are nice, but let's take it up a notch by categorizing products. Many ecommerce solutions allow products to be categorized in different ways, such as a product being marked for fashion, power tools, and so on. Starting with a one-to-one relationship between product and categories will cause major code changes later if we need to start supporting multiple categories. Let's set up a category association that will allow us to start off tracking a single category per product, but easily support more later as we grow our features.
For now, categories will contain only textual information. Our first order of business is to decide where categories live in the application. We have our Catalog context, which manages the exhibition of our products. Product categorization is a natural fit here. Phoenix is also smart enough to generate code inside an existing context, which makes adding new resources to a context a breeze. Run the following command at your project root:
Sometimes it may be tricky to determine if two resources belong to the same context or not. In those cases, prefer distinct contexts per resource and refactor later if necessary. Otherwise you can easily end-up with large contexts of loosely related entities. Also keep in mind that the fact two resources are related does not necessarily mean they belong to the same context, otherwise you would quickly end-up with one large context, as the majority of resources in an application are connected to each other. To sum it up: if you are unsure, you should prefer separate modules (contexts).

$ mix phx.gen.context Catalog Category categories \
title:string:unique

You are generating into an existing context.
...
Would you like to proceed? [Yn] y
* creating lib/hello/catalog/category.ex
* creating priv/repo/migrations/20210203192325_create_categories.exs
* injecting lib/hello/catalog.ex
* injecting test/hello/catalog_test.exs
* injecting test/support/fixtures/catalog_fixtures.ex

Remember to update your repository by running migrations:

    $ mix ecto.migrate

This time around, we used mix phx.gen.context, which is just like mix phx.gen.html, except it doesn't generate the web files for us. Since we already have controllers and templates for managing products, we can integrate the new category features into our existing web form and product show page. We can see we now have a new Category schema alongside our product schema at lib/hello/catalog/category.ex, and Phoenix told us it was injecting new functions in our existing Catalog context for the category functionality. The injected functions will look very familiar to our product functions, with new functions like create_category, list_categories, and so on. Before we migrate up, we need to do a second bit of code generation. Our category schema is great for representing an individual category in the system, but we need to support a many-to-many relationship between products and categories. Fortunately, ecto allows us to do this simply with a join table, so let's generate that now with the ecto.gen.migration command:
$ mix ecto.gen.migration create_product_categories

* creating priv/repo/migrations/20210203192958_create_product_categories.exs

Next, let's open up the new migration file and add the following code to the change function:

defmodule Hello.Repo.Migrations.CreateProductCategories do
  use Ecto.Migration

  def change do
    create table(:product_categories, primary_key: false) do
      add :product_id, references(:products, on_delete: :delete_all)
      add :category_id, references(:categories, on_delete: :delete_all)
    end

    create index(:product_categories, [:product_id])
    create index(:product_categories, [:category_id])
    create unique_index(:product_categories, [:product_id, :category_id])
  end
end
We created a product_categories table and used the primary_key: false option since our join table does not need a primary key. Next we defined our :product_id and :category_id foreign key fields, and passed on_delete: :delete_all to ensure the database prunes our join table records if a linked product or category is deleted. By using a database constraint, we enforce data integrity at the database level, rather than relying on ad-hoc and error-prone application logic.
Next, we created indexes for our foreign keys, and a unique index to ensure a product cannot have duplicate categories. With our migrations in place, we can migrate up.
$ mix ecto.migrate

18:20:36.489 [info]  == Running 20210222231834 Hello.Repo.Migrations.CreateCategories.change/0 forward

18:20:36.493 [info]  create table categories

18:20:36.508 [info]  create index categories_title_index

18:20:36.512 [info]  == Migrated 20210222231834 in 0.0s

18:20:36.547 [info]  == Running 20210222231930 Hello.Repo.Migrations.CreateProductCategories.change/0 forward

18:20:36.547 [info]  create table product_categories

18:20:36.557 [info]  create index product_categories_product_id_index

18:20:36.558 [info]  create index product_categories_category_id_index

18:20:36.560 [info]  create index product_categories_product_id_category_id_index

18:20:36.562 [info]  == Migrated 20210222231930 in 0.0s

Now that we have a Catalog.Product schema and a join table to associate products and categories, we're nearly ready to start wiring up our new features. Before we dive in, we first need real categories to select in our web UI. Let's quickly seed some new categories in the application. Add the following code to your seeds file in priv/repo/seeds.exs:
for title <- ["Home Improvement", "Power Tools", "Gardening", "Books"] do
  {:ok, _} = Hello.Catalog.create_category(%{title: title})
end
We simply enumerate over a list of category titles and use the generated create_category/1 function of our catalog context to persist the new records. We can run the seeds with mix run:
$ mix run priv/repo/seeds.exs

[debug] QUERY OK db=3.1ms decode=1.1ms queue=0.7ms idle=2.2ms
INSERT INTO "categories" ("title","inserted_at","updated_at") VALUES ($1,$2,$3) RETURNING "id" ["Home Improvement", ~N[2021-02-03 19:39:53], ~N[2021-02-03 19:39:53]]
[debug] QUERY OK db=1.2ms queue=1.3ms idle=12.3ms
INSERT INTO "categories" ("title","inserted_at","updated_at") VALUES ($1,$2,$3) RETURNING "id" ["Power Tools", ~N[2021-02-03 19:39:53], ~N[2021-02-03 19:39:53]]
[debug] QUERY OK db=1.1ms queue=1.1ms idle=15.1ms
INSERT INTO "categories" ("title","inserted_at","updated_at") VALUES ($1,$2,$3) RETURNING "id" ["Gardening", ~N[2021-02-03 19:39:53], ~N[2021-02-03 19:39:53]]
[debug] QUERY OK db=2.4ms queue=1.0ms idle=17.6ms
INSERT INTO "categories" ("title","inserted_at","updated_at") VALUES ($1,$2,$3) RETURNING "id" ["Books", ~N[2021-02-03 19:39:53], ~N[2021-02-03 19:39:53]]

Perfect. Before we integrate categories in the web layer, we need to let our context know how to associate products and categories. First, open up lib/hello/catalog/product.ex and add the following association:
+ alias Hello.Catalog.Category

  schema "products" do
    field :description, :string
    field :price, :decimal
    field :title, :string
    field :views, :integer

+   many_to_many :categories, Category, join_through: "product_categories", on_replace: :delete

    timestamps()
  end

We used Ecto.Schema's many_to_many macro to let Ecto know how to associate our product to multiple categories thru the "product_categories" join table. We also used the on_replace: :delete option to declare that any existing join records should be deleted when we are changing our categories.
With our schema associations set up, we can implement the selection of categories in our product form. To do so, we need to translate the user input of catalog IDs from the front-end to our many-to-many association. Fortunately Ecto makes this a breeze now that our schema is set up. Open up your catalog context and make the following changes:
- def get_product!(id), do: Repo.get!(Product, id)
+ def get_product!(id) do
+   Product |> Repo.get(id) |> Repo.preload(:categories)
+ end

  def create_product(attrs \\ %{}) do
    %Product{}
-   |> Product.changeset(attrs)
+   |> change_product(attrs)
    |> Repo.insert()
  end

  def update_product(%Product{} = product, attrs) do
    product
-   |> Product.changeset(attrs)
+   |> change_product(attrs)
    |> Repo.update()
  end

  def change_product(%Product{} = product, attrs \\ %{}) do
-   Product.changeset(product, attrs)
+   categories = list_categories_by_id(attrs["category_ids"])

+   product
+   |> Repo.preload(:categories)
+   |> Product.changeset(attrs)
+   |> Ecto.Changeset.put_assoc(:categories, categories)
  end

+ def list_categories_by_id(nil), do: []
+ def list_categories_by_id(category_ids) do
+   Repo.all(from c in Category, where: c.id in ^category_ids)
+ end
First, we added Repo.preload to preload our categories when we fetch a product. This will allow us to reference product.categories in our controllers, templates, and anywhere else we want to make use of category information. Next, we modified our create_product and update_product functions to call into our existing change_product function to produce a changeset. Within change_product we added a lookup to find all categories if the "category_ids" attribute is present. Then we preloaded categories and called Ecto.Changeset.put_assoc to place the fetched categories into the changeset. Finally, we implemented the list_categories_by_id/1 function to query the categories matching the category IDs, or return an empty list if no "category_ids" attribute is present. Now our create_product and update_product functions receive a changeset with the category associations all ready to go once we attempt an insert or update against our repo.
Next, let's expose our new feature to the web by adding the category input to our product form. To keep our form template tidy, let's write a new function to wrap up the details of rendering a category select input for our product. Open up your ProductView in lib/hello_web/views/product_view.ex and key this in:
defmodule HelloWeb.ProductView do
  use HelloWeb, :view

  def category_select(f, changeset) do
    existing_ids = changeset |> Ecto.Changeset.get_change(:categories, []) |> Enum.map(& &1.data.id)

    category_opts =
      for cat <- Hello.Catalog.list_categories(),
          do: [key: cat.title, value: cat.id, selected: cat.id in existing_ids]

    multiple_select(f, :category_ids, category_opts)
  end
end
We added a new category_select/2 function which uses Phoenix.HTML's multiple_select/3 to generate a multiple select tag. We calculated the existing category IDs from our changeset, then used those values when we generate the select options for the input tag. We did this by enumerating over all of our categories and returning the appropriate key, value, and selected values. We marked an option as selected if the category ID was found in those category IDs in our changeset.
With our category_select function in place, we can open up lib/hello_web/templates/product/form.html.heex and add:
  ...
  <%= label f, :views %>
  <%= number_input f, :views %>
  <%= error_tag f, :views %>

+ <%= category_select f, @changeset %>

  <div>
    <%= submit "Save" %>
  </div>
We added a category_select above our save button. Now let's try it out. Next, let's show the product's categories in the product show template. Add the following code to lib/hello_web/templates/product/show.html.heex:
  ...
+ <li>
+   <strong>Categories:</strong>
+   <%= for cat <- @product.categories do %>
+     <%= cat.title %>
+     <br>
+   <% end %>
+ </li>
</ul>
Now if we start the server with mix phx.server and visit http://localhost:4000/products/new, we'll see the new category multiple select input. Enter some valid product details, select a category or two, and click save.
Title: Elixir Flashcards
Description: Flash card set for the Elixir programming language
Price: 5.000000
Views: 0
Categories:
Education
Books
It's not much to look at yet, but it works! We added relationships within our context complete with data integrity enforced by the database. Not bad. Let's keep building!
Cross-context dependencies
Now that we have the beginnings of our product catalog features, let's begin to work on the other main features of our application – carting products from the catalog. In order to properly track products that have been added to a user's cart, we'll need a new place to persist this information, along with point-in-time product information like the price at time of carting. This is necessary so we can detect product price changes in the future. We know what we need to build, but now we need to decide where the cart functionality lives in our application.
If we take a step back and think about the isolation of our application, the exhibition of products in our catalog distinctly differs from the responsibilities of managing a user's cart. A product catalog shouldn't care about the rules of a our shopping cart system, and vice-versa. There's a clear need here for a separate context to handle the new cart responsibilities. Let's call it ShoppingCart.
Let's create a ShoppingCart context to handle basic cart duties. Before we write code, let's imagine we have the following feature requirements:
	Add products to a user's cart from the product show page
	Store point-in-time product price information at time of carting
	Store and update quantities in cart
	Calculate and display sum of cart prices

From the description, it's clear we need a Cart resource for storing the user's cart, along with a CartItem to track products in the cart. With our plan set, let's get to work. Run the following command to generate our new context:
$ mix phx.gen.context ShoppingCart Cart carts user_uuid:uuid:unique

* creating lib/hello/shopping_cart/cart.ex
* creating priv/repo/migrations/20210205203128_create_carts.exs
* creating lib/hello/shopping_cart.ex
* injecting lib/hello/shopping_cart.ex
* creating test/hello/shopping_cart_test.exs
* injecting test/hello/shopping_cart_test.exs
* creating test/support/fixtures/shopping_cart_fixtures.ex
* injecting test/support/fixtures/shopping_cart_fixtures.ex

Some of the generated database columns are unique. Please provide
unique implementations for the following fixture function(s) in
test/support/fixtures/shopping_cart_fixtures.ex:

    def unique_cart_user_uuid do
      raise "implement the logic to generate a unique cart user_uuid"
    end

Remember to update your repository by running migrations:

    $ mix ecto.migrate

We generated our new context ShoppingCart, with a new ShoppingCart.Cart schema to tie a user to their cart which holds cart items. We don't have real users yet, so for now our cart will be tracked by an anonymous user UUID that we'll add to our plug session in a moment. With our cart in place, let's generate our cart items:
$ mix phx.gen.context ShoppingCart CartItem cart_items \
cart_id:references:carts product_id:references:products \
price_when_carted:decimal quantity:integer

You are generating into an existing context.
...
Would you like to proceed? [Yn] y
* creating lib/hello/shopping_cart/cart_item.ex
* creating priv/repo/migrations/20210205213410_create_cart_items.exs
* injecting lib/hello/shopping_cart.ex
* injecting test/hello/shopping_cart_test.exs
* injecting test/support/fixtures/shopping_cart_fixtures.ex

Remember to update your repository by running migrations:

    $ mix ecto.migrate


We generated a new resource inside our ShoppingCart named CartItem. This schema and table will hold references to a cart and product, along with the price at the time we added the item to our cart, and the quantity the user wishes to purchase. Let's touch up the generated migration file in priv/repo/migrations/*_create_cart_items.ex:
    create table(:cart_items) do
-     add :price_when_carted, :decimal
+     add :price_when_carted, :decimal, precision: 15, scale: 6, null: false
      add :quantity, :integer
-     add :cart_id, references(:carts, on_delete: :nothing)
+     add :cart_id, references(:carts, on_delete: :delete_all)
-     add :product_id, references(:products, on_delete: :nothing)
+     add :product_id, references(:products, on_delete: :delete_all)

      timestamps()
    end

    create index(:cart_items, [:cart_id])
    create index(:cart_items, [:product_id])
+   create unique_index(:cart_items, [:cart_id, :product_id])
We used the :delete_all strategy again to enforce data integrity. This way, when a cart or product is deleted from the application, we don't have to rely on application code in our ShoppingCart or Catalog contexts to worry about cleaning up the records. This keeps our application code decoupled and the data integrity enforcement where it belongs – in the database. We also added a unique constraint to ensure a duplicate product is not allowed to be added to a cart. With our database tables in place, we can now migrate up:
$ mix ecto.migrate

16:59:51.941 [info]  == Running 20210205203342 Hello.Repo.Migrations.CreateCarts.change/0 forward

16:59:51.945 [info]  create table carts

16:59:51.949 [info]  create index carts_user_uuid_index

16:59:51.952 [info]  == Migrated 20210205203342 in 0.0s

16:59:51.988 [info]  == Running 20210205213410 Hello.Repo.Migrations.CreateCartItems.change/0 forward

16:59:51.988 [info]  create table cart_items

16:59:51.998 [info]  create index cart_items_cart_id_index

16:59:52.000 [info]  create index cart_items_product_id_index

16:59:52.001 [info]  create index cart_items_cart_id_product_id_index

16:59:52.002 [info]  == Migrated 20210205213410 in 0.0s

Our database is ready to go with new carts and cart_items tables, but now we need to map that back into application code. You may be wondering how we can mix database foreign keys across different tables and how that relates to the context pattern of isolated, grouped functionality. Let's jump in and discuss the approaches and their tradeoffs.
Cross-context data
So far, we've done a great job isolating the two main contexts of our application from each other, but now we have a necessary dependency to handle.
Our Catalog.Product resource serves to keep the responsibilities of representing a product inside the catalog, but ultimately for an item to exist in the cart, a product from the catalog must be present. Given this, our ShoppingCart context will have a data dependency on the Catalog context. With that in mind, we have two options. One is to expose APIs on the Catalog context that allows us to efficiently fetch product data for use in the ShoppingCart system, which we would manually stitch together. Or we can use database joins to fetch the dependent data. Both are valid options given your tradeoffs and application size, but joining data from the database when you have a hard data dependency is just fine for a large class of applications and is the approach we will take here.
Now that we know where our data dependencies exist, let's add our schema associations so we can tie shopping cart items to products. First, let's make a quick change to our cart schema in lib/hello/shopping_cart/cart.ex to associate a cart to its items:
  schema "carts" do
    field :user_uuid, Ecto.UUID

+   has_many :items, Hello.ShoppingCart.CartItem

    timestamps()
  end
Now that our cart is associated to the items we place in it, let's set up the cart item associations inside lib/hello/shopping_cart/cart_item.ex:
  schema "cart_items" do
-   field :cart_id, :id
-   field :product_id, :id
    field :price_when_carted, :decimal
    field :quantity, :integer

+   belongs_to :cart, Hello.ShoppingCart.Cart
+   belongs_to :product, Hello.Catalog.Product

    timestamps()
  end

  @doc false
  def changeset(cart_item, attrs) do
    cart_item
    |> cast(attrs, [:price_when_carted, :quantity])
    |> validate_required([:price_when_carted, :quantity])
+   |> validate_number(:quantity, greater_than_or_equal_to: 0, less_than: 100)
  end
First, we replaced the cart_id field with a standard belongs_to pointing at our ShoppingCart.Cart schema. Next, we replaced our product_id field by adding our first cross-context data dependency with a belongs_to for the Catalog.Product schema. Here, we intentionally coupled the data boundaries because it provides exactly what we need. An isolated context API with the bare minimum knowledge necessary to reference a product in our system. Next, we added a new validation to our changeset. With validate_number/3, we ensure any quantity provided by user input is between 0 and 100.
With our schemas in place, we can start integrating the new data structures and ShoppingCart context APIs into our web-facing features.
Adding Shopping Cart functions
As we mentioned before, the context generators are only a starting point for our application. We can and should write well-named, purpose built functions to accomplish the goals of our context. We have a few new features to implement. First, we need to ensure every user of our application is granted a cart if one does not yet exist. From there, we can then allow users to add items to their cart, update item quantities, and calculate cart totals. Let's get started!
We won't focus on a real user authentication system at this point, but by the time we're done, you'll be able to naturally integrate one with what we've written here. To simulate a current user session, open up your lib/hello_web/router.ex and key this in:
  pipeline :browser do
    plug :accepts, ["html"]
    plug :fetch_session
    plug :fetch_live_flash
    plug :put_root_layout, {HelloWeb.LayoutView, :root}
    plug :protect_from_forgery
    plug :put_secure_browser_headers
+   plug :fetch_current_user
+   plug :fetch_current_cart
  end

+ defp fetch_current_user(conn, _) do
+   if user_uuid = get_session(conn, :current_uuid) do
+     assign(conn, :current_uuid, user_uuid)
+   else
+     new_uuid = Ecto.UUID.generate()
+
+     conn
+     |> assign(:current_uuid, new_uuid)
+     |> put_session(:current_uuid, new_uuid)
+   end
+ end

+ alias Hello.ShoppingCart
+
+ def fetch_current_cart(conn, _opts) do
+   if cart = ShoppingCart.get_cart_by_user_uuid(conn.assigns.current_uuid) do
+     assign(conn, :cart, cart)
+   else
+     {:ok, new_cart} = ShoppingCart.create_cart(conn.assigns.current_uuid)
+     assign(conn, :cart, new_cart)
+   end
+ end
We added a new :fetch_current_user and :fetch_current_cart plug to our browser pipeline to run on all browser-based requests. Next, we implemented the fetch_current_user plug which simply checks the session for a user UUID that was previously added. If we find one, we add a current_uuid assign to the connection and we're done. In the case we haven't yet identified this visitor, we generate a unique UUID with Ecto.UUID.generate(), then we place that value in the current_uuid assign, along with a new session value to identify this visitor on future requests. A random, unique ID isn't much to represent a user, but it's enough for us to track and identify a visitor across requests, which is all we need for now. Later as our application becomes more complete, you'll be ready to migrate to a complete user authentication solution. With a guaranteed current user, we then implemented the fetch_current_cart plug which either finds a cart for the user UUID or creates a cart for the current user and assigns the result in the connection assigns. We'll need to implement our ShoppingCart.get_cart_by_user_uuid/1 and modify the create cart function to accept a UUID, but let's add our routes first.
We'll need to implement a cart controller for handling cart operations like viewing a cart, updating quantities, and initiating the checkout process, as well as a cart items controller for adding and removing individual items to and from the cart. Add the following routes to your router in lib/hello_web/router.ex:
  scope "/", HelloWeb do
    pipe_through :browser

    get "/", PageController, :index
    resources "/products", ProductController

+   resources "/cart_items", CartItemController, only: [:create, :delete]

+   get "/cart", CartController, :show
+   put "/cart", CartController, :update
  end
We added a resources declaration for a CartItemController, which will wire up the routes for a create and delete action for adding and remove individual cart items. Next, we added two new routes pointing at a CartController. The first route, a GET request, will map to our show action, to show the cart contents. The second route, a PUT request, will handle the submission of a form for updating our cart quantities.
With our routes in place, let's add the ability to add an item to our cart from the product show page. Create a new file at lib/hello_web/controllers/cart_item_controller.ex and key this in:
defmodule HelloWeb.CartItemController do
  use HelloWeb, :controller

  alias Hello.{ShoppingCart, Catalog}

  def create(conn, %{"product_id" => product_id}) do
    product = Catalog.get_product!(product_id)

    case ShoppingCart.add_item_to_cart(conn.assigns.cart, product) do
      {:ok, _item} ->
        conn
        |> put_flash(:info, "Item added to your cart")
        |> redirect(to: Routes.cart_path(conn, :show))

      {:error, _changeset} ->
        conn
        |> put_flash(:error, "There was an error adding the item to your cart")
        |> redirect(to: Routes.cart_path(conn, :show))
    end
  end

  def delete(conn, %{"id" => product_id}) do
    {:ok, _cart} = ShoppingCart.remove_item_from_cart(conn.assigns.cart, product_id)
    redirect(conn, to: Routes.cart_path(conn, :show))
  end
end
We defined a new CartItemController with the create and delete actions that we declared in our router. For create, we first lookup the product in the catalog with Catalog.get_product!/1, then we call a ShoppingCart.add_item_to_cart/2 function which we'll implement in a moment. If successful, we show a flash successful message and redirect to the cart show page; else, we show a flash error message and redirect to the cart show page. For delete, we'll call a remove_item_from_cart function which we'll implement on our ShoppingCart context  and then redirect back to the cart show page. We haven't implemented these two shopping cart functions yet, but notice how their names scream their intent: add_item_to_cart and remove_item_from_cart make it obvious what we are accomplishing here. It also allows us to spec out our web layer and context APIs without thinking about all the implementation details at once.
Let's implement the new interface for the ShoppingCart context API in lib/hello/shopping_cart.ex:
  alias Hello.Catalog
  alias Hello.ShoppingCart.{Cart, CartItem}

  def get_cart_by_user_uuid(user_uuid) do
    Repo.one(
      from(c in Cart,
        where: c.user_uuid == ^user_uuid,
        left_join: i in assoc(c, :items),
        left_join: p in assoc(i, :product),
        order_by: [asc: i.inserted_at],
        preload: [items: {i, product: p}]
      )
    )
  end

- def create_cart(attrs \\ %{}) do
-   %Cart{}
-   |> Cart.changeset(attrs)
+ def create_cart(user_uuid) do
+   %Cart{user_uuid: user_uuid}
+   |> Cart.changeset(%{})
    |> Repo.insert()
+   |> case do
+     {:ok, cart} -> {:ok, reload_cart(cart)}
+     {:error, changeset} -> {:error, changeset}
+   end
  end

  defp reload_cart(%Cart{} = cart), do: get_cart_by_user_uuid(cart.user_uuid)

  def add_item_to_cart(%Cart{} = cart, %Catalog.Product{} = product) do
    %CartItem{quantity: 1, price_when_carted: product.price}
    |> CartItem.changeset(%{})
    |> Ecto.Changeset.put_assoc(:cart, cart)
    |> Ecto.Changeset.put_assoc(:product, product)
    |> Repo.insert(
      on_conflict: [inc: [quantity: 1]],
      conflict_target: [:cart_id, :product_id]
    )
  end

  def remove_item_from_cart(%Cart{} = cart, product_id) do
    {1, _} =
      Repo.delete_all(
        from(i in CartItem,
          where: i.cart_id == ^cart.id,
          where: i.product_id == ^product_id
        )
      )

    {:ok, reload_cart(cart)}
  end
We started by implementing  get_cart_by_user_uuid/1 which fetches our cart and joins the cart items, and their products so that we have the full cart populated with all preloaded data. Next, we modified our create_cart function to accept a user UUID instead of attributes, which we used to populate the user_uuid field. If the insert is successful, we reload the cart contents by calling a private reload_cart/1 function, which simply calls get_cart_by_user_uuid/1 to refetch data. Next, we wrote our new add_item_to_cart/2 function which accepts a cart struct and a product struct from the catalog. We used an upsert operation against our repo to either insert a new cart item into the database, or increase the quantity by one if it already exists in the cart. This is accomplished via the on_conflict and conflict_target options, which tells our repo how to handle an insert conflict. Next, we implemented remove_item_from_cart/2 where we simply issue a Repo.delete_all call with a query to delete the cart item in our cart that matches the product ID. Finally, we reload the cart contents by calling reload_cart/1.
With our new cart functions in place, we can now expose the "Add to cart" button on the product catalog show page. Open up your template in lib/hello_web/templates/product/show.html.heex and make the following changes:
<h1>Show Product</h1>

+<%= link "Add to cart",
+  to: Routes.cart_item_path(@conn, :create, product_id: @product.id),
+  method: :post %>
...
The link functions from Phoenix.HTML accepts a :method option to issue an HTTP verb when clicked, instead of the default GET request. With this link in place, the "Add to cart" link will issue a POST request, which will be matched by the route we defined in router which dispatches to the CartItemController.create/2 function.
Let's try it out. Start your server with mix phx.server and visit a product page. If we try clicking the add to cart link, we'll be greeted by an error page with the following logs in the console:
[info] POST /cart_items
[debug] Processing with HelloWeb.CartItemController.create/2
  Parameters: %{"_method" => "post", "product_id" => "1", ...}
  Pipelines: [:browser]
INSERT INTO "cart_items" ...
[info] Sent 302 in 24ms
[info] GET /cart
[debug] Processing with HelloWeb.CartController.show/2
  Parameters: %{}
  Pipelines: [:browser]
[debug] QUERY OK source="carts" db=1.9ms idle=1798.5ms

[error] #PID<0.856.0> running HelloWeb.Endpoint (connection #PID<0.841.0>, stream id 5) terminated
Server: localhost:4000 (http)
Request: GET /cart
** (exit) an exception was raised:
    ** (UndefinedFunctionError) function HelloWeb.CartController.init/1 is undefined
       (module HelloWeb.CartController is not available)
       ...
It's working! Kind of. If we follow the logs, we see our POST to the /cart_items path. Next, we can see our ShoppingCart.add_item_to_cart function successfully inserted a row into the cart_items table, and then we issued a redirect to /cart. Before our error, we also see a query to the carts table, which means we're fetching the current user's cart. So far so good. We know our CartItem controller and new ShoppingCart context functions are doing their jobs, but we've hit our next unimplemented feature when the router attempts to dispatch to a non-existent cart controller. Let's create the cart controller, view, and template to display and manage user carts.
Create a new file at lib/hello_web/controllers/cart_controller.ex and key this in:
defmodule HelloWeb.CartController do
  use HelloWeb, :controller

  alias Hello.ShoppingCart

  def show(conn, _params) do
    render(conn, "show.html", changeset: ShoppingCart.change_cart(conn.assigns.cart))
  end
end
We defined a new cart controller to handle the get "/cart" route. For showing a cart, we render a "show.html" template which we'll create in moment. We know we need to allow the cart items to be changed by quantity updates, so right away we know we'll need a cart changeset. Fortunately, the context generator included a ShoppingChart.change_cart/1 function, which we'll use. We pass it our cart struct which is already in the connection assigns thanks to the fetch_current_cart plug we defined in the router.
Next, we can implement the view and template. Create a new view file at lib/hello_web/views/cart_view.ex with the following content:
defmodule HelloWeb.CartView do
  use HelloWeb, :view

  alias Hello.ShoppingCart

  def currency_to_str(%Decimal{} = val), do: "$#{Decimal.round(val, 2)}"
end
We created a view to render our show.html template and aliased our ShoppingCart context so it will be in scope for our template. We'll need to display the cart prices like product item price, cart total, etc, so we defined a currency_to_str/1 which takes our decimal struct, rounds it properly for display, and prepends a USD dollar sign.
Next we can create the template at lib/hello_web/templates/cart/show.html.heex:
<h1>My Cart</h1>

<%= if @cart.items == [] do %>
  Your cart is empty
<% else %>
  <%= form_for @changeset, Routes.cart_path(@conn, :update), fn f -> %>
    <ul>
      <%= for item_form <- inputs_for(f, :items), item = item_form.data do %>
        <li>
          <%= hidden_inputs_for(item_form) %>
          <%= item.product.title %>
          <%= number_input item_form, :quantity %>
          <%= currency_to_str(ShoppingCart.total_item_price(item)) %>
        </li>
      <% end %>
    </ul>

    <%= submit "update cart" %>
  <% end %>

  <b>Total</b>: <%= currency_to_str(ShoppingCart.total_cart_price(@cart)) %>
<% end %>
We started by showing the empty cart message if our preloaded cart.items is empty. If we have items, we use form_for to take our cart changeset that we assigned in the CartController.show/2 action and create a form which maps to our cart controller update/2 action. Within the form, we use Phoenix.HTML.Form.inputs_for/2 to render inputs for the nested cart items. For each item form input, we use hidden_inputs_for/1 which will render out the item ID as a hidden input tag. This will allow us to map item inputs back together when the form is submitted. Next, we display the product title for the item in the cart, followed by a number input for the item quantity. We finish the item form by converting the item price to string. We haven't written the ShoppingCart.total_item_price/1 function yet, but again we employed the idea of clear, descriptive public interfaces for our contexts. After rendering inputs for all the cart items, we show an "update cart" submit button, along with the total price of the entire cart. This is accomplished with another new ShoppingCart.total_cart_price/1 function which we'll implement in a moment.
We're almost ready to try out our cart page, but first we need to implement our new currency calculation functions. Open up your shopping cart context at lib/hello/shopping_cart.ex and add these new functions:
  def total_item_price(%CartItem{} = item) do
    Decimal.mult(item.product.price, item.quantity)
  end

  def total_cart_price(%Cart{} = cart) do
    Enum.reduce(cart.items, 0, fn item, acc ->
      item
      |> total_item_price()
      |> Decimal.add(acc)
    end)
  end
We implemented total_item_price/1 which accepts a %CartItem{} struct. To calculate the total price, we simply take the preloaded product's price and multiply it by the item's quantity. We used Decimal.mult/2 to take our decimal currency struct and multiply it with proper precision. Similarly for calculating the total cart price, we implemented a total_cart_price/1 function which accepts the cart and sums the preloaded product prices for items in the cart. We again make use of the Decimal functions to add our decimal structs together.
Now that we can calculate price totals, let's try it out! Visit http://localhost:4000/cart and you should already see your first item in the cart. Going back to the same product and clicking "add to cart" will show our upsert in action. Your quantity should now be two. Nice work!
Our cart page is almost complete, but submitting the form will yield yet another error.
Request: POST /cart
** (exit) an exception was raised:
    ** (UndefinedFunctionError) function HelloWeb.CartController.update/2 is undefined or private
Let's head back to our CartController at lib/hello_web/controllers/cart_controller.ex and implement the update action:
  def update(conn, %{"cart" => cart_params}) do
    case ShoppingCart.update_cart(conn.assigns.cart, cart_params) do
      {:ok, cart} ->
        redirect(conn, to: Routes.cart_path(conn, :show))

      {:error, _changeset} ->
        conn
        |> put_flash(:error, "There was an error updating your cart")
        |> redirect(to: Routes.cart_path(conn, :show))
    end
  end
We started by plucking out the cart params from the form submit. Next, we call our existing ShoppingCart.update_cart/2 function which was added by the context generator. We'll need to make some changes to this function, but the interface is good as is. If the update is successful, we redirect back to the cart page, otherwise we show a flash error message and send the user back to the cart page to fix any mistakes. Out-of-the-box, our ShoppingCart.update_cart/2 function only concerned itself with casting the cart params into a changeset and updates it against our repo. For our purposes, we now need it to handled nested cart item associations, and most importantly, business logic for how to handle quantity updates like zero-quantity items being removed from the cart.
Head back over to your shopping cart context in lib/hello/shopping_cart.ex and replace your update_cart/2 function with the following implementation:
  def update_cart(%Cart{} = cart, attrs) do
    changeset =
      cart
      |> Cart.changeset(attrs)
      |> Ecto.Changeset.cast_assoc(:items, with: &CartItem.changeset/2)

    Ecto.Multi.new()
    |> Ecto.Multi.update(:cart, changeset)
    |> Ecto.Multi.delete_all(:discarded_items, fn %{cart: cart} ->
      from(i in CartItem, where: i.cart_id == ^cart.id and i.quantity == 0)
    end)
    |> Repo.transaction()
    |> case do
      {:ok, %{cart: cart}} -> {:ok, cart}
      {:error, :cart, changeset, _changes_so_far} -> {:error, changeset}
    end
  end
We started much like how our out-of-the-box code started – we take the cart struct and cast the user input to a cart changeset, except this time we use Ecto.Changeset.cast_assoc/3 to cast the nested item data into CartItem changesets. Remember the hidden_inputs_for/1 call in our cart form template? That hidden ID data is what allows Ecto's cast_assoc to map item data back to existing item associations in the cart. Next we use Ecto.Multi.new/0, which you may not have seen before. Ecto's Multi is a feature that allows lazily defining a chain of named operations to eventually execute inside a database transaction. Each operation in the multi chain receives the values from the previous steps and executes until a failed step is encountered. When an operation fails, the transaction is rolled back and an error is returned, otherwise the transaction is committed.
For our multi operations, we start by issuing an update of our cart, which we named :cart. After the cart update is issued, we perform a multi delete_all operation, which takes the updated cart and applies our zero-quantity logic. We prune any items in the cart with zero quantity by returning an ecto query that finds all cart items for this cart with an empty quantity. Calling Repo.transaction/1 with our multi will execute the operations in a new transaction and we return the success or failure result to the caller just like the original function.
Let's head back to the browser and try it out. Add a few products to your cart, update the quantities, and watch the values changes along with the price calculations. Setting any quantity to 0 will also remove the item. Pretty neat!
Adding an Orders context
With our Catalog and ShoppingCart contexts, we're seeing first-hand how our well-considered modules and function names are yielding clear and maintainable code. Our last order of business is to allow the user to initiate the checkout process. We won't go as far as integrating payment processing or order fulfillment, but we'll get you started in that direction. Like before, we need to decide where code for completing an order should live. Is it part of the catalog? Clearly not, but what about the shopping cart? Shopping carts are related to orders – after all the user has to add items in order to purchase any products, but should the order checkout process be grouped here?
If we stop and consider the order process, we'll see that orders involve related, but distinctly different data from the cart contents. Also, business rules around the checkout process are much different than carting. For example, we may allow a user to add a back-ordered item to their cart, but we could not allow an order with no inventory to be completed. Additionally, we need to capture point-in-time product information when an order is completed, such as the price of the items at payment transaction time. This is essential because a product price may change in the future, but the line items in our order must always record and display what we charged at time of purchase. For these reasons, we can start to see ordering can reasonably stand on its own with its own data concerns and business rules.
Naming wise, Orders clearly defines the scope of our context, so let's get started by again taking advantage of the context generators. Run the following command in your console:
$ mix phx.gen.html Orders Order orders user_uuid:uuid total_price:decimal

* creating lib/hello_web/controllers/order_controller.ex
* creating lib/hello_web/templates/order/edit.html.heex
* creating lib/hello_web/templates/order/form.html.heex
* creating lib/hello_web/templates/order/index.html.heex
* creating lib/hello_web/templates/order/new.html.heex
* creating lib/hello_web/templates/order/show.html.heex
* creating lib/hello_web/views/order_view.ex
* creating test/hello_web/controllers/order_controller_test.exs
* creating lib/hello/orders/order.ex
* creating priv/repo/migrations/20210209214612_create_orders.exs
* creating lib/hello/orders.ex
* injecting lib/hello/orders.ex
* creating test/hello/orders_test.exs
* injecting test/hello/orders_test.exs
* creating test/support/fixtures/orders_fixtures.ex
* injecting test/support/fixtures/orders_fixtures.ex

Add the resource to your browser scope in lib/hello_web/router.ex:

    resources "/orders", OrderController


Remember to update your repository by running migrations:

    $ mix ecto.migrate

We generated a Orders context along with HTML controllers, views, etc. We added a user_uuid field to associate our placeholder current user to an order, along with a total_price column. With our starting point in place, let's open up the newly created migration in priv/repo/migrations/*_create_orders.exs and make the following changes:
  def change do
    create table(:orders) do
      add :user_uuid, :uuid
-     add :total_price, :decimal
+     add :total_price, :decimal, precision: 15, scale: 6, null: false

      timestamps()
    end
  end
Like we did previously, we gave appropriate precision and scale options for our decimal column which will allow us to store currency without precision loss. We also added a not-null constraint to enforce all orders to have a price.
The orders table alone doesn't hold much information, but we know we'll need to store point-in-time product price information of all the items in the order. For that, we'll add an additional struct for this context named LineItem. Line items will capture the price of the product at payment transaction time. Please run the following command:
$ mix phx.gen.context Orders LineItem order_line_items \
price:decimal quantity:integer \
order_id:references:orders product_id:references:products

You are generating into an existing context.
Would you like to proceed? [Yn] y
* creating lib/hello/orders/line_item.ex
* creating priv/repo/migrations/20210209215050_create_order_line_items.exs
* injecting lib/hello/orders.ex
* injecting test/hello/orders_test.exs
* injecting test/support/fixtures/orders_fixtures.ex

Remember to update your repository by running migrations:

    $ mix ecto.migrate

We used the phx.gen.context command to generate the LineItem Ecto schema and inject supporting functions into our orders context. Like before, let's modify the migration in priv/repo/migrations/*_create_order_line_items.exs and make the following decimal field changes:
  def change do
    create table(:order_line_items) do
-     add :price, :decimal
+     add :price, :decimal, precision: 15, scale: 6, null: false
      add :quantity, :integer
      add :order_id, references(:orders, on_delete: :nothing)
      add :product_id, references(:products, on_delete: :nothing)

      timestamps()
    end

    create index(:order_line_items, [:order_id])
    create index(:order_line_items, [:product_id])
  end
With our migration in place, let's wire up our orders and line items associations in lib/hello/orders/order.ex:
  schema "orders" do
    field :total_price, :decimal
    field :user_uuid, Ecto.UUID

+   has_many :line_items, Hello.Orders.LineItem
+   has_many :products, through: [:line_items, :product]

    timestamps()
  end
We used has_many :line_items to associate orders and line items, just like we've seen before. Next, we used the :through feature of has_many, which allows us to instruct ecto how to associate resources across another relationship. In this case, we can associate products of an order by finding all products through associated line items. Next, let's wire up the association in the other direction in lib/hello/orders/line_item.ex:
  schema "order_line_items" do
    field :price, :decimal
    field :quantity, :integer
-   field :order_id, :id
-   field :product_id, :id

+   belongs_to :order, Hello.Orders.Order
+   belongs_to :product, Hello.Catalog.Product

    timestamps()
  end
We used belongs_to to associate line items to orders and products. With our associations in place, we can start integrating the web interface into our order process. Open up your router lib/hello_web/router.ex and add the following line:
  scope "/", HelloWeb do
    pipe_through :browser

    ...
+   resources "/orders", OrderController, only: [:create, :show]
  end
We wired up create and show routes for our generated OrderController, since these are the only actions we need at the moment. With our routes in place, we can now migrate up:
$ mix ecto.migrate

17:14:37.715 [info]  == Running 20210209214612 Hello.Repo.Migrations.CreateOrders.change/0 forward

17:14:37.720 [info]  create table orders

17:14:37.755 [info]  == Migrated 20210209214612 in 0.0s

17:14:37.784 [info]  == Running 20210209215050 Hello.Repo.Migrations.CreateOrderLineItems.change/0 forward

17:14:37.785 [info]  create table order_line_items

17:14:37.795 [info]  create index order_line_items_order_id_index

17:14:37.796 [info]  create index order_line_items_product_id_index

17:14:37.798 [info]  == Migrated 20210209215050 in 0.0s

Before we render information about our orders, we need to ensure our order data is fully populated and can be looked up by a current user. Open up your orders context in lib/hello/orders.ex and replace your get_order!/1 function by a new get_order!/2 definition:
  def get_order!(user_uuid, id) do
    Order
    |> Repo.get_by!(id: id, user_uuid: user_uuid)
    |> Repo.preload([line_items: [:product]])
  end
We rewrote the function to accept a user UUID and query our repo for an order matching the user's ID for a given order ID. Then we populated the order by preloading our line item and product associations.
To complete an order, our cart page can issue a POST to the OrderController.create action, but we need to implement the operations and logic to actually complete an order. Like before, we'll start at the web interface by rewriting the create function in lib/hello_web/controllers/order_controller.ex:
  def create(conn, _) do
    case Orders.complete_order(conn.assigns.cart) do
      {:ok, order} ->
        conn
        |> put_flash(:info, "Order created successfully.")
        |> redirect(to: Routes.order_path(conn, :show, order))

      {:error, _reason} ->
        conn
        |> put_flash(:error, "There was an error processing your order")
        |> redirect(to: Routes.cart_path(conn, :show))
    end
  end
We rewrote the create action to call an as-yet-implemented Orders.complete_order/1 function. The code that phoenix generated had a generic Orders.create_order/1 call. Our code is technically "creating" an order, but it's important to step back and consider the naming of your interfaces. The act of completing an order is extremely important in our system. Money changes hands in a transaction, physical goods could be automatically shipped, etc. Such an operation deserves a better, more obvious function name, such as complete_order. If the order is completed successfully we redirect to the show page, otherwise a flash error is shown as we redirect back to the cart page.
Now we can implement our Orders.complete_order/1 function. To complete an order, our job will require a few operations:
	A new order record must be persisted with the total price of the order
	All items in the cart must be transformed into new order line items records
with quantity and point-in-time product price information
	After successful order insert (and eventual payment), items must be pruned
from the cart

From our requirements alone, we can start to see why a generic create_order function doesn't cut it. Let's implement this new function in lib/hello/orders.ex:
  alias Hello.ShoppingCart
  alias Hello.Orders.LineItem

  def complete_order(%ShoppingCart.Cart{} = cart) do
    line_items =
      Enum.map(cart.items, fn item ->
        %{product_id: item.product_id, price: item.product.price, quantity: item.quantity}
      end)

    order =
      Ecto.Changeset.change(%Order{},
        user_uuid: cart.user_uuid,
        total_price: ShoppingCart.total_cart_price(cart),
        line_items: line_items
      )

    Ecto.Multi.new()
    |> Ecto.Multi.insert(:order, order)
    |> Ecto.Multi.run(:prune_cart, fn _repo, _changes ->
      ShoppingCart.prune_cart_items(cart)
    end)
    |> Repo.transaction()
    |> case do
      {:ok, %{order: order}} -> {:ok, order}
      {:error, name, value, _changes_so_far} -> {:error, {name, value}}
    end
  end
We started by mapping the %ShoppingCart.CartItem{}'s in our shopping cart into a map of order line items structs. The job of the order line item record is to capture the price of the product at payment transaction time, so we reference the product's price here. Next, we create a bare order changeset with Ecto.Changeset.change/2 and associate our user UUID, set our total price calculation, and place our order line items in the changeset. With a fresh order changeset ready to be inserted, we can again make use of Ecto.Multi to execute our operations in a database transaction. We start by inserting the order, followed by a run operation. The Ecto.Multi.run/3 function allows us to run any code in the function which must either succeed with {:ok, result} or error, which halts and rolls back the transaction. Here, we simply can call into our shopping cart context and ask it to prune all items in a cart. Running the transaction will execute the multi as before and we return the result to the caller.
To close out our order completion, we need to implement the ShoppingCart.prune_cart_items/1 function in lib/hello/shopping_cart.ex:
  def prune_cart_items(%Cart{} = cart) do
    {_, _} = Repo.delete_all(from(i in CartItem, where: i.cart_id == ^cart.id))
    {:ok, reload_cart(cart)}
  end
Our new function accepts the cart struct and issues a Repo.delete_all which accepts a query of all items for the provided cart. We return a success result by simply reloading the pruned cart to the caller. With our context complete, we now need to show the user their completed order. Head back to your order controller and modify the show/2 action:
  def show(conn, %{"id" => id}) do
-   order = Orders.get_order!(id)
+   order = Orders.get_order!(conn.assigns.current_uuid, id)
    render(conn, "show.html", order: order)
  end
We tweaked the show action to pass our conn.assigns.current_uuid to get_order! which authorizes orders to be viewable only by the owner of the order. Next, we can replace the order show template in lib/hello_web/templates/order/show.html.heex:
<h1>Thank you for your order!</h1>

<ul>
  <li>
    <strong>User uuid:</strong>
    <%= @order.user_uuid %>
  </li>

  <%= for item <- @order.line_items do %>
    <li>
      <%= item.product.title %>
      (<%= item.quantity %>) - <%= HelloWeb.CartView.currency_to_str(item.price) %>
    </li>
  <% end %>

  <li>
    <strong>Total price:</strong>
    <%= HelloWeb.CartView.currency_to_str(@order.total_price) %>
  </li>

</ul>

<span><%= link "Back", to: Routes.cart_path(@conn, :show) %></span>
To show our completed order, we displayed the order's user, followed by the line item listing with product title, quantity, and the price we "transacted" when completing the order, along with the total price.
Our last addition will be to add the "complete order" button to our cart page to allow completing an order. Add the following button to the bottom of the cart show template in lib/hello_web/templates/cart/show.html.heex:
  <b>Total</b>: <%= currency_to_str(ShoppingCart.total_cart_price(@cart)) %>

+ <%= link "complete order", to: Routes.order_path(@conn, :create), method: :post %>
<% end %>
We added a link with method: :post to send a POST request to our OrderController.create action. If we head back to our cart page at http://localhost:4000/cart and complete an order, we'll be greeted by our rendered template:
Thank you for your order!

User uuid: 08964c7c-908c-4a55-bcd3-9811ad8b0b9d
Metaprogramming Elixir (2) - $15.00
Total price: $30.00
Nice work! We haven't added payments, but we can already see how our ShoppingCart and Orders context splitting is driving us towards a maintainable solution. With our cart items separated from our order line items, we are well equipped in the future to add payment transactions, cart price detection, and more.
Great work!
FAQ
Returning Ecto structures from context APIs
As we explored the context API, you might have wondered:
If one of the goals of our context is to encapsulate Ecto Repo access, why does create_user/1 return an Ecto.Changeset struct when we fail to create a user?

Although Changesets are part of Ecto, they are not tied to the database, and they can be used to map data from and to any source, which makes it a general and useful data structure for tracking field changes, perform validations, and generate error messages.
For those reasons, %Ecto.Changeset{} is a good choice to model the data changes between your contexts and your web layer. Regardless if you are talking to an API or the database.
Finally, note that your controllers and views are not hardcoded to work exclusively with Ecto either. Instead, Phoenix defines protocols such as Phoenix.Param and Phoenix.HTML.FormData, which allow any library to extend how Phoenix generates URL parameters or renders forms. Conveniently for us, the phoenix_ecto project implements those protocols, but you could as well bring your own data structures and implement them yourself.

There are currently a number of built-in Phoenix-specific and Ecto-specific Mix tasks available to us within a newly-generated application. We can also create our own application specific tasks.
Note to learn more about mix, you can read Elixir's official Introduction to Mix.

Phoenix tasks
$ mix help --search "phx"
mix local.phx          # Updates the Phoenix project generator locally
mix phx                # Prints Phoenix help information
mix phx.digest         # Digests and compresses static files
mix phx.digest.clean   # Removes old versions of static assets.
mix phx.gen.auth       # Generates authentication logic for a resource
mix phx.gen.cert       # Generates a self-signed certificate for HTTPS testing
mix phx.gen.channel    # Generates a Phoenix channel
mix phx.gen.context    # Generates a context with functions around an Ecto schema
mix phx.gen.embedded   # Generates an embedded Ecto schema file
mix phx.gen.html       # Generates controller, views, and context for an HTML resource
mix phx.gen.json       # Generates controller, views, and context for a JSON resource
mix phx.gen.live       # Generates LiveView, templates, and context for a resource
mix phx.gen.notifier   # Generates a notifier that delivers emails by default
mix phx.gen.presence   # Generates a Presence tracker
mix phx.gen.schema     # Generates an Ecto schema and migration file
mix phx.gen.secret     # Generates a secret
mix phx.gen.socket     # Generates a Phoenix socket handler
mix phx.new            # Creates a new Phoenix application
mix phx.new.ecto       # Creates a new Ecto project within an umbrella project
mix phx.new.web        # Creates a new Phoenix web project within an umbrella project
mix phx.routes         # Prints all routes
mix phx.server         # Starts applications and their servers

We have seen all of these at one point or another in the guides, but having all the information about them in one place seems like a good idea.
We will cover all Phoenix Mix tasks, except phx.new, phx.new.ecto, and phx.new.web, which are part of the Phoenix installer. You can learn more about them or any other task by calling mix help TASK.
mix phx.gen.html
Phoenix offers the ability to generate all the code to stand up a complete HTML resource — Ecto migration, Ecto context, controller with all the necessary actions, view, and templates. This can be a tremendous time saver. Let's take a look at how to make this happen.
The mix phx.gen.html task takes the following arguments: the module name of the context, the module name of the schema, the resource name, and a list of column_name:type attributes. The module name we pass in must conform to the Elixir rules of module naming, following proper capitalization.
$ mix phx.gen.html Blog Post posts body:string word_count:integer
* creating lib/hello_web/controllers/post_controller.ex
* creating lib/hello_web/templates/post/edit.html.heex
* creating lib/hello_web/templates/post/form.html.heex
* creating lib/hello_web/templates/post/index.html.heex
* creating lib/hello_web/templates/post/new.html.heex
* creating lib/hello_web/templates/post/show.html.heex
* creating lib/hello_web/views/post_view.ex
* creating test/hello_web/controllers/post_controller_test.exs
* creating lib/hello/blog/post.ex
* creating priv/repo/migrations/20211001233016_create_posts.exs
* creating lib/hello/blog.ex
* injecting lib/hello/blog.ex
* creating test/hello/blog_test.exs
* injecting test/hello/blog_test.exs
* creating test/support/fixtures/blog_fixtures.ex
* injecting test/support/fixtures/blog_fixtures.ex

When mix phx.gen.html is done creating files, it helpfully tells us that we need to add a line to our router file as well as run our Ecto migrations.
Add the resource to your browser scope in lib/hello_web/router.ex:

    resources "/posts", PostController

Remember to update your repository by running migrations:

    $ mix ecto.migrate

Important: If we don't do this, we will see the following warnings in our logs, and our application will error when trying to execute the function.
$ mix phx.server
Compiling 17 files (.ex)

warning: function HelloWeb.Router.Helpers.post_path/3 is undefined or private
  lib/hello_web/controllers/post_controller.ex:22:

If we don't want to create a context or schema for our resource we can use the --no-context flag. Note that this still requires a context module name as a parameter.
$ mix phx.gen.html Blog Post posts body:string word_count:integer --no-context
* creating lib/hello_web/controllers/post_controller.ex
* creating lib/hello_web/templates/post/edit.html.heex
* creating lib/hello_web/templates/post/form.html.heex
* creating lib/hello_web/templates/post/index.html.heex
* creating lib/hello_web/templates/post/new.html.heex
* creating lib/hello_web/templates/post/show.html.heex
* creating lib/hello_web/views/post_view.ex
* creating test/hello_web/controllers/post_controller_test.exs

It will tell us we need to add a line to our router file, but since we skipped the context, it won't mention anything about ecto.migrate.
Add the resource to your browser scope in lib/hello_web/router.ex:

    resources "/posts", PostController

Similarly, if we want a context created without a schema for our resource we can use the --no-schema flag.
$ mix phx.gen.html Blog Post posts body:string word_count:integer --no-schema
* creating lib/hello_web/controllers/post_controller.ex
* creating lib/hello_web/templates/post/edit.html.heex
* creating lib/hello_web/templates/post/form.html.heex
* creating lib/hello_web/templates/post/index.html.heex
* creating lib/hello_web/templates/post/new.html.heex
* creating lib/hello_web/templates/post/show.html.heex
* creating lib/hello_web/views/post_view.ex
* creating test/hello_web/controllers/post_controller_test.exs
* creating lib/hello/blog.ex
* injecting lib/hello/blog.ex
* creating test/hello/blog_test.exs
* injecting test/hello/blog_test.exs
* creating test/support/fixtures/blog_fixtures.ex
* injecting test/support/fixtures/blog_fixtures.ex

It will tell us we need to add a line to our router file, but since we skipped the schema, it won't mention anything about ecto.migrate.
mix phx.gen.json
Phoenix also offers the ability to generate all the code to stand up a complete JSON resource — Ecto migration, Ecto schema, controller with all the necessary actions and view. This command will not create any template for the app.
The mix phx.gen.json task takes the following arguments: the module name of the context, the module name of the schema, the resource name, and a list of column_name:type attributes. The module name we pass in must conform to the Elixir rules of module naming, following proper capitalization.
$ mix phx.gen.json Blog Post posts title:string content:string
* creating lib/hello_web/controllers/post_controller.ex
* creating lib/hello_web/views/post_view.ex
* creating test/hello_web/controllers/post_controller_test.exs
* creating lib/hello_web/views/changeset_view.ex
* creating lib/hello_web/controllers/fallback_controller.ex
* creating lib/hello/blog/post.ex
* creating priv/repo/migrations/20170906153323_create_posts.exs
* creating lib/hello/blog.ex
* injecting lib/hello/blog.ex
* creating test/hello/blog/blog_test.exs
* injecting test/hello/blog/blog_test.exs
* creating test/support/fixtures/blog_fixtures.ex
* injecting test/support/fixtures/blog_fixtures.ex

When mix phx.gen.json is done creating files, it helpfully tells us that we need to add a line to our router file as well as run our Ecto migrations.
Add the resource to your :api scope in lib/hello_web/router.ex:

    resources "/posts", PostController, except: [:new, :edit]

Remember to update your repository by running migrations:

    $ mix ecto.migrate

Important: If we don't do this, we'll get the following warning in our logs and the application will error when attempting to load the page:
$ mix phx.server
Compiling 19 files (.ex)

warning: function HelloWeb.Router.Helpers.post_path/3 is undefined or private
  lib/hello_web/controllers/post_controller.ex:18

mix phx.gen.json also supports --no-context, --no-schema, and others, as in mix phx.gen.html.
mix phx.gen.context
If we don't need a complete HTML/JSON resource and only need a context, we can use the mix phx.gen.context task. It will generate a context, a schema, a migration and a test case.
The mix phx.gen.context task takes the following arguments: the module name of the context, the module name of the schema, the resource name, and a list of column_name:type attributes.
$ mix phx.gen.context Accounts User users name:string age:integer
* creating lib/hello/accounts/user.ex
* creating priv/repo/migrations/20170906161158_create_users.exs
* creating lib/hello/accounts.ex
* injecting lib/hello/accounts.ex
* creating test/hello/accounts/accounts_test.exs
* injecting test/hello/accounts/accounts_test.exs
* creating test/support/fixtures/accounts_fixtures.ex
* injecting test/support/fixtures/accounts_fixtures.ex

Note: If we need to namespace our resource we can simply namespace the first argument of the generator.

$ mix phx.gen.context Admin.Accounts User users name:string age:integer
* creating lib/hello/admin/accounts/user.ex
* creating priv/repo/migrations/20170906161246_create_users.exs
* creating lib/hello/admin/accounts.ex
* injecting lib/hello/admin/accounts.ex
* creating test/hello/admin/accounts_test.exs
* injecting test/hello/admin/accounts_test.exs
* creating test/support/fixtures/admin/accounts_fixtures.ex
* injecting test/support/fixtures/admin/accounts_fixtures.ex

mix phx.gen.schema
If we don't need a complete HTML/JSON resource and are not interested in generating or altering a context we can use the mix phx.gen.schema task. It will generate a schema, and a migration.
The mix phx.gen.schema task takes the following arguments: the module name of the schema (which may be namespaced), the resource name, and a list of column_name:type attributes.
$ mix phx.gen.schema Accounts.Credential credentials email:string:unique user_id:references:users
* creating lib/hello/accounts/credential.ex
* creating priv/repo/migrations/20170906162013_create_credentials.exs

mix phx.gen.auth
Phoenix also offers the ability to generate all of the code to stand up a complete authentication system — Ecto migration, phoenix context, controllers, templates, etc. This can be a huge time saver, allowing you to quickly add authentication to your system and shift your focus back to the primary problems your application is trying to solve.
The mix phx.gen.auth task takes the following arguments: the module name of the context, the module name of the schema, and a plural version of the schema name used to generate database tables and route helpers.
Here is an example version of the command:
$ mix phx.gen.auth Accounts User users
* creating priv/repo/migrations/20201205184926_create_users_auth_tables.exs
* creating lib/hello/accounts/user_notifier.ex
* creating lib/hello/accounts/user.ex
* creating lib/hello/accounts/user_token.ex
* creating lib/hello_web/controllers/user_auth.ex
* creating test/hello_web/controllers/user_auth_test.exs
* creating lib/hello_web/views/user_confirmation_view.ex
* creating lib/hello_web/templates/user_confirmation/new.html.heex
* creating lib/hello_web/templates/user_confirmation/edit.html.heex
* creating lib/hello_web/controllers/user_confirmation_controller.ex
* creating test/hello_web/controllers/user_confirmation_controller_test.exs
* creating lib/hello_web/templates/layout/_user_menu.html.heex
* creating lib/hello_web/templates/user_registration/new.html.heex
* creating lib/hello_web/controllers/user_registration_controller.ex
* creating test/hello_web/controllers/user_registration_controller_test.exs
* creating lib/hello_web/views/user_registration_view.ex
* creating lib/hello_web/views/user_reset_password_view.ex
* creating lib/hello_web/controllers/user_reset_password_controller.ex
* creating test/hello_web/controllers/user_reset_password_controller_test.exs
* creating lib/hello_web/templates/user_reset_password/edit.html.heex
* creating lib/hello_web/templates/user_reset_password/new.html.heex
* creating lib/hello_web/views/user_session_view.ex
* creating lib/hello_web/controllers/user_session_controller.ex
* creating test/hello_web/controllers/user_session_controller_test.exs
* creating lib/hello_web/templates/user_session/new.html.heex
* creating lib/hello_web/views/user_settings_view.ex
* creating lib/hello_web/templates/user_settings/edit.html.heex
* creating lib/hello_web/controllers/user_settings_controller.ex
* creating test/hello_web/controllers/user_settings_controller_test.exs
* creating lib/hello/accounts.ex
* injecting lib/hello/accounts.ex
* creating test/hello/accounts_test.exs
* injecting test/hello/accounts_test.exs
* creating test/support/fixtures/accounts_fixtures.ex
* injecting test/support/fixtures/accounts_fixtures.ex
* injecting test/support/conn_case.ex
* injecting config/test.exs
* injecting mix.exs
* injecting lib/hello_web/router.ex
* injecting lib/hello_web/router.ex - imports
* injecting lib/hello_web/router.ex - plug
* injecting lib/hello_web/templates/layout/root.html.heex

When mix phx.gen.auth is done creating files, it helpfully tells us that we need to re-fetch our dependencies as well as run our Ecto migrations.
Please re-fetch your dependencies with the following command:

    mix deps.get

Remember to update your repository by running migrations:

  $ mix ecto.migrate

Once you are ready, visit "/users/register"
to create your account and then access to "/dev/mailbox" to
see the account confirmation email.

A more complete walk-through of how to get started with this generator is available in the mix phx.gen.auth authentication guide.
mix phx.gen.channel and mix phx.gen.socket
This task will generate a basic Phoenix channel, the socket to power the channel (if you haven't created one yet), as well a test case for it. It takes the module name for the channel as the only argument:
$ mix phx.gen.channel Room
* creating lib/hello_web/channels/room_channel.ex
* creating test/hello_web/channels/room_channel_test.exs

If your application does not have a UserSocket yet, it will ask if you want to create one:
The default socket handler - HelloWeb.UserSocket - was not found
in its default location.

Do you want to create it? [Y/n]

By pressing confirming, a channel will be created, then you need to connect the socket in your endpoint:
Add the socket handler to your `lib/hello_web/endpoint.ex`, for example:

    socket "/socket", HelloWeb.UserSocket,
      websocket: true,
      longpoll: false

For the front-end integration, you need to import the `user_socket.js`
in your `assets/js/app.js` file:

    import "./user_socket.js"

In case a UserSocket already exists or you decide to not create one, the channel generator will tell you to add it to the Socket manually:
Add the channel to your `lib/hello_web/channels/user_socket.ex` handler, for example:

    channel "rooms:lobby", HelloWeb.RoomChannel

You can also create a socket any time by invoking mix phx.gen.socket.
mix phx.gen.presence
This task will generate a presence tracker. The module name can be passed as an argument,
Presence is used if no module name is passed.
$ mix phx.gen.presence Presence
* lib/hello_web/channels/presence.ex

Add your new module to your supervision tree,
in lib/hello/application.ex:

    children = [
      ...
      HelloWeb.Presence
    ]

mix phx.routes
This task has a single purpose, to show us all the routes defined for a given router. We saw it used extensively in the routing guide.
If we don't specify a router for this task, it will default to the router Phoenix generated for us.
$ mix phx.routes
page_path  GET  /  TaskTester.PageController.index/2

We can also specify an individual router if we have more than one for our application.
$ mix phx.routes TaskTesterWeb.Router
page_path  GET  /  TaskTesterWeb.PageController.index/2

mix phx.server
This is the task we use to get our application running. It takes no arguments at all. If we pass any in, they will be silently ignored.
$ mix phx.server
[info] Running TaskTesterWeb.Endpoint with Cowboy on port 4000 (http)

It will silently ignore our DoesNotExist argument:
$ mix phx.server DoesNotExist
[info] Running TaskTesterWeb.Endpoint with Cowboy on port 4000 (http)

If we would like to start our application and also have an IEx session open to it, we can run the Mix task within iex like this, iex -S mix phx.server.
$ iex -S mix phx.server
Erlang/OTP 17 [erts-6.4] [source] [64-bit] [smp:8:8] [async-threads:10] [hipe] [kernel-poll:false] [dtrace]

[info] Running TaskTesterWeb.Endpoint with Cowboy on port 4000 (http)
Interactive Elixir (1.0.4) - press Ctrl+C to exit (type h() ENTER for help)
iex(1)>

mix phx.digest
This task does two things, it creates a digest for our static assets and then compresses them.
"Digest" here refers to an MD5 digest of the contents of an asset which gets added to the filename of that asset. This creates a sort of fingerprint for it. If the digest doesn't change, browsers and CDNs will use a cached version. If it does change, they will re-fetch the new version.
Before we run this task let's inspect the contents of two directories in our hello application.
First priv/static/ which should look similar to this:
├── images
│   └── phoenix.png
└── robots.txt

And then assets/ which should look similar to this:
├── css
│   └── app.css
├── js
│   └── app.js
└── vendor
    └── phoenix.js

All of these files are our static assets. Now let's run the mix phx.digest task.
$ mix phx.digest
Check your digested files at 'priv/static'.

We can now do as the task suggests and inspect the contents of priv/static/ directory. We'll see that all files from assets/ have been copied over to priv/static/ and also each file now has a couple of versions. Those versions are:
	the original file
	a compressed file with gzip
	a file containing the original file name and its digest
	a compressed file containing the file name and its digest

We can optionally determine which files should be gzipped by using the :gzippable_exts option in the config file:
config :phoenix, :gzippable_exts, ~w(.js .css)
Note: We can specify a different output folder where mix phx.digest will put processed files. The first argument is the path where the static files are located.

$ mix phx.digest priv/static/ -o www/public/
Check your digested files at 'www/public/'

Note: You can use mix phx.digest.clean to prune stale versions of the assets. If you want to remove all produced files, run mix phx.digest.clean --all.

Ecto tasks
Newly generated Phoenix applications now include Ecto and Postgrex as dependencies by default (which is to say, unless we use mix phx.new with the --no-ecto flag). With those dependencies come Mix tasks to take care of common Ecto operations. Let's see which tasks we get out of the box.
$ mix help --search "ecto"
mix ecto               # Prints Ecto help information
mix ecto.create        # Creates the repository storage
mix ecto.drop          # Drops the repository storage
mix ecto.dump          # Dumps the repository database structure
mix ecto.gen.migration # Generates a new migration for the repo
mix ecto.gen.repo      # Generates a new repository
mix ecto.load          # Loads previously dumped database structure
mix ecto.migrate       # Runs the repository migrations
mix ecto.migrations    # Displays the repository migration status
mix ecto.reset         # Alias defined in mix.exs
mix ecto.rollback      # Rolls back the repository migrations
mix ecto.setup         # Alias defined in mix.exs

Note: We can run any of the tasks above with the --no-start flag to execute the task without starting the application.
mix ecto.create
This task will create the database specified in our repo. By default it will look for the repo named after our application (the one generated with our app unless we opted out of Ecto), but we can pass in another repo if we want.
Here's what it looks like in action.
$ mix ecto.create
The database for Hello.Repo has been created.

There are a few things that can go wrong with ecto.create. If our Postgres database doesn't have a "postgres" role (user), we'll get an error like this one.
$ mix ecto.create
** (Mix) The database for Hello.Repo couldn't be created, reason given: psql: FATAL:  role "postgres" does not exist

We can fix this by creating the "postgres" role in the psql console with the permissions needed to log in and create a database.
=# CREATE ROLE postgres LOGIN CREATEDB;
CREATE ROLE

If the "postgres" role does not have permission to log in to the application, we'll get this error.
$ mix ecto.create
** (Mix) The database for Hello.Repo couldn't be created, reason given: psql: FATAL:  role "postgres" is not permitted to log in

To fix this, we need to change the permissions on our "postgres" user to allow login.
=# ALTER ROLE postgres LOGIN;
ALTER ROLE

If the "postgres" role does not have permission to create a database, we'll get this error.
$ mix ecto.create
** (Mix) The database for Hello.Repo couldn't be created, reason given: ERROR:  permission denied to create database

To fix this, we need to change the permissions on our "postgres" user in the psql console  to allow database creation.
=# ALTER ROLE postgres CREATEDB;
ALTER ROLE

If the "postgres" role is using a password different from the default "postgres", we'll get this error.
$ mix ecto.create
** (Mix) The database for Hello.Repo couldn't be created, reason given: psql: FATAL:  password authentication failed for user "postgres"

To fix this, we can change the password in the environment specific configuration file. For the development environment the password used can be found at the bottom of the config/dev.exs file.
Finally, if we happen to have another repo called OurCustom.Repo that we want to create the database for, we can run this.
$ mix ecto.create -r OurCustom.Repo
The database for OurCustom.Repo has been created.

mix ecto.drop
This task will drop the database specified in our repo. By default it will look for the repo named after our application (the one generated with our app unless we opted out of Ecto). It will not prompt us to check if we're sure we want to drop the database, so do exercise caution.
$ mix ecto.drop
The database for Hello.Repo has been dropped.

If we happen to have another repo that we want to drop the database for, we can specify it with the -r flag.
$ mix ecto.drop -r OurCustom.Repo
The database for OurCustom.Repo has been dropped.

mix ecto.gen.repo
Many applications require more than one data store. For each data store, we'll need a new repo, and we can generate them automatically with ecto.gen.repo.
If we name our repo OurCustom.Repo, this task will create it here lib/our_custom/repo.ex.
$ mix ecto.gen.repo -r OurCustom.Repo
* creating lib/our_custom
* creating lib/our_custom/repo.ex
* updating config/config.exs
Don't forget to add your new repo to your supervision tree
(typically in lib/hello/application.ex):

    {OurCustom.Repo, []}

Notice that this task has updated config/config.exs. If we take a look, we'll see this extra configuration block for our new repo.
. . .
config :hello, OurCustom.Repo,
  username: "user",
  password: "pass",
  hostname: "localhost",
  database: "hello_repo",
. . .
Of course, we'll need to change the login credentials to match what our database expects. We'll also need to change the config for other environments.
We certainly should follow the instructions and add our new repo to our supervision tree. In our Hello application, we would open up lib/hello/application.ex, and add our repo as a worker to the children list.
. . .
children = [
  # Start the Ecto repository
  Hello.Repo,
  # Our custom repo
  OurCustom.Repo,
  # Start the endpoint when the application starts
  HelloWeb.Endpoint,
]
. . .
mix ecto.gen.migration
Migrations are a programmatic, repeatable way to affect changes to a database schema. Migrations are also just modules, and we can create them with the ecto.gen.migration task. Let's walk through the steps to create a migration for a new comments table.
We simply need to invoke the task with a snake_case version of the module name that we want. Preferably, the name will describe what we want the migration to do.
$ mix ecto.gen.migration add_comments_table
* creating priv/repo/migrations
* creating priv/repo/migrations/20150318001628_add_comments_table.exs

Notice that the migration's filename begins with a string representation of the date and time the file was created.
Let's take a look at the file ecto.gen.migration has generated for us at priv/repo/migrations/20150318001628_add_comments_table.exs.
defmodule Hello.Repo.Migrations.AddCommentsTable do
  use Ecto.Migration

  def change do
  end
end
Notice that there is a single function change/0 which will handle both forward migrations and rollbacks. We'll define the schema changes that we want using Ecto's handy DSL, and Ecto will figure out what to do depending on whether we are rolling forward or rolling back. Very nice indeed.
What we want to do is create a comments table with a body column, a word_count column, and timestamp columns for inserted_at and updated_at.
. . .
def change do
  create table(:comments) do
    add :body, :string
    add :word_count, :integer
    timestamps()
  end
end
. . .
Again, we can run this task with the -r flag and another repo if we need to.
$ mix ecto.gen.migration -r OurCustom.Repo add_users
* creating priv/repo/migrations
* creating priv/repo/migrations/20150318172927_add_users.exs

For more information on how to modify your database schema please refer to the
Ecto's migration DSL docs.
For example, to alter an existing schema see the documentation on Ecto’s
alter/2 function.
That's it! We're ready to run our migration.
mix ecto.migrate
Once we have our migration module ready, we can simply run mix ecto.migrate to have our changes applied to the database.
$ mix ecto.migrate
[info] == Running Hello.Repo.Migrations.AddCommentsTable.change/0 forward
[info] create table comments
[info] == Migrated in 0.1s

When we first run ecto.migrate, it will create a table for us called schema_migrations. This will keep track of all the migrations which we run by storing the timestamp portion of the migration's filename.
Here's what the schema_migrations table looks like.
hello_dev=# select * from schema_migrations;
version        |     inserted_at
---------------+---------------------
20150317170448 | 2015-03-17 21:07:26
20150318001628 | 2015-03-18 01:45:00
(2 rows)

When we roll back a migration, ecto.rollback will remove the record representing this migration from schema_migrations.
By default, ecto.migrate will execute all pending migrations. We can exercise more control over which migrations we run by specifying some options when we run the task.
We can specify the number of pending migrations we would like to run with the -n or --step options.
$ mix ecto.migrate -n 2
[info] == Running Hello.Repo.Migrations.CreatePost.change/0 forward
[info] create table posts
[info] == Migrated in 0.0s
[info] == Running Hello.Repo.Migrations.AddCommentsTable.change/0 forward
[info] create table comments
[info] == Migrated in 0.0s

The --step option will behave the same way.
mix ecto.migrate --step 2

The --to option will run all migrations up to and including given version.
mix ecto.migrate --to 20150317170448

mix ecto.rollback
The ecto.rollback task will reverse the last migration we have run, undoing the schema changes. ecto.migrate and ecto.rollback are mirror images of each other.
$ mix ecto.rollback
[info] == Running Hello.Repo.Migrations.AddCommentsTable.change/0 backward
[info] drop table comments
[info] == Migrated in 0.0s

ecto.rollback will handle the same options as ecto.migrate, so -n, --step, -v, and --to will behave as they do for ecto.migrate.
Creating our own Mix task
As we've seen throughout this guide, both Mix itself and the dependencies we bring in to our application provide a number of really useful tasks for free. Since neither of these could possibly anticipate all our individual application's needs, Mix allows us to create our own custom tasks. That's exactly what we are going to do now.
The first thing we need to do is create a mix/tasks/ directory inside of lib/. This is where any of our application specific Mix tasks will go.
$ mkdir -p lib/mix/tasks/

Inside that directory, let's create a new file, hello.greeting.ex, that looks like this.
defmodule Mix.Tasks.Hello.Greeting do
  use Mix.Task

  @shortdoc "Sends a greeting to us from Hello Phoenix"

  @moduledoc """
  This is where we would put any long form documentation and doctests.
  """

  @impl Mix.Task
  def run(_args) do
    Mix.shell().info("Greetings from the Hello Phoenix Application!")
  end

  # We can define other functions as needed here.
end
Let's take a quick look at the moving parts involved in a working Mix task.
The first thing we need to do is name our module. All tasks must be defined in the Mix.Tasks namespace. We'd like to invoke this as mix hello.greeting, so we complete the module name with
Hello.Greeting.
The use Mix.Task line brings in functionality from Mix that makes this module behave as a Mix task.
The @shortdoc module attribute holds a string which will describe our task when users invoke mix help.
@moduledoc serves the same function that it does in any module. It's where we can put long-form documentation and doctests, if we have any.
The run/1 function is the critical heart of any Mix task. It's the function that does all the work when users invoke our task. In ours, all we do is send a greeting from our app, but we can implement our run/1 function to do whatever we need it to. Note that Mix.shell().info/1 is the preferred way to print text back out to the user.
Of course, our task is just a module, so we can define other private functions as needed to support our run/1 function.
Now that we have our task module defined, our next step is to compile the application.
$ mix compile
Compiled lib/tasks/hello.greeting.ex
Generated hello.app

Now our new task should be visible to mix help.
$ mix help --search hello
mix hello.greeting # Sends a greeting to us from Hello Phoenix

Notice that mix help displays the text we put into the @shortdoc along with the name of our task.
So far, so good, but does it work?
$ mix hello.greeting
Greetings from the Hello Phoenix Application!

Indeed it does.
If you want to make your new Mix task to use your application's infrastructure, you need to make sure the application is started and configure when Mix task is being executed. This is particularly useful if you need to access your database from within the Mix task. Thankfully, Mix makes it really easy for us via the @requirements module attribute:
  @requirements ["app.config"]

  @impl Mix.Task
  def run(_args) do
    Mix.shell().info("Now I have access to Repo and other goodies!")
    Mix.shell().info("Greetings from the Hello Phoenix Application!")
  end

In this guide, we will show you how to instrument and report
on :telemetry events in your Phoenix application.
te·lem·e·try - the process of recording and transmitting
the readings of an instrument.

As you follow along with this guide, we will introduce you to
the core concepts of Telemetry, you will initialize a
reporter to capture your application's events as they occur,
and we will guide you through the steps to properly
instrument your own functions using :telemetry. Let's take
a closer look at how Telemetry works in your application.
Overview
The [:telemetry] library allows you to emit events at various stages of an application's lifecycle. You can then respond to these events by, among other things, aggregating them as metrics and sending the metrics data to a reporting destination.
Telemetry stores events by their name in an ETS table, along with the handler for each event. Then, when a given event is executed, Telemetry looks up its handler and invokes it.
Phoenix's Telemetry tooling provides you with a supervisor that uses Telemetry.Metrics to define the list of Telemetry events to handle and how to handle those events, i.e. how to structure them as a certain type of metric. This supervisor works together with Telemetry reporters to respond to the specified Telemetry events by aggregating them as the appropriate metric and sending them to the correct reporting destination.
The Telemetry supervisor
Since v1.5, new Phoenix applications are generated with a
Telemetry supervisor. This module is responsible for
managing the lifecycle of your Telemetry processes. It also
defines a metrics/0 function, which returns a list of
Telemetry.Metrics
that you define for your application.
By default, the supervisor also starts
:telemetry_poller.
By simply adding :telemetry_poller as a dependency, you
can receive VM-related events on a specified interval.
If you are coming from an older version of Phoenix, install
the :telemetry_metrics and :telemetry_poller packages:
{:telemetry_metrics, "~> 0.6"},
{:telemetry_poller, "~> 1.0"}
and create your Telemetry supervisor at
lib/my_app_web/telemetry.ex:
# lib/my_app_web/telemetry.ex
defmodule MyAppWeb.Telemetry do
  use Supervisor
  import Telemetry.Metrics

  def start_link(arg) do
    Supervisor.start_link(__MODULE__, arg, name: __MODULE__)
  end

  def init(_arg) do
    children = [
      {:telemetry_poller, measurements: periodic_measurements(), period: 10_000}
      # Add reporters as children of your supervision tree.
      # {Telemetry.Metrics.ConsoleReporter, metrics: metrics()}
    ]

    Supervisor.init(children, strategy: :one_for_one)
  end

  def metrics do
    [
      # Phoenix Metrics
      summary("phoenix.endpoint.stop.duration",
        unit: {:native, :millisecond}
      ),
      summary("phoenix.router_dispatch.stop.duration",
        tags: [:route],
        unit: {:native, :millisecond}
      ),
      # VM Metrics
      summary("vm.memory.total", unit: {:byte, :kilobyte}),
      summary("vm.total_run_queue_lengths.total"),
      summary("vm.total_run_queue_lengths.cpu"),
      summary("vm.total_run_queue_lengths.io")
    ]
  end

  defp periodic_measurements do
    [
      # A module, function and arguments to be invoked periodically.
      # This function must call :telemetry.execute/3 and a metric must be added above.
      # {MyApp, :count_users, []}
    ]
  end
end
Make sure to replace MyApp by your actual application name.
Then add to your main application's supervision tree
(usually in lib/my_app/application.ex):
children = [
  MyApp.Repo,
  MyAppWeb.Telemetry,
  MyAppWeb.Endpoint,
  ...
]
Telemetry Events
Many Elixir libraries (including Phoenix) are already using
the :telemetry package as a
way to give users more insight into the behavior of their
applications, by emitting events at key moments in the
application lifecycle.
A Telemetry event is made up of the following:
	name - A string (e.g. "my_app.worker.stop") or a
list of atoms that uniquely identifies the event.

	measurements - A map of atom keys (e.g. :duration)
and numeric values.

	metadata - A map of key-value pairs that can be used
for tagging metrics.


A Phoenix Example
Here is an example of an event from your endpoint:
	[:phoenix, :endpoint, :stop] - dispatched by
Plug.Telemetry, one of the default plugs in your endpoint, whenever the response is
sent	Measurement: %{duration: native_time}

	Metadata: %{conn: Plug.Conn.t}




This means that after each request, Plug, via :telemetry,
will emit a "stop" event, with a measurement of how long it
took to get the response:
:telemetry.execute([:phoenix, :endpoint, :stop], %{duration: duration}, %{conn: conn})
Phoenix Telemetry Events
A full list of all Phoenix telemetry events can be found in Phoenix.Logger
Metrics
Metrics are aggregations of Telemetry events with a
specific name, providing a view of the system's behaviour
over time.
― Telemetry.Metrics

The Telemetry.Metrics package provides a common interface
for defining metrics. It exposes a set of five metric type functions that are responsible for structuring a given Telemetry event as a particular measurement.
The package does not perform any aggregation of the measurements itself. Instead, it provides a reporter with the Telemetry event-as-measurement definition and the reporter uses that definition to perform aggregations and report them.
We will discuss
reporters in the next section.
Let's take a look at some examples.
Using Telemetry.Metrics, you can define a counter metric,
which counts how many HTTP requests were completed:
Telemetry.Metrics.counter("phoenix.endpoint.stop.duration")
or you could use a distribution metric to see how many
requests were completed in particular time buckets:
Telemetry.Metrics.distribution("phoenix.endpoint.stop.duration")
This ability to introspect HTTP requests is really powerful --
and this is but one of many telemetry events emitted by
the Phoenix framework! We'll discuss more of these events,
as well as specific patterns for extracting valuable data
from Phoenix/Plug events in the
Phoenix Metrics section later in this
guide.
The full list of :telemetry events emitted from Phoenix,
along with their measurements and metadata, is available in
the "Instrumentation" section of the Phoenix.Logger module
documentation.

An Ecto Example
Like Phoenix, Ecto ships with built-in Telemetry events.
This means that you can gain introspection into your web
and database layers using the same tools.
Here is an example of a Telemetry event executed by Ecto when an Ecto repository starts:
	[:ecto, :repo, :init] - dispatched by Ecto.Repo	Measurement: %{system_time: native_time}

	Metadata: %{repo: Ecto.Repo, opts: Keyword.t()}




This means that whenever the Ecto.Repo starts, it will emit an event, via :telemetry,
with a measurement of the time at start-up.
:telemetry.execute([:ecto, :repo, :init], %{system_time: System.system_time()}, %{repo: repo, opts: opts})
Additional Telemetry events are executed by Ecto adapters.
One such adapter-specific event is the [:my_app, :repo, :query] event.
For instance, if you want to graph query execution time, you can use the Telemetry.Metrics.summary/2 function to instruct your reporter to calculate statistics of the [:my_app, :repo, :query] event, like maximum, mean, percentiles etc.:
Telemetry.Metrics.summary("my_app.repo.query.query_time",
  unit: {:native, :millisecond}
)
Or you could use the Telemetry.Metrics.distribution/2 function to define a histogram for another adapter-specific event: [:my_app, :repo, :query, :queue_time], thus visualizing how long queries spend queued:
Telemetry.Metrics.distribution("my_app.repo.query.queue_time",
  unit: {:native, :millisecond}
)
You can learn more about Ecto Telemetry in the "Telemetry
Events" section of the
Ecto.Repo module
documentation.

So far we have seen some of the Telemetry events common to
Phoenix applications, along with some examples of their
various measurements and metadata. With all of this data
just waiting to be consumed, let's talk about reporters.
Reporters
Reporters subscribe to Telemetry events using the common
interface provided by Telemetry.Metrics. They then
aggregate the measurements (data) into metrics to provide
meaningful information about your application.
For example, if the following Telemetry.Metrics.summary/2 call is added to the metrics/0 function of your Telemetry supervisor:
summary("phoenix.endpoint.stop.duration",
  unit: {:native, :millisecond}
)
Then the reporter will attach a listener for the "phoenix.endpoint.stop.duration" event and will respond to this event by calculating a summary metric with the given event metadata and reporting on that metric to the appropriate source.
Phoenix.LiveDashboard
For developers interested in real-time visualizations for
their Telemetry metrics, you may be interested in installing
LiveDashboard.
LiveDashboard acts as a Telemetry.Metrics reporter to render
your data as beautiful, real-time charts on the dashboard.
Telemetry.Metrics.ConsoleReporter
Telemetry.Metrics ships with a ConsoleReporter that can
be used to print events and metrics to the terminal. You can
use this reporter to experiment with the metrics discussed in
this guide.
Uncomment or add the following to this list of children in
your Telemetry supervision tree (usually in
lib/my_app_web/telemetry.ex):
{Telemetry.Metrics.ConsoleReporter, metrics: metrics()}
There are numerous reporters available, for services like
StatsD, Prometheus, and more. You can find them by
searching for "telemetry_metrics" on hex.pm.

Phoenix Metrics
Earlier we looked at the "stop" event emitted by
Plug.Telemetry, and used it to count the number of HTTP
requests. In reality, it's only somewhat helpful to be
able to see just the total number of requests. What if you
wanted to see the number of requests per route, or per route
and method?
Let's take a look at another event emitted during the HTTP
request lifecycle, this time from Phoenix.Router:
	[:phoenix, :router_dispatch, :stop] - dispatched by
Phoenix.Router after successfully dispatching to a matched
route	Measurement: %{duration: native_time}

	Metadata: %{conn: Plug.Conn.t, route: binary, plug: module, plug_opts: term, path_params: map, pipe_through: [atom]}




Let's start by grouping these events by route. Add the
following (if it does not already exist) to the metrics/0
function of your Telemetry supervisor (usually in
lib/my_app_web/telemetry.ex):
# lib/my_app_web/telemetry.ex
def metrics do
  [
    ...metrics...
    summary("phoenix.router_dispatch.stop.duration",
      tags: [:route],
      unit: {:native, :millisecond}
    )
  ]
end
Restart your server, and then make requests to a page or two.
In your terminal, you should see the ConsoleReporter print
logs for the Telemetry events it received as a result of
the metrics definitions you provided.
The log line for each request contains the specific route
for that request. This is due to specifying the :tags
option for the summary metric, which takes care of our first
requirement; we can use :tags to group metrics by route.
Note that reporters will necessarily handle tags differently
depending on the underlying service in use.
Looking more closely at the Router "stop" event, you can see
that the Plug.Conn struct representing the request is
present in the metadata, but how do you access the
properties in conn?
Fortunately, Telemetry.Metrics provides the following
options to help you classify your events:
	:tags - A list of metadata keys for grouping;

	:tag_values - A function which transforms the metadata
into the desired shape; Note that this function is called
for each event, so it's important to keep it fast if the
rate of events is high.


Learn about all the available metrics options in the
Telemetry.Metrics module documentation.

Let's find out how to extract more tags from events that
include a conn in their metadata.
Extracting tag values from Plug.Conn
Let's add another metric for the route event, this time to
group by route and method:
summary("phoenix.router_dispatch.stop.duration",
  tags: [:method, :route],
  tag_values: &get_and_put_http_method/1,
  unit: {:native, :millisecond}
)
We've introduced the :tag_values option here, because we
need to perform a transformation on the event metadata in
order to get to the values we need.
Add the following private function to your Telemetry module
to lift the :method value from the Plug.Conn struct:
# lib/my_app_web/telemetry.ex
defp get_and_put_http_method(%{conn: %{method: method}} = metadata) do
  Map.put(metadata, :method, method)
end
Restart your server and make some more requests. You should
begin to see logs with tags for both the HTTP method and the
route.
Note the :tags and :tag_values options can be applied to
all Telemetry.Metrics types.
Renaming value labels using tag values
Sometimes when displaying a metric, the value label may need to be transformed
to improve readability. Take for example the following metric that displays the
duration of the each LiveView's mount/3 callback by connected? status.
summary("phoenix.live_view.mount.stop.duration",
  unit: {:native, :millisecond},
  tags: [:view, :connected?],
  tag_values: &live_view_metric_tag_values/1
)
The following function lifts metadata.socket.view and
metadata.socket.connected? to be top-level keys on metadata, as we did in
the previous example.
# lib/my_app_web/telemetry.ex
defp live_view_metric_tag_values(metadata) do
  metadata
  |> Map.put(:view, metadata.socket.view)
  |> Map.put(:connected?, metadata.socket.connected?)
end
However, when rendering these metrics in LiveDashboard, the value label is
output as "Elixir.Phoenix.LiveDashboard.MetricsLive true".
To make the value label easier to read, we can update our private function to
generate more user friendly names. We'll run the value of the :view through
inspect/1 to remove the Elixir. prefix and call another private function to
convert the connected? boolean into human readable text.
# lib/my_app_web/telemetry.ex
defp live_view_metric_tag_values(metadata) do
  metadata
  |> Map.put(:view, inspect(metadata.socket.view))
  |> Map.put(:connected?, get_connection_status(metadata.socket))
end

defp get_connection_status(%{connected?: true}), do: "Connected"
defp get_connection_status(%{connected?: false}), do: "Disconnected"
Now the value label will be rendered like "Phoenix.LiveDashboard.MetricsLive Connected".
Hopefully, this gives you some inspiration on how to use the :tag_values
option. Just remember to keep this function fast since it is called on every
event.
Periodic measurements
You might want to periodically measure key-value pairs within
your application. Fortunately the
:telemetry_poller
package provides a mechanism for custom measurements,
which is useful for retrieving process information or for
performing custom measurements periodically.
Add the following to the list in your Telemetry supervisor's
periodic_measurements/0 function, which is a private
function that returns a list of measurements to take on a
specified interval.
# lib/my_app_web/telemetry.ex
defp periodic_measurements do
  [
    {MyApp, :measure_users, []},
    {:process_info,
      event: [:my_app, :my_server],
      name: MyApp.MyServer,
      keys: [:message_queue_len, :memory]}
  ]
end
where MyApp.measure_users/0 could be written like this:
# lib/my_app.ex
defmodule MyApp do
  def measure_users do
    :telemetry.execute([:my_app, :users], %{total: MyApp.users_count()}, %{})
  end
end
Now with measurements in place, you can define the metrics for the
events above:
# lib/my_app_web/telemetry.ex
def metrics do
  [
    ...metrics...
    # MyApp Metrics
    last_value("my_app.users.total"),
    last_value("my_app.my_server.memory", unit: :byte),
    last_value("my_app.my_server.message_queue_len")
    summary("my_app.my_server.call.stop.duration"),
    counter("my_app.my_server.call.exception")
  ]
end
You will implement MyApp.MyServer in the
Custom Events section.

Libraries using Telemetry
Telemetry is quickly becoming the de-facto standard for
package instrumentation in Elixir. Here is a list of
libraries currently emitting :telemetry events.
Library authors are actively encouraged to send a PR adding
their own (in alphabetical order, please):
	Absinthe - Events
	Broadway - Events
	Ecto - Events
	Oban - Events
	Phoenix - Events
	Plug - Events
	Tesla - Events

Custom Events
If you need custom metrics and instrumentation in your
application, you can utilize the :telemetry package
(https://hexdocs.pm/telemetry) just like your favorite
frameworks and libraries.
Here is an example of a simple GenServer that emits telemetry
events. Create this file in your app at
lib/my_app/my_server.ex:
# lib/my_app/my_server.ex
defmodule MyApp.MyServer do
  @moduledoc """
  An example GenServer that runs arbitrary functions and emits telemetry events when called.
  """
  use GenServer

  # A common prefix for :telemetry events
  @prefix [:my_app, :my_server, :call]

  def start_link(fun) do
    GenServer.start_link(__MODULE__, fun, name: __MODULE__)
  end

  @doc """
  Runs the function contained within this server.

  ## Events

  The following events may be emitted:

    * `[:my_app, :my_server, :call, :start]` - Dispatched
      immediately before invoking the function. This event
      is always emitted.

      * Measurement: `%{system_time: system_time}`

      * Metadata: `%{}`

    * `[:my_app, :my_server, :call, :stop]` - Dispatched
      immediately after successfully invoking the function.

      * Measurement: `%{duration: native_time}`

      * Metadata: `%{}`

    * `[:my_app, :my_server, :call, :exception]` - Dispatched
      immediately after invoking the function, in the event
      the function throws or raises.

      * Measurement: `%{duration: native_time}`

      * Metadata: `%{kind: kind, reason: reason, stacktrace: stacktrace}`
  """
  def call!, do: GenServer.call(__MODULE__, :called)

  @impl true
  def init(fun) when is_function(fun, 0), do: {:ok, fun}

  @impl true
  def handle_call(:called, _from, fun) do
    # Wrap the function invocation in a "span"
    result = telemetry_span(fun)

    {:reply, result, fun}
  end

  # Emits telemetry events related to invoking the function
  defp telemetry_span(fun) do
    start_time = emit_start()

    try do
      fun.()
    catch
      kind, reason ->
        stacktrace = System.stacktrace()
        duration = System.monotonic_time() - start_time
        emit_exception(duration, kind, reason, stacktrace)
        :erlang.raise(kind, reason, stacktrace)
    else
      result ->
        duration = System.monotonic_time() - start_time
        emit_stop(duration)
        result
    end
  end

  defp emit_start do
    start_time_mono = System.monotonic_time()

    :telemetry.execute(
      @prefix ++ [:start],
      %{system_time: System.system_time()},
      %{}
    )

    start_time_mono
  end

  defp emit_stop(duration) do
    :telemetry.execute(
      @prefix ++ [:stop],
      %{duration: duration},
      %{}
    )
  end

  defp emit_exception(duration, kind, reason, stacktrace) do
    :telemetry.execute(
      @prefix ++ [:exception],
      %{duration: duration},
      %{
        kind: kind,
        reason: reason,
        stacktrace: stacktrace
      }
    )
  end
end
and add it to your application's supervisor tree (usually in
lib/my_app/application.ex), giving it a function to invoke
when called:
# lib/my_app/application.ex
children = [
  # Start a server that greets the world
  {MyApp.MyServer, fn -> "Hello, world!" end},
]
Now start an IEx session and call the server:
iex> MyApp.MyServer.call!
and you should see something like the following output:
[Telemetry.Metrics.ConsoleReporter] Got new event!
Event name: my_app.my_server.call.stop
All measurements: %{duration: 4000}
All metadata: %{}

Metric measurement: #Function<2.111777250/1 in Telemetry.Metrics.maybe_convert_measurement/2> (summary)
With value: 0.004 millisecond
Tag values: %{}

"Hello, world!"



  

  
    
    Asset Management - Phoenix v1.6.9
    
    

    



  
  

    
Asset Management
    

Beside producing HTML, most web applications have various assets (JavaScript, CSS, images, fonts and so on).
From Phoenix v1.6, new applications use esbuild to prepare assets via the Elixir esbuild wrapper. This direct integration with esbuild means that newly generated applications do not have dependencies on Node.js or an external build system (e.g. Webpack).
Your JavaScript is typically placed at "assets/js/app.js" and esbuild will extract it to "priv/static/assets/app.js". In development, this is done automatically via the esbuild watcher. In production, this is done by running mix assets.deploy.
esbuild can also handle your CSS files. For this, there is typically an import "../css/app.css" at the top of your "assets/js/app.js". We will explore alternatives below.
Finally, all other assets, that usually don't have to be preprocessed, go directly to "priv/static".
Third-party JS packages
If you want to import JavaScript dependencies, you have two options to add them to your application:
	Vendor those dependencies inside your project and import them in your "assets/js/app.js" using a relative path:
import topbar from "../vendor/topbar"

	Call npm install topbar --save inside your assets directory and esbuild will be able to automatically pick them up:
import topbar from "topbar"


CSS
esbuild has basic support for CSS. If you import a .css file at the top of your main .js file, esbuild will also bundle it, and write it to the same directory as your final app.js. That's what Phoenix does by default:
import "../css/app.css"
However, if you want to use a CSS framework, you will need to use a separate tool. Here are some options to do so:
	Use standalone Tailwind or standalone SASS. Both similar to esbuild.

	You can use esbuild plugins (requires npm). See the "Esbuild plugins" section below


Don't forget to remove the import "../css/app.css" from your JavaScript file when doing so.
Images, fonts, and external files
If you reference an external file in your CSS or JavaScript files, esbuild will attempt to validate and manage them, unless told otherwise.
For example, imagine you want to reference priv/static/images/bg.png, served at /images/bg.png, from your CSS file:
body {
  background-image: url(/images/bg.png);
}
The above may fail with the following message:
error: Could not resolve "/images/bg.png" (mark it as external to exclude it from the bundle)
Given the images are already managed by Phoenix, you need to mark all resources from /images (and also /fonts) as external, as the error message says. This is what Phoenix does by default for new apps since v1.6.1+. In your config/config.exs, you will find:
args: ~w(js/app.js --bundle --target=es2017 --outdir=../priv/static/assets --external:/fonts/* --external:/images/*),
If you need to reference other directories, you need to update the arguments above accordingly. Note running mix phx.digest will create digested files for all of the assets in priv/static, so your images and fonts are still cache-busted.
Esbuild plugins
Phoenix's default configuration of esbuild (via the Elixir wrapper) does not allow you to use esbuild plugins. If you want to use an esbuild plugin, for example to compile SASS files to CSS, you can replace the default build system with a custom build script.
The following is an example of a custom build using esbuild via Node.JS. First of all, you'll need to install Node.js in development and make it available for your production build step.
Then you'll need to add esbuild to your Node.js packages and the Phoenix packages. Inside the assets directory, run:
$ npm install esbuild --save-dev
$ npm install ../deps/phoenix ../deps/phoenix_html ../deps/phoenix_live_view --save

or, for Yarn:
$ yarn add --dev esbuild
$ yarn add ../deps/phoenix ../deps/phoenix_html ../deps/phoenix_live_view

Next, add a custom Javascript build script. We'll call the example assets/build.js:
const esbuild = require('esbuild')

const args = process.argv.slice(2)
const watch = args.includes('--watch')
const deploy = args.includes('--deploy')

const loader = {
  // Add loaders for images/fonts/etc, e.g. { '.svg': 'file' }
}

const plugins = [
  // Add and configure plugins here
]

let opts = {
  entryPoints: ['js/app.js'],
  bundle: true,
  target: 'es2017',
  outdir: '../priv/static/assets',
  logLevel: 'info',
  loader,
  plugins
}

if (watch) {
  opts = {
    ...opts,
    watch,
    sourcemap: 'inline'
  }
}

if (deploy) {
  opts = {
    ...opts,
    minify: true
  }
}

const promise = esbuild.build(opts)

if (watch) {
  promise.then(_result => {
    process.stdin.on('close', () => {
      process.exit(0)
    })

    process.stdin.resume()
  })
}
This script covers following use cases:
	node build.js: builds for development & testing (useful on CI)
	node build.js --watch: like above, but watches for changes continuously
	node build.js --deploy: builds minified assets for production

Modify config/dev.exs so that the script runs whenever you change files, replacing the existing :esbuild configuration under watchers:
config :hello, HelloWeb.Endpoint,
  ...
  watchers: [
    node: ["build.js", "--watch", cd: Path.expand("../assets", __DIR__)]
  ],
  ...
Modify the aliases task in mix.exs to install npm packages during mix setup and use the new esbuild on mix assets.deploy:
  defp aliases do
    [
      setup: ["deps.get", "ecto.setup", "cmd --cd assets npm install"],
      ...,
      "assets.deploy": ["cmd --cd assets node build.js --deploy", "phx.digest"]
    ]
  end
Finally, remove the esbuild configuration from config/config.exs and remove the dependency from the deps function in your mix.exs, and you are done!
Removing esbuild
If you are writing an API, or for some other reason you do not need to serve any assets, you can disable asset management completely.
	Remove the esbuild configuration in config/config.exs and config/dev.exs,
	Remove the assets.deploy task defined in mix.exs,
	Remove the esbuild dependency from mix.exs,
	Unlock the esbuild dependency:

$ mix deps.unlock esbuild

The mix phx.gen.auth command generates a flexible, pre-built authentication system into your Phoenix app. This simple generator allows you to quickly move past the task of adding authentication to your codebase and stay focused on the real-world problem your application is trying to solve.
Getting started
Let's start by running the following command from the root of our app (or apps/my_app_web in an umbrella app):
$ mix phx.gen.auth Accounts User users

This creates an Accounts context with an Accounts.User schema module. The final argument is the plural version of the schema module, which is used for generating database table names and route helpers. The mix phx.gen.auth generator is similar to mix phx.gen.html except it does not accept a list of additional fields to add to the schema, and it generates many more context functions.
Since this generator installed additional dependencies in mix.exs, let's fetch those:
$ mix deps.get

Now we need to verify the database connection details for the development and test environments in config/ so the migrator and tests can run properly. Then run the following to create the database:
$ mix ecto.setup

Let's run the tests to make sure our new authentication system works as expected.
$ mix test

And finally, let's start our Phoenix server and try it out.
$ mix phx.server

Developer responsibilities
Since Phoenix generates this code into your application instead of building these modules into Phoenix itself, you now have complete freedom to modify the authentication system, so it works best with your use case. The one caveat with using a generated authentication system is it will not be updated after it's been generated. Therefore, as improvements are made to the output of mix phx.gen.auth, it becomes your responsibility to determine if these changes need to be ported into your application. Security-related and other important improvements will be explicitly and clearly marked in the CHANGELOG.md file and upgrade notes.
Generated code
The following are notes about the generated authentication system.
Password hashing
The password hashing mechanism defaults to bcrypt for Unix systems and pbkdf2 for Windows systems. Both systems use the Comeonin interface.
The password hashing mechanism can be overridden with the --hashing-lib option. The following values are supported:
	bcrypt - bcrypt_elixir
	pbkdf2 - pbkdf2_elixir
	argon2 - argon2_elixir

We recommend developers to consider using argon2, which is the most robust of all 3. The downside is that argon2 is quite CPU and memory intensive, and you will need more powerful instances to run your applications on.
For more information about choosing these libraries, see the Comeonin project.
Forbidding access
The generated code ships with an authentication module with a handful of plugs that fetch the current user, require authentication and so on. For instance, in an app named Demo which had mix phx.gen.auth Accounts User users run on it, you will find a module named DemoWeb.UserAuth with plugs such as:
	fetch_current_user - fetches the current user information if available
	require_authenticated_user - must be invoked after fetch_current_user and requires that a current user exists and is authenticated
	redirect_if_user_is_authenticated - used for the few pages that must not be available to authenticated users

Confirmation
The generated functionality ships with an account confirmation mechanism, where users have to confirm their account, typically by email. However, the generated code does not forbid users from using the application if their accounts have not yet been confirmed. You can add this functionality by customizing the require_authenticated_user in the Auth module to check for the confirmed_at field (and any other property you desire).
Notifiers
The generated code is not integrated with any system to send SMSes or emails for confirming accounts, resetting passwords, etc. Instead, it simply logs a message to the terminal. It is your responsibility to integrate with the proper system after generation.
Tracking sessions
All sessions and tokens are tracked in a separate table. This allows you to track how many sessions are active for each account. You could even expose this information to users if desired.
Note that whenever the password changes (either via reset password or directly), all tokens are deleted, and the user has to log in again on all devices.
User Enumeration attacks
A user enumeration attack allows someone to check if an email is registered in the application. The generated authentication code does not attempt to protect from such checks. For instance, when you register an account, if the email is already registered, the code will notify the user the email is already registered.
If your application is sensitive to enumeration attacks, you need to implement your own workflows, which tends to be very different from most applications, as you need to carefully balance security and user experience.
Furthermore, if you are concerned about enumeration attacks, beware of timing attacks too. For example, registering a new account typically involves additional work (such as writing to the database, sending emails, etc) compared to when an account already exists. Someone could measure the time taken to execute those additional tasks to enumerate emails. This applies to all endpoints (registration, confirmation, password recovery, etc.) that may send email, in-app notifications, etc.
Case sensitiveness
The email lookup is made to be case-insensitive. Case-insensitive lookups are the default in MySQL and MSSQL. In SQLite3 we use COLLATE NOCASE in the column definition to support it. In PostgreSQL, we use the citext extension.
Note citext is part of PostgreSQL itself and is bundled with it in most operating systems and package managers. mix phx.gen.auth takes care of creating the extension and no extra work is necessary in the majority of cases. If by any chance your package manager splits citext into a separate package, you will get an error while migrating, and you can most likely solve it by installing the postgres-contrib package.
Concurrent tests
The generated tests run concurrently if you are using a database that supports concurrent tests, which is the case of PostgreSQL.
More about mix phx.gen.auth
Check out mix phx.gen.auth for more details, such as using a different password hashing library, customizing the web module namespace, generating binary id type, configuring the default options, and using custom table names.
Additional resources
The following links have more information regarding the motivation and design of the code this generates.
	José Valim's blog post - An upcoming authentication solution for Phoenix
	The original phx_gen_auth repo (for Phoenix 1.5 applications) - This is a great resource to see discussions around decisions that have been made in earlier versions of the project.
	Original pull request on bare Phoenix app
	Original design spec

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Channels are an exciting part of Phoenix that enable soft real-time communication with and between millions of connected clients.
Some possible use cases include:
	Chat rooms and APIs for messaging apps
	Breaking news, like "a goal was scored" or "an earthquake is coming"
	Tracking trains, trucks, or race participants on a map
	Events in multiplayer games
	Monitoring sensors and controlling lights
	Notifying a browser that a page's CSS or JavaScript has changed (this is handy in development)

Conceptually, Channels are pretty simple.
First, clients connect to the server using some transport, like WebSocket. Once connected, they join one or more topics. For example, to interact with a public chat room clients may join a topic called public_chat, and to receive updates from a product with ID 7, they may need to join a topic called product_updates:7.
Clients can push messages to the topics they've joined, and can also receive messages from them. The other way around, Channel servers receive messages from their connected clients, and can push messages to them too.
Servers are able to broadcast messages to all clients subscribed to a certain topic. This is illustrated in the following diagram:
                                                                  +----------------+
                                                     +--Topic X-->| Mobile Client  |
                                                     |            +----------------+
                              +-------------------+  |
+----------------+            |                   |  |            +----------------+
| Browser Client |--Topic X-->| Phoenix Server(s) |--+--Topic X-->| Desktop Client |
+----------------+            |                   |  |            +----------------+
                              +-------------------+  |
                                                     |            +----------------+
                                                     +--Topic X-->|   IoT Client   |
                                                                  +----------------+
Broadcasts work even if the application runs on several nodes/computers. That is, if two clients have their socket connected to different application nodes and are subscribed to the same topic T, both of them will receive messages broadcasted to T. That is possible thanks to an internal PubSub mechanism.
Channels can support any kind of client: a browser, native app, smart watch, embedded device, or anything else that can connect to a network.
All the client needs is a suitable library; see the Client Libraries section below.
Each client library communicates using one of the "transports" that Channels understand.
Currently, that's either Websockets or long polling, but other transports may be added in the future.
Unlike stateless HTTP connections, Channels support long-lived connections, each backed by a lightweight BEAM process, working in parallel and maintaining its own state.
This architecture scales well; Phoenix Channels can support millions of subscribers with reasonable latency on a single box, passing hundreds of thousands of messages per second.
And that capacity can be multiplied by adding more nodes to the cluster.
The Moving Parts
Although Channels are simple to use from a client perspective, there are a number of components involved in routing messages to clients across a cluster of servers.
Let's take a look at them.
Overview
To start communicating, a client connects to a node (a Phoenix server) using a transport (e.g., Websockets or long polling) and joins one or more channels using that single network connection.
One channel server lightweight process is created per client, per topic. Each channel holds onto the %Phoenix.Socket{} and can maintain any state it needs within its socket.assigns.
Once the connection is established, each incoming message from a client is routed, based on its topic, to the correct channel server.
If the channel server asks to broadcast a message, that message is sent to the local PubSub, which sends it out to any clients connected to the same server and subscribed to that topic.
If there are other nodes in the cluster, the local PubSub also forwards the message to their PubSubs, which send it out to their own subscribers.
Because only one message has to be sent per additional node, the performance cost of adding nodes is negligible, while each new node supports many more subscribers.
The message flow looks something like this:
                                 Channel   +-------------------------+      +--------+
                                  route    | Sending Client, Topic 1 |      | Local  |
                              +----------->|     Channel.Server      |----->| PubSub |--+
+----------------+            |            +-------------------------+      +--------+  |
| Sending Client |-Transport--+                                                  |      |
+----------------+                         +-------------------------+           |      |
                                           | Sending Client, Topic 2 |           |      |
                                           |     Channel.Server      |           |      |
                                           +-------------------------+           |      |
                                                                                 |      |
                                           +-------------------------+           |      |
+----------------+                         | Browser Client, Topic 1 |           |      |
| Browser Client |<-------Transport--------|     Channel.Server      |<----------+      |
+----------------+                         +-------------------------+                  |
                                                                                        |
                                                                                        |
                                                                                        |
                                           +-------------------------+                  |
+----------------+                         |  Phone Client, Topic 1  |                  |
|  Phone Client  |<-------Transport--------|     Channel.Server      |<-+               |
+----------------+                         +-------------------------+  |   +--------+  |
                                                                        |   | Remote |  |
                                           +-------------------------+  +---| PubSub |<-+
+----------------+                         |  Watch Client, Topic 1  |  |   +--------+  |
|  Watch Client  |<-------Transport--------|     Channel.Server      |<-+               |
+----------------+                         +-------------------------+                  |
                                                                                        |
                                                                                        |
                                           +-------------------------+      +--------+  |
+----------------+                         |   IoT Client, Topic 1   |      | Remote |  |
|   IoT Client   |<-------Transport--------|     Channel.Server      |<-----| PubSub |<-+
+----------------+                         +-------------------------+      +--------+
Endpoint
In your Phoenix app's Endpoint module, a socket declaration specifies which socket handler will receive connections on a given URL.
socket "/socket", HelloWeb.UserSocket,
  websocket: true,
  longpoll: false
Phoenix comes with two default transports: websocket and longpoll. You can configure them directly via the socket declaration.
Socket Handlers
On the client side, you will establish a socket connection to the route above:
let socket = new Socket("/socket", {params: {token: window.userToken}})
On the server, Phoenix will invoke HelloWeb.UserSocket.connect/2, passing your parameters and the initial socket state. Within the socket, you can authenticate and identify a socket connection and set default socket assigns. The socket is also where you define your channel routes.
Channel Routes
Channel routes match on the topic string and dispatch matching requests to the given Channel module.
The star character * acts as a wildcard matcher, so in the following example route, requests for room:lobby and room:123 would both be dispatched to the RoomChannel. In your UserSocket, you would have:
channel "room:*", HelloWeb.RoomChannel
Channels
Channels handle events from clients, so they are similar to Controllers, but there are two key differences. Channel events can go both directions - incoming and outgoing. Channel connections also persist beyond a single request/response cycle. Channels are the highest level abstraction for real-time communication components in Phoenix.
Each Channel will implement one or more clauses of each of these four callback functions - join/3, terminate/2, handle_in/3, and handle_out/3.
Topics
Topics are string identifiers - names that the various layers use in order to make sure messages end up in the right place. As we saw above, topics can use wildcards. This allows for a useful "topic:subtopic" convention. Often, you'll compose topics using record IDs from your application layer, such as "users:123".
Messages
The Phoenix.Socket.Message module defines a struct with the following keys which denotes a valid message. From the Phoenix.Socket.Message docs.
	topic - The string topic or "topic:subtopic" pair namespace, such as "messages" or "messages:123"
	event - The string event name, for example "phx_join"
	payload - The message payload
	ref - The unique string ref

PubSub
PubSub is provided by the Phoenix.PubSub module. Interested parties can receive events by subscribing to topics. Other processes can broadcast events to certain topics.
This is useful to broadcast messages on channel and also for application development in general. For instance, letting all connected live views to know that a new comment has been added to a post.
The PubSub system takes care of getting messages from one node to another so that they can be sent to all subscribers across the cluster.
By default, this is done using Phoenix.PubSub.PG2, which uses native BEAM messaging.
If your deployment environment does not support distributed Elixir or direct communication between servers, Phoenix also ships with a Redis Adapter that uses Redis to exchange PubSub data. Please see the Phoenix.PubSub docs for more information.
Client Libraries
Any networked device can connect to Phoenix Channels as long as it has a client library.
The following libraries exist today, and new ones are always welcome.
Official
Phoenix ships with a JavaScript client that is available when generating a new Phoenix project. The documentation for the JavaScript module is available at https://hexdocs.pm/phoenix/js/; the code is in multiple js files.
3rd Party
	Swift (iOS)	SwiftPhoenix


	Java (Android)	JavaPhoenixChannels


	Kotlin (Android)	JavaPhoenixClient


	C#	PhoenixSharp


	Elixir	phoenix_gen_socket_client


	GDScript (Godot Game Engine)	GodotPhoenixChannels



Tying it all together
Let's tie all these ideas together by building a simple chat application. Make sure you created a new Phoenix application and now we are ready to generate the UserSocket.
Generating a socket
Let's invoke the socket generator to get started:
$ mix phx.gen.socket User

It will create two files, the client code in assets/js/user_socket.js and the server counter-part in lib/hello_web/channels/user_socket.ex. After running, the generator will also ask to add the following line to lib/hello_web/endpoint.ex:
defmodule HelloWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :hello

  socket "/socket", HelloWeb.UserSocket,
    websocket: true,
    longpoll: false

  ...
end
The generator also asks us to import the client code, we will do that later.
Next, we will configure our socket to ensure messages get routed to the correct channel. To do that, we'll uncomment the "room:*" channel definition:
defmodule HelloWeb.UserSocket do
  use Phoenix.Socket

  ## Channels
  channel "room:*", HelloWeb.RoomChannel
  ...
Now, whenever a client sends a message whose topic starts with "room:", it will be routed to our RoomChannel. Next, we'll define a HelloWeb.RoomChannel module to manage our chat room messages.
Joining Channels
The first priority of your channels is to authorize clients to join a given topic. For authorization, we must implement join/3 in lib/hello_web/channels/room_channel.ex.
defmodule HelloWeb.RoomChannel do
  use Phoenix.Channel

  def join("room:lobby", _message, socket) do
    {:ok, socket}
  end

  def join("room:" <> _private_room_id, _params, _socket) do
    {:error, %{reason: "unauthorized"}}
  end
end
For our chat app, we'll allow anyone to join the "room:lobby" topic, but any other room will be considered private and special authorization, say from a database, will be required.
(We won't worry about private chat rooms for this exercise, but feel free to explore after we finish.)
With our channel in place, let's get the client and server talking.
The generated assets/js/user_socket.js defines a simple client based on the socket implementation that ships with Phoenix.
We can use that library to connect to our socket and join our channel, we just need to set our room name to "room:lobby" in that file.
// assets/js/user_socket.js
// ...
socket.connect()

// Now that you are connected, you can join channels with a topic:
let channel = socket.channel("room:lobby", {})
channel.join()
  .receive("ok", resp => { console.log("Joined successfully", resp) })
  .receive("error", resp => { console.log("Unable to join", resp) })

export default socket
After that, we need to make sure assets/js/user_socket.js gets imported into our application JavaScript file. To do that, uncomment this line in assets/js/app.js.
// ...
import "./user_socket.js"
Save the file and your browser should auto refresh, thanks to the Phoenix live reloader. If everything worked, we should see "Joined successfully" in the browser's JavaScript console. Our client and server are now talking over a persistent connection. Now let's make it useful by enabling chat.
In lib/hello_web/templates/page/index.html.heex, we'll replace the existing code with a container to hold our chat messages, and an input field to send them:
<div id="messages" role="log" aria-live="polite"></div>
<input id="chat-input" type="text">
Now let's add a couple of event listeners to assets/js/user_socket.js:
// ...
let channel           = socket.channel("room:lobby", {})
let chatInput         = document.querySelector("#chat-input")
let messagesContainer = document.querySelector("#messages")

chatInput.addEventListener("keypress", event => {
  if(event.key === 'Enter'){
    channel.push("new_msg", {body: chatInput.value})
    chatInput.value = ""
  }
})

channel.join()
  .receive("ok", resp => { console.log("Joined successfully", resp) })
  .receive("error", resp => { console.log("Unable to join", resp) })

export default socket
All we had to do is detect that enter was pressed and then push an event over the channel with the message body. We named the event "new_msg". With this in place, let's handle the other piece of a chat application, where we listen for new messages and append them to our messages container.
// ...
let channel           = socket.channel("room:lobby", {})
let chatInput         = document.querySelector("#chat-input")
let messagesContainer = document.querySelector("#messages")

chatInput.addEventListener("keypress", event => {
  if(event.key === 'Enter'){
    channel.push("new_msg", {body: chatInput.value})
    chatInput.value = ""
  }
})

channel.on("new_msg", payload => {
  let messageItem = document.createElement("p")
  messageItem.innerText = `[${Date()}] ${payload.body}`
  messagesContainer.appendChild(messageItem)
})

channel.join()
  .receive("ok", resp => { console.log("Joined successfully", resp) })
  .receive("error", resp => { console.log("Unable to join", resp) })

export default socket
We listen for the "new_msg" event using channel.on, and then append the message body to the DOM. Now let's handle the incoming and outgoing events on the server to complete the picture.
Incoming Events
We handle incoming events with handle_in/3. We can pattern match on the event names, like "new_msg", and then grab the payload that the client passed over the channel. For our chat application, we simply need to notify all other room:lobby subscribers of the new message with broadcast!/3.
defmodule HelloWeb.RoomChannel do
  use Phoenix.Channel

  def join("room:lobby", _message, socket) do
    {:ok, socket}
  end

  def join("room:" <> _private_room_id, _params, _socket) do
    {:error, %{reason: "unauthorized"}}
  end

  def handle_in("new_msg", %{"body" => body}, socket) do
    broadcast!(socket, "new_msg", %{body: body})
    {:noreply, socket}
  end
end
broadcast!/3 will notify all joined clients on this socket's topic and invoke their handle_out/3 callbacks. handle_out/3 isn't a required callback, but it allows us to customize and filter broadcasts before they reach each client. By default, handle_out/3 is implemented for us and simply pushes the message on to the client. Hooking into outgoing events allows for powerful message customization and filtering. Let's see how.
Intercepting Outgoing Events
We won't implement this for our application, but imagine our chat app allowed users to ignore messages about new users joining a room. We could implement that behavior like this, where we explicitly tell Phoenix which outgoing event we want to intercept and then define a handle_out/3 callback for those events. (Of course, this assumes that we have an Accounts context with an ignoring_user?/2 function, and that we pass a user in via the assigns map). It is important to note that the handle_out/3 callback will be called for every recipient of a message, so more expensive operations like hitting the database should be considered carefully before being included in handle_out/3.
intercept ["user_joined"]

def handle_out("user_joined", msg, socket) do
  if Accounts.ignoring_user?(socket.assigns[:user], msg.user_id) do
    {:noreply, socket}
  else
    push(socket, "user_joined", msg)
    {:noreply, socket}
  end
end
That's all there is to our basic chat app. Fire up multiple browser tabs and you should see your messages being pushed and broadcasted to all windows!
Using Token Authentication
When we connect, we'll often need to authenticate the client. Fortunately, this is a 4-step process with Phoenix.Token.
Step 1 - Assign a Token in the Connection
Let's say we have an authentication plug in our app called OurAuth. When OurAuth authenticates a user, it sets a value for the :current_user key in conn.assigns. Since the current_user exists, we can simply assign the user's token in the connection for use in the layout. We can wrap that behavior up in a private function plug, put_user_token/2. This could also be put in its own module as well. To make this all work, we just add OurAuth and put_user_token/2 to the browser pipeline.
pipeline :browser do
  ...
  plug OurAuth
  plug :put_user_token
end

defp put_user_token(conn, _) do
  if current_user = conn.assigns[:current_user] do
    token = Phoenix.Token.sign(conn, "user socket", current_user.id)
    assign(conn, :user_token, token)
  else
    conn
  end
end
Now our conn.assigns contains the current_user and user_token.
Step 2 - Pass the Token to the JavaScript
Next, we need to pass this token to JavaScript. We can do so inside a script tag in web/templates/layout/app.html.heex right above the app.js script, as follows:
<script>window.userToken = "<%= assigns[:user_token] %>";</script>
<script src={Routes.static_path(@conn, "/assets/app.js")}></script>
Step 3 - Pass the Token to the Socket Constructor and Verify
We also need to pass the :params to the socket constructor and verify the user token in the connect/3 function. To do so, edit web/channels/user_socket.ex, as follows:
def connect(%{"token" => token}, socket, _connect_info) do
  # max_age: 1209600 is equivalent to two weeks in seconds
  case Phoenix.Token.verify(socket, "user socket", token, max_age: 1209600) do
    {:ok, user_id} ->
      {:ok, assign(socket, :current_user, user_id)}
    {:error, reason} ->
      :error
  end
end
In our JavaScript, we can use the token set previously when constructing the Socket:
let socket = new Socket("/socket", {params: {token: window.userToken}})
We used Phoenix.Token.verify/4 to verify the user token provided by the client. Phoenix.Token.verify/4 returns either {:ok, user_id} or {:error, reason}. We can pattern match on that return in a case statement. With a verified token, we set the user's id as the value to :current_user in the socket. Otherwise, we return :error.
Step 4 - Connect to the socket in JavaScript
With authentication set up, we can connect to sockets and channels from JavaScript.
let socket = new Socket("/socket", {params: {token: window.userToken}})
socket.connect()
Now that we are connected, we can join channels with a topic:
let channel = socket.channel("topic:subtopic", {})
channel.join()
  .receive("ok", resp => { console.log("Joined successfully", resp) })
  .receive("error", resp => { console.log("Unable to join", resp) })

export default socket
Note that token authentication is preferable since it's transport agnostic and well-suited for long running-connections like channels, as opposed to using sessions or authentication approaches.
Fault Tolerance and Reliability Guarantees
Servers restart, networks split, and clients lose connectivity. In order to design robust systems, we need to understand how Phoenix responds to these events and what guarantees it offers.
Handling Reconnection
Clients subscribe to topics, and Phoenix stores those subscriptions in an in-memory ETS table. If a channel crashes, the clients will need to reconnect to the topics they had previously subscribed to. Fortunately, the Phoenix JavaScript client knows how to do this. The server will notify all the clients of the crash. This will trigger each client's Channel.onError callback. The clients will attempt to reconnect to the server using an exponential backoff strategy. Once they reconnect, they'll attempt to rejoin the topics they had previously subscribed to. If they are successful, they'll start receiving messages from those topics as before.
Resending Client Messages
Channel clients queue outgoing messages into a PushBuffer, and send them to the server when there is a connection. If no connection is available, the client holds on to the messages until it can establish a new connection. With no connection, the client will hold the messages in memory until it establishes a connection, or until it receives a timeout event. The default timeout is set to 5000 milliseconds. The client won't persist the messages in the browser's local storage, so if the browser tab closes, the messages will be gone.
Resending Server Messages
Phoenix uses an at-most-once strategy when sending messages to clients. If the client is offline and misses the message, Phoenix won't resend it. Phoenix doesn't persist messages on the server. If the server restarts, unsent messages will be gone. If our application needs stronger guarantees around message delivery, we'll need to write that code ourselves. Common approaches involve persisting messages on the server and having clients request missing messages. For an example, see Chris McCord's Phoenix training: client code and server code.
Example Application
To see an example of the application we just built, checkout the project phoenix_chat_example.
You can also see a live demo at https://phoenixchat.herokuapp.com/.

Presence
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Channels guide.

Phoenix Presence is a feature which allows you to register process information on a topic and replicate it transparently across a cluster. It's a combination of both a server-side and client-side library, which makes it simple to implement. A simple use-case would be showing which users are currently online in an application.
Phoenix Presence is special for a number of reasons. It has no single point of failure, no single source of truth, relies entirely on the standard library with no operational dependencies and self-heals.
Setting up
We are going to use Presence to track which users are connected on the server and send updates to the client as users join and leave. We will deliver those updates via Phoenix Channels. Therefore, let's create a RoomChannel, as we did in the channels guides:
$ mix phx.gen.channel Room

Follow the steps after the generator and you are ready to start tracking presence.
The Presence generator
To get started with Presence, we'll first need to generate a presence module. We can do this with the mix phx.gen.presence task:
$ mix phx.gen.presence
* creating lib/hello_web/channels/presence.ex

Add your new module to your supervision tree,
in lib/hello/application.ex:

    children = [
      ...
      HelloWeb.Presence,
    ]

You're all set! See the Phoenix.Presence docs for more details:
https://hexdocs.pm/phoenix/Phoenix.Presence.html

If we open up the lib/hello_web/channels/presence.ex file, we will see the following line:
use Phoenix.Presence,
  otp_app: :hello,
  pubsub_server: Hello.PubSub
This sets up the module for presence, defining the functions we require for tracking presences. As mentioned in the generator task, we should add this module to our supervision tree in
application.ex:
children = [
  ...
  HelloWeb.Presence,
]
Next, we will create the channel that we'll communicate presence over. After a user joins, we can push the list of presences down the channel and then track the connection. We can also provide a map of additional information to track.
defmodule HelloWeb.RoomChannel do
  use Phoenix.Channel
  alias HelloWeb.Presence

  def join("room:lobby", %{"name" => name}, socket) do
    send(self(), :after_join)
    {:ok, assign(socket, :name, name)}
  end

  def handle_info(:after_join, socket) do
    {:ok, _} =
      Presence.track(socket, socket.assigns.name, %{
        online_at: inspect(System.system_time(:second))
      })

    push(socket, "presence_state", Presence.list(socket))
    {:noreply, socket}
  end
end
Finally, we can use the client-side Presence library included in phoenix.js to manage the state and presence diffs that come down the socket. It listens for the "presence_state" and "presence_diff" events and provides a simple callback for you to handle the events as they happen, with the onSync callback.
The onSync callback allows you to easily react to presence state changes, which most often results in re-rendering an updated list of active users. You can use the list method to format and return each individual presence based on the needs of your application.
To iterate users, we use the presences.list() function which accepts a callback. The callback will be called for each presence item with 2 arguments, the presence id and a list of metas (one for each presence for that presence id). We use this to display the users and the number of devices they are online with.
We can see presence working by adding the following to assets/js/app.js:
import {Socket, Presence} from "phoenix"

let socket = new Socket("/socket", {params: {token: window.userToken}})
let channel = socket.channel("room:lobby", {name: window.location.search.split("=")[1])
let presence = new Presence(channel)

function renderOnlineUsers(presence) {
  let response = ""

  presence.list((id, {metas: [first, ...rest]}) => {
    let count = rest.length + 1
    response += `<br>${id} (count: ${count})</br>`
  })

  document.querySelector("main[role=main]").innerHTML = response
}

socket.connect()

presence.onSync(() => renderOnlineUsers(presence))

channel.join()
We can ensure this is working by opening 3 browser tabs. If we navigate to http://localhost:4000/?name=Alice on two browser tabs and http://localhost:4000/?name=Bob then we should see:
Alice (count: 2)
Bob (count: 1)
If we close one of the Alice tabs, then the count should decrease to 1. If we close another tab, the user should disappear from the list entirely.
Making it safe
In our initial implementation, we are passing the name of the user as part of the URL. However, in many systems, you want to allow only logged in users to access the presence functionality. To do so, you should set up token authentication, as detailed in the token authentication section of the channels guide.
With token authentication, you should access socket.assigns.user_id, set in UserSocket, instead of socket.assigns.name set from parameters.

Introduction to Testing
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Testing has become integral to the software development process, and the ability to easily write meaningful tests is an indispensable feature for any modern web framework. Phoenix takes this seriously, providing support files to make all the major components of the framework easy to test. It also generates test modules with real-world examples alongside any generated modules to help get us going.
Elixir ships with a built-in testing framework called ExUnit. ExUnit strives to be clear and explicit, keeping magic to a minimum. Phoenix uses ExUnit for all of its testing, and we will use it here as well.
Running tests
When Phoenix generates a web application for us, it also includes tests. To run them, simply type mix test:
$ mix test
....

Finished in 0.09 seconds
3 tests, 0 failures

Randomized with seed 652656

We already have three tests!
In fact, we already have a directory structure completely set up for testing, including a test helper and support files.
test
├── hello_web
│   ├── channels
│   ├── controllers
│   │   └── page_controller_test.exs
│   └── views
│       ├── error_view_test.exs
│       ├── layout_view_test.exs
│       └── page_view_test.exs
├── support
│   ├── channel_case.ex
│   ├── conn_case.ex
│   └── data_case.ex
└── test_helper.exs

The test cases we get for free include test/hello_web/controllers/page_controller_test.exs, test/hello_web/views/error_view_test.exs, and test/hello_web/views/page_view_test.exs. They are testing our controllers and views. If you haven't read the guides for controllers and views, now is a good time.
Understanding test modules
We are going to use the next sections to get acquainted with Phoenix testing structure. We will start with the three test files generated by Phoenix.
The first test file we'll look at is test/hello_web/controllers/page_controller_test.exs.
defmodule HelloWeb.PageControllerTest do
  use HelloWeb.ConnCase

  test "GET /", %{conn: conn} do
    conn = get(conn, "/")
    assert html_response(conn, 200) =~ "Welcome to Phoenix!"
  end
end
There are a couple of interesting things happening here.
Our test files simply define modules. At the top of each module, you will find a line such as:
use HelloWeb.ConnCase
If you were to write an Elixir library, outside of Phoenix, instead of use HelloWeb.ConnCase you would write use ExUnit.Case. However, Phoenix already ships with a bunch of functionality for testing controllers and HelloWeb.ConnCase builds on top of ExUnit.Case to bring these functionalities in. We will explore the HelloWeb.ConnCase module soon.
Then we define each test using the test/3 macro. The test/3 macro receives three arguments: the test name, the testing context that we are pattern matching on, and the contents of the test. In this test, we access the root page of our application by a "GET" HTTP request on the path "/" with the get/2 macro. Then we assert that the rendered page contains the string "Welcome to Phoenix!".
When writing tests in Elixir, we use assertions to check that something is true. In our case, assert html_response(conn, 200) =~ "Welcome to Phoenix!" is doing a couple things:
	It asserts that conn has rendered a response
	It asserts that the response has the 200 status code (which means OK in HTTP parlance)
	It asserts that the type of the response is HTML
	It asserts that the result of html_response(conn, 200), which is an HTML response, has the string "Welcome to Phoenix!" in it

However, from where does the conn we use on get and html_response come from? To answer this question, let's take a look at HelloWeb.ConnCase.
The ConnCase
If you open up test/support/conn_case.ex, you will find this (with comments removed):
defmodule HelloWeb.ConnCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      # Import conveniences for testing with connections
      import Plug.Conn
      import Phoenix.ConnTest
      alias HelloWeb.Router.Helpers, as: Routes

      # The default endpoint for testing
      @endpoint HelloWeb.Endpoint
    end
  end

  setup tags do
    pid = Ecto.Adapters.SQL.Sandbox.start_owner!(Demo.Repo, shared: not tags[:async])
    on_exit(fn -> Ecto.Adapters.SQL.Sandbox.stop_owner(pid) end)
    %{conn: Phoenix.ConnTest.build_conn()}
  end
end
There is a lot to unpack here.
The second line says this is a case template. This is a ExUnit feature that allows developers to replace the built-in use ExUnit.Case by their own case. This line is pretty much what allows us to write use HelloWeb.ConnCase at the top of our controller tests.
Now that we have made this module a case template, we can define callbacks that are invoked on certain occasions. The using callback defines code to be injected on every module that calls use HelloWeb.ConnCase. In this case, we import Plug.Conn, so all of the connection helpers available in controllers are also available in tests, and then imports Phoenix.ConnTest. You can consult these modules to learn all functionality available.
Then it aliases the module with all path helpers, so we can easily generate URLs in our tests. Finally, it sets the @endpoint module attribute with the name of our endpoint.
Then our case template defines a setup block. The setup block will be called before test. Most of the setup block is on setting up the SQL Sandbox, which we will talk about it later. In the last line of the setup block, we will find this:
%{conn: Phoenix.ConnTest.build_conn()}
The last line of setup can return test metadata that will be available in each test. The metadata we are passing forward here is a newly built Plug.Conn. In our test, we extract the connection out of this metadata at the very beginning of our test:
test "GET /", %{conn: conn} do
And that's where the connection comes from! At first, the testing structure does come with a bit of indirection, but this indirection pays off as our test suite grows, since it allows us to cut down the amount of boilerplate.
View tests
The other test files in our application are responsible for testing our views.
The error view test case, test/hello_web/views/error_view_test.exs, illustrates a few interesting things of its own.
defmodule HelloWeb.ErrorViewTest do
  use HelloWeb.ConnCase, async: true

  # Bring render/3 and render_to_string/3 for testing custom views
  import Phoenix.View

  test "renders 404.html" do
    assert render_to_string(HelloWeb.ErrorView, "404.html", []) ==
           "Not Found"
  end

  test "renders 500.html" do
    assert render_to_string(HelloWeb.ErrorView, "500.html", []) ==
           "Internal Server Error"
  end
end
HelloWeb.ErrorViewTest sets async: true which means that this test case will be run in parallel with other test cases. While individual tests within the case still run serially, this can greatly increase overall test speeds.
It also imports Phoenix.View in order to use the render_to_string/3 function. With that, all the assertions can be simple string equality tests.
The page view case, test/hello_web/views/page_view_test.exs, does not contain any tests by default, but it is here for us when we need to add functions to our HelloWeb.PageView module.
defmodule HelloWeb.PageViewTest do
  use HelloWeb.ConnCase, async: true
end
Running tests per directory/file
Now that we have an idea what our tests are doing, let's look at different ways to run them.
As we saw near the beginning of this guide, we can run our entire suite of tests with mix test.
$ mix test
....

Finished in 0.2 seconds
3 tests, 0 failures

Randomized with seed 540755

If we would like to run all the tests in a given directory, test/hello_web/controllers for instance, we can pass the path to that directory to mix test.
$ mix test test/hello_web/controllers/
.

Finished in 0.2 seconds
1 tests, 0 failures

Randomized with seed 652376

In order to run all the tests in a specific file, we can pass the path to that file into mix test.
$ mix test test/hello_web/views/error_view_test.exs
...

Finished in 0.2 seconds
2 tests, 0 failures

Randomized with seed 220535

And we can run a single test in a file by appending a colon and a line number to the filename.
Let's say we only wanted to run the test for the way HelloWeb.ErrorView renders 500.html. The test begins on line 11 of the file, so this is how we would do it.
$ mix test test/hello_web/views/error_view_test.exs:11
Including tags: [line: "11"]
Excluding tags: [:test]

.

Finished in 0.1 seconds
2 tests, 0 failures, 1 excluded

Randomized with seed 288117

We chose to run this specifying the first line of the test, but actually, any line of that test will do. These line numbers would all work - :11, :12, or :13.
Running tests using tags
ExUnit allows us to tag our tests individually or for the whole module. We can then choose to run only the tests with a specific tag, or we can exclude tests with that tag and run everything else.
Let's experiment with how this works.
First, we'll add a @moduletag to test/hello_web/views/error_view_test.exs.
defmodule HelloWeb.ErrorViewTest do
  use HelloWeb.ConnCase, async: true

  @moduletag :error_view_case
  ...
end
If we use only an atom for our module tag, ExUnit assumes that it has a value of true. We could also specify a different value if we wanted.
defmodule HelloWeb.ErrorViewTest do
  use HelloWeb.ConnCase, async: true

  @moduletag error_view_case: "some_interesting_value"
  ...
end
For now, let's leave it as a simple atom @moduletag :error_view_case.
We can run only the tests from the error view case by passing --only error_view_case into mix test.
$ mix test --only error_view_case
Including tags: [:error_view_case]
Excluding tags: [:test]

...

Finished in 0.1 seconds
3 tests, 0 failures, 1 excluded

Randomized with seed 125659

Note: ExUnit tells us exactly which tags it is including and excluding for each test run. If we look back to the previous section on running tests, we'll see that line numbers specified for individual tests are actually treated as tags.

$ mix test test/hello_web/views/error_view_test.exs:11
Including tags: [line: "11"]
Excluding tags: [:test]

.

Finished in 0.2 seconds
2 tests, 0 failures, 1 excluded

Randomized with seed 364723

Specifying a value of true for error_view_case yields the same results.
$ mix test --only error_view_case:true
Including tags: [error_view_case: "true"]
Excluding tags: [:test]

...

Finished in 0.1 seconds
3 tests, 0 failures, 1 excluded

Randomized with seed 833356

Specifying false as the value for error_view_case, however, will not run any tests because no tags in our system match error_view_case: false.
$ mix test --only error_view_case:false
Including tags: [error_view_case: "false"]
Excluding tags: [:test]



Finished in 0.1 seconds
3 tests, 0 failures, 3 excluded

Randomized with seed 622422
The --only option was given to "mix test" but no test executed

We can use the --exclude flag in a similar way. This will run all of the tests except those in the error view case.
$ mix test --exclude error_view_case
Excluding tags: [:error_view_case]

.

Finished in 0.2 seconds
3 tests, 0 failures, 2 excluded

Randomized with seed 682868

Specifying values for a tag works the same way for --exclude as it does for --only.
We can tag individual tests as well as full test cases. Let's tag a few tests in the error view case to see how this works.
defmodule HelloWeb.ErrorViewTest do
  use HelloWeb.ConnCase, async: true

  @moduletag :error_view_case

  # Bring render/3 and render_to_string/3 for testing custom views
  import Phoenix.View

  @tag individual_test: "yup"
  test "renders 404.html" do
    assert render_to_string(HelloWeb.ErrorView, "404.html", []) ==
           "Not Found"
  end

  @tag individual_test: "nope"
  test "renders 500.html" do
    assert render_to_string(HelloWeb.ErrorView, "500.html", []) ==
           "Internal Server Error"
  end
end
If we would like to run only tests tagged as individual_test, regardless of their value, this will work.
$ mix test --only individual_test
Including tags: [:individual_test]
Excluding tags: [:test]

..

Finished in 0.1 seconds
3 tests, 0 failures, 1 excluded

Randomized with seed 813729

We can also specify a value and run only tests with that.
$ mix test --only individual_test:yup
Including tags: [individual_test: "yup"]
Excluding tags: [:test]

.

Finished in 0.1 seconds
3 tests, 0 failures, 2 excluded

Randomized with seed 770938

Similarly, we can run all tests except for those tagged with a given value.
$ mix test --exclude individual_test:nope
Excluding tags: [individual_test: "nope"]

...

Finished in 0.2 seconds
3 tests, 0 failures, 1 excluded

Randomized with seed 539324

We can be more specific and exclude all the tests from the error view case except the one tagged with individual_test that has the value "yup".
$ mix test --exclude error_view_case --include individual_test:yup
Including tags: [individual_test: "yup"]
Excluding tags: [:error_view_case]

..

Finished in 0.2 seconds
3 tests, 0 failures, 1 excluded

Randomized with seed 61472

Finally, we can configure ExUnit to exclude tags by default. The default ExUnit configuration is done in the test/test_helper.exs file:
ExUnit.start(exclude: [error_view_case: true])

Ecto.Adapters.SQL.Sandbox.mode(Hello.Repo, :manual)
Now when we run mix test, it only runs one spec from our page_controller_test.exs.
$ mix test
Excluding tags: [error_view_case: true]

.

Finished in 0.2 seconds
3 tests, 0 failures, 2 excluded

Randomized with seed 186055

We can override this behavior with the --include flag, telling mix test to include tests tagged with error_view_case.
$ mix test --include error_view_case
Including tags: [:error_view_case]
Excluding tags: [error_view_case: true]

....

Finished in 0.2 seconds
3 tests, 0 failures

Randomized with seed 748424

This technique can be very useful to control very long running tests, which you may only want to run in CI or in specific scenarios.
Randomization
Running tests in random order is a good way to ensure that our tests are truly isolated. If we notice that we get sporadic failures for a given test, it may be because a previous test changes the state of the system in ways that aren't cleaned up afterward, thereby affecting the tests which follow. Those failures might only present themselves if the tests are run in a specific order.
ExUnit will randomize the order tests run in by default, using an integer to seed the randomization. If we notice that a specific random seed triggers our intermittent failure, we can re-run the tests with that same seed to reliably recreate that test sequence in order to help us figure out what the problem is.
$ mix test --seed 401472
....

Finished in 0.2 seconds
3 tests, 0 failures

Randomized with seed 401472

Concurrency and partitioning
As we have seen, ExUnit allows developers to run tests concurrently. This allows developers to use all of the power in their machine to run their test suites as fast as possible. Couple this with Phoenix performance, most test suites compile and run in a fraction of the time compared to other frameworks.
While developers usually have powerful machines available to them during development, this may not always be the case in your Continuous Integration servers. For this reason, ExUnit also supports out of the box test partitioning in test environments. If you open up your config/test.exs, you will find the database name set to:
database: "hello_test#{System.get_env("MIX_TEST_PARTITION")}",
By default, the MIX_TEST_PARTITION environment variable has no value, and therefore it has no effect. But in your CI server, you can, for example, split your test suite across machines by using four distinct commands:
$ MIX_TEST_PARTITION=1 mix test --partitions 4
$ MIX_TEST_PARTITION=2 mix test --partitions 4
$ MIX_TEST_PARTITION=3 mix test --partitions 4
$ MIX_TEST_PARTITION=4 mix test --partitions 4

That's all you need to do and ExUnit and Phoenix will take care of all rest, including setting up the database for each distinct partition with a distinct name.
Going further
While ExUnit is a simple test framework, it provides a really flexible and robust test runner through the mix test command. We recommend you to run mix help test or read the docs online
We've seen what Phoenix gives us with a newly generated app. Furthermore, whenever you generate a new resource, Phoenix will generate all appropriate tests for that resource too. For example, you can create a complete scaffold with schema, context, controllers, and views by running the following command at the root of your application:
$ mix phx.gen.html Blog Post posts title body:text
* creating lib/demo_web/controllers/post_controller.ex
* creating lib/demo_web/templates/post/edit.html.heex
* creating lib/demo_web/templates/post/form.html.heex
* creating lib/demo_web/templates/post/index.html.heex
* creating lib/demo_web/templates/post/new.html.heex
* creating lib/demo_web/templates/post/show.html.heex
* creating lib/demo_web/views/post_view.ex
* creating test/demo_web/controllers/post_controller_test.exs
* creating lib/demo/blog/post.ex
* creating priv/repo/migrations/20200215122336_create_posts.exs
* creating lib/demo/blog.ex
* injecting lib/demo/blog.ex
* creating test/demo/blog_test.exs
* injecting test/demo/blog_test.exs

Add the resource to your browser scope in lib/demo_web/router.ex:

    resources "/posts", PostController


Remember to update your repository by running migrations:

    $ mix ecto.migrate


Now let's follow the directions and add the new resources route to our lib/hello_web/router.ex file and run the migrations.
When we run mix test again, we see that we now have nineteen tests!
$ mix test
................

Finished in 0.1 seconds
19 tests, 0 failures

Randomized with seed 537537

At this point, we are at a great place to transition to the rest of the testing guides, in which we'll examine these tests in much more detail, and add some of our own.

Testing Contexts
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Introduction to Testing guide.

Requirement: This guide expects that you have gone through the Contexts guide.

At the end of the Introduction to Testing guide, we generated an HTML resource for posts using the following command:
$ mix phx.gen.html Blog Post posts title body:text

This gave us a number of modules for free, including a Blog context and a Post schema, alongside their respective test files. As we have learned in the Context guide, the Blog context is simply a module with functions to a particular area of our business domain, while Post schema maps to a particular table in our database.
In this guide, we are going to explore the tests generated for our contexts and schemas. Before we do anything else, let's run mix test to make sure our test suite runs cleanly.
$ mix test
................

Finished in 0.6 seconds
19 tests, 0 failures

Randomized with seed 638414

Great. We've got nineteen tests and they are all passing!
Testing posts
If you open up test/hello/blog_test.exs, you will see a file with the following:
defmodule Hello.BlogTest do
  use Hello.DataCase

  alias Hello.Blog

  describe "posts" do
    alias Hello.Blog.Post

    @valid_attrs %{body: "some body", title: "some title"}
    @update_attrs %{body: "some updated body", title: "some updated title"}
    @invalid_attrs %{body: nil, title: nil}

    def post_fixture(attrs \\ %{}) do
      {:ok, post} =
        attrs
        |> Enum.into(@valid_attrs)
        |> Blog.create_post()

      post
    end

    test "list_posts/0 returns all posts" do
      post = post_fixture()
      assert Blog.list_posts() == [post]
    end

    ...
As the top of the file we import Hello.DataCase, which as we will see soon, it is similar to HelloWeb.ConnCase. While HelloWeb.ConnCase sets up helpers for working with connections, which is useful when testing controllers and views, Hello.DataCase provides functionality for working with contexts and schemas.
Next, we define an alias, so we can refer to Hello.Blog simply as Blog.
Then we start a describe "posts" block. A describe block is a feature in ExUnit that allows us to group similar tests. The reason why we have grouped all post related tests together is because contexts in Phoenix are capable of grouping multiple schemas together. For example, if we ran this command:
$ mix phx.gen.html Blog Comment comments post_id:references:posts body:text

We will get a bunch of new functions in the Hello.Blog context, plus a whole new describe "comments" block in our test file.
The tests defined for our context are very straight-forward. They call the functions in our context and assert on their results. As you can see, some of those tests even create entries in the database:
test "create_post/1 with valid data creates a post" do
  assert {:ok, %Post{} = post} = Blog.create_post(@valid_attrs)
  assert post.body == "some body"
  assert post.title == "some title"
end
At this point, you may wonder: how can Phoenix make sure the data created in one of the tests do not affect other tests? We are glad you asked. To answer this question, let's talk about the DataCase.
The DataCase
If you open up test/support/data_case.ex, you will find the following:
defmodule Hello.DataCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      alias Hello.Repo

      import Ecto
      import Ecto.Changeset
      import Ecto.Query
      import Hello.DataCase
    end
  end

  setup tags do
    pid = Ecto.Adapters.SQL.Sandbox.start_owner!(Demo.Repo, shared: not tags[:async])
    on_exit(fn -> Ecto.Adapters.SQL.Sandbox.stop_owner(pid) end)
    :ok
  end

  def errors_on(changeset) do
    ...
  end
end
Hello.DataCase is another ExUnit.CaseTemplate. In the using block, we can see all of the aliases and imports DataCase brings into our tests. The setup chunk for DataCase is very similar to the one from ConnCase. As we can see, most of the setup block revolves around setting up a SQL Sandbox.
The SQL Sandbox is precisely what allows our tests to write to the database without affecting any of the other tests. In a nutshell, at the beginning of every test, we start a transaction in the database. When the test is over, we automatically rollback the transaction, effectively erasing all of the data created in the test.
Furthermore, the SQL Sandbox allows multiple tests to run concurrently, even if they talk to the database. This feature is provided for PostgreSQL databases and it can be used to further speed up your contexts and controllers tests by adding a async: true flag when using them:
use Hello.DataCase, async: true
There are some considerations you need to have in mind when running asynchronous tests with the sandbox, so please refer to the Ecto.Adapters.SQL.Sandbox for more information.
Finally at the end of the of the DataCase module we can find a function named errors_on with some examples of how to use it. This function is used for testing any validation we may want to add to our schemas. Let's give it a try by adding our own validations and then testing them.
Testing schemas
When we generate our HTML Post resource, Phoenix generated a Blog context and a Post schema. It generated a test file for the context, but no test file for the schema. However, this doesn't mean we don't need to test the schema, it just means we did not have to test the schema so far.
You may be wondering then: when do we test the context directly and when do we test the schema directly? The answer to this question is the same answer to the question of when do we add code to a context and when do we add it to the schema?
The general guideline is to keep all side-effect free code in the schema. In other words, if you are simply working with data structures, schemas and changesets, put it in the schema. The context will typically have the code that creates and updates schemas and then write them to a database or an API.
We'll be adding additional validations to the schema module, so that's a great opportunity to write some schema specific tests. Open up lib/hello/blog/post.ex and add the following validation to def changeset:
def changeset(post, attrs) do
  post
  |> cast(attrs, [:title, :body])
  |> validate_required([:title, :body])
  |> validate_length(:title, min: 2)
end
The new validation says the title needs to have at least 2 characters. Let's write a test for this. Create a new file at test/hello/blog/post_test.exs with this:
defmodule Hello.Blog.PostTest do
  use Hello.DataCase, async: true
  alias Hello.Blog.Post

  test "title must be at least two characters long" do
    changeset = Post.changeset(%Post{}, %{title: "I"})
    assert %{title: ["should be at least 2 character(s)"]} = errors_on(changeset)
  end
end
And that's it. As our business domain grows, we have well-defined places to test our contexts and schemas.

Testing Controllers
    

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Introduction to Testing guide.

At the end of the Introduction to Testing guide, we generated an HTML resource for posts using the following command:
$ mix phx.gen.html Blog Post posts title body:text

This gave us a number of modules for free, including a PostController and the associated tests. We are going to explore those tests to learn more about testing controllers in general. At the end of the guide, we will generate a JSON resource, and explore how our API tests look like.
HTML controller tests
If you open up test/hello_web/controllers/post_controller_test.exs, you will find the following:
defmodule HelloWeb.PostControllerTest do
  use HelloWeb.ConnCase

  alias Hello.Blog

  @create_attrs %{body: "some body", title: "some title"}
  @update_attrs %{body: "some updated body", title: "some updated title"}
  @invalid_attrs %{body: nil, title: nil}

  def fixture(:post) do
    {:ok, post} = Blog.create_post(@create_attrs)
    post
  end

  ...
Similar to the PageControllerTest that ships with our application, this controller tests uses use HelloWeb.ConnCase to setup the testing structure. Then, as usual, it defines some aliases, some module attributes to use throughout testing, and then it starts a series of describe blocks, each of them to test a different controller action.
The index action
The first describe block is for the index action. The action itself is implemented like this in lib/hello_web/controllers/post_controller.ex:
def index(conn, _params) do
  posts = Blog.list_posts()
  render(conn, "index.html", posts: posts)
end
It gets all posts and renders the "index.html" template. The template can be found in lib/hello_web/templates/page/index.html.heex.
The test looks like this:
describe "index" do
  test "lists all posts", %{conn: conn} do
    conn = get(conn, Routes.post_path(conn, :index))
    assert html_response(conn, 200) =~ "Listing Posts"
  end
end
The test for the index page is quite straight-forward. It uses the get/2 helper to make a request to the "/posts" page, returned by Routes.post_path(conn, :index), then we assert we got a successful HTML response and match on its contents.
The create action
The next test we will look at is the one for the create action. The create action implementation is this:
def create(conn, %{"post" => post_params}) do
  case Blog.create_post(post_params) do
    {:ok, post} ->
      conn
      |> put_flash(:info, "Post created successfully.")
      |> redirect(to: Routes.post_path(conn, :show, post))

    {:error, %Ecto.Changeset{} = changeset} ->
      render(conn, "new.html", changeset: changeset)
  end
end
Since there are two possible outcomes for the create, we will have at least two tests:
describe "create post" do
  test "redirects to show when data is valid", %{conn: conn} do
    conn = post(conn, Routes.post_path(conn, :create), post: @create_attrs)

    assert %{id: id} = redirected_params(conn)
    assert redirected_to(conn) == Routes.post_path(conn, :show, id)

    conn = get(conn, Routes.post_path(conn, :show, id))
    assert html_response(conn, 200) =~ "Show Post"
  end

  test "renders errors when data is invalid", %{conn: conn} do
    conn = post(conn, Routes.post_path(conn, :create), post: @invalid_attrs)
    assert html_response(conn, 200) =~ "New Post"
  end
end
The first test starts with a post/2 request. That's because once the form in the /posts/new page is submitted, it becomes a POST request to the create action. Because we have supplied valid attributes, the post should have been successfully created and we should have redirected to the show action of the new post. This new page will have an address like /posts/ID, where ID is the identifier of the post in the database.
We then use redirected_params(conn) to get the ID of the post and then match that we indeed redirected to the show action. Finally, we do request a get request to the page we redirected to, allowing us to verify that the post was indeed created.
For the second test, we simply test the failure scenario. If any invalid attribute is given, it should re-render the "New Post" page.
One common question is: how many failure scenarios do you test at the controller level? For example, in the Testing Contexts guide, we introduced a validation to the title field of the post:
def changeset(post, attrs) do
  post
  |> cast(attrs, [:title, :body])
  |> validate_required([:title, :body])
  |> validate_length(:title, min: 2)
end
In other words, creating a post can fail for the following reasons:
	the title is missing
	the body is missing
	the title is present but is less than 2 characters

Should we test all of these possible outcomes in our controller tests?
The answer is no. All of the different rules and outcomes should be verified in your context and schema tests. The controller works as the integration layer. In the controller tests we simply want to verify, in broad strokes, that we handle both success and failure scenarios.
The test for update follows a similar structure as create, so let's skip to the delete test.
The delete action
The delete action looks like this:
def delete(conn, %{"id" => id}) do
  post = Blog.get_post!(id)
  {:ok, _post} = Blog.delete_post(post)

  conn
  |> put_flash(:info, "Post deleted successfully.")
  |> redirect(to: Routes.post_path(conn, :index))
end
The test is written like this:
  describe "delete post" do
    setup [:create_post]

    test "deletes chosen post", %{conn: conn, post: post} do
      conn = delete(conn, Routes.post_path(conn, :delete, post))
      assert redirected_to(conn) == Routes.post_path(conn, :index)
      assert_error_sent 404, fn ->
        get(conn, Routes.post_path(conn, :show, post))
      end
    end
  end

  defp create_post(_) do
    post = fixture(:post)
    %{post: post}
  end
First of all, setup is used to declare that the create_post function should run before every test in this describe block. The create_post function simply creates a post and stores it in the test metadata. This allows us to, in the first line of the test, match on both the post and the connection:
test "deletes chosen post", %{conn: conn, post: post} do
The test uses delete/2 to delete the post and then asserts that we redirected to the index page. Finally, we check that it is no longer possible to access the show page of the deleted post:
assert_error_sent 404, fn ->
  get(conn, Routes.post_path(conn, :show, post))
end
assert_error_sent is a testing helper provided by Phoenix.ConnTest. In this case, it verifies that:
	An exception was raised
	The exception has a status code equivalent to 404 (which stands for Not Found)

This pretty much mimics how Phoenix handles exceptions. For example, when we access /posts/12345 where 12345 is an ID that does not exist, we will invoke our show action:
def show(conn, %{"id" => id}) do
  post = Blog.get_post!(id)
  render(conn, "show.html", post: post)
end
When an unknown post ID is given to Blog.get_post!/1, it raises an Ecto.NotFoundError. If your application raises any exception during a web request, Phoenix translates those requests into proper HTTP response codes. In this case, 404.
We could, for example, have written this test as:
assert_raise Ecto.NotFoundError, fn ->
  get(conn, Routes.post_path(conn, :show, post))
end
However, you may prefer the implementation Phoenix generates by default as it ignores the specific details of the failure, and instead verifies what the browser would actually receive.
The tests for new, edit, and show actions are simpler variations of the tests we have seen so far. You can check the action implementation and their respective tests yourself. Now we are ready to move to JSON controller tests.
JSON controller tests
So far we have been working with a generated HTML resource. However, let's take a look at how our tests look like when we generate a JSON resource.
First of all, run this command:
$ mix phx.gen.json News Article articles title body

We chose a very similar concept to the Blog context <-> Post schema, except we are using a different name, so we can study these concepts in isolation.
After you run the command above, do not forget to follow the final steps output by the generator. Once all is done, we should run mix test and now have 33 passing tests:
$ mix test
................

Finished in 0.6 seconds
33 tests, 0 failures

Randomized with seed 618478

You may have noticed that this time the scaffold controller has generated fewer tests. Previously it generated 16 (we went from 3 to 19) and now it generated 14 (we went from 19 to 33). That's because JSON APIs do not need to expose the new and edit actions. We can see this is the case in the resource we have added to the router at the end of the mix phx.gen.json command:
resources "/articles", ArticleController, except: [:new, :edit]
new and edit are only necessary for HTML because they basically exist to assist users in creating and updating resources. Besides having less actions, we will notice the controller and view tests and implementations for JSON are drastically different from the HTML ones.
The only thing that is pretty much the same between HTML and JSON is the contexts and the schema, which, once you think about it, it makes total sense. After all, your business logic should remain the same, regardless if you are exposing it as HTML or JSON.
With the differences in hand, let's take a look at the controller tests.
The index action
Open up test/hello_web/controllers/article_controller_test.exs. The initial structure is quite similar to post_controller_test.exs. So let's take a look at the tests for the index action. The index action itself is implemented in lib/hello_web/controllers/article_controller.ex like this:
def index(conn, _params) do
  articles = News.list_articles()
  render(conn, "index.json", articles: articles)
end
The action gets all articles and renders index.json. Since we are talking about JSON, we don't have a index.json.eex template. Instead, the code that converts articles into JSON can be found directly in the ArticleView module, defined at lib/hello_web/views/article_view.ex like this:
defmodule HelloWeb.ArticleView do
  use HelloWeb, :view
  alias HelloWeb.ArticleView

  def render("index.json", %{articles: articles}) do
    %{data: render_many(articles, ArticleView, "article.json")}
  end

  def render("show.json", %{article: article}) do
    %{data: render_one(article, ArticleView, "article.json")}
  end

  def render("article.json", %{article: article}) do
    %{id: article.id,
      title: article.title,
      body: article.body}
  end
end
We talked about render_many in the Views and templates guide. All we need to know for now is that all JSON replies have a "data" key with either a list of posts (for index) or a single post inside of it.
Let's take a look at the test for the index action then:
describe "index" do
  test "lists all articles", %{conn: conn} do
    conn = get(conn, Routes.article_path(conn, :index))
    assert json_response(conn, 200)["data"] == []
  end
end
It simply accesses the index path, asserts we got a JSON response with status 200 and that it contains a "data" key with an empty list, as we have no articles to return.
That was quite boring. Let's look at something more interesting.
The create action
The create action is defined like this:
def create(conn, %{"article" => article_params}) do
  with {:ok, %Article{} = article} <- News.create_article(article_params) do
    conn
    |> put_status(:created)
    |> put_resp_header("location", Routes.article_path(conn, :show, article))
    |> render("show.json", article: article)
  end
end
As we can see, it checks if an article could be created. If so, it sets the status code to :created (which translates to 201), it sets a "location" header with the location of the article, and then renders "show.json" with the article.
This is precisely what the first test for the create action verifies:
describe "create" do
  test "renders article when data is valid", %{conn: conn} do
    conn = post(conn, Routes.article_path(conn, :create), article: @create_attrs)
    assert %{"id" => id} = json_response(conn, 201)["data"]

    conn = get(conn, Routes.article_path(conn, :show, id))

    assert %{
             "id" => id,
             "body" => "some body",
             "title" => "some title"
           } = json_response(conn, 200)["data"]
  end
The test uses post/2 to create a new article and then we verify that the article returned a JSON response, with status 201, and that it had a "data" key in it. We pattern match the "data" on %{"id" => id}, which allows us to extract the ID of the new article. Then we perform a get/2 request on the show route and verify that the article was successfully created.
Inside describe "create", we will find another test, which handles the failure scenario. Can you spot the failure scenario in the create action? Let's recap it:
def create(conn, %{"article" => article_params}) do
  with {:ok, %Article{} = article} <- News.create_article(article_params) do
The with special form that ships as part of Elixir allows us to check explicitly for the happy paths. In this case, we are interested only in the scenarios where News.create_article(article_params) returns {:ok, article}, if it returns anything else, the other value will simply be returned directly and none of the contents inside the do/end block will be executed. In other words, if News.create_article/1 returns {:error, changeset}, we will simply return {:error, changeset} from the action.
However, this introduces an issue. Our actions do not know how to handle the {:error, changeset} result by default. Luckily, we can teach Phoenix Controllers to handle it with the Action Fallback controller. At the top of ArticleController, you will find:
  action_fallback HelloWeb.FallbackController
This line says: if any action does not return a %Plug.Conn{}, we want to invoke FallbackController with the result. You will find HelloWeb.FallbackController at lib/hello_web/controllers/fallback_controller.ex and it looks like this:
defmodule HelloWeb.FallbackController do
  use HelloWeb, :controller

  def call(conn, {:error, %Ecto.Changeset{} = changeset}) do
    conn
    |> put_status(:unprocessable_entity)
    |> put_view(HelloWeb.ChangesetView)
    |> render("error.json", changeset: changeset)
  end

  def call(conn, {:error, :not_found}) do
    conn
    |> put_status(:not_found)
    |> put_view(HelloWeb.ErrorView)
    |> render(:"404")
  end
end
You can see how the first clause of the call/2 function handles the {:error, changeset} case, setting the status code to unprocessable entity (422), and then rendering "error.json" from the changeset view with the failed changeset.
With this in mind, let's look at our second test for create:
test "renders errors when data is invalid", %{conn: conn} do
  conn = post(conn, Routes.article_path(conn, :create), article: @invalid_attrs)
  assert json_response(conn, 422)["errors"] != %{}
end
It simply posts to the create path with invalid parameters. This makes it return a JSON response, with status code 422, and a response with a non-empty "errors" key.
The action_fallback can be extremely useful to reduce boilerplate when designing APIs. You can learn more about the "Action Fallback" in the Controllers guide.
The delete action
Finally, the last action we will study is the delete action for JSON. Its implementation looks like this:
def delete(conn, %{"id" => id}) do
  article = News.get_article!(id)

  with {:ok, %Article{}} <- News.delete_article(article) do
    send_resp(conn, :no_content, "")
  end
end
The new action simply attempts to delete the article and, if it succeeds, it returns an empty response with status code :no_content (204).
The test looks like this:
describe "delete article" do
  setup [:create_article]

  test "deletes chosen article", %{conn: conn, article: article} do
    conn = delete(conn, Routes.article_path(conn, :delete, article))
    assert response(conn, 204)

    assert_error_sent 404, fn ->
      get(conn, Routes.article_path(conn, :show, article))
    end
  end
end

defp create_article(_) do
  article = fixture(:article)
  %{article: article}
end
It setups a new article, then in the test it invokes the delete path to delete it, asserting on a 204 response, which is neither JSON nor HTML. Then it verifies that we can no longer access said article.
That's all!
Now that we understand how the scaffolded code and their tests work for both HTML and JSON APIs, we are prepared to move forward in building and maintaining our web applications!

Requirement: This guide expects that you have gone through the introductory guides and got a Phoenix application up and running.

Requirement: This guide expects that you have gone through the Introduction to Testing guide.

Requirement: This guide expects that you have gone through the Channels guide.

In the Channels guide, we saw that a "Channel" is a layered system with different components. Given this, there would be cases when writing unit tests for our Channel functions may not be enough. We may want to verify that its different moving parts are working together as we expect. This integration testing would assure us that we correctly defined our channel route, the channel module, and its callbacks; and that the lower-level layers such as the PubSub and Transport are configured correctly and are working as intended.
Generating channels
As we progress through this guide, it would help to have a concrete example we could work off of. Phoenix comes with a Mix task for generating a basic channel and tests. These generated files serve as a good reference for writing channels and their corresponding tests. Let's go ahead and generate our Channel:
$ mix phx.gen.channel Room
* creating lib/hello_web/channels/room_channel.ex
* creating test/hello_web/channels/room_channel_test.exs

Add the channel to your `lib/hello_web/channels/user_socket.ex` handler, for example:

    channel "room:lobby", HelloWeb.RoomChannel

This creates a channel, its test and instructs us to add a channel route in lib/hello_web/channels/user_socket.ex. It is important to add the channel route or our channel won't function at all!
The ChannelCase
Open up test/hello_web/channels/room_channel_test.exs and you will find this:
defmodule HelloWeb.RoomChannelTest do
  use HelloWeb.ChannelCase
Similar to ConnCase and DataCase, we now have a ChannelCase. All three of them have been generated for us when we started our Phoenix application. Let's take a look at it. Open up test/support/channel_case.ex:
defmodule HelloWeb.ChannelCase do
  use ExUnit.CaseTemplate

  using do
    quote do
      # Import conveniences for testing with channels
      import Phoenix.ChannelTest

      # The default endpoint for testing
      @endpoint HelloWeb.Endpoint
    end
  end

  setup tags do
    pid = Ecto.Adapters.SQL.Sandbox.start_owner!(Demo.Repo, shared: not tags[:async])
    on_exit(fn -> Ecto.Adapters.SQL.Sandbox.stop_owner(pid) end)
    :ok
  end
end
It is very straight-forward. It sets up a case template that imports all of Phoenix.ChannelTest on use. In the setup block, it starts the SQL Sandbox, which we discussed in the Testing contexts guide.
Subscribe and joining
Now that we know that Phoenix provides with a custom Test Case just for channels and what it
provides, we can move on to understanding the rest of test/hello_web/channels/room_channel_test.exs.
First off, is the setup block:
setup do
  {:ok, _, socket} =
    UserSocket
    |> socket("user_id", %{some: :assign})
    |> subscribe_and_join(RoomChannel, "room:lobby")

  %{socket: socket}
end
The setup block sets up a Phoenix.Socket based on the UserSocket module, which you can find at lib/hello_web/channels/user_socket.ex. Then it says we want to subscribe and join the RoomChannel, accessible as "room:lobby" in the UserSocket. At the end of the test, we return the %{socket: socket} as metadata, so we can reuse it on every test.
In a nutshell, subscribe_and_join/3 emulates the client joining a channel and subscribes the test process to the given topic. This is a necessary step since clients need to join a channel before they can send and receive events on that channel.
Testing a synchronous reply
The first test block in our generated channel test looks like:
test "ping replies with status ok", %{socket: socket} do
  ref = push(socket, "ping", %{"hello" => "there"})
  assert_reply ref, :ok, %{"hello" => "there"}
end
This tests the following code in our MyAppWeb.RoomChannel:
# Channels can be used in a request/response fashion
# by sending replies to requests from the client
def handle_in("ping", payload, socket) do
  {:reply, {:ok, payload}, socket}
end
As is stated in the comment above, we see that a reply is synchronous since it mimics the request/response pattern we are familiar with in HTTP. This synchronous reply is best used when we only want to send an event back to the client when we are done processing the message on the server. For example, when we save something to the database and then send a message to the client only once that's done.
In the test "ping replies with status ok", %{socket: socket} do line, we see that we have the map %{socket: socket}. This gives us access to the socket in the setup block.
We emulate the client pushing a message to the channel with push/3. In the line ref = push(socket, "ping", %{"hello" => "there"}), we push the event "ping" with the payload %{"hello" => "there"} to the channel. This triggers the handle_in/3 callback we have for the "ping" event in our channel. Note that we store the ref since we need that on the next line for asserting the reply. With assert_reply ref, :ok, %{"hello" => "there"}, we assert that the server sends a synchronous reply :ok, %{"hello" => "there"}. This is how we check that the handle_in/3 callback for the "ping" was triggered.
Testing a Broadcast
It is common to receive messages from the client and broadcast to everyone subscribed to a current topic. This common pattern is simple to express in Phoenix and is one of the generated handle_in/3 callbacks in our MyAppWeb.RoomChannel.
def handle_in("shout", payload, socket) do
  broadcast(socket, "shout", payload)
  {:noreply, socket}
end
Its corresponding test looks like:
test "shout broadcasts to room:lobby", %{socket: socket} do
  push(socket, "shout", %{"hello" => "all"})
  assert_broadcast "shout", %{"hello" => "all"}
end
We notice that we access the same socket that is from the setup block. How handy! We also do the same push/3 as we did in the synchronous reply test. So we push the "shout" event with the payload %{"hello" => "all"}.
Since the handle_in/3 callback for the "shout" event just broadcasts the same event and payload, all subscribers in the "room:lobby" should receive the message. To check that, we do assert_broadcast "shout", %{"hello" => "all"}.
NOTE: assert_broadcast/3 tests that the message was broadcast in the PubSub system. For testing if a client receives a message, use assert_push/3.
Testing an asynchronous push from the server
The last test in our MyAppWeb.RoomChannelTest verifies that broadcasts from the server are pushed to the client. Unlike the previous tests discussed, we are indirectly testing that our channel's handle_out/3 callback is triggered. This handle_out/3 is defined in our MyApp.RoomChannel as:
def handle_out(event, payload, socket) do
  push(socket, event, payload)
  {:noreply, socket}
end
Since the handle_out/3 event is only triggered when we call broadcast/3 from our channel, we will need to emulate that in our test. We do that by calling broadcast_from or broadcast_from!. Both serve the same purpose with the only difference of broadcast_from! raising an error when broadcast fails.
The line broadcast_from!(socket, "broadcast", %{"some" => "data"}) will trigger our handle_out/3 callback above which pushes the same event and payload back to the client. To test this, we do assert_push "broadcast", %{"some" => "data"}.
That's it. Now you are ready to develop and fully test real-time applications. To learn more about other functionality provided when testing channels, check out the documentation for Phoenix.ChannelTest.

Introduction to Deployment
    

Once we have a working application, we're ready to deploy it. If you're not quite finished with your own application, don't worry. Just follow the Up and Running Guide to create a basic application to work with.
When preparing an application for deployment, there are three main steps:
	Handling of your application secrets
	Compiling your application assets
	Starting your server in production

In this guide, we will learn how to get the production environment running locally. You can use the same techniques in this guide to run your application in production, but depending on your deployment infrastructure, extra steps will be necessary.
As an example of deploying to other infrastructures, we also discuss four different approaches in our guides: using Elixir's releases with mix release, using Gigalixir, using Fly, and using Heroku. We've also included links to deploying Phoenix on other platforms under Community Deployment Guides. Finally, the release guide has a sample Dockerfile you can use if you prefer to deploy with container technologies.
Let's explore those steps above one by one.
Handling of your application secrets
All Phoenix applications have data that must be kept secure, for example, the username and password for your production database, and the secret Phoenix uses to sign and encrypt important information. The general recommendation is to keep those in environment variables and load them into your application. This is done in config/runtime.exs (formerly config/prod.secret.exs or config/releases.exs), which is responsible for loading secrets and configuration from environment variables.
Therefore, you need to make sure the proper relevant variables are set in production:
$ mix phx.gen.secret
REALLY_LONG_SECRET
$ export SECRET_KEY_BASE=REALLY_LONG_SECRET
$ export DATABASE_URL=ecto://USER:PASS@HOST/database

Do not copy those values directly, set SECRET_KEY_BASE according to the result of mix phx.gen.secret and DATABASE_URL according to your database address.
If for some reason you do not want to rely on environment variables, you can hard code the secrets in your config/runtime.exs but make sure not to check the file into your version control system.
With your secret information properly secured, it is time to configure assets!
Before taking this step, we need to do one bit of preparation. Since we will be readying everything for production, we need to do some setup in that environment by getting our dependencies and compiling.
$ mix deps.get --only prod
$ MIX_ENV=prod mix compile

Compiling your application assets
This step is required only if you have compilable assets like JavaScript and stylesheets. By default, Phoenix uses esbuild but everything is encapsulated in a single mix assets.deploy task defined in your mix.exs:
$ MIX_ENV=prod mix assets.deploy
Check your digested files at "priv/static".

And that is it! The Mix task by default builds the assets and then generates digests with a cache manifest file so Phoenix can quickly serve assets in production.
Note: if you run the task above in your local machine, it will generate many digested assets in priv/static. You can prune them by running mix phx.digest.clean --all.

Keep in mind that, if you by any chance forget to run the steps above, Phoenix will show an error message:
$ PORT=4001 MIX_ENV=prod mix phx.server
10:50:18.732 [info] Running MyAppWeb.Endpoint with Cowboy on http://example.com
10:50:18.735 [error] Could not find static manifest at "my_app/_build/prod/lib/foo/priv/static/cache_manifest.json". Run "mix phx.digest" after building your static files or remove the configuration from "config/prod.exs".

The error message is quite clear: it says Phoenix could not find a static manifest. Just run the commands above to fix it or, if you are not serving or don't care about assets at all, you can just remove the cache_static_manifest configuration from config/prod.exs.
Starting your server in production
To run Phoenix in production, we need to set the PORT and MIX_ENV environment variables when invoking mix phx.server:
$ PORT=4001 MIX_ENV=prod mix phx.server
10:59:19.136 [info] Running MyAppWeb.Endpoint with Cowboy on http://example.com

To run in detached mode so that the Phoenix server does not stop and continues to run even if you close the terminal:
$ PORT=4001 MIX_ENV=prod elixir --erl "-detached" -S mix phx.server

In case you get an error message, please read it carefully, and open up a bug report if it is still not clear how to address it.
You can also run your application inside an interactive shell:
$ PORT=4001 MIX_ENV=prod iex -S mix phx.server
10:59:19.136 [info] Running MyAppWeb.Endpoint with Cowboy on http://example.com

Putting it all together
The previous sections give an overview about the main steps required to deploy your Phoenix application. In practice, you will end-up adding steps of your own as well. For example, if you are using a database, you will also want to run mix ecto.migrate before starting the server to ensure your database is up to date.
Overall, here is a script you can use as a starting point:
# Initial setup
$ mix deps.get --only prod
$ MIX_ENV=prod mix compile

# Compile assets
$ MIX_ENV=prod mix assets.deploy

# Custom tasks (like DB migrations)
$ MIX_ENV=prod mix ecto.migrate

# Finally run the server
$ PORT=4001 MIX_ENV=prod mix phx.server

And that's it. Next, you can use one of our official guides to deploy:
	with Elixir's releases
	to Gigalixir, an Elixir-centric Platform as a Service (PaaS)
	to Fly.io, a PaaS that deploys your servers close to your users with built-in distribution support
	and to Heroku, one of the most popular PaaS.

Community Deployment Guides
	Render has first class support for Phoenix applications. There are guides for hosting Phoenix with Mix releases, Distillery, and as a Distributed Elixir Cluster.

Deploying with Releases
    

What we'll need
The only thing we'll need for this guide is a working Phoenix application. For those of us who need a simple application to deploy, please follow the Up and Running guide.
Goals
Our main goal for this guide is to package your Phoenix application into a self-contained directory that includes the Erlang VM, Elixir, all of your code and dependencies. This package can then be dropped into a production machine.
Releases, assemble!
To assemble a release, you will need at least Elixir v1.9, however this guide assumes you are using Elixir v1.12 or later to take advantage of latest releases improvements:
$ elixir -v
1.12.0

If you are not familiar with Elixir releases yet, we recommend you to read Elixir's excellent docs before continuing.
Once that is done, you can assemble a release by going through all of the steps in our general deployment guide with mix release at the end. Let's recap.
First set the environment variables:
$ mix phx.gen.secret
REALLY_LONG_SECRET
$ export SECRET_KEY_BASE=REALLY_LONG_SECRET
$ export DATABASE_URL=ecto://USER:PASS@HOST/database

Then load dependencies to compile code and assets:
# Initial setup
$ mix deps.get --only prod
$ MIX_ENV=prod mix compile

# Compile assets
$ MIX_ENV=prod mix assets.deploy

And now run mix phx.gen.release:
$ mix phx.gen.release
==> my_app
* creating rel/overlays/bin/server
* creating rel/overlays/bin/server.bat
* creating rel/overlays/bin/migrate
* creating rel/overlays/bin/migrate.bat
* creating lib/my_app/release.ex

Your application is ready to be deployed in a release!

    # To start your system
    _build/dev/rel/my_app/bin/my_app start

    # To start your system with the Phoenix server running
    _build/dev/rel/my_app/bin/server

    # To run migrations
    _build/dev/rel/my_app/bin/migrate

Once the release is running:

    # To connect to it remotely
    _build/dev/rel/my_app/bin/my_app remote

    # To stop it gracefully (you may also send SIGINT/SIGTERM)
    _build/dev/rel/my_app/bin/my_app stop

To list all commands:

    _build/dev/rel/my_app/bin/my_app


The phx.gen.release task generated a few files for us to assist in releases. First, it created server and migrate overlay scripts for conveniently running the phoenix server inside a release or invoking migrations from a release. The files in the rel/overlays directory are copied into every release environment. Next, it generated a release.ex file which is used to invoked Ecto migrations without a dependency on mix itself.
Note: If you are a docker user, you can pass the --docker flag to mix phx.gen.release to generate a Dockerfile ready for deployment.
Next, we can invoke mix release to build the release:
$ MIX_ENV=prod mix release
Generated my_app app
* assembling my_app-0.1.0 on MIX_ENV=prod
* using config/runtime.exs to configure the release at runtime

Release created at _build/prod/rel/my_app!

    # To start your system
    _build/prod/rel/my_app/bin/my_app start

...

You can start the release by calling _build/prod/rel/my_app/bin/my_app start, or boot your webserver by calling _build/prod/rel/my_app/bin/server, where you have to replace my_app by your current application name.
Now you can get all of the files under the _build/prod/rel/my_app directory, package it, and run it in any production machine with the same OS and architecture as the one that assembled the release. For more details, check the docs for mix release.
But before we finish this guide, there is one more feature from releases that most Phoenix application will use, so let's talk about that.
Ecto migrations and custom commands
A common need in production systems is to execute custom commands required to set up the production environment. One of such commands is precisely migrating the database. Since we don't have Mix, a build tool, inside releases, which are a production artifact, we need to bring said commands directly into the release.
The phx.gen.release command created the following release.ex file in your project lib/my_app/release.ex, with the following content:
defmodule MyApp.Release do
  @app :my_app

  def migrate do
    load_app()

    for repo <- repos() do
      {:ok, _, _} = Ecto.Migrator.with_repo(repo, &Ecto.Migrator.run(&1, :up, all: true))
    end
  end

  def rollback(repo, version) do
    load_app()
    {:ok, _, _} = Ecto.Migrator.with_repo(repo, &Ecto.Migrator.run(&1, :down, to: version))
  end

  defp repos do
    Application.fetch_env!(@app, :ecto_repos)
  end

  defp load_app do
    Application.load(@app)
  end
end
Where you replace the first two lines by your application names.
Now you can assemble a new release with MIX_ENV=prod mix release and you can invoke any code, including the functions in the module above, by calling the eval command:
$ _build/prod/rel/my_app/bin/my_app eval "MyApp.Release.migrate"

And that's it! If you peek inside the migrate script, you'll see it wraps exactly this invocation.
You can use this approach to create any custom command to run in production. In this case, we used load_app, which calls Application.load/1 to load the current application without starting it. However, you may want to write a custom command that starts the whole application. In such cases, Application.ensure_all_started/1 must be used. Keep in mind, starting the application will start all processes for the current application, including the Phoenix endpoint. This can be circumvented by changing your supervision tree to not start certain children under certain conditions. For example, in the release commands file you could do:
defp start_app do
  load_app()
  Application.put_env(@app, :minimal, true)
  Application.ensure_all_started(@app)
end
And then in your application you check Application.get_env(@app, :minimal) and start only part of the children when it is set.
Containers
Elixir releases work well with container technologies, such as Docker. The idea is that you assemble the release inside the Docker container and then build an image based on the release artifacts.
If you call mix phx.gen.release --docker you'll see a new file with these contents:
# Find eligible builder and runner images on Docker Hub. We use Ubuntu/Debian instead of
# Alpine to avoid DNS resolution issues in production.
#
# https://hub.docker.com/r/hexpm/elixir/tags?page=1&name=ubuntu
# https://hub.docker.com/_/ubuntu?tab=tags
#
#
# This file is based on these images:
#
#   - https://hub.docker.com/r/hexpm/elixir/tags - for the build image
#   - https://hub.docker.com/_/debian?tab=tags&page=1&name=bullseye-20210902-slim - for the release image
#   - https://pkgs.org/ - resource for finding needed packages
#   - Ex: hexpm/elixir:1.12.0-erlang-24.0.1-debian-bullseye-20210902-slim
#
ARG ELIXIR_VERSION=1.12.0
ARG OTP_VERSION=24.0.1
ARG DEBIAN_VERSION=bullseye-20210902-slim

ARG BUILDER_IMAGE="hexpm/elixir:${ELIXIR_VERSION}-erlang-${OTP_VERSION}-debian-${DEBIAN_VERSION}"
ARG RUNNER_IMAGE="debian:${DEBIAN_VERSION}"

FROM ${BUILDER_IMAGE} as builder

# install build dependencies
RUN apt-get update -y && apt-get install -y build-essential git \
    && apt-get clean && rm -f /var/lib/apt/lists/*_*

# prepare build dir
WORKDIR /app

# install hex + rebar
RUN mix local.hex --force && \
    mix local.rebar --force

# set build ENV
ENV MIX_ENV="prod"

# install mix dependencies
COPY mix.exs mix.lock ./
RUN mix deps.get --only $MIX_ENV
RUN mkdir config

# copy compile-time config files before we compile dependencies
# to ensure any relevant config change will trigger the dependencies
# to be re-compiled.
COPY config/config.exs config/${MIX_ENV}.exs config/
RUN mix deps.compile

COPY priv priv

# note: if your project uses a tool like https://purgecss.com/,
# which customizes asset compilation based on what it finds in
# your Elixir templates, you will need to move the asset compilation
# step down so that `lib` is available.
COPY assets assets

# compile assets
RUN mix assets.deploy

# Compile the release
COPY lib lib

RUN mix compile

# Changes to config/runtime.exs don't require recompiling the code
COPY config/runtime.exs config/

COPY rel rel
RUN mix release

# start a new build stage so that the final image will only contain
# the compiled release and other runtime necessities
FROM ${RUNNER_IMAGE}

RUN apt-get update -y && apt-get install -y libstdc++6 openssl libncurses5 locales \
  && apt-get clean && rm -f /var/lib/apt/lists/*_*

# Set the locale
RUN sed -i '/en_US.UTF-8/s/^# //g' /etc/locale.gen && locale-gen

ENV LANG en_US.UTF-8
ENV LANGUAGE en_US:en
ENV LC_ALL en_US.UTF-8

WORKDIR "/app"
RUN chown nobody /app

# set runner ENV
ENV MIX_ENV="prod"

# Only copy the final release from the build stage
COPY --from=builder --chown=nobody:root /app/_build/${MIX_ENV}/rel/my_app ./

USER nobody

CMD /app/bin/server
Where my_app is the name of your app. At the end, you will have an application in /app ready to run as /app/bin/server.
A few points about configuring a containerized application:
	If you run your app in a container, the Endpoint needs to be configured to listen on a "public" :ip address (like 0.0.0.0) so that the app can be reached from outside the container. Whether the host should publish the container's ports to its own public IP or to localhost depends on your needs.
	The more configuration you can provide at runtime (using config/runtime.exs), the more reusable your images will be across environments. In particular, secrets like database credentials and API keys should not be compiled into the image, but rather should be provided when creating containers based on that image. This is why the Endpoint's :secret_key_base is configured in config/runtime.exs by default.
	If possible, any environment variables that are needed at runtime should be read in config/runtime.exs, not scattered throughout your code. Having them all visible in one place will make it easier to ensure the containers get what they need, especially if the person doing the infrastructure work does not work on the Elixir code. Libraries in particular should never directly read environment variables; all their configuration should be handed to them by the top-level application, preferably without using the application environment.

Deploying on Gigalixir
    

What we'll need
The only thing we'll need for this guide is a working Phoenix application. For those of us who need a simple application to deploy, please follow the Up and Running guide.
Goals
Our main goal for this guide is to get a Phoenix application running on Gigalixir.
Steps
Let's separate this process into a few steps, so we can keep track of where we are.
	Initialize Git repository
	Install the Gigalixir CLI
	Sign up for Gigalixir
	Create and set up Gigalixir application
	Provision a database
	Make our project ready for Gigalixir
	Deploy time!
	Useful Gigalixir commands

Initializing Git repository
If you haven't already, we'll need to commit our files to git. We can do so by running the following commands in our project directory:
$ git init
$ git add .
$ git commit -m "Initial commit"

Installing the Gigalixir CLI
Follow the instructions here to install the command-line interface for your platform.
Signing up for Gigalixir
We can sign up for an account at gigalixir.com or with the CLI. Let's use the CLI.
$ gigalixir signup

Gigalixir’s free tier does not require a credit card and comes with 1 app instance and 1 PostgreSQL database for free, but please consider upgrading to a paid plan if you are running a production application.
Next, let's login
$ gigalixir login

And verify
$ gigalixir account

Creating and setting up our Gigalixir application
There are three different ways to deploy a Phoenix app on Gigalixir: with mix, with Elixir's releases, or with Distillery. In this guide, we'll be using Mix because it is the easiest to get up and running, but you won't be able to connect a remote observer or hot upgrade. For more information, see Mix vs Distillery vs Elixir Releases. If you want to deploy with another method, follow the Getting Started Guide.
Creating a Gigalixir application
Let's create a Gigalixir application
$ gigalixir create -n "your_app_name"

Note: the app name cannot be changed afterwards. A random name is used if you do not provide one.
Verify the app was created
$ gigalixir apps

Verify that a git remote was created 
$ git remote -v

Specifying versions
The buildpacks we use default to Elixir, Erlang, and Node.js versions that are quite old and it's generally a good idea to run the same version in production as you do in development, so let's do that.
$ echo 'elixir_version=1.10.3' > elixir_buildpack.config
$ echo 'erlang_version=22.3' >> elixir_buildpack.config
$ echo 'node_version=12.16.3' > phoenix_static_buildpack.config

Phoenix v1.6 uses esbuild to compile your assets, but all Gigalixir images come with npm, so we will configure npm directly to deploy our assets. Add a assets/package.json file if you don't have any with the following:
{
  "scripts": {
    "deploy": "cd .. && mix assets.deploy && rm -f _build/esbuild*"
  }
}
Finally, don't forget to commit:
$ git add elixir_buildpack.config phoenix_static_buildpack.config assets/package.json
$ git commit -m "Set Elixir, Erlang, and Node version"

Making our Project ready for Gigalixir
There's nothing we need to do to get our app running on Gigalixir, but for a production app, you probably want to enforce SSL. To do that, see Force SSL
You may also want to use SSL for your database connection. For that, uncomment the line ssl: true in your Repo config.
Provisioning a database
Let's provision a database for our app
$ gigalixir pg:create --free

Verify the database was created
$ gigalixir pg

Verify that a DATABASE_URL and POOL_SIZE were created
$ gigalixir config

Deploy Time!
Our project is now ready to be deployed on Gigalixir.
$ git push gigalixir

Check the status of your deploy and wait until the app is Healthy
$ gigalixir ps

Run migrations
$ gigalixir run mix ecto.migrate

Check your app logs
$ gigalixir logs

If everything looks good, let's take a look at your app running on Gigalixir
$ gigalixir open

Useful Gigalixir Commands
Open a remote console
$ gigalixir account:ssh_keys:add "$(cat ~/.ssh/id_rsa.pub)"
$ gigalixir ps:remote_console

To open a remote observer, see Remote Observer
To set up clustering, see Clustering Nodes
To hot upgrade, see Hot Upgrades
For custom domains, scaling, jobs and other features, see the Gigalixir Documentation
Troubleshooting
See Troubleshooting
Also, don't hesitate to email help@gigalixir.com or request an invitation and join the #gigalixir channel on Slack.

Deploying on Fly.io
    

What we'll need
The only thing we'll need for this guide is a working Phoenix application. For those of us who need a simple application to deploy, please follow the Up and Running guide.
You can just:
$ mix phx.new my_app

Goals
The main goal for this guide is to get a Phoenix application running on Fly.io.
Sections
Let's separate this process into a few steps, so we can keep track of where we are.
	Install the Fly.io CLI
	Sign up for Fly.io
	Deploy the app to Fly.io
	Extra Fly.io tips
	Helpful Fly.io resources

Installing the Fly.io CLI
Follow the instructions here to install Flyctl, the command-line interface for the Fly.io platform.
Sign up for Fly.io
We can sign up for an account using the CLI.
$ fly auth signup

Or sign in.
$ flyctl auth login

Fly has a free tier for most applications. A credit card is required when setting up an account to help prevent abuse. See the pricing page for more details.
Deploy the app to Fly.io
To tell Fly about your application, run fly launch in the directory with your source code. This creates and configures a Fly.io app.
$ fly launch

This scans your source, detects the Phoenix project, and runs mix phx.gen.release --docker for you! This creates a Dockerfile for you.
The fly launch command walks you through a few questions.
	You can name the app or have it generate a random name for you.
	Choose an organization (defaults to personal). Organizations are a way of sharing applications and resources between Fly.io users.
	Choose a region to deploy to. Defaults to the nearest Fly.io region. You can check out the complete list of regions here.
	Sets up a Postgres DB for you.
	Builds the Dockerfile.
	Deploys your application!

The fly launch command also created a fly.toml file for you. This is where you can set ENV values and other config.
Storing secrets on Fly.io
You may also have some secrets you'd like to set on your app.
Use fly secrets to configure those.
$ fly secrets set MY_SECRET_KEY=my_secret_value

Deploying again
When you want to deploy changes to your application, use fly deploy.
$ fly deploy

Note: On Apple Silicon (M1) computers, docker runs cross-platform builds using qemu which might not always work. If you get a segmentation fault error like the following:
 => [build  7/17] RUN mix deps.get --only
 => => # qemu: uncaught target signal 11 (Segmentation fault) - core dumped
You can use fly's remote builder by adding the --remote-only flag:
$ fly deploy --remote-only

You can always check on the status of a deploy
$ fly status

Check your app logs
$ fly logs

If everything looks good, open your app on Fly
$ fly open

Extra Fly.io tips
Getting an IEx shell into a running node
Elixir supports getting a IEx shell into a running production node.
There are a couple prerequisites, we first need to establish an SSH Shell to our machine on Fly.io.
This step sets up a root certificate for your account and then issues a certificate.
$ fly ssh establish
$ fly ssh issue

With SSH configured, let's open a console.
$ fly ssh console
Connecting to my-app-1234.internal... complete
/ #

If all has gone smoothly, then you have a shell into the machine! Now we just need to launch our remote IEx shell. The deployment Dockerfile was configured to pull our application into /app. So the command for an app named my_app looks like this:
$ app/bin/my_app remote
Erlang/OTP 23 [erts-11.2.1] [source] [64-bit] [smp:1:1] [ds:1:1:10] [async-threads:1]

Interactive Elixir (1.11.2) - press Ctrl+C to exit (type h() ENTER for help)
iex(my_app@fdaa:0:1da8:a7b:ac4:b204:7e29:2)1>

Now we have a running IEx shell into our node! You can safely disconnect using CTRL+C, CTRL+C.
Clustering your application
Elixir and the BEAM have the incredible ability to be clustered together and pass messages seamlessly between nodes. This portion of the guide walks you through clustering your Elixir application.
There are 2 parts to getting clustering quickly setup on Fly.io.
	Installing and using libcluster
	Scaling the application to multiple instances

Adding libcluster
The widely adopted library libcluster helps here.
There are multiple strategies that libcluster can use to find and connect with other nodes. The strategy we'll use on Fly.io is DNSPoll.
After installing libcluster, add it to the application like this:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    topologies = Application.get_env(:libcluster, :topologies) || []

    children = [
      # ...
      # setup for clustering
      {Cluster.Supervisor, [topologies, [name: MyApp.ClusterSupervisor]]}
    ]

    # ...
  end

  # ...
end
Our next step is to add the topologies configuration to config/runtime.exs.
  app_name =
    System.get_env("FLY_APP_NAME") ||
      raise "FLY_APP_NAME not available"

  config :libcluster,
    topologies: [
      fly6pn: [
        strategy: Cluster.Strategy.DNSPoll,
        config: [
          polling_interval: 5_000,
          query: "#{app_name}.internal",
          node_basename: app_name
        ]
      ]
    ]
This configures libcluster to use the DNSPoll strategy and look for other deployed apps using the $FLY_APP_NAME on the .internal private network.
Controlling the name for our node
We need to control the naming of our Elixir nodes. To help them connect up, we'll name them using this pattern: your-fly-app-name@the.ipv6.address.on.fly. To do this, we'll generate the release config.
$ mix release.init

Then edit the generated rel/env.sh.eex file and add the following lines:
ip=$(grep fly-local-6pn /etc/hosts | cut -f 1)
export RELEASE_DISTRIBUTION=name
export RELEASE_NODE=$FLY_APP_NAME@$ip
After making the change, deploy your app!
$ fly deploy

For our app to be clustered, we have to have multiple instances. Next we'll add an additional node instance.
Running multiple instances
There are two ways to run multiple instances.
	Scale our application to have multiple instances in one region.
	Add an instance to another region (multiple regions).

Let's first start with a baseline of our single deployment.
$ fly status
...
Instances
ID       VERSION REGION DESIRED STATUS  HEALTH CHECKS      RESTARTS CREATED
f9014bf7 26      sea    run     running 1 total, 1 passing 0        1h8m ago

Scaling in a single region
Let's scale up to 2 instances in our current region.
$ fly scale count 2
Count changed to 2

Checking the status, we can see what happened.
$ fly status
...
Instances
ID       VERSION REGION DESIRED STATUS  HEALTH CHECKS      RESTARTS CREATED
eb4119d3 27      sea    run     running 1 total, 1 passing 0        39s ago
f9014bf7 27      sea    run     running 1 total, 1 passing 0        1h13m ago

We now have two instances in the same region.
Let's make sure they are clustered together. We can check the logs:
$ fly logs
...
app[eb4119d3] sea [info] 21:50:21.924 [info] [libcluster:fly6pn] connected to :"my-app-1234@fdaa:0:1da8:a7b:ac2:f901:4bf7:2"
...

But that's not as rewarding as seeing it from inside a node. From an IEx shell, we can ask the node we're connected to, what other nodes it can see.
$ fly ssh console
$ /app/bin/my_app remote

iex(my-app-1234@fdaa:0:1da8:a7b:ac2:f901:4bf7:2)1> Node.list
[:"my-app-1234@fdaa:0:1da8:a7b:ac4:eb41:19d3:2"]
The IEx prompt is included to help show the IP address of the node we are connected to. Then getting the Node.list returns the other node. Our two instances are connected and clustered!
Scaling to multiple regions
Fly makes it easy to deploy instances closer to your users. Through the magic of DNS, users are directed to the nearest region where your application is located. You can read more about Fly.io regions here.
Starting back from our baseline of a single instance running in sea which is Seattle, Washington (US), let's add the region ewr which is Parsippany, NJ (US). This puts an instance on both coasts of the US.
$ fly regions add ewr
Region Pool:
ewr
sea
Backup Region:
iad
lax
sjc
vin

Looking at the status shows that we're only in 1 region because our count is set to 1.
$ fly status
...
Instances
ID       VERSION REGION DESIRED STATUS  HEALTH CHECKS      RESTARTS CREATED
cdf6c422 29      sea    run     running 1 total, 1 passing 0        58s ago

Let's add a 2nd instance and see it deploy to ewr.
$ fly scale count 2
Count changed to 2

Now the status shows we have two instances spread across 2 regions!
$ fly status
...
Instances
ID       VERSION REGION DESIRED STATUS  HEALTH CHECKS      RESTARTS CREATED
0a8e6666 30      ewr    run     running 1 total, 1 passing 0        16s ago
cdf6c422 30      sea    run     running 1 total, 1 passing 0        6m47s ago

Let's ensure they are clustered together.
$ fly ssh console
$ /app/bin/my_app remote

iex(my-app-1234@fdaa:0:1da8:a7b:ac2:cdf6:c422:2)1> Node.list
[:"my-app-1234@fdaa:0:1da8:a7b:ab2:a8e:6666:2"]
We have two instances of our application deployed to the West and East coasts of the North American continent and they are clustered together! Our users will automatically be directed to the server nearest them.
The Fly.io platform has built-in distribution support making it easy to cluster distributed Elixir nodes in multiple regions.
Helpful Fly.io resources
Open the Dashboard for your account
$ fly dashboard

Deploy your application
$ fly deploy

Show the status of your deployed application
$ fly status

Access and tail the logs
$ fly logs

Scaling your application up or down
$ fly scale count 2

Refer to the Fly.io Elixir documentation for additional information.
Working with Fly.io applications covers things like:
	Status and logs
	Custom domains
	Certificates

Troubleshooting
See Troubleshooting
Visit the Fly.io Community to find solutions and ask questions.

What we'll need
The only thing we'll need for this guide is a working Phoenix application. For those of us who need a simple application to deploy, please follow the Up and Running guide.
Goals
Our main goal for this guide is to get a Phoenix application running on Heroku.
Limitations
Heroku is a great platform and Elixir performs well on it. However, you may run into limitations if you plan to leverage advanced features provided by Elixir and Phoenix, such as:
	Connections are limited.
	Heroku limits the number of simultaneous connections as well as the duration of each connection. It is common to use Elixir for real-time apps which need lots of concurrent, persistent connections, and Phoenix is capable of handling over 2 million connections on a single server.


	Distributed clustering is not possible.
	Heroku firewalls dynos off from one another. This means things like distributed Phoenix channels and distributed tasks will need to rely on something like Redis instead of Elixir's built-in distribution.


	In-memory state such as those in Agents, GenServers, and ETS will be lost every 24 hours.
	Heroku restarts dynos every 24 hours regardless of whether the node is healthy.


	The built-in observer can't be used with Heroku.
	Heroku does allow for connection into your dyno, but you won't be able to use the observer to watch the state of your dyno.



If you are just getting started, or you don't expect to use the features above, Heroku should be enough for your needs. For instance, if you are migrating an existing application running on Heroku to Phoenix, keeping a similar set of features, Elixir will perform just as well or even better than your current stack.
If you want a platform-as-a-service without these limitations, try Gigalixir. If you would rather deploy to a cloud platform, such as EC2, Google Cloud, etc, consider using mix release.
Steps
Let's separate this process into a few steps, so we can keep track of where we are.
	Initialize Git repository
	Sign up for Heroku
	Install the Heroku Toolbelt
	Create and set up Heroku application
	Make our project ready for Heroku
	Deploy time!
	Useful Heroku commands

Initializing Git repository
Git is a popular decentralized revision control system and is also used to deploy apps to Heroku.
Before we can push to Heroku, we'll need to initialize a local Git repository and commit our files to it. We can do so by running the following commands in our project directory:
$ git init
$ git add .
$ git commit -m "Initial commit"

Heroku offers some great information on how it is using Git here.
Signing up for Heroku
Signing up to Heroku is very simple, just head over to https://signup.heroku.com/ and fill in the form.
The Free plan will give us one web dyno and one worker dyno, as well as a PostgreSQL and Redis instance for free.
These are meant to be used for testing and development, and come with some limitations. In order to run a production application, please consider upgrading to a paid plan.
Installing the Heroku Toolbelt
Once we have signed up, we can download the correct version of the Heroku Toolbelt for our system here.
The Heroku CLI, part of the Toolbelt, is useful to create Heroku applications, list currently running dynos for an existing application, tail logs or run one-off commands (mix tasks for instance).
Create and Set Up Heroku Application
There are two different ways to deploy a Phoenix app on Heroku. We could use Heroku buildpacks or their container stack. The difference between these two approaches is in how we tell Heroku to treat our build. In buildpack case, we need to update our apps configuration on Heroku to use Phoenix/Elixir specific buildpacks. On container approach, we have more control on how we want to set up our app, and we can define our container image using Dockerfile and heroku.yml. This section will explore the buildpack approach. In order to use Dockerfile, it is often recommended to convert our app to use releases, which we will describe later on.
Create Application
A buildpack is a convenient way of packaging framework and/or runtime support. Phoenix requires 2 buildpacks to run on Heroku, the first adds basic Elixir support and the second adds Phoenix specific commands.
With the Toolbelt installed, let's create the Heroku application. We will do so using the latest available version of the Elixir buildpack:
$ heroku create --buildpack hashnuke/elixir
Creating app... done, ⬢ mysterious-meadow-6277
Setting buildpack to hashnuke/elixir... done
https://mysterious-meadow-6277.herokuapp.com/ | https://git.heroku.com/mysterious-meadow-6277.git

Note: the first time we use a Heroku command, it may prompt us to log in. If this happens, just enter the email and password you specified during signup.

Note: the name of the Heroku application is the random string after "Creating" in the output above (mysterious-meadow-6277). This will be unique, so expect to see a different name from "mysterious-meadow-6277".

Note: the URL in the output is the URL to our application. If we open it in our browser now, we will get the default Heroku welcome page.

Note: if we hadn't initialized our Git repository before we ran the heroku create command, we wouldn't have our Heroku remote repository properly set up at this point. We can set that up manually by running: heroku git:remote -a [our-app-name].

The buildpack uses a predefined Elixir and Erlang version, but to avoid surprises when deploying, it is best to explicitly list the Elixir and Erlang version we want in production to be the same we are using during development or in your continuous integration servers. This is done by creating a config file named elixir_buildpack.config in the root directory of your project with your target version of Elixir and Erlang:
# Elixir version
elixir_version=1.12.2

# Erlang version
# https://github.com/HashNuke/heroku-buildpack-elixir-otp-builds/blob/master/otp-versions
erlang_version=24.0.3

# Invoke assets.deploy defined in your mix.exs to deploy assets with esbuild
# Note we nuke the esbuild executable from the image
hook_post_compile="eval mix assets.deploy && rm -f _build/esbuild*"
Finally, let's tell the build pack how to start our webserver. Create a file named Procfile at the root of your project:
web: mix phx.server
Optional: Node, npm, and the Phoenix Static buildpack
By default, Phoenix uses esbuild and manages all assets for you. However, if you are using node and npm, you will need to install the Phoenix Static buildpack to handle them:
$ heroku buildpacks:add https://github.com/gjaldon/heroku-buildpack-phoenix-static.git
Buildpack added. Next release on mysterious-meadow-6277 will use:
  1. https://github.com/HashNuke/heroku-buildpack-elixir.git
  2. https://github.com/gjaldon/heroku-buildpack-phoenix-static.git

When using this buildpack, you want to delegate all asset bundling to npm. So you must remove the hook_post_compile configuration from your elixir_buildpack.config and move it to the deploy script of your assets/package.json. Something like this:
{
  ...
  "scripts": {
    "deploy": "cd .. && mix assets.deploy && rm -f _build/esbuild*"
  }
  ...
}
The Phoenix Static buildpack uses a predefined Node.js version, but to avoid surprises when deploying, it is best to explicitly list the Node.js version we want in production to be the same we are using during development or in your continuous integration servers. This is done by creating a config file named phoenix_static_buildpack.config in the root directory of your project with your target version of Node.js:
# Node.js version
node_version=10.20.1
Please refer to the configuration section for full details. You can make your own custom build script, but for now we will use the default one provided.
Finally, note that since we are using multiple buildpacks, you might run into an issue where the sequence is out of order (the Elixir buildpack needs to run before the Phoenix Static buildpack). Heroku's docs explain this better, but you will need to make sure the Phoenix Static buildpack comes last.
Making our Project ready for Heroku
Every new Phoenix project ships with a config file config/runtime.exs (formerly config/prod.secret.exs) which loads configuration and secrets from environment variables. This aligns well with Heroku best practices, so the only work left for us to do is to configure URLs and SSL.
First let's tell Phoenix to use our Heroku URL and enforce we only use the SSL version of the website. Also, bind to the port requested by Heroku in the $PORT environment variable. Find the url line in your config/prod.exs:
url: [host: "example.com", port: 80],
... and replace it with this (don't forget to replace mysterious-meadow-6277 with your application name):
url: [scheme: "https", host: "mysterious-meadow-6277.herokuapp.com", port: 443],
force_ssl: [rewrite_on: [:x_forwarded_proto]],
Then open up your config/runtime.xs (formerly config/prod.secret.exs) and uncomment the # ssl: true, line in your repository configuration. It will look like this:
config :hello, Hello.Repo,
  ssl: true,
  url: database_url,
  pool_size: String.to_integer(System.get_env("POOL_SIZE") || "10")
Finally, if you plan on using websockets, then we will need to decrease the timeout for the websocket transport in lib/hello_web/endpoint.ex. If you do not plan on using websockets, then leaving it set to false is fine. You can find further explanation of the options available at the documentation.
defmodule HelloWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :hello

  socket "/socket", HelloWeb.UserSocket,
    websocket: [timeout: 45_000]

  ...
end
This ensures that any idle connections are closed by Phoenix before they reach Heroku's 55-second timeout window.
Creating Environment Variables in Heroku
The DATABASE_URL config var is automatically created by Heroku when we add the Heroku Postgres add-on. We can create the database via the Heroku toolbelt:
$ heroku addons:create heroku-postgresql:hobby-dev

Now we set the POOL_SIZE config var:
$ heroku config:set POOL_SIZE=18

This value should be just under the number of available connections, leaving a couple open for migrations and mix tasks. The hobby-dev database allows 20 connections, so we set this number to 18. If additional dynos will share the database, reduce the POOL_SIZE to give each dyno an equal share.
When running a mix task later (after we have pushed the project to Heroku) you will also want to limit its pool size like so:
$ heroku run "POOL_SIZE=2 mix hello.task"

So that Ecto does not attempt to open more than the available connections.
We still have to create the SECRET_KEY_BASE config based on a random string. First, use mix phx.gen.secret to get a new secret:
$ mix phx.gen.secret
xvafzY4y01jYuzLm3ecJqo008dVnU3CN4f+MamNd1Zue4pXvfvUjbiXT8akaIF53

Your random string will be different; don't use this example value.
Now set it in Heroku:
$ heroku config:set SECRET_KEY_BASE="xvafzY4y01jYuzLm3ecJqo008dVnU3CN4f+MamNd1Zue4pXvfvUjbiXT8akaIF53"
Setting config vars and restarting mysterious-meadow-6277... done, v3
SECRET_KEY_BASE: xvafzY4y01jYuzLm3ecJqo008dVnU3CN4f+MamNd1Zue4pXvfvUjbiXT8akaIF53

Deploy Time!
Our project is now ready to be deployed on Heroku.
Let's commit all our changes:
$ git add elixir_buildpack.config
$ git commit -a -m "Use production config from Heroku ENV variables and decrease socket timeout"

And deploy:
$ git push heroku master
Counting objects: 55, done.
Delta compression using up to 8 threads.
Compressing objects: 100% (49/49), done.
Writing objects: 100% (55/55), 48.48 KiB | 0 bytes/s, done.
Total 55 (delta 1), reused 0 (delta 0)
remote: Compressing source files... done.
remote: Building source:
remote:
remote: -----> Multipack app detected
remote: -----> Fetching custom git buildpack... done
remote: -----> elixir app detected
remote: -----> Checking Erlang and Elixir versions
remote:        WARNING: elixir_buildpack.config wasn't found in the app
remote:        Using default config from Elixir buildpack
remote:        Will use the following versions:
remote:        * Stack cedar-14
remote:        * Erlang 17.5
remote:        * Elixir 1.0.4
remote:        Will export the following config vars:
remote:        * Config vars DATABASE_URL
remote:        * MIX_ENV=prod
remote: -----> Stack changed, will rebuild
remote: -----> Fetching Erlang 17.5
remote: -----> Installing Erlang 17.5 (changed)
remote:
remote: -----> Fetching Elixir v1.0.4
remote: -----> Installing Elixir v1.0.4 (changed)
remote: -----> Installing Hex
remote: 2015-07-07 00:04:00 URL:https://s3.amazonaws.com/s3.hex.pm/installs/1.0.0/hex.ez [262010/262010] ->
"/app/.mix/archives/hex.ez" [1]
remote: * creating /app/.mix/archives/hex.ez
remote: -----> Installing rebar
remote: * creating /app/.mix/rebar
remote: -----> Fetching app dependencies with mix
remote: Running dependency resolution
remote: Dependency resolution completed successfully
remote: [...]
remote: -----> Compiling
remote: [...]
remote: Generated phoenix_heroku app
remote: [...]
remote: Consolidated protocols written to _build/prod/consolidated
remote: -----> Creating .profile.d with env vars
remote: -----> Fetching custom git buildpack... done
remote: -----> Phoenix app detected
remote:
remote: -----> Loading configuration and environment
remote:        Loading config...
remote:        [...]
remote:        Will export the following config vars:
remote:        * Config vars DATABASE_URL
remote:        * MIX_ENV=prod
remote:
remote: -----> Compressing... done, 82.1MB
remote: -----> Launching... done, v5
remote:        https://mysterious-meadow-6277.herokuapp.com/ deployed to Heroku
remote:
remote: Verifying deploy... done.
To https://git.heroku.com/mysterious-meadow-6277.git
 * [new branch]      master -> master

Typing heroku open in the terminal should launch a browser with the Phoenix welcome page opened. In the event that you are using Ecto to access a database, you will also need to run migrations after the first deploy:
$ heroku run "POOL_SIZE=2 mix ecto.migrate"

And that's it!
Deploying to Heroku using the container stack
Create Heroku application
Set the stack of your app to container, this allows us to use Dockerfile to define our app setup.
$ heroku create
Creating app... done, ⬢ mysterious-meadow-6277
$ heroku stack:set container

Add a new heroku.yml file to your root folder. In this file you can define addons used by your app, how to build the image and what configs are passed to the image. You can learn more about Heroku's heroku.yml options here. Here is a sample:
setup:
  addons:
    - plan: heroku-postgresql
      as: DATABASE
build:
  docker:
    web: Dockerfile
  config:
    MIX_ENV: prod
    SECRET_KEY_BASE: $SECRET_KEY_BASE
    DATABASE_URL: $DATABASE_URL
Set up releases and Dockerfile
Now we need to define a Dockerfile at the root folder of your project that contains your application. We recommend to use releases when doing so, as the release will allow us to build a container with only the parts of Erlang and Elixir we actually use. Follow the releases docs. At the end of the guide, there is a sample Dockerfile file you can use.
Once you have the image definition set up, you can push your app to heroku and you can see it starts building the image and deploy it.
Useful Heroku Commands
We can look at the logs of our application by running the following command in our project directory:
$ heroku logs # use --tail if you want to tail them

We can also start an IEx session attached to our terminal for experimenting in our app's environment:
$ heroku run "POOL_SIZE=2 iex -S mix"

In fact, we can run anything using the heroku run command, like the Ecto migration task from above:
$ heroku run "POOL_SIZE=2 mix ecto.migrate"

Connecting to your dyno
Heroku gives you the ability to connect to your dyno with an IEx shell which allows running Elixir code such as database queries.
	Modify the web process in your Procfile to run a named node:  web: elixir --sname server -S mix phx.server

	Redeploy to Heroku
	Connect to the dyno with heroku ps:exec (if you have several applications on the same repository you will need to specify the app name or the remote name with --app APP_NAME or --remote REMOTE_NAME)
	Launch an iex session with iex --sname console --remsh server

You have an iex session into your dyno!
Troubleshooting
Compilation Error
Occasionally, an application will compile locally, but not on Heroku. The compilation error on Heroku will look something like this:
remote: == Compilation error on file lib/postgrex/connection.ex ==
remote: could not compile dependency :postgrex, "mix compile" failed. You can recompile this dependency with "mix deps.compile postgrex", update it with "mix deps.update postgrex" or clean it with "mix deps.clean postgrex"
remote: ** (CompileError) lib/postgrex/connection.ex:207: Postgrex.Connection.__struct__/0 is undefined, cannot expand struct Postgrex.Connection
remote:     (elixir) src/elixir_map.erl:58: :elixir_map.translate_struct/4
remote:     (stdlib) lists.erl:1353: :lists.mapfoldl/3
remote:     (stdlib) lists.erl:1354: :lists.mapfoldl/3
remote:
remote:
remote:  !     Push rejected, failed to compile elixir app
remote:
remote: Verifying deploy...
remote:
remote: !   Push rejected to mysterious-meadow-6277.
remote:
To https://git.heroku.com/mysterious-meadow-6277.git

This has to do with stale dependencies which are not getting recompiled properly. It's possible to force Heroku to recompile all dependencies on each deploy, which should fix this problem. The way to do it is to add a new file called elixir_buildpack.config at the root of the application. The file should contain this line:
always_rebuild=true
Commit this file to the repository and try to push again to Heroku.
Connection Timeout Error
If you are constantly getting connection timeouts while running heroku run this could mean that your internet provider has blocked port number 5000:
heroku run "POOL_SIZE=2 mix myapp.task"
Running POOL_SIZE=2 mix myapp.task on mysterious-meadow-6277... !
ETIMEDOUT: connect ETIMEDOUT 50.19.103.36:5000

You can overcome this by adding detached option to run command:
heroku run:detached "POOL_SIZE=2 mix ecto.migrate"
Running POOL_SIZE=2 mix ecto.migrate on mysterious-meadow-6277... done, run.8089 (Free)

Phoenix has a view called ErrorView which lives in lib/hello_web/views/error_view.ex. The purpose of ErrorView is to handle errors in a general way, from one centralized location.
The ErrorView
For new applications, the ErrorView view looks like this:
defmodule HelloWeb.ErrorView do
  use HelloWeb, :view

  # If you want to customize a particular status code
  # for a certain format, you may uncomment below.
  # def render("500.html", _assigns) do
  #   "Internal Server Error"
  # end

  # By default, Phoenix returns the status message from
  # the template name. For example, "404.html" becomes
  # "Not Found".
  def template_not_found(template, _assigns) do
    Phoenix.Controller.status_message_from_template(template)
  end
end
Before we dive into this, let's see what the rendered 404 Not Found message looks like in a browser. In the development environment, Phoenix will debug errors by default, showing us a very informative debugging page. What we want here, however, is to see what page the application would serve in production. In order to do that, we need to set debug_errors: false in config/dev.exs.
import Config

config :hello, HelloWeb.Endpoint,
  http: [port: 4000],
  debug_errors: false,
  code_reloader: true,
  . . .
After modifying our config file, we need to restart our server in order for this change to take effect. After restarting the server, let's go to http://localhost:4000/such/a/wrong/path for a running local application and see what we get.
Ok, that's not very exciting. We get the bare string "Not Found", displayed without any markup or styling.
The first question is, where does that error string come from? The answer is right in ErrorView.
def template_not_found(template, _assigns) do
  Phoenix.Controller.status_message_from_template(template)
end
Great, so we have this template_not_found/2 function that takes a template and an assigns map, which we ignore. The template_not_found/2 is invoked whenever a Phoenix.View view attempts to render a template, but no template is found.
In other words, to provide custom error pages, we could simply define a proper render/2 function clause in HelloWeb.ErrorView.
  def render("404.html", _assigns) do
    "Page Not Found"
  end
But we can do even better.
Phoenix generates an ErrorView for us, but it doesn't give us a lib/hello_web/templates/error directory. Let's create one now. Inside our new directory, let's add a template named404.html.heex and give it some markup – a mixture of our application layout and a new <div> with our message to the user.
<!DOCTYPE html>
<html lang="en">
  <head>
    <meta charset="utf-8"/>
    <meta http-equiv="X-UA-Compatible" content="IE=edge"/>
    <meta name="viewport" content="width=device-width, initial-scale=1.0"/>
    <title>Welcome to Phoenix!</title>
    <link rel="stylesheet" href="/css/app.css"/>
    <script defer type="text/javascript" src="/js/app.js"></script>
  </head>
  <body>
    <header>
      <section class="container">
        <nav>
          <ul>
            <li><a href="https://hexdocs.pm/phoenix/overview.html">Get Started</a></li>
          </ul>
        </nav>
        <a href="https://phoenixframework.org/" class="phx-logo">
          <img src="/images/phoenix.png" alt="Phoenix Framework Logo"/>
        </a>
      </section>
    </header>
    <main class="container">
      <section class="phx-hero">
        <p>Sorry, the page you are looking for does not exist.</p>
      </section>
    </main>
  </body>
</html>
After you define the template file, remember to remove the equivalent render/2 clause for that template, as otherwise the function overrides the template. Let's do so for the 404.html clause we have previously introduced in lib/hello_web/views/error_view.ex:
- def render("404.html", _assigns) do
-  "Page Not Found"
- end
Now, when we go back to http://localhost:4000/such/a/wrong/path, we should see a much nicer error page. It is worth noting that we did not render our 404.html.heex template through our application layout, even though we want our error page to have the look and feel of the rest of our site. This is to avoid circular errors. For example, what happens if our application failed due to an error in the layout? Attempting to render the layout again will just trigger another error. So ideally we want to minimize the amount of dependencies and logic in our error templates, sharing only what is necessary.
Custom exceptions
Elixir provides a macro called defexception/1 for defining custom exceptions. Exceptions are represented as structs, and structs need to be defined inside of modules.
In order to create a custom exception, we need to define a new module. Conventionally, this will have "Error" in the name. Inside that module, we need to define a new exception with defexception/1, the file lib/hello_web.ex seems like a good place for it.
defmodule HelloWeb.SomethingNotFoundError do
  defexception [:message]
end
You can raise your new exception like this:
raise HelloWeb.SomethingNotFoundError, "oops"
By default, Plug and Phoenix will treat all exceptions as 500 errors. However, Plug provides a protocol called Plug.Exception where we are able to customize the status and add actions that exception structs can return on the debug error page.
If we wanted to supply a status of 404 for an HelloWeb.SomethingNotFoundError error, we could do it by defining an implementation for the Plug.Exception protocol like this, in lib/hello_web.ex:
defimpl Plug.Exception, for: HelloWeb.SomethingNotFoundError do
  def status(_exception), do: 404
  def actions(_exception), do: []
end
Alternatively, you could define a plug_status field directly in the exception struct:
defmodule HelloWeb.SomethingNotFoundError do
  defexception [:message, plug_status: 404]
end
However, implementing the Plug.Exception protocol by hand can be convenient in certain occasions, such as when providing actionable errors.
Actionable errors
Exception actions are functions that can be triggered by the error page, and they're basically a list of maps defining a label and a handler to be executed.
They are rendered in the error page as a collection of buttons and follow the format of:
[
  %{
    label: String.t(),
    handler: {module(), function :: atom(), args :: []}
  }
]
If we wanted to return some actions for an HelloWeb.SomethingNotFoundError we would implement Plug.Exception like this:
defimpl Plug.Exception, for: HelloWeb.SomethingNotFoundError do
  def status(_exception), do: 404

  def actions(_exception) do
    [
      %{
        label: "Run seeds",
        handler: {Code, :eval_file, "priv/repo/seeds.exs"}
      }
    ]
  end
end

To prepare an application to serve requests over SSL, we need to add a little bit of configuration and two environment variables. In order for SSL to actually work, we'll need a key file and certificate file from a certificate authority. The environment variables that we'll need are paths to those two files.
The configuration consists of a new https: key for our endpoint whose value is a keyword list of port, path to the key file, and path to the cert (PEM) file. If we add the otp_app: key whose value is the name of our application, Plug will begin to look for them at the root of our application. We can then put those files in our priv directory and set the paths to priv/our_keyfile.key and priv/our_cert.crt.
Here's an example configuration from config/prod.exs.
import Config

config :hello, HelloWeb.Endpoint,
  http: [port: {:system, "PORT"}],
  url: [host: "example.com"],
  cache_static_manifest: "priv/static/cache_manifest.json",
  https: [
    port: 443,
    cipher_suite: :strong,
    otp_app: :hello,
    keyfile: System.get_env("SOME_APP_SSL_KEY_PATH"),
    certfile: System.get_env("SOME_APP_SSL_CERT_PATH"),
    # OPTIONAL Key for intermediate certificates:
    cacertfile: System.get_env("INTERMEDIATE_CERTFILE_PATH")
  ]

Without the otp_app: key, we need to provide absolute paths to the files wherever they are on the filesystem in order for Plug to find them.
Path.expand("../../../some/path/to/ssl/key.pem", __DIR__)
The options under the https: key are passed to the Plug adapter, typically Plug.Cowboy, which in turn uses Plug.SSL to select the TLS socket options. Please refer to the documentation for Plug.SSL.configure/1 for more information on the available options and their defaults. The Plug HTTPS Guide and the Erlang/OTP ssl documentation also provide valuable information.
SSL in Development
If you would like to use HTTPS in development, a self-signed certificate can be generated by running: mix phx.gen.cert. This requires Erlang/OTP 20 or later.
With your self-signed certificate, your development configuration in config/dev.exs can be updated to run an HTTPS endpoint:
config :my_app, MyAppWeb.Endpoint,
  ...
  https: [
    port: 4001,
    cipher_suite: :strong,
    keyfile: "priv/cert/selfsigned_key.pem",
    certfile: "priv/cert/selfsigned.pem"
  ]
This can replace your http configuration, or you can run HTTP and HTTPS servers on different ports.
Force SSL
In many cases, you'll want to force all incoming requests to use SSL by redirecting HTTP to HTTPS. This can be accomplished by setting the :force_ssl option in your endpoint configuration. It expects a list of options which are forwarded to Plug.SSL. By default, it sets the "strict-transport-security" header in HTTPS requests, forcing browsers to always use HTTPS. If an unsafe (HTTP) request is sent, it redirects to the HTTPS version using the :host specified in the :url configuration. For example:
config :my_app, MyAppWeb.Endpoint,
  force_ssl: [rewrite_on: [:x_forwarded_proto]]
To dynamically redirect to the host of the current request, set :host in the :force_ssl configuration to nil.
config :my_app, MyAppWeb.Endpoint,
  force_ssl: [rewrite_on: [:x_forwarded_proto], host: nil]
In these examples, the rewrite_on: key specifies the HTTP header used by a reverse proxy or load balancer in front of the application to indicate whether the request was received over HTTP or HTTPS. For more information on the implications of offloading TLS to an external element, in particular relating to secure cookies, refer to the Plug HTTPS Guide. Keep in mind that the options passed to Plug.SSL in that document should be set using the force_ssl: endpoint option in a Phoenix application.
HSTS
HSTS or "strict-transport-security" is a mechanism that allows a website to declare itself as only accessible via a secure connection (HTTPS). It was introduced to prevent man-in-the-middle attacks that strip SSL/TLS. It causes web browsers to redirect from HTTP to HTTPS and refuse to connect unless the connection uses SSL/TLS.
With force_ssl: [hsts: true] set, the Strict-Transport-Security header is set with a max age that defines the length of time the policy is valid for. Modern web browsers will respond to this by redirecting from HTTP to HTTPS for the standard case but it does have other consequences. RFC6797 which defines HSTS also specifies that the browser should keep track of the policy of a host and apply it until it expires. It also specifies that traffic on any port other than 80 is assumed to be encrypted as per the policy.
This can result in unexpected behaviour if you access your application on localhost, for example https://localhost:4000, as from that point forward any traffic coming from localhost will be expected to be encrypted, except port 80 which will be redirected to port 443. This has the potential to disrupt traffic to any other local servers or proxies that you may be running on your computer. Other applications or proxies on localhost will refuse to work unless the traffic is encrypted.
If you do inadvertently turn on HSTS for localhost, you may need to reset the cache on your browser before it will accept any HTTP traffic from localhost. For Chrome, you need to Empty Cache and Hard Reload which is available from the reload menu that appears when you click and hold the reload icon from the Developer Tools Panel. For Safari, you will need to clear your cache, remove the entry from ~/Library/Cookies/HSTS.plist (or delete that file entirely) and restart Safari. Alternately, you can set the :expires option on force_ssl to 0 which should expired the entry to turn off HSTS. More information on the options for HSTS are available at Plug.SSL.

This is the documentation for the Phoenix project.
By default, Phoenix applications depend on the following packages
across these categories.
General
	Ecto - a language integrated query and
database wrapper

	ExUnit - Elixir's built-in test framework

	Gettext - Internationalization and
localization through gettext

	Phoenix - the Phoenix web framework
(these docs)

	Phoenix PubSub - a distributed
pub/sub system with presence support

	Phoenix HTML - conveniences for
working with HTML in Phoenix

	Phoenix View - a set of functions
for building Phoenix.View and working with template languages such
as Elixir's own EEx

	Phoenix LiveView - rich,
real-time user experiences with server-rendered HTML

	Phoenix LiveDashboard -
real-time performance monitoring and debugging tools for Phoenix
developers

	Plug - a specification and conveniences
for composable modules in between web applications

	Swoosh - a library for composing,
delivering and testing emails, also used by mix phx.gen.auth

	Telemetry Metrics - common
interface for defining metrics based on Telemetry events


To get started, see our overview guides.

          Returns the configured JSON encoding library for Phoenix.



      


      
        
          plug_init_mode()

        


          Returns the :plug_init_mode that controls when plugs are
initialized.

Returns the configured JSON encoding library for Phoenix.
To customize the JSON library, including the following
in your config/config.exs:
config :phoenix, :json_library, Jason

Returns the :plug_init_mode that controls when plugs are
initialized.
We recommend to set it to :runtime in development for
compilation time improvements. It must be :compile in
production (the default).
This option is passed as the :init_mode to Plug.Builder.compile/3.

Defines a Phoenix Channel.
Channels provide a means for bidirectional communication from clients that
integrate with the Phoenix.PubSub layer for soft-realtime functionality.
Topics & Callbacks
Every time you join a channel, you need to choose which particular topic you
want to listen to. The topic is just an identifier, but by convention it is
often made of two parts: "topic:subtopic". Using the "topic:subtopic"
approach pairs nicely with the Phoenix.Socket.channel/3 allowing you to
match on all topics starting with a given prefix by using a splat (the *
character) as the last character in the topic pattern:
channel "room:*", MyAppWeb.RoomChannel
Any topic coming into the router with the "room:" prefix would dispatch
to MyAppWeb.RoomChannel in the above example. Topics can also be pattern
matched in your channels' join/3 callback to pluck out the scoped pattern:
# handles the special `"lobby"` subtopic
def join("room:lobby", _payload, socket) do
  {:ok, socket}
end

# handles any other subtopic as the room ID, for example `"room:12"`, `"room:34"`
def join("room:" <> room_id, _payload, socket) do
  {:ok, socket}
end
Authorization
Clients must join a channel to send and receive PubSub events on that channel.
Your channels must implement a join/3 callback that authorizes the socket
for the given topic. For example, you could check if the user is allowed to
join that particular room.
To authorize a socket in join/3, return {:ok, socket}.
To refuse authorization in join/3, return {:error, reply}.
Incoming Events
After a client has successfully joined a channel, incoming events from the
client are routed through the channel's handle_in/3 callbacks. Within these
callbacks, you can perform any action. Typically you'll either forward a
message to all listeners with broadcast!/3, or push a message directly down
the socket with push/3. Incoming callbacks must return the socket to
maintain ephemeral state.
Here's an example of receiving an incoming "new_msg" event from one client,
and broadcasting the message to all topic subscribers for this socket.
def handle_in("new_msg", %{"uid" => uid, "body" => body}, socket) do
  broadcast!(socket, "new_msg", %{uid: uid, body: body})
  {:noreply, socket}
end
General message payloads are received as maps, and binary data payloads are
passed as a {:binary, data} tuple:
def handle_in("file_chunk", {:binary, chunk}, socket) do
  ...
  {:reply, :ok, socket}
end
You can also push a message directly down the socket, in the form of a map,
or a tagged {:binary, data} tuple:
# client asks for their current rank, push sent directly as a new event.
def handle_in("current_rank", _, socket) do
  push(socket, "current_rank", %{val: Game.get_rank(socket.assigns[:user])})
  push(socket, "photo", {:binary, File.read!(socket.assigns.photo_path)})
  {:noreply, socket}
end
Replies
In addition to pushing messages out when you receive a handle_in event,
you can also reply directly to a client event for request/response style
messaging. This is useful when a client must know the result of an operation
or to simply ack messages.
For example, imagine creating a resource and replying with the created record:
def handle_in("create:post", attrs, socket) do
  changeset = Post.changeset(%Post{}, attrs)

  if changeset.valid? do
    post = Repo.insert!(changeset)
    response = MyAppWeb.PostView.render("show.json", %{post: post})
    {:reply, {:ok, response}, socket}
  else
    response = MyAppWeb.ChangesetView.render("errors.json", %{changeset: changeset})
    {:reply, {:error, response}, socket}
  end
end
Alternatively, you may just want to ack the status of the operation:
def handle_in("create:post", attrs, socket) do
  changeset = Post.changeset(%Post{}, attrs)

  if changeset.valid? do
    Repo.insert!(changeset)
    {:reply, :ok, socket}
  else
    {:reply, :error, socket}
  end
end
Like binary pushes, binary data is also supported with replies via a {:binary, data} tuple:
{:reply, {:ok, {:binary, bin}}, socket}
Intercepting Outgoing Events
When an event is broadcasted with broadcast/3, each channel subscriber can
choose to intercept the event and have their handle_out/3 callback triggered.
This allows the event's payload to be customized on a socket by socket basis
to append extra information, or conditionally filter the message from being
delivered. If the event is not intercepted with Phoenix.Channel.intercept/1,
then the message is pushed directly to the client:
intercept ["new_msg", "user_joined"]

# for every socket subscribing to this topic, append an `is_editable`
# value for client metadata.
def handle_out("new_msg", msg, socket) do
  push(socket, "new_msg", Map.merge(msg,
    %{is_editable: User.can_edit_message?(socket.assigns[:user], msg)}
  ))
  {:noreply, socket}
end

# do not send broadcasted `"user_joined"` events if this socket's user
# is ignoring the user who joined.
def handle_out("user_joined", msg, socket) do
  unless User.ignoring?(socket.assigns[:user], msg.user_id) do
    push(socket, "user_joined", msg)
  end
  {:noreply, socket}
end
Broadcasting to an external topic
In some cases, you will want to broadcast messages without the context of
a socket. This could be for broadcasting from within your channel to an
external topic, or broadcasting from elsewhere in your application like a
controller or another process. Such can be done via your endpoint:
# within channel
def handle_in("new_msg", %{"uid" => uid, "body" => body}, socket) do
  ...
  broadcast_from!(socket, "new_msg", %{uid: uid, body: body})
  MyAppWeb.Endpoint.broadcast_from!(self(), "room:superadmin",
    "new_msg", %{uid: uid, body: body})
  {:noreply, socket}
end

# within controller
def create(conn, params) do
  ...
  MyAppWeb.Endpoint.broadcast!("room:" <> rid, "new_msg", %{uid: uid, body: body})
  MyAppWeb.Endpoint.broadcast!("room:superadmin", "new_msg", %{uid: uid, body: body})
  redirect(conn, to: "/")
end
Terminate
On termination, the channel callback terminate/2 will be invoked with
the error reason and the socket.
If we are terminating because the client left, the reason will be
{:shutdown, :left}. Similarly, if we are terminating because the
client connection was closed, the reason will be {:shutdown, :closed}.
If any of the callbacks return a :stop tuple, it will also
trigger terminate with the reason given in the tuple.
terminate/2, however, won't be invoked in case of errors nor in
case of exits. This is the same behaviour as you find in Elixir
abstractions like GenServer and others. Similar to GenServer,
it would also be possible :trap_exit to guarantee that terminate/2
is invoked. This practice is not encouraged though.
Typically speaking, if you want to clean something up, it is better to
monitor your channel process and do the clean up from another process.
All channel callbacks including join/3 are called from within the
channel process. Therefore, self() in any of them returns the PID to
be monitored.
Exit reasons when stopping a channel
When the channel callbacks return a :stop tuple, such as:
{:stop, :shutdown, socket}
{:stop, {:error, :enoent}, socket}
the second argument is the exit reason, which follows the same behaviour as
standard GenServer exits.
You have three options to choose from when shutting down a channel:
	:normal - in such cases, the exit won't be logged and linked processes
do not exit

	:shutdown or {:shutdown, term} - in such cases, the exit won't be
logged and linked processes exit with the same reason unless they're
trapping exits

	any other term - in such cases, the exit will be logged and linked
processes exit with the same reason unless they're trapping exits


Subscribing to external topics
Sometimes you may need to programmatically subscribe a socket to external
topics in addition to the internal socket.topic. For example,
imagine you have a bidding system where a remote client dynamically sets
preferences on products they want to receive bidding notifications on.
Instead of requiring a unique channel process and topic per
preference, a more efficient and simple approach would be to subscribe a
single channel to relevant notifications via your endpoint. For example:
defmodule MyAppWeb.Endpoint.NotificationChannel do
  use Phoenix.Channel

  def join("notification:" <> user_id, %{"ids" => ids}, socket) do
    topics = for product_id <- ids, do: "product:#{product_id}"

    {:ok, socket
          |> assign(:topics, [])
          |> put_new_topics(topics)}
  end

  def handle_in("watch", %{"product_id" => id}, socket) do
    {:reply, :ok, put_new_topics(socket, ["product:#{id}"])}
  end

  def handle_in("unwatch", %{"product_id" => id}, socket) do
    {:reply, :ok, MyAppWeb.Endpoint.unsubscribe("product:#{id}")}
  end

  defp put_new_topics(socket, topics) do
    Enum.reduce(topics, socket, fn topic, acc ->
      topics = acc.assigns.topics
      if topic in topics do
        acc
      else
        :ok = MyAppWeb.Endpoint.subscribe(topic)
        assign(acc, :topics, [topic | topics])
      end
    end)
  end
end
Note: the caller must be responsible for preventing duplicate subscriptions.
After calling subscribe/1 from your endpoint, the same flow applies to
handling regular Elixir messages within your channel. Most often, you'll
simply relay the %Phoenix.Socket.Broadcast{} event and payload:
alias Phoenix.Socket.Broadcast
def handle_info(%Broadcast{topic: _, event: event, payload: payload}, socket) do
  push(socket, event, payload)
  {:noreply, socket}
end
Hibernation
From Erlang/OTP 20, channels automatically hibernate to save memory
after 15_000 milliseconds of inactivity. This can be customized by
passing the :hibernate_after option to use Phoenix.Channel:
use Phoenix.Channel, hibernate_after: 60_000
You can also set it to :infinity to fully disable it.
Shutdown
You can configure the shutdown of each channel used when your application
is shutting down by setting the :shutdown value on use:
use Phoenix.Channel, shutdown: 5_000
It defaults to 5_000.
Logging
By default, channel "join" and "handle_in" events are logged, using
the level :info and :debug, respectively. Logs can be customized per
event type or disabled by setting the :log_join and :log_handle_in
options when using Phoenix.Channel. For example, the following
configuration logs join events as :info, but disables logging for
incoming events:
use Phoenix.Channel, log_join: :info, log_handle_in: false

          Invoked when the channel process is about to exit.

          broadcast!(socket, event, message)

        


          Same as broadcast/3, but raises if broadcast fails.



      


      
        
          broadcast(socket, event, message)

        


          Broadcast an event to all subscribers of the socket topic.

          Same as broadcast_from/3, but raises if broadcast fails.

          Broadcast event from pid to all subscribers of the socket topic.



      


      
        
          intercept(events)

        


          Defines which Channel events to intercept for handle_out/3 callbacks.

          @type payload() :: map() | {:binary, binary()}

          @type reply() :: status :: atom() | {status :: atom(), response :: payload()}

          @type socket_ref() ::
  {transport_pid :: Pid, serializer :: module(), topic :: binary(),
   ref :: binary(), join_ref :: binary()}

          @callback code_change(old_vsn, Phoenix.Socket.t(), extra :: term()) ::
  {:ok, Phoenix.Socket.t()} | {:error, reason :: term()}
when old_vsn: term() | {:down, term()}

          @callback handle_call(
  msg :: term(),
  from :: {pid(), tag :: term()},
  socket :: Phoenix.Socket.t()
) ::
  {:reply, response :: term(), Phoenix.Socket.t()}
  | {:noreply, Phoenix.Socket.t()}
  | {:stop, reason :: term(), Phoenix.Socket.t()}


      


Handle regular GenServer call messages.
See GenServer.handle_call/3.

          @callback handle_cast(msg :: term(), socket :: Phoenix.Socket.t()) ::
  {:noreply, Phoenix.Socket.t()} | {:stop, reason :: term(), Phoenix.Socket.t()}


      


Handle regular GenServer cast messages.
See GenServer.handle_cast/2.

          @callback handle_in(
  event :: String.t(),
  payload :: payload(),
  socket :: Phoenix.Socket.t()
) ::
  {:noreply, Phoenix.Socket.t()}
  | {:noreply, Phoenix.Socket.t(), timeout() | :hibernate}
  | {:reply, reply(), Phoenix.Socket.t()}
  | {:stop, reason :: term(), Phoenix.Socket.t()}
  | {:stop, reason :: term(), reply(), Phoenix.Socket.t()}


      


Handle incoming events.

  
  example

  
  Example


def handle_in("ping", payload, socket) do
  {:reply, {:ok, payload}, socket}
end

  



  
    
      
      Link to this callback
    
    handle_info(msg, socket)


      
       
       View Source
     


      (optional)

  


  

      

          @callback handle_info(msg :: term(), socket :: Phoenix.Socket.t()) ::
  {:noreply, Phoenix.Socket.t()} | {:stop, reason :: term(), Phoenix.Socket.t()}


      


Handle regular Elixir process messages.
See GenServer.handle_info/2.

          @callback handle_out(
  event :: String.t(),
  payload :: payload(),
  socket :: Phoenix.Socket.t()
) ::
  {:noreply, Phoenix.Socket.t()}
  | {:noreply, Phoenix.Socket.t(), timeout() | :hibernate}
  | {:stop, reason :: term(), Phoenix.Socket.t()}


      


Intercepts outgoing events.
See intercept/1.

          @callback join(topic :: binary(), payload :: payload(), socket :: Phoenix.Socket.t()) ::
  {:ok, Phoenix.Socket.t()}
  | {:ok, reply :: payload(), Phoenix.Socket.t()}
  | {:error, reason :: map()}


      


Handle channel joins by topic.
To authorize a socket, return {:ok, socket} or {:ok, reply, socket}. To
refuse authorization, return {:error, reason}.

  
  example

  
  Example


def join("room:lobby", payload, socket) do
  if authorized?(payload) do
    {:ok, socket}
  else
    {:error, %{reason: "unauthorized"}}
  end
end

          @callback terminate(
  reason :: :normal | :shutdown | {:shutdown, :left | :closed | term()},
  Phoenix.Socket.t()
) :: term()


      


Invoked when the channel process is about to exit.
See GenServer.terminate/2.

Same as broadcast/3, but raises if broadcast fails.

  



  
    
      
      Link to this function

Broadcast an event to all subscribers of the socket topic.
The event's message must be a serializable map or a tagged {:binary, data}
tuple where data is binary data.

  
  examples

  
  Examples


iex> broadcast(socket, "new_message", %{id: 1, content: "hello"})
:ok

iex> broadcast(socket, "new_message", {:binary, "hello"})
:ok

    broadcast_from!(socket, event, message)

Same as broadcast_from/3, but raises if broadcast fails.

    broadcast_from(socket, event, message)

Broadcast event from pid to all subscribers of the socket topic.
The channel that owns the socket will not receive the published
message. The event's message must be a serializable map or a tagged
{:binary, data} tuple where data is binary data.

  
  examples

  
  Examples


iex> broadcast_from(socket, "new_message", %{id: 1, content: "hello"})
:ok

iex> broadcast_from(socket, "new_message", {:binary, "hello"})
:ok

Defines which Channel events to intercept for handle_out/3 callbacks.
By default, broadcasted events are pushed directly to the client, but
intercepting events gives your channel a chance to customize the event
for the client to append extra information or filter the message from being
delivered.
Note: intercepting events can introduce significantly more overhead if a
large number of subscribers must customize a message since the broadcast will
be encoded N times instead of a single shared encoding across all subscribers.

  
  examples

  
  Examples


intercept ["new_msg"]

def handle_out("new_msg", payload, socket) do
  push(socket, "new_msg", Map.merge(payload,
    is_editable: User.can_edit_message?(socket.assigns[:user], payload)
  ))
  {:noreply, socket}
end
handle_out/3 callbacks must return one of:
{:noreply, Socket.t} |
{:noreply, Socket.t, timeout | :hibernate} |
{:stop, reason :: term, Socket.t}

Sends event to the socket.
The event's message must be a serializable map or a tagged {:binary, data}
tuple where data is binary data.

  
  examples

  
  Examples


iex> push(socket, "new_message", %{id: 1, content: "hello"})
:ok

iex> push(socket, "new_message", {:binary, "hello"})
:ok

          @spec reply(socket_ref(), reply()) :: :ok


      


Replies asynchronously to a socket push.
Useful when you need to reply to a push that can't otherwise be handled using
the {:reply, {status, payload}, socket} return from your handle_in
callbacks. reply/2 will be used in the rare cases you need to perform work in
another process and reply when finished by generating a reference to the push
with socket_ref/1.
Note: In such cases, a socket_ref should be generated and
passed to the external process, so the socket itself is not leaked outside
the channel. The socket holds information such as assigns and transport
configuration, so it's important to not copy this information outside of the
channel that owns it.

  
  examples

  
  Examples


def handle_in("work", payload, socket) do
  Worker.perform(payload, socket_ref(socket))
  {:noreply, socket}
end

def handle_info({:work_complete, result, ref}, socket) do
  reply(ref, {:ok, result})
  {:noreply, socket}
end

          @spec socket_ref(Phoenix.Socket.t()) :: socket_ref()


      


Generates a socket_ref for an async reply.
See reply/2 for example usage.

Controllers are used to group common functionality in the same
(pluggable) module.
For example, the route:
get "/users/:id", MyAppWeb.UserController, :show
will invoke the show/2 action in the MyAppWeb.UserController:
defmodule MyAppWeb.UserController do
  use MyAppWeb, :controller

  def show(conn, %{"id" => id}) do
    user = Repo.get(User, id)
    render(conn, "show.html", user: user)
  end
end
An action is a regular function that receives the connection
and the request parameters as arguments. The connection is a
Plug.Conn struct, as specified by the Plug library.
Options
When used, the controller supports the following options:
	:namespace - sets the namespace to properly inflect
the layout view. By default it uses the base alias
in your controller name

	:put_default_views - controls whether the default view
and layout should be set or not


Connection
A controller by default provides many convenience functions for
manipulating the connection, rendering templates, and more.
Those functions are imported from two modules:
	Plug.Conn - a collection of low-level functions to work with
the connection

	Phoenix.Controller - functions provided by Phoenix
to support rendering, and other Phoenix specific behaviour


If you want to have functions that manipulate the connection
without fully implementing the controller, you can import both
modules directly instead of use Phoenix.Controller.
Plug pipeline
As with routers, controllers also have their own plug pipeline.
However, different from routers, controllers have a single pipeline:
defmodule MyAppWeb.UserController do
  use MyAppWeb, :controller

  plug :authenticate, usernames: ["jose", "eric", "sonny"]

  def show(conn, params) do
    # authenticated users only
  end

  defp authenticate(conn, options) do
    if get_session(conn, :username) in options[:usernames] do
      conn
    else
      conn |> redirect(to: "/") |> halt()
    end
  end
end
The :authenticate plug will be invoked before the action. If the
plug calls Plug.Conn.halt/1 (which is by default imported into
controllers), it will halt the pipeline and won't invoke the action.
Guards
plug/2 in controllers supports guards, allowing a developer to configure
a plug to only run in some particular action:
plug :authenticate, [usernames: ["jose", "eric", "sonny"]] when action in [:show, :edit]
plug :authenticate, [usernames: ["admin"]] when not action in [:index]
The first plug will run only when action is show or edit. The second plug will
always run, except for the index action.
Those guards work like regular Elixir guards and the only variables accessible
in the guard are conn, the action as an atom and the controller as an
alias.
Controllers are plugs
Like routers, controllers are plugs, but they are wired to dispatch
to a particular function which is called an action.
For example, the route:
get "/users/:id", UserController, :show
will invoke UserController as a plug:
UserController.call(conn, :show)
which will trigger the plug pipeline and which will eventually
invoke the inner action plug that dispatches to the show/2
function in UserController.
As controllers are plugs, they implement both init/1 and
call/2, and it also provides a function named action/2
which is responsible for dispatching the appropriate action
after the plug stack (and is also overridable).
Overriding action/2 for custom arguments
Phoenix injects an action/2 plug in your controller which calls the
function matched from the router. By default, it passes the conn and params.
In some cases, overriding the action/2 plug in your controller is a
useful way to inject arguments into your actions that you would otherwise
need to repeatedly fetch off the connection. For example, imagine if you
stored a conn.assigns.current_user in the connection and wanted quick
access to the user for every action in your controller:
def action(conn, _) do
  args = [conn, conn.params, conn.assigns.current_user]
  apply(__MODULE__, action_name(conn), args)
end

def index(conn, _params, user) do
  videos = Repo.all(user_videos(user))
  # ...
end

def delete(conn, %{"id" => id}, user) do
  video = Repo.get!(user_videos(user), id)
  # ...
end
Rendering and layouts
One of the main features provided by controllers is the ability
to perform content negotiation and render templates based on
information sent by the client. Read render/3 to learn more.
It is also important not to confuse Phoenix.Controller.render/3
with Phoenix.View.render/3. The former expects
a connection and relies on content negotiation while the latter is
connection-agnostic and typically invoked from your views.

          @spec accepts(Plug.Conn.t(), [binary()]) :: Plug.Conn.t() | no_return()


      


Performs content negotiation based on the available formats.
It receives a connection, a list of formats that the server
is capable of rendering and then proceeds to perform content
negotiation based on the request information. If the client
accepts any of the given formats, the request proceeds.
If the request contains a "_format" parameter, it is
considered to be the format desired by the client. If no
"_format" parameter is available, this function will parse
the "accept" header and find a matching format accordingly.
This function is useful when you may want to serve different
content-types (such as JSON and HTML) from the same routes.
However, if you always have distinct routes, you can also
disable content negotiation and simply hardcode your format
of choice in your route pipelines:
plug :put_format, "html"
It is important to notice that browsers have historically
sent bad accept headers. For this reason, this function will
default to "html" format whenever:
	the accepted list of arguments contains the "html" format

	the accept header specified more than one media type preceded
or followed by the wildcard media type "*/*"


This function raises Phoenix.NotAcceptableError, which is rendered
with status 406, whenever the server cannot serve a response in any
of the formats expected by the client.

  
  examples

  
  Examples


accepts/2 can be invoked as a function:
iex> accepts(conn, ["html", "json"])
or used as a plug:
plug :accepts, ["html", "json"]
plug :accepts, ~w(html json)

  
  custom-media-types

  
  Custom media types


It is possible to add custom media types to your Phoenix application.
The first step is to teach Plug about those new media types in
your config/config.exs file:
config :mime, :types, %{
  "application/vnd.api+json" => ["json-api"]
}
The key is the media type, the value is a list of formats the
media type can be identified with. For example, by using
"json-api", you will be able to use templates with extension
"index.json-api" or to force a particular format in a given
URL by sending "?_format=json-api".
After this change, you must recompile plug:
$ mix deps.clean mime --build
$ mix deps.get

And now you can use it in accepts too:
plug :accepts, ["html", "json-api"]

Registers the plug to call as a fallback to the controller action.
A fallback plug is useful to translate common domain data structures
into a valid %Plug.Conn{} response. If the controller action fails to
return a %Plug.Conn{}, the provided plug will be called and receive
the controller's %Plug.Conn{} as it was before the action was invoked
along with the value returned from the controller action.

  
  examples

  
  Examples


defmodule MyController do
  use Phoenix.Controller

  action_fallback MyFallbackController

  def show(conn, %{"id" => id}, current_user) do
    with {:ok, post} <- Blog.fetch_post(id),
         :ok <- Authorizer.authorize(current_user, :view, post) do

      render(conn, "show.json", post: post)
    end
  end
end
In the above example, with is used to match only a successful
post fetch, followed by valid authorization for the current user.
In the event either of those fail to match, with will not invoke
the render block and instead return the unmatched value. In this case,
imagine Blog.fetch_post/2 returned {:error, :not_found} or
Authorizer.authorize/3 returned {:error, :unauthorized}. For cases
where these data structures serve as return values across multiple
boundaries in our domain, a single fallback module can be used to
translate the value into a valid response. For example, you could
write the following fallback controller to handle the above values:
defmodule MyFallbackController do
  use Phoenix.Controller

  def call(conn, {:error, :not_found}) do
    conn
    |> put_status(:not_found)
    |> put_view(MyErrorView)
    |> render(:"404")
  end

  def call(conn, {:error, :unauthorized}) do
    conn
    |> put_status(403)
    |> put_view(MyErrorView)
    |> render(:"403")
  end
end

          @spec action_name(Plug.Conn.t()) :: atom()


      


Returns the action name as an atom, raises if unavailable.

          @spec allow_jsonp(Plug.Conn.t(), Keyword.t()) :: Plug.Conn.t()


      


A plug that may convert a JSON response into a JSONP one.
In case a JSON response is returned, it will be converted
to a JSONP as long as the callback field is present in
the query string. The callback field itself defaults to
"callback", but may be configured with the callback option.
In case there is no callback or the response is not encoded
in JSON format, it is a no-op.
Only alphanumeric characters and underscore are allowed in the
callback name. Otherwise an exception is raised.

  
  examples

  
  Examples


# Will convert JSON to JSONP if callback=someFunction is given
plug :allow_jsonp

# Will convert JSON to JSONP if cb=someFunction is given
plug :allow_jsonp, callback: "cb"

Clears all flash messages.

          @spec controller_module(Plug.Conn.t()) :: atom()


      


Returns the controller module as an atom, raises if unavailable.

Returns the current request path with its default query parameters:
iex> current_path(conn)
"/users/123?existing=param"
See current_path/2 to override the default parameters.
The path is normalized based on the conn.script_name and
conn.path_info. For example, "/foo//bar/" will become "/foo/bar".
If you want the original path, use conn.request_path instead.

Returns the current path with the given query parameters.
You may also retrieve only the request path by passing an
empty map of params.

  
  examples

  
  Examples


iex> current_path(conn)
"/users/123?existing=param"

iex> current_path(conn, %{new: "param"})
"/users/123?new=param"

iex> current_path(conn, %{filter: %{status: ["draft", "published"]}})
"/users/123?filter[status][]=draft&filter[status][]=published"

iex> current_path(conn, %{})
"/users/123"
The path is normalized based on the conn.script_name and
conn.path_info. For example, "/foo//bar/" will become "/foo/bar".
If you want the original path, use conn.request_path instead.

    current_url(conn)

Returns the current request url with its default query parameters:
iex> current_url(conn)
"https://www.example.com/users/123?existing=param"
See current_url/2 to override the default parameters.

Returns the current request URL with query params.
The path will be retrieved from the currently requested path via
current_path/1. The scheme, host and others will be received from
the URL configuration in your Phoenix endpoint. The reason we don't
use the host and scheme information in the request is because most
applications are behind proxies and the host and scheme may not
actually reflect the host and scheme accessed by the client. If you
want to access the url precisely as requested by the client, see
Plug.Conn.request_url/1.

  
  examples

  
  Examples


iex> current_url(conn)
"https://www.example.com/users/123?existing=param"

iex> current_url(conn, %{new: "param"})
"https://www.example.com/users/123?new=param"

iex> current_url(conn, %{})
"https://www.example.com/users/123"

  
  custom-url-generation

  
  Custom URL Generation


In some cases, you'll need to generate a request's URL, but using a
different scheme, different host, etc. This can be accomplished in
two ways.
If you want to do so in a case-by-case basis, you can define a custom
function that gets the endpoint URI configuration and changes it accordingly.
For example, to get the current URL always in HTTPS format:
def current_secure_url(conn, params \\ %{}) do
  cur_uri  = MyAppWeb.Endpoint.struct_url()
  cur_path = Phoenix.Controller.current_path(conn, params)

  MyAppWeb.Router.Helpers.url(%URI{cur_uri | scheme: "https"}) <> cur_path
end
However, if you want all generated URLs to always have a certain schema,
host, etc, you may use put_router_url/2.

Deletes the CSRF token from the process dictionary.
Note: The token is deleted only after a response has been sent.

          @spec endpoint_module(Plug.Conn.t()) :: atom()


      


Returns the endpoint module as an atom, raises if unavailable.

Fetches the flash storage.

Gets or generates a CSRF token.
If a token exists, it is returned, otherwise it is generated and stored
in the process dictionary.

Returns a map of previously set flash messages or an empty map.

  
  examples

  
  Examples


iex> get_flash(conn)
%{}

iex> conn = put_flash(conn, :info, "Welcome Back!")
iex> get_flash(conn)
%{"info" => "Welcome Back!"}

Returns a message from flash by key (or nil if no message is available for key).

  
  examples

  
  Examples


iex> conn = put_flash(conn, :info, "Welcome Back!")
iex> get_flash(conn, :info)
"Welcome Back!"

Returns the request format, such as "json", "html".
This format is used when rendering a template as an atom.
For example, render(conn, :foo) will render "foo.FORMAT"
where the format is the one set here. The default format
is typically set from the negotiation done in accepts/2.

          @spec html(Plug.Conn.t(), iodata()) :: Plug.Conn.t()


      


Sends html response.

  
  examples

  
  Examples


iex> html(conn, "<html><head>...")

          @spec json(Plug.Conn.t(), term()) :: Plug.Conn.t()


      


Sends JSON response.
It uses the configured :json_library under the :phoenix
application for :json to pick up the encoder module.

  
  examples

  
  Examples


iex> json(conn, %{id: 123})

          @spec layout(Plug.Conn.t()) :: {atom(), String.t() | atom()} | false


      


Retrieves the current layout.

          @spec layout_formats(Plug.Conn.t()) :: [String.t()]


      


Retrieves current layout formats.

  



  
    
      
      Link to this function

Merges a map into the flash.
Returns the updated connection.

  
  examples

  
  Examples


iex> conn = merge_flash(conn, info: "Welcome Back!")
iex> get_flash(conn, :info)
"Welcome Back!"

    protect_from_forgery(conn, opts \\ [])

Enables CSRF protection.
Currently used as a wrapper function for Plug.CSRFProtection
and mainly serves as a function plug in YourApp.Router.
Check get_csrf_token/0 and delete_csrf_token/0 for
retrieving and deleting CSRF tokens.

    put_flash(conn, key, message)

Persists a value in flash.
Returns the updated connection.

  
  examples

  
  Examples


iex> conn = put_flash(conn, :info, "Welcome Back!")
iex> get_flash(conn, :info)
"Welcome Back!"

    put_format(conn, format)

Puts the format in the connection.
This format is used when rendering a template as an atom.
For example, render(conn, :foo) will render "foo.FORMAT"
where the format is the one set here. The default format
is typically set from the negotiation done in accepts/2.
See get_format/1 for retrieval.

          @spec put_layout(
  Plug.Conn.t(),
  {atom(), binary() | atom()} | atom() | binary() | false
) ::
  Plug.Conn.t()


      


Stores the layout for rendering.
The layout must be a tuple, specifying the layout view and the layout
name, or false. In case a previous layout is set, put_layout also
accepts the layout name to be given as a string or as an atom. If a
string, it must contain the format. Passing an atom means the layout
format will be found at rendering time, similar to the template in
render/3. It can also be set to false. In this case, no layout
would be used.

  
  examples

  
  Examples


iex> layout(conn)
false

iex> conn = put_layout conn, {AppView, "application.html"}
iex> layout(conn)
{AppView, "application.html"}

iex> conn = put_layout conn, "print.html"
iex> layout(conn)
{AppView, "print.html"}

iex> conn = put_layout conn, :print
iex> layout(conn)
{AppView, :print}
Raises Plug.Conn.AlreadySentError if conn is already sent.

  



  
    
      
      Link to this function
    
    put_layout_formats(conn, formats)

          @spec put_layout_formats(Plug.Conn.t(), [String.t()]) :: Plug.Conn.t()


      


Sets which formats have a layout when rendering.

  
  examples

  
  Examples


iex> layout_formats(conn)
["html"]

iex> put_layout_formats(conn, ["html", "mobile"])
iex> layout_formats(conn)
["html", "mobile"]
Raises Plug.Conn.AlreadySentError if conn is already sent.

          @spec put_new_layout(Plug.Conn.t(), {atom(), binary() | atom()} | false) ::
  Plug.Conn.t()


      


Stores the layout for rendering if one was not stored yet.
Raises Plug.Conn.AlreadySentError if conn is already sent.

  



  
    
      
      Link to this function
    
    put_new_view(conn, module)

          @spec put_new_view(Plug.Conn.t(), atom()) :: Plug.Conn.t()


      


Stores the view for rendering if one was not stored yet.
Raises Plug.Conn.AlreadySentError if conn is already sent.

  



  
    
      
      Link to this function
    
    put_root_layout(conn, layout)

          @spec put_root_layout(
  Plug.Conn.t(),
  {atom(), binary() | atom()} | atom() | binary() | false
) ::
  Plug.Conn.t()


      


Stores the root layout for rendering.
Like put_layout/2, the layout must be a tuple,
specifying the layout view and the layout name, or false.
In case a previous layout is set, put_root_layout also
accepts the layout name to be given as a string or as an atom. If a
string, it must contain the format. Passing an atom means the layout
format will be found at rendering time, similar to the template in
render/3. It can also be set to false. In this case, no layout
would be used.

  
  examples

  
  Examples


iex> root_layout(conn)
false

iex> conn = put_root_layout conn, {AppView, "root.html"}
iex> root_layout(conn)
{AppView, "root.html"}

iex> conn = put_root_layout conn, "bare.html"
iex> root_layout(conn)
{AppView, "bare.html"}

iex> conn = put_root_layout conn, :bare
iex> root_layout(conn)
{AppView, :bare}
Raises Plug.Conn.AlreadySentError if conn is already sent.

    put_router_url(conn, uri)

Puts the url string or %URI{} to be used for route generation.
This function overrides the default URL generation pulled
from the %Plug.Conn{}'s endpoint configuration.

  
  examples

  
  Examples


Imagine your application is configured to run on "example.com"
but after the user signs in, you want all links to use
"some_user.example.com". You can do so by setting the proper
router url configuration:
def put_router_url_by_user(conn) do
  put_router_url(conn, get_user_from_conn(conn).account_name <> ".example.com")
end
Now when you call Routes.some_route_url(conn, ...), it will use
the router url set above. Keep in mind that, if you want to generate
routes to the current domain, it is preferred to use
Routes.some_route_path helpers, as those are always relative.

    put_secure_browser_headers(conn, headers \\ %{})

Put headers that improve browser security.
It sets the following headers:
	x-frame-options - set to SAMEORIGIN to avoid clickjacking
through iframes unless in the same origin
	x-content-type-options - set to nosniff. This requires
script and style tags to be sent with proper content type
	x-xss-protection - set to "1; mode=block" to improve XSS
protection on both Chrome and IE
	x-download-options - set to noopen to instruct the browser
not to open a download directly in the browser, to avoid
HTML files rendering inline and accessing the security
context of the application (like critical domain cookies)
	x-permitted-cross-domain-policies - set to none to restrict
Adobe Flash Player’s access to data
	cross-origin-window-policy - set to deny to avoid window
control attacks

A custom headers map may also be given to be merged with defaults.
It is recommended for custom header keys to be in lowercase, to avoid sending
duplicate keys in a request.
Additionally, responses with mixed-case headers served over HTTP/2 are not
considered valid by common clients, resulting in dropped responses.

Puts the URL or %URI{} to be used for the static url generation.
Using this function on a %Plug.Conn{} struct tells static_url/2 to use
the given information for URL generation instead of the the %Plug.Conn{}'s
endpoint configuration (much like put_router_url/2 but for static URLs).

  



  
    
      
      Link to this function
    
    put_view(conn, module)


      
       
       View Source
     


  


  

      

          @spec put_view(Plug.Conn.t(), atom()) :: Plug.Conn.t()


      


Stores the view for rendering.
Raises Plug.Conn.AlreadySentError if conn is already sent.

  



  
    
      
      Link to this function
    
    redirect(conn, opts)


      
       
       View Source
     


  


  

Sends redirect response to the given url.
For security, :to only accepts paths. Use the :external
option to redirect to any URL.
The response will be sent with the status code defined within
the connection, via Plug.Conn.put_status/2. If no status
code is set, a 302 response is sent.

  
  examples

  
  Examples


iex> redirect(conn, to: "/login")

iex> redirect(conn, external: "https://elixir-lang.org")

  



    

  
    
      
      Link to this function
    
    render(conn, template_or_assigns \\ [])


      
       
       View Source
     


  


  

      

          @spec render(Plug.Conn.t(), Keyword.t() | map() | binary() | atom()) :: Plug.Conn.t()


      


Render the given template or the default template
specified by the current action with the given assigns.
See render/3 for more information.

  



  
    
      
      Link to this function
    
    render(conn, template, assigns)


      
       
       View Source
     


  


  

      

          @spec render(
  Plug.Conn.t(),
  binary() | atom(),
  Keyword.t() | map() | binary() | atom()
) :: Plug.Conn.t()


      


Renders the given template and assigns based on the conn information.
Once the template is rendered, the template format is set as the response
content type (for example, an HTML template will set "text/html" as response
content type) and the data is sent to the client with default status of 200.

  
  arguments

  
  Arguments


	conn - the Plug.Conn struct

	template - which may be an atom or a string. If an atom, like :index,
it will render a template with the same format as the one returned by
get_format/1. For example, for an HTML request, it will render
the "index.html" template. If the template is a string, it must contain
the extension too, like "index.json"

	assigns - a dictionary with the assigns to be used in the view. Those
assigns are merged and have higher precedence than the connection assigns
(conn.assigns)



  
  examples

  
  Examples


defmodule MyAppWeb.UserController do
  use Phoenix.Controller

  def show(conn, _params) do
    render(conn, "show.html", message: "Hello")
  end
end
The example above renders a template "show.html" from the MyAppWeb.UserView
and sets the response content type to "text/html".
In many cases, you may want the template format to be set dynamically based
on the request. To do so, you can pass the template name as an atom (without
the extension):
def show(conn, _params) do
  render(conn, :show, message: "Hello")
end
In order for the example above to work, we need to do content negotiation with
the accepts plug before rendering. You can do so by adding the following to your
pipeline (in the router):
plug :accepts, ["html"]

  
  views

  
  Views


By default, Controllers render templates in a view with a similar name to the
controller. For example, MyAppWeb.UserController will render templates inside
the MyAppWeb.UserView. This information can be changed any time by using the
put_view/2 function:
def show(conn, _params) do
  conn
  |> put_view(MyAppWeb.SpecialView)
  |> render(:show, message: "Hello")
end
put_view/2 can also be used as a plug:
defmodule MyAppWeb.UserController do
  use Phoenix.Controller

  plug :put_view, MyAppWeb.SpecialView

  def show(conn, _params) do
    render(conn, :show, message: "Hello")
  end
end

  
  layouts

  
  Layouts


Templates are often rendered inside layouts. By default, Phoenix
will render layouts for html requests. For example:
defmodule MyAppWeb.UserController do
  use Phoenix.Controller

  def show(conn, _params) do
    render(conn, "show.html", message: "Hello")
  end
end
will render the  "show.html" template inside an "app.html"
template specified in MyAppWeb.LayoutView. put_layout/2 can be used
to change the layout, similar to how put_view/2 can be used to change
the view.
layout_formats/1 and put_layout_formats/2 can be used to configure
which formats support/require layout rendering (defaults to "html" only).

  



  
    
      
      Link to this function
    
    root_layout(conn)


      
       
       View Source
     


  


  

      

          @spec root_layout(Plug.Conn.t()) :: {atom(), String.t() | atom()} | false


      


Retrieves the current root layout.

  



  
    
      
      Link to this function
    
    router_module(conn)


      
       
       View Source
     


  


  

      

          @spec router_module(Plug.Conn.t()) :: atom()


      


Returns the router module as an atom, raises if unavailable.

  



  
    
      
      Link to this function
    
    scrub_params(conn, required_key)


      
       
       View Source
     


  


  

      

          @spec scrub_params(Plug.Conn.t(), String.t()) :: Plug.Conn.t()


      


Scrubs the parameters from the request.
This process is two-fold:
	Checks to see if the required_key is present
	Changes empty parameters of required_key (recursively) to nils

This function is useful for removing empty strings sent
via HTML forms. If you are providing an API, there
is likely no need to invoke scrub_params/2.
If the required_key is not present, it will
raise Phoenix.MissingParamError.

  
  examples

  
  Examples


iex> scrub_params(conn, "user")

  



    

  
    
      
      Link to this function
    
    send_download(conn, kind, opts \\ [])


      
       
       View Source
     


  


  

Sends the given file or binary as a download.
The second argument must be {:binary, contents}, where
contents will be sent as download, or{:file, path},
where path is the filesystem location of the file to
be sent. Be careful to not interpolate the path from
external parameters, as it could allow traversal of the
filesystem.
The download is achieved by setting "content-disposition"
to attachment. The "content-type" will also be set based
on the extension of the given filename but can be customized
via the :content_type and :charset options.

  
  options

  
  Options


	:filename - the filename to be presented to the user
as download
	:content_type - the content type of the file or binary
sent as download. It is automatically inferred from the
filename extension
	:disposition - specifies disposition type
(:attachment or :inline). If :attachment was used,
user will be prompted to save the file. If :inline was used,
the browser will attempt to open the file.
Defaults to :attachment.
	:charset - the charset of the file, such as "utf-8".
Defaults to none
	:offset - the bytes to offset when reading. Defaults to 0
	:length - the total bytes to read. Defaults to :all
	:encode - encodes the filename using URI.encode_www_form/1.
Defaults to true. When false, disables encoding. If you
disable encoding, you need to guarantee there are no special
characters in the filename, such as quotes, newlines, etc.
Otherwise you can expose your application to security attacks


  
  examples

  
  Examples


To send a file that is stored inside your application priv
directory:
path = Application.app_dir(:my_app, "priv/prospectus.pdf")
send_download(conn, {:file, path})
When using {:file, path}, the filename is inferred from the
given path but may also be set explicitly.
To allow the user to download contents that are in memory as
a binary or string:
send_download(conn, {:binary, "world"}, filename: "hello.txt")
See Plug.Conn.send_file/3 and Plug.Conn.send_resp/3 if you
would like to access the low-level functions used to send files
and responses via Plug.

  



  
    
      
      Link to this function
    
    status_message_from_template(template)


      
       
       View Source
     


  


  

Generates a status message from the template name.

  
  examples

  
  Examples


iex> status_message_from_template("404.html")
"Not Found"
iex> status_message_from_template("whatever.html")
"Internal Server Error"

  



  
    
      
      Link to this function
    
    text(conn, data)


      
       
       View Source
     


  


  

      

          @spec text(Plug.Conn.t(), String.Chars.t()) :: Plug.Conn.t()


      


Sends text response.

  
  examples

  
  Examples


iex> text(conn, "hello")

iex> text(conn, :implements_to_string)

  



  
    
      
      Link to this function
    
    view_module(conn)


      
       
       View Source
     


  


  

      

          @spec view_module(Plug.Conn.t()) :: atom()


      


Retrieves the current view.

  



  
    
      
      Link to this function
    
    view_template(conn)


      
       
       View Source
     


  


  

      

          @spec view_template(Plug.Conn.t()) :: binary() | nil


      


Returns the template name rendered in the view as a string
(or nil if no template was rendered).

Phoenix.Digester.Compressor behaviour
    



      
Defines the Phoenix.Digester.Compressor behaviour for
implementing static file compressors.
A custom compressor expects 2 functions to be implemented.
By default, Phoenix uses only Phoenix.Digester.Gzip to compress
static files, but additional compressors can be defined and added
to the digest process.
Example
If you wanted to compress files using an external brotli compression
library, you could define a new module implementing the behaviour and add the
module to the list of configured Phoenix static compressors.
defmodule MyApp.BrotliCompressor do
  @behaviour Phoenix.Digester.Compressor

  def compress_file(file_path, content) do
    valid_extension = Path.extname(file_path) in Application.fetch_env!(:phoenix, :gzippable_exts)
    compressed_content = :brotli.encode(content)

    if valid_extension && byte_size(compressed_content) < byte_size(content) do
      {:ok, compressed_content}
    else
      :error
    end
  end

  def file_extensions do
    [".br"]
  end
end

# config/config.exs
config :phoenix,
  static_compressors: [Phoenix.Digester.Gzip, MyApp.BrotliCompressor],
  # ...

      Link to this callback
    
    compress_file(t, binary)

          @callback compress_file(Path.t(), binary()) :: {:ok, binary()} | :error


      



  



  
    
      
      Link to this callback
    
    file_extensions()

          @callback file_extensions() :: [String.t(), ...]

Phoenix.Digester.Gzip 
    



      
Gzip compressor for Phoenix.Digester

          Callback implementation for Phoenix.Digester.Compressor.compress_file/2.

          Callback implementation for Phoenix.Digester.Compressor.file_extensions/0.

      Link to this function
    
    compress_file(file_path, content)

Callback implementation for Phoenix.Digester.Compressor.compress_file/2.

Callback implementation for Phoenix.Digester.Compressor.file_extensions/0.

Defines a Phoenix endpoint.
The endpoint is the boundary where all requests to your
web application start. It is also the interface your
application provides to the underlying web servers.
Overall, an endpoint has three responsibilities:
	to provide a wrapper for starting and stopping the
endpoint as part of a supervision tree

	to define an initial plug pipeline for requests
to pass through

	to host web specific configuration for your
application


Endpoints
An endpoint is simply a module defined with the help
of Phoenix.Endpoint. If you have used the mix phx.new
generator, an endpoint was automatically generated as
part of your application:
defmodule YourAppWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :your_app

  # plug ...
  # plug ...

  plug YourApp.Router
end
Endpoints must be explicitly started as part of your application
supervision tree. Endpoints are added by default
to the supervision tree in generated applications. Endpoints can be
added to the supervision tree as follows:
children = [
  YourAppWeb.Endpoint
]
Endpoint configuration
All endpoints are configured in your application environment.
For example:
config :your_app, YourAppWeb.Endpoint,
  secret_key_base: "kjoy3o1zeidquwy1398juxzldjlksahdk3"
Endpoint configuration is split into two categories. Compile-time
configuration means the configuration is read during compilation
and changing it at runtime has no effect. The compile-time
configuration is mostly related to error handling.
Runtime configuration, instead, is accessed during or
after your application is started and can be read through the
config/2 function:
YourAppWeb.Endpoint.config(:port)
YourAppWeb.Endpoint.config(:some_config, :default_value)
Dynamic configuration
For dynamically configuring the endpoint, such as loading data
from environment variables or configuration files, Phoenix invokes
the init/2 callback on the endpoint, passing the atom :supervisor
as the first argument and the endpoint configuration as second.
All of Phoenix configuration, except the Compile-time configuration
below can be set dynamically from the init/2 callback.
Compile-time configuration
	:code_reloader - when true, enables code reloading functionality.
For the list of code reloader configuration options see
Phoenix.CodeReloader.reload/1. Keep in mind code reloading is
based on the file-system, therefore it is not possible to run two
instances of the same app at the same time with code reloading in
development, as they will race each other and only one will effectively
recompile the files. In such cases, tweak your config files so code
reloading is enabled in only one of the apps or set the MIX_BUILD
environment variable to give them distinct build directories

	:debug_errors - when true, uses Plug.Debugger functionality for
debugging failures in the application. Recommended to be set to true
only in development as it allows listing of the application source
code during debugging. Defaults to false

	:force_ssl - ensures no data is ever sent via HTTP, always redirecting
to HTTPS. It expects a list of options which are forwarded to Plug.SSL.
By default it sets the "strict-transport-security" header in HTTPS requests,
forcing browsers to always use HTTPS. If an unsafe request (HTTP) is sent,
it redirects to the HTTPS version using the :host specified in the :url
configuration. To dynamically redirect to the host of the current request,
set :host in the :force_ssl configuration to nil


Runtime configuration
	:adapter - which webserver adapter to use for serving web requests.
See the "Adapter configuration" section below

	:cache_static_manifest - a path to a json manifest file that contains
static files and their digested version. This is typically set to
"priv/static/cache_manifest.json" which is the file automatically generated
by mix phx.digest. It can be either: a string containing a file system path
or a tuple containing the application name and the path within that application.

	:cache_static_manifest_latest - a map of the static files pointing to their
digest version. This is automatically loaded from cache_static_manifest on
boot. However, if you have your own static handling mechanism, you may want to
set this value explicitly. This is used by projects such as LiveView to
detect if the client is running on the latest version of all assets.

	:cache_manifest_skip_vsn - when true, skips the appended query string
"?vsn=d" when generatic paths to static assets. This query string is used
by Plug.Static to set long expiry dates, therefore, you should set this
option to true only if you are not using Plug.Static to serve assets,
for example, if you are using a CDN. If you are setting this option, you
should also consider passing --no-vsn to mix phx.digest. Defaults to
false.

	:check_origin - configure the default :check_origin setting for
transports. See socket/3 for options. Defaults to true.

	:secret_key_base - a secret key used as a base to generate secrets
for encrypting and signing data. For example, cookies and tokens
are signed by default, but they may also be encrypted if desired.
Defaults to nil as it must be set per application

	:server - when true, starts the web server when the endpoint
supervision tree starts. Defaults to false. The mix phx.server
task automatically sets this to true

	:url - configuration for generating URLs throughout the app.
Accepts the :host, :scheme, :path and :port options. All
keys except :path can be changed at runtime. Defaults to:
[host: "localhost", path: "/"]
The :port option requires either an integer or string. The :host
option requires a string.
The :scheme option accepts "http" and "https" values. Default value
is inferred from top level :http or :https option. It is useful
when hosting Phoenix behind a load balancer or reverse proxy and
terminating SSL there.
The :path option can be used to override root path. Useful when hosting
Phoenix behind a reverse proxy with URL rewrite rules

	:static_url - configuration for generating URLs for static files.
It will fallback to url if no option is provided. Accepts the same
options as url

	:watchers - a set of watchers to run alongside your server. It
expects a list of tuples containing the executable and its arguments.
Watchers are guaranteed to run in the application directory, but only
when the server is enabled (unless :force_watchers configuration is
set to true). For example, the watcher below will run the "watch" mode
of the webpack build tool when the server starts. You can configure it
to whatever build tool or command you want:
[
  node: [
    "node_modules/webpack/bin/webpack.js",
    "--mode",
    "development",
    "--watch",
    "--watch-options-stdin"
  ]
]
The :cd and :env options can be given at the end of the list to customize
the watcher:
[node: [..., cd: "assets", env: [{"TAILWIND_MODE", "watch"}]]]
A watcher can also be a module-function-args tuple that will be invoked accordingly:
[another: {Mod, :fun, [arg1, arg2]}]

	:force_watchers - when true, forces your watchers to start
even when the :server option is set to false.

	:live_reload - configuration for the live reload option.
Configuration requires a :patterns option which should be a list of
file patterns to watch. When these files change, it will trigger a reload.
If you are using a tool like pow in development,
you may need to set the :url option appropriately.
live_reload: [
  url: "ws://localhost:4000",
  patterns: [
    ~r{priv/static/.*(js|css|png|jpeg|jpg|gif)$},
    ~r{web/views/.*(ex)$},
    ~r{web/templates/.*(eex)$}
  ]
]

	:pubsub_server - the name of the pubsub server to use in channels
and via the Endpoint broadcast functions. The PubSub server is typically
started in your supervision tree.

	:render_errors - responsible for rendering templates whenever there
is a failure in the application. For example, if the application crashes
with a 500 error during a HTML request, render("500.html", assigns)
will be called in the view given to :render_errors. Defaults to:
[view: MyApp.ErrorView, accepts: ~w(html), layout: false, log: :debug]
The default format is used when none is set in the connection


Adapter configuration
Phoenix allows you to choose which webserver adapter to use. The default
is Phoenix.Endpoint.Cowboy2Adapter which can be configured via the
following top-level options.
	:http - the configuration for the HTTP server. It accepts all options
as defined by Plug.Cowboy. Defaults
to false

	:https - the configuration for the HTTPS server. It accepts all options
as defined by Plug.Cowboy. Defaults
to false

	:drainer - a drainer process that triggers when your application is
shutting down to wait for any on-going request to finish. It accepts all
options as defined by Plug.Cowboy.Drainer.
Defaults to [], which will start a drainer process for each configured endpoint,
but can be disabled by setting it to false.


Endpoint API
In the previous section, we have used the config/2 function that is
automatically generated in your endpoint. Here's a list of all the functions
that are automatically defined in your endpoint:
	for handling paths and URLs: struct_url/0, url/0, path/1,
static_url/0,static_path/1, and static_integrity/1

	for broadcasting to channels: broadcast/3, broadcast!/3,
broadcast_from/4, broadcast_from!/4, local_broadcast/3,
and local_broadcast_from/4

	for configuration: start_link/1, config/2, and config_change/2

	as required by the Plug behaviour: Plug.init/1 and Plug.call/2

          broadcast!(topic, event, msg)

        


          Broadcasts a msg as event in the given topic to all nodes.



      


      
        
          broadcast(topic, event, msg)

        


          Broadcasts a msg as event in the given topic to all nodes.



      


      
        
          broadcast_from!(from, topic, event, msg)

        


          Broadcasts a msg from the given from as event in the given topic to all nodes.



      


      
        
          broadcast_from(from, topic, event, msg)

        


          Broadcasts a msg from the given from as event in the given topic to all nodes.



      


      
        
          config key, default

        


          Access the endpoint configuration given by key.



      


      
        
          config_change(changed, removed)

        


          Reload the endpoint configuration on application upgrades.



      


      
        
          host()

        


          Returns the host from the :url configuration.



      


      
        
          init(atom, config)

        


          Initialize the endpoint configuration.



      


      
        
          local_broadcast(topic, event, msg)

        


          Broadcasts a msg as event in the given topic within the current node.



      


      
        
          local_broadcast_from(from, topic, event, msg)

        


          Broadcasts a msg from the given from as event in the given topic within the current node.



      


      
        
          path(path)

        


          Generates the path information when routing to this endpoint.



      


      
        
          script_name()

        


          Returns the script name from the :url configuration.



      


      
        
          start_link(keyword)

        


          Starts the endpoint supervision tree.



      


      
        
          static_integrity(path)

        


          Generates an integrity hash to a static file in priv/static.



      


      
        
          static_lookup(path)

        


          Generates a two item tuple containing the static_path and static_integrity.



      


      
        
          static_path(path)

        


          Generates a route to a static file in priv/static.



      


      
        
          static_url()

        


          Generates the static URL without any path information.



      


      
        
          struct_url()

        


          Generates the endpoint base URL, but as a URI struct.



      


      
        
          subscribe(topic, opts)

        


          Subscribes the caller to the given topic.



      


      
        
          unsubscribe(topic)

        


          Unsubscribes the caller from the given topic.



      


      
        
          url()

        


          Generates the endpoint base URL without any path information.

          server?(otp_app, endpoint)

        


          Checks if Endpoint's web server has been configured to start.



      


      
        
          socket(path, module, opts \\ [])

        


          Defines a websocket/longpoll mount-point for a socket.



      


  


      


      
        
          
            
            Anchor for this section
          
Types
        

        


  
    
      
      Link to this type
    
    event()


      
       
       View Source
     


  


  

      

          @type event() :: String.t()

          @type msg() :: map() | {:binary, binary()}

          @type topic() :: String.t()

      Link to this callback
    
    broadcast!(topic, event, msg)

          @callback broadcast!(topic(), event(), msg()) :: :ok | no_return()


      


Broadcasts a msg as event in the given topic to all nodes.
Raises in case of failures.

  



  
    
      
      Link to this callback
    
    broadcast(topic, event, msg)

          @callback broadcast(topic(), event(), msg()) :: :ok | {:error, term()}


      


Broadcasts a msg as event in the given topic to all nodes.

    broadcast_from!(from, topic, event, msg)

          @callback broadcast_from!(from :: pid(), topic(), event(), msg()) :: :ok | no_return()


      


Broadcasts a msg from the given from as event in the given topic to all nodes.
Raises in case of failures.

          @callback broadcast_from(from :: pid(), topic(), event(), msg()) :: :ok | {:error, term()}


      


Broadcasts a msg from the given from as event in the given topic to all nodes.

          @callback config(key :: atom(), default :: term()) :: term()


      


Access the endpoint configuration given by key.

          @callback config_change(changed :: term(), removed :: term()) :: term()


      


Reload the endpoint configuration on application upgrades.

          @callback host() :: String.t()


      


Returns the host from the :url configuration.

          @callback init(:supervisor, config :: Keyword.t()) :: {:ok, Keyword.t()}


      


Initialize the endpoint configuration.
Invoked when the endpoint supervisor starts, allows dynamically
configuring the endpoint from system environment or other runtime sources.

          @callback local_broadcast(topic(), event(), msg()) :: :ok


      


Broadcasts a msg as event in the given topic within the current node.

    local_broadcast_from(from, topic, event, msg)

          @callback local_broadcast_from(from :: pid(), topic(), event(), msg()) :: :ok


      


Broadcasts a msg from the given from as event in the given topic within the current node.

          @callback path(path :: String.t()) :: String.t()


      


Generates the path information when routing to this endpoint.

          @callback script_name() :: [String.t()]


      


Returns the script name from the :url configuration.

    start_link(keyword)

          @callback start_link(keyword()) :: Supervisor.on_start()


      


Starts the endpoint supervision tree.
Starts endpoint's configuration cache and possibly the servers for
handling requests.

  



  
    
      
      Link to this callback
    
    static_integrity(path)


      
       
       View Source
     


  


  

      

          @callback static_integrity(path :: String.t()) :: String.t() | nil


      


Generates an integrity hash to a static file in priv/static.

  



  
    
      
      Link to this callback
    
    static_lookup(path)


      
       
       View Source
     


  


  

      

          @callback static_lookup(path :: String.t()) ::
  {String.t(), String.t()} | {String.t(), nil}


      


Generates a two item tuple containing the static_path and static_integrity.

  



  
    
      
      Link to this callback
    
    static_path(path)


      
       
       View Source
     


  


  

      

          @callback static_path(path :: String.t()) :: String.t()


      


Generates a route to a static file in priv/static.

  



  
    
      
      Link to this callback
    
    static_url()


      
       
       View Source
     


  


  

      

          @callback static_url() :: String.t()


      


Generates the static URL without any path information.

  



  
    
      
      Link to this callback
    
    struct_url()


      
       
       View Source
     


  


  

      

          @callback struct_url() :: URI.t()


      


Generates the endpoint base URL, but as a URI struct.

  



  
    
      
      Link to this callback
    
    subscribe(topic, opts)


      
       
       View Source
     


  


  

      

          @callback subscribe(topic(), opts :: Keyword.t()) :: :ok | {:error, term()}


      


Subscribes the caller to the given topic.
See Phoenix.PubSub.subscribe/3 for options.

  



  
    
      
      Link to this callback
    
    unsubscribe(topic)


      
       
       View Source
     


  


  

      

          @callback unsubscribe(topic()) :: :ok | {:error, term()}


      


Unsubscribes the caller from the given topic.

  



  
    
      
      Link to this callback
    
    url()


      
       
       View Source
     


  


  

      

          @callback url() :: String.t()


      


Generates the endpoint base URL without any path information.

      Link to this function
    
    server?(otp_app, endpoint)


      
       
       View Source
     


  


  

Checks if Endpoint's web server has been configured to start.
	otp_app - The OTP app running the endpoint, for example :my_app
	endpoint - The endpoint module, for example MyAppWeb.Endpoint


  
  examples

  
  Examples


iex> Phoenix.Endpoint.server?(:my_app, MyAppWeb.Endpoint)
true

  



    

  
    
      
      Link to this macro
    
    socket(path, module, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Defines a websocket/longpoll mount-point for a socket.

  
  options

  
  Options


	:websocket - controls the websocket configuration.
Defaults to true. May be false or a keyword list
of options. See "Common configuration" and
"WebSocket configuration" for the whole list

	:longpoll - controls the longpoll configuration.
Defaults to false. May be true or a keyword list
of options. See "Common configuration" and
"Longpoll configuration" for the whole list


If your socket is implemented using Phoenix.Socket,
you can also pass to each transport above all options
accepted on use Phoenix.Socket. An option given here
will override the value in use Phoenix.Socket.

  
  examples

  
  Examples


socket "/ws", MyApp.UserSocket

socket "/ws/admin", MyApp.AdminUserSocket,
  longpoll: true,
  websocket: [compress: true]

  
  path-params

  
  Path params


It is possible to include variables in the path, these will be
available in the params that are passed to the socket.
socket "/ws/:user_id", MyApp.UserSocket,
  websocket: [path: "/project/:project_id"]

  
  common-configuration

  
  Common configuration


The configuration below can be given to both :websocket and
:longpoll keys:
	:path - the path to use for the transport. Will default
 to the transport name ("/websocket" or "/longpoll")

	:serializer - a list of serializers for messages. See
Phoenix.Socket for more information

	:transport_log - if the transport layer itself should log and,
if so, the level

	:check_origin - if the transport should check the origin of requests when
the origin header is present. May be true, false, a list of hosts that
are allowed, or a function provided as MFA tuple. Defaults to :check_origin
setting at endpoint configuration.
If true, the header is checked against :host in YourAppWeb.Endpoint.config(:url)[:host].
If false, your app is vulnerable to Cross-Site WebSocket Hijacking (CSWSH)
attacks. Only use in development, when the host is truly unknown or when
serving clients that do not send the origin header, such as mobile apps.
You can also specify a list of explicitly allowed origins. Wildcards are
supported.
check_origin: [
  "https://example.com",
  "//another.com:888",
  "//*.other.com"
]
Or to accept any origin matching the request connection's host, port, and scheme:
check_origin: :conn
Or a custom MFA function:
check_origin: {MyAppWeb.Auth, :my_check_origin?, []}
The MFA is invoked with the request %URI{} as the first argument,
followed by arguments in the MFA list, and must return a boolean.

	:code_reloader - enable or disable the code reloader. Defaults to your
endpoint configuration

	:connect_info - a list of keys that represent data to be copied from
the transport to be made available in the user socket connect/3 callback
The valid keys are:
	:peer_data - the result of Plug.Conn.get_peer_data/1

	:trace_context_headers - a list of all trace context headers. Supported
headers are defined by the W3C Trace Context Specification.
These headers are necessary for libraries such as OpenTelemetry to extract
trace propagation information to know this request is part of a larger trace
in progress.

	:x_headers - all request headers that have an "x-" prefix

	:uri - a %URI{} with information from the conn

	:user_agent - the value of the "user-agent" request header

	{:session, session_config} - the session information from Plug.Conn.
The session_config is an exact copy of the arguments given to Plug.Session.
This requires the "_csrf_token" to be given as request parameter with
the value of URI.encode_www_form(Plug.CSRFProtection.get_csrf_token())
when connecting to the socket. It can also be a MFA to allow loading
config in runtime {MyAppWeb.Auth, :get_session_config, []}. Otherwise
the session will be nil.


Arbitrary keywords may also appear following the above valid keys, which
is useful for passing custom connection information to the socket.
For example:
  socket "/socket", AppWeb.UserSocket,
      websocket: [
        connect_info: [:peer_data, :trace_context_headers, :x_headers, :uri, session: [store: :cookie]]
      ]
With arbitrary keywords:
  socket "/socket", AppWeb.UserSocket,
      websocket: [
        connect_info: [:uri, custom_value: "abcdef"]
      ]



  
  websocket-configuration

  
  Websocket configuration


The following configuration applies only to :websocket.
	:timeout - the timeout for keeping websocket connections
open after it last received data, defaults to 60_000ms

	:max_frame_size - the maximum allowed frame size in bytes,
defaults to "infinity"

	:fullsweep_after - the maximum number of garbage collections
before forcing a fullsweep for the socket process. You can set
it to 0 to force more frequent cleanups of your websocket
transport processes. Setting this option requires Erlang/OTP 24

	:compress - whether to enable per message compression on
all data frames, defaults to false

	:subprotocols - a list of supported websocket subprotocols.
Used for handshake Sec-WebSocket-Protocol response header, defaults to nil.
For example:
subprotocols: ["sip", "mqtt"]

	:error_handler - custom error handler for connection errors.
If Phoenix.Socket.connect/3 returns an {:error, reason} tuple,
the error handler will be called with the error reason. For WebSockets,
the error handler must be a MFA tuple that receives a Plug.Conn, the
error reason, and returns a Plug.Conn with a response. For example:
error_handler: {MySocket, :handle_error, []}
and a {:error, :rate_limit} return may be handled on MySocket as:
def handle_error(conn, :rate_limit), do: Plug.Conn.send_resp(conn, 429, "Too many requests")



  
  longpoll-configuration

  
  Longpoll configuration


The following configuration applies only to :longpoll:
	:window_ms - how long the client can wait for new messages
in its poll request, defaults to 10_000ms.

	:pubsub_timeout_ms - how long a request can wait for the
pubsub layer to respond, defaults to 2000ms.

	:crypto - options for verifying and signing the token, accepted
by Phoenix.Token. By default tokens are valid for 2 weeks

Instrumenter to handle logging of various instrumentation events.
Instrumentation
Phoenix uses the :telemetry library for instrumentation. The following events
are published by Phoenix with the following measurements and metadata:
	[:phoenix, :endpoint, :start] - dispatched by Plug.Telemetry in your endpoint,
usually after code reloading
	Measurement: %{system_time: system_time}
	Metadata: %{conn: Plug.Conn.t, options: Keyword.t}
	Options: %{log: Logger.level | false}

	Disable logging: In your endpoint plug Plug.Telemetry, ..., log: Logger.level | false

	Configure log level dynamically: plug Plug.Telemetry, ..., log: {Mod, Fun, Args}


	[:phoenix, :endpoint, :stop] - dispatched by Plug.Telemetry in your
endpoint whenever the response is sent
	Measurement: %{duration: native_time}
	Metadata: %{conn: Plug.Conn.t, options: Keyword.t}
	Options: %{log: Logger.level | false}

	Disable logging: In your endpoint plug Plug.Telemetry, ..., log: Logger.level | false

	Configure log level dynamically: plug Plug.Telemetry, ..., log: {Mod, Fun, Args}


	[:phoenix, :router_dispatch, :start] - dispatched by Phoenix.Router
before dispatching to a matched route
	Measurement: %{system_time: System.system_time}
	Metadata: %{conn: Plug.Conn.t, route: binary, plug: module, plug_opts: term, path_params: map, pipe_through: [atom], log: Logger.level | false}

	Disable logging: Pass log: false to the router macro, for example: get("/page", PageController, :index, log: false)
	Configure log level dynamically: get("/page", PageController, :index, log: {Mod, Fun, Args})


	[:phoenix, :router_dispatch, :exception] - dispatched by Phoenix.Router
after exceptions on dispatching a route
	Measurement: %{duration: native_time}
	Metadata: %{conn: Plug.Conn.t, kind: :throw | :error | :exit, reason: term(), stacktrace: Exception.stacktrace()}

	Disable logging: This event is not logged


	[:phoenix, :router_dispatch, :stop] - dispatched by Phoenix.Router
after successfully dispatching a matched route
	Measurement: %{duration: native_time}
	Metadata: %{conn: Plug.Conn.t, route: binary, plug: module, plug_opts: term, path_params: map, pipe_through: [atom], log: Logger.level | false}

	Disable logging: This event is not logged


	[:phoenix, :error_rendered] - dispatched at the end of an error view being rendered
	Measurement: %{duration: native_time}
	Metadata: %{conn: Plug.Conn.t, status: Plug.Conn.status, kind: Exception.kind, reason: term, stacktrace: Exception.stacktrace}
	Disable logging: Set render_errors: [log: false] on your endpoint configuration


	[:phoenix, :socket_connected] - dispatched by Phoenix.Socket, at the end of a socket connection
	Measurement: %{duration: native_time}
	Metadata: %{endpoint: atom, transport: atom, params: term, connect_info: map, vsn: binary, user_socket: atom, result: :ok | :error, serializer: atom, log: Logger.level | false}

	Disable logging: use Phoenix.Socket, log: false or socket "/foo", MySocket, websocket: [log: false] in your endpoint


	[:phoenix, :channel_joined] - dispatched at the end of a channel join
	Measurement: %{duration: native_time}
	Metadata: %{result: :ok | :error, params: term, socket: Phoenix.Socket.t}

	Disable logging: This event cannot be disabled


	[:phoenix, :channel_handled_in] - dispatched at the end of a channel handle in
	Measurement: %{duration: native_time}
	Metadata: %{event: binary, params: term, socket: Phoenix.Socket.t}
	Disable logging: This event cannot be disabled



To see an example of how Phoenix LiveDashboard uses these events to create
metrics, visit https://hexdocs.pm/phoenix_live_dashboard/metrics.html.
Parameter filtering
When logging parameters, Phoenix can filter out sensitive parameters
such as passwords and tokens. Parameters to be filtered can be
added via the :filter_parameters option:
config :phoenix, :filter_parameters, ["password", "secret"]
With the configuration above, Phoenix will filter any parameter
that contains the terms password or secret. The match is
case sensitive.
Phoenix's default is ["password"].
Phoenix can filter all parameters by default and selectively keep
parameters. This can be configured like so:
config :phoenix, :filter_parameters, {:keep, ["id", "order"]}
With the configuration above, Phoenix will filter all parameters,
except those that match exactly id or order. If a kept parameter
matches, all parameters nested under that one will also be kept.
Dynamic log level
In some cases you may wish to set the log level dynamically
on a per-request basis. To do so, set the :log option to
a tuple, {Mod, Fun, Args}. The Plug.Conn.t() for the
request will be prepended to the provided list of arguments.
When invoked, your function must return a
Logger.level() or false to
disable logging for the request.
For example, in your Endpoint you might do something like this:
  # lib/my_app_web/endpoint.ex
  plug Plug.Telemetry,
    event_prefix: [:phoenix, :endpoint],
    log: {__MODULE__, :log_level, []}

  # Disables logging for routes like /status/*
  def log_level(%{path_info: ["status" | _]}), do: false
  def log_level(_), do: :info
Disabling
When you are using custom logging system it is not always desirable to enable
Phoenix.Logger by default. You can always disable this in general by:
config :phoenix, :logger, false

Conveniences for inflecting and working with names in Phoenix.

      Functions
    


      
        
          camelize(value)

        


          Converts a string to camel case.



      


      
        
          camelize(value, atom)

        


      


      
        
          humanize(atom)

        


          Converts an attribute/form field into its humanize version.



      


      
        
          resource_name(alias, suffix \\ "")

        


          Extracts the resource name from an alias.



      


      
        
          underscore(value)

        


          Converts a string to underscore case.



      


      
        
          unsuffix(value, suffix)

        


          Removes the given suffix from the name if it exists.

      Link to this function
    
    camelize(value)


      
       
       View Source
     


  


  

      

          @spec camelize(String.t()) :: String.t()


      


Converts a string to camel case.
Takes an optional :lower flag to return lowerCamelCase.

  
  examples

  
  Examples


iex> Phoenix.Naming.camelize("my_app")
"MyApp"

iex> Phoenix.Naming.camelize("my_app", :lower)
"myApp"
In general, camelize can be thought of as the reverse of
underscore, however, in some cases formatting may be lost:
Phoenix.Naming.underscore "SAPExample"  #=> "sap_example"
Phoenix.Naming.camelize   "sap_example" #=> "SapExample"

  



  
    
      
      Link to this function
    
    camelize(value, atom)


      
       
       View Source
     


  


  

      

          @spec camelize(String.t(), :lower) :: String.t()


      



  



  
    
      
      Link to this function
    
    humanize(atom)


      
       
       View Source
     


  


  

      

          @spec humanize(atom() | String.t()) :: String.t()


      


Converts an attribute/form field into its humanize version.

  
  examples

  
  Examples


iex> Phoenix.Naming.humanize(:username)
"Username"
iex> Phoenix.Naming.humanize(:created_at)
"Created at"
iex> Phoenix.Naming.humanize("user_id")
"User"

  



    

  
    
      
      Link to this function
    
    resource_name(alias, suffix \\ "")


      
       
       View Source
     


  


  

      

          @spec resource_name(String.Chars.t(), String.t()) :: String.t()


      


Extracts the resource name from an alias.

  
  examples

  
  Examples


iex> Phoenix.Naming.resource_name(MyApp.User)
"user"

iex> Phoenix.Naming.resource_name(MyApp.UserView, "View")
"user"

  



  
    
      
      Link to this function
    
    underscore(value)


      
       
       View Source
     


  


  

      

          @spec underscore(String.t()) :: String.t()


      


Converts a string to underscore case.

  
  examples

  
  Examples


iex> Phoenix.Naming.underscore("MyApp")
"my_app"
In general, underscore can be thought of as the reverse of
camelize, however, in some cases formatting may be lost:
Phoenix.Naming.underscore "SAPExample"  #=> "sap_example"
Phoenix.Naming.camelize   "sap_example" #=> "SapExample"

  



  
    
      
      Link to this function
    
    unsuffix(value, suffix)


      
       
       View Source
     


  


  

      

          @spec unsuffix(String.t(), String.t()) :: String.t()


      


Removes the given suffix from the name if it exists.

  
  examples

  
  Examples


iex> Phoenix.Naming.unsuffix("MyApp.User", "View")
"MyApp.User"

iex> Phoenix.Naming.unsuffix("MyApp.UserView", "View")
"MyApp.User"

A protocol that converts data structures into URL parameters.
This protocol is used by URL helpers and other parts of the
Phoenix stack. For example, when you write:
user_path(conn, :edit, @user)
Phoenix knows how to extract the :id from @user thanks
to this protocol.
By default, Phoenix implements this protocol for integers, binaries, atoms,
and structs. For structs, a key :id is assumed, but you may provide a
specific implementation.
Nil values cannot be converted to param.
Custom parameters
In order to customize the parameter for any struct,
one can simply implement this protocol.
However, for convenience, this protocol can also be
derivable. For example:
defmodule User do
  @derive Phoenix.Param
  defstruct [:id, :username]
end
By default, the derived implementation will also use
the :id key. In case the user does not contain an
:id key, the key can be specified with an option:
defmodule User do
  @derive {Phoenix.Param, key: :username}
  defstruct [:username]
end
will automatically use :username in URLs.
When using Ecto, you must call @derive before
your schema call:
@derive {Phoenix.Param, key: :username}
schema "users" do

      Link to this type
    
    t()


      
       
       View Source
     


  


  

      

          @type t() :: term()

      Link to this function
    
    to_param(term)


      
       
       View Source
     


  


  

      

          @spec to_param(term()) :: String.t()

Provides Presence tracking to processes and channels.
This behaviour provides presence features such as fetching
presences for a given topic, as well as handling diffs of
join and leave events as they occur in real-time. Using this
module defines a supervisor and a module that implements the
Phoenix.Tracker behaviour that uses Phoenix.PubSub to
broadcast presence updates.
In case you want to use only a subset of the functionality
provided by Phoenix.Presence, such as tracking processes
but without broadcasting updates, we recommend that you look
at the Phoenix.Tracker functionality from the phoenix_pubsub
project.
Example Usage
Start by defining a presence module within your application
which uses Phoenix.Presence and provide the :otp_app which
holds your configuration, as well as the :pubsub_server.
defmodule MyAppWeb.Presence do
  use Phoenix.Presence,
    otp_app: :my_app,
    pubsub_server: MyApp.PubSub
end
The :pubsub_server must point to an existing pubsub server
running in your application, which is included by default as
MyApp.PubSub for new applications.
Next, add the new supervisor to your supervision tree in
lib/my_app/application.ex. It must be after the PubSub child
and before the endpoint:
children = [
  ...
  {Phoenix.PubSub, name: MyApp.PubSub},
  MyAppWeb.Presence,
  MyAppWeb.Endpoint
]
Once added, presences can be tracked in your channel after joining:
defmodule MyAppWeb.MyChannel do
  use MyAppWeb, :channel
  alias MyAppWeb.Presence

  def join("some:topic", _params, socket) do
    send(self(), :after_join)
    {:ok, assign(socket, :user_id, ...)}
  end

  def handle_info(:after_join, socket) do
    {:ok, _} = Presence.track(socket, socket.assigns.user_id, %{
      online_at: inspect(System.system_time(:second))
    })

    push(socket, "presence_state", Presence.list(socket))
    {:noreply, socket}
  end
end
In the example above, Presence.track is used to register this channel's process as a
presence for the socket's user ID, with a map of metadata.
Next, the current presence information for
the socket's topic is pushed to the client as a "presence_state" event.
Finally, a diff of presence join and leave events will be sent to the
client as they happen in real-time with the "presence_diff" event.
The diff structure will be a map of :joins and :leaves of the form:
%{
  joins: %{"123" => %{metas: [%{status: "away", phx_ref: ...}]}},
  leaves: %{"456" => %{metas: [%{status: "online", phx_ref: ...}]}}
},
See list/1 for more information on the presence data structure.
Fetching Presence Information
Presence metadata should be minimized and used to store small,
ephemeral state, such as a user's "online" or "away" status.
More detailed information, such as user details that need to be fetched
from the database, can be achieved by overriding the fetch/2 function.
The fetch/2 callback is triggered when using list/1 and on
every update, and it serves as a mechanism to fetch presence information
a single time, before broadcasting the information to all channel subscribers.
This prevents N query problems and gives you a single place to group
isolated data fetching to extend presence metadata.
The function must return a map of data matching the outlined Presence
data structure, including the :metas key, but can extend the map of
information to include any additional information. For example:
def fetch(_topic, presences) do
  users = presences |> Map.keys() |> Accounts.get_users_map()

  for {key, %{metas: metas}} <- presences, into: %{} do
    {key, %{metas: metas, user: users[String.to_integer(key)]}}
  end
end
Where Account.get_users_map/1 could be implemented like:
def get_users_map(ids) do
  query =
    from u in User,
      where: u.id in ^ids,
      select: {u.id, u}

  query |> Repo.all() |> Enum.into(%{})
end
The fetch/2 function above fetches all users from the database who
have registered presences for the given topic. The presences
information is then extended with a :user key of the user's
information, while maintaining the required :metas field from the
original presence data.
Testing with Presence
Every time the fetch callback is invoked, it is done from a separate
process. Given those processes run asynchronously, it is often necessary
to guarantee they have been shutdown at the end of every test. This can
be done by using ExUnit's on_exit hook plus fetchers_pids function:
on_exit(fn ->
  for pid <- MyAppWeb.Presence.fetchers_pids() do
    ref = Process.monitor(pid)
    assert_receive {:DOWN, ^ref, _, _, _}, 1000
  end
end)

          fetch(topic, presences)

        


          Extend presence information with additional data.



      


      
        
          get_by_key(arg1, key)

        


          Returns the map of presence metadata for a socket/topic-key pair.



      


      
        
          list(arg1)

        


          Returns presences for a socket/topic.



      


      
        
          track(socket, key, meta)

        


          Track a channel's process as a presence.



      


      
        
          track(pid, topic, key, meta)

        


          Track an arbitrary process as a presence.



      


      
        
          untrack(socket, key)

        


          Stop tracking a channel's process.



      


      
        
          untrack(pid, topic, key)

        


          Stop tracking a process.



      


      
        
          update(socket, key, meta)

        


          Update a channel presence's metadata.



      


      
        
          update(pid, topic, key, meta)

        


          Update a process presence's metadata.

      Link to this type
    
    presence()


      
       
       View Source
     


  


  

      

          @type presence() :: %{key: String.t(), meta: map()}


      



  



  
    
      
      Link to this type

          @type presences() :: %{required(String.t()) => %{metas: [map()]}}

          @type topic() :: String.t()

      Link to this callback
    
    fetch(topic, presences)

          @callback fetch(topic(), presences()) :: presences()


      


Extend presence information with additional data.
When list/1 is used to list all presences of the given topic, this
callback is triggered once to modify the result before it is broadcasted to
all channel subscribers. This avoids N query problems and provides a single
place to extend presence metadata. You must return a map of data matching the
original result, including the :metas key, but can extend the map to include
any additional information.
The default implementation simply passes presences through unchanged.

  
  example

  
  Example


def fetch(_topic, presences) do
  query =
    from u in User,
      where: u.id in ^Map.keys(presences),
      select: {u.id, u}

  users = query |> Repo.all() |> Enum.into(%{})
  for {key, %{metas: metas}} <- presences, into: %{} do
    {key, %{metas: metas, user: users[key]}}
  end
end

  



  
    
      
      Link to this callback
    
    get_by_key(arg1, key)

          @callback get_by_key(Phoenix.Socket.t() | topic(), key :: String.t()) :: presences()


      


Returns the map of presence metadata for a socket/topic-key pair.

  
  examples

  
  Examples


Uses the same data format as list/1, but only
returns metadata for the presences under a topic and key pair. For example,
a user with key "user1", connected to the same chat room "room:1" from two
devices, could return:
iex> MyPresence.get_by_key("room:1", "user1")
[%{name: "User 1", metas: [%{device: "Desktop"}, %{device: "Mobile"}]}]
Like list/1, the presence metadata is passed to the fetch
callback of your presence module to fetch any additional information.

  



  
    
      
      Link to this callback
    
    list(arg1)


      
       
       View Source
     


  


  

      

          @callback list(Phoenix.Socket.t() | topic()) :: presences()


      


Returns presences for a socket/topic.

  
  presence-data-structure

  
  Presence data structure


The presence information is returned as a map with presences grouped
by key, cast as a string, and accumulated metadata, with the following form:
%{key => %{metas: [%{phx_ref: ..., ...}, ...]}}
For example, imagine a user with id 123 online from two
different devices, as well as a user with id 456 online from
just one device. The following presence information might be returned:
%{"123" => %{metas: [%{status: "away", phx_ref: ...},
                     %{status: "online", phx_ref: ...}]},
  "456" => %{metas: [%{status: "online", phx_ref: ...}]}}
The keys of the map will usually point to a resource ID. The value
will contain a map with a :metas key containing a list of metadata
for each resource. Additionally, every metadata entry will contain a
:phx_ref key which can be used to uniquely identify metadata for a
given key. In the event that the metadata was previously updated,
a :phx_ref_prev key will be present containing the previous
:phx_ref value.

  



  
    
      
      Link to this callback
    
    track(socket, key, meta)


      
       
       View Source
     


  


  

      

          @callback track(socket :: Phoenix.Socket.t(), key :: String.t(), meta :: map()) ::
  {:ok, ref :: binary()} | {:error, reason :: term()}


      


Track a channel's process as a presence.
Tracked presences are grouped by key, cast as a string. For example, to
group each user's channels together, use user IDs as keys. Each presence can
be associated with a map of metadata to store small, ephemeral state, such as
a user's online status. To store detailed information, see fetch/2.

  
  example

  
  Example


alias MyApp.Presence
def handle_info(:after_join, socket) do
  {:ok, _} = Presence.track(socket, socket.assigns.user_id, %{
    online_at: inspect(System.system_time(:second))
  })
  {:noreply, socket}
end

  



  
    
      
      Link to this callback
    
    track(pid, topic, key, meta)


      
       
       View Source
     


  


  

      

          @callback track(pid(), topic(), key :: String.t(), meta :: map()) ::
  {:ok, ref :: binary()} | {:error, reason :: term()}


      


Track an arbitrary process as a presence.
Same with track/3, except track any process by topic and key.

  



  
    
      
      Link to this callback
    
    untrack(socket, key)


      
       
       View Source
     


  


  

      

          @callback untrack(socket :: Phoenix.Socket.t(), key :: String.t()) :: :ok


      


Stop tracking a channel's process.

  



  
    
      
      Link to this callback
    
    untrack(pid, topic, key)


      
       
       View Source
     


  


  

      

          @callback untrack(pid(), topic(), key :: String.t()) :: :ok


      


Stop tracking a process.

  



  
    
      
      Link to this callback
    
    update(socket, key, meta)


      
       
       View Source
     


  


  

      

          @callback update(
  socket :: Phoenix.Socket.t(),
  key :: String.t(),
  meta :: map() | (map() -> map())
) ::
  {:ok, ref :: binary()} | {:error, reason :: term()}


      


Update a channel presence's metadata.
Replace a presence's metadata by passing a new map or a function that takes
the current map and returns a new one.

  



  
    
      
      Link to this callback
    
    update(pid, topic, key, meta)


      
       
       View Source
     


  


  

      

          @callback update(pid(), topic(), key :: String.t(), meta :: map() | (map() -> map())) ::
  {:ok, ref :: binary()} | {:error, reason :: term()}


      


Update a process presence's metadata.
Same as update/3, but with an arbitrary process.

Defines a Phoenix router.
The router provides a set of macros for generating routes
that dispatch to specific controllers and actions. Those
macros are named after HTTP verbs. For example:
defmodule MyAppWeb.Router do
  use Phoenix.Router

  get "/pages/:page", PageController, :show
end
The get/3 macro above accepts a request to /pages/hello and dispatches
it to PageController's show action with %{"page" => "hello"} in
params.
Phoenix's router is extremely efficient, as it relies on Elixir
pattern matching for matching routes and serving requests.
Routing
get/3, post/3, put/3 and other macros named after HTTP verbs are used
to create routes.
The route:
get "/pages", PageController, :index
matches a GET request to /pages and dispatches it to the index action in
PageController.
get "/pages/:page", PageController, :show
matches /pages/hello and dispatches to the show action with
%{"page" => "hello"} in params.
defmodule PageController do
  def show(conn, params) do
    # %{"page" => "hello"} == params
  end
end
Partial and multiple segments can be matched. For example:
get "/api/v:version/pages/:id", PageController, :show
matches /api/v1/pages/2 and puts %{"version" => "1", "id" => "2"} in
params. Only the trailing part of a segment can be captured.
Routes are matched from top to bottom. The second route here:
get "/pages/:page", PageController, :show
get "/pages/hello", PageController, :hello
will never match /pages/hello because /pages/:page matches that first.
Routes can use glob-like patterns to match trailing segments.
get "/pages/*page", PageController, :show
matches /pages/hello/world and puts the globbed segments in params["page"].
GET /pages/hello/world
%{"page" => ["hello", "world"]} = params
Globs cannot have prefixes nor suffixes, but can be mixed with variables:
 get "/pages/he:page/*rest", PageController, :show
matches
GET /pages/hello
 %{"page" => "llo", "rest" => []} = params

GET /pages/hey/there/world
 %{"page" => "y", "rest" => ["there" "world"]} = params
Helpers
Phoenix automatically generates a module Helpers inside your router
which contains named helpers to help developers generate and keep
their routes up to date.
Helpers are automatically generated based on the controller name.
For example, the route:
get "/pages/:page", PageController, :show
will generate the following named helper:
MyAppWeb.Router.Helpers.page_path(conn_or_endpoint, :show, "hello")
"/pages/hello"

MyAppWeb.Router.Helpers.page_path(conn_or_endpoint, :show, "hello", some: "query")
"/pages/hello?some=query"

MyAppWeb.Router.Helpers.page_url(conn_or_endpoint, :show, "hello")
"http://example.com/pages/hello"

MyAppWeb.Router.Helpers.page_url(conn_or_endpoint, :show, "hello", some: "query")
"http://example.com/pages/hello?some=query"
If the route contains glob-like patterns, parameters for those have to be given as
list:
MyAppWeb.Router.Helpers.page_path(conn_or_endpoint, :show, ["hello", "world"])
"/pages/hello/world"
The URL generated in the named URL helpers is based on the configuration for
:url, :http and :https. However, if for some reason you need to manually
control the URL generation, the url helpers also allow you to pass in a URI
struct:
uri = %URI{scheme: "https", host: "other.example.com"}
MyAppWeb.Router.Helpers.page_url(uri, :show, "hello")
"https://other.example.com/pages/hello"
The named helper can also be customized with the :as option. Given
the route:
get "/pages/:page", PageController, :show, as: :special_page
the named helper will be:
MyAppWeb.Router.Helpers.special_page_path(conn, :show, "hello")
"/pages/hello"
Scopes and Resources
It is very common in Phoenix applications to namespace all of your
routes under the application scope:
scope "/", MyAppWeb do
  get "/pages/:id", PageController, :show
end
The route above will dispatch to MyAppWeb.PageController. This syntax
is not only convenient for developers, since we don't have to repeat
the MyAppWeb. prefix on all routes, but it also allows Phoenix to put
less pressure on the Elixir compiler. If instead we had written:
get "/pages/:id", MyAppWeb.PageController, :show
The Elixir compiler would infer that the router depends directly on
MyAppWeb.PageController, which is not true. By using scopes, Phoenix
can properly hint to the Elixir compiler the controller is not an
actual dependency of the router. This provides more efficient
compilation times.
Scopes allow us to scope on any path or even on the helper name:
scope "/api/v1", MyAppWeb, as: :api_v1 do
  get "/pages/:id", PageController, :show
end
For example, the route above will match on the path "/api/v1/pages/:id"
and the named route will be api_v1_page_path, as expected from the
values given to scope/2 option.
Phoenix also provides a resources/4 macro that allows developers
to generate "RESTful" routes to a given resource:
defmodule MyAppWeb.Router do
  use Phoenix.Router

  resources "/pages", PageController, only: [:show]
  resources "/users", UserController, except: [:delete]
end
Finally, Phoenix ships with a mix phx.routes task that nicely
formats all routes in a given router. We can use it to verify all
routes included in the router above:
$ mix phx.routes
page_path  GET    /pages/:id       PageController.show/2
user_path  GET    /users           UserController.index/2
user_path  GET    /users/:id/edit  UserController.edit/2
user_path  GET    /users/new       UserController.new/2
user_path  GET    /users/:id       UserController.show/2
user_path  POST   /users           UserController.create/2
user_path  PATCH  /users/:id       UserController.update/2
           PUT    /users/:id       UserController.update/2

One can also pass a router explicitly as an argument to the task:
$ mix phx.routes MyAppWeb.Router

Check scope/2 and resources/4 for more information.
Pipelines and plugs
Once a request arrives at the Phoenix router, it performs
a series of transformations through pipelines until the
request is dispatched to a desired end-point.
Such transformations are defined via plugs, as defined
in the Plug specification.
Once a pipeline is defined, it can be piped through per scope.
For example:
defmodule MyAppWeb.Router do
  use Phoenix.Router

  pipeline :browser do
    plug :fetch_session
    plug :accepts, ["html"]
  end

  scope "/" do
    pipe_through :browser

    # browser related routes and resources
  end
end
Phoenix.Router imports functions from both Plug.Conn and Phoenix.Controller
to help define plugs. In the example above, fetch_session/2
comes from Plug.Conn while accepts/2 comes from Phoenix.Controller.
Note that router pipelines are only invoked after a route is found.
No plug is invoked in case no matches were found.

          connect(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a connect request to the given path.



      


      
        
          delete(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a delete request to the given path.



      


      
        
          forward(path, plug, plug_opts \\ [], router_opts \\ [])

        


          Forwards a request at the given path to a plug.



      


      
        
          get(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a get request to the given path.



      


      
        
          head(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a head request to the given path.



      


      
        
          match(verb, path, plug, plug_opts, options \\ [])

        


          Generates a route match based on an arbitrary HTTP method.



      


      
        
          options(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a options request to the given path.



      


      
        
          patch(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a patch request to the given path.



      


      
        
          pipe_through(pipes)

        


          Defines a list of plugs (and pipelines) to send the connection through.



      


      
        
          pipeline(plug, list)

        


          Defines a plug pipeline.



      


      
        
          plug(plug, opts \\ [])

        


          Defines a plug inside a pipeline.



      


      
        
          post(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a post request to the given path.



      


      
        
          put(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a put request to the given path.



      


      
        
          resources(path, controller)

        


          See resources/4.



      


      
        
          resources(path, controller, opts)

        


          See resources/4.



      


      
        
          resources(path, controller, opts, list)

        


          Defines "RESTful" routes for a resource.



      


      
        
          route_info(router, method, path, host)

        


          Returns the compile-time route info and runtime path params for a request.



      


      
        
          routes(router)

        


          Returns all routes information from the given router.



      


      
        
          scope(options, list)

        


          Defines a scope in which routes can be nested.



      


      
        
          scope(path, options, list)

        


          Define a scope with the given path.



      


      
        
          scope(path, alias, options, list)

        


          Defines a scope with the given path and alias.



      


      
        
          scoped_alias(router_module, alias)

        


          Returns the full alias with the current scope's aliased prefix.



      


      
        
          trace(path, plug, plug_opts, options \\ [])

        


          Generates a route to handle a trace request to the given path.

      Link to this macro
    
    connect(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a connect request to the given path.
connect("/events/:id", EventController, :action)
See match/5 for options.

  



    

  
    
      
      Link to this macro
    
    delete(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a delete request to the given path.
delete("/events/:id", EventController, :action)
See match/5 for options.

  



    

    

  
    
      
      Link to this macro
    
    forward(path, plug, plug_opts \\ [], router_opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Forwards a request at the given path to a plug.
All paths that match the forwarded prefix will be sent to
the forwarded plug. This is useful for sharing a router between
applications or even breaking a big router into smaller ones.
The router pipelines will be invoked prior to forwarding the
connection.
However, we don't advise forwarding to another endpoint.
The reason is that plugs defined by your app and the forwarded
endpoint would be invoked twice, which may lead to errors.

  
  examples

  
  Examples


scope "/", MyApp do
  pipe_through [:browser, :admin]

  forward "/admin", SomeLib.AdminDashboard
  forward "/api", ApiRouter
end

  



    

  
    
      
      Link to this macro
    
    get(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a get request to the given path.
get("/events/:id", EventController, :action)
See match/5 for options.

  



    

  
    
      
      Link to this macro
    
    head(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a head request to the given path.
head("/events/:id", EventController, :action)
See match/5 for options.

  



    

  
    
      
      Link to this macro
    
    match(verb, path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route match based on an arbitrary HTTP method.
Useful for defining routes not included in the builtin macros.
The catch-all verb, :*, may also be used to match all HTTP methods.

  
  options

  
  Options


	:as - configures the named helper exclusively. If false, does not generate
a helper.
	:alias - configure if the scope alias should be applied to the route.
Defaults to true, disables scoping if false.
	:log - the level to log the route dispatching under,
may be set to false. Defaults to :debug
	:host - a string containing the host scope, or prefix host scope,
ie "foo.bar.com", "foo."
	:private - a map of private data to merge into the connection
when a route matches
	:assigns - a map of data to merge into the connection when a route matches
	:metadata - a map of metadata used by the telemetry events and returned by
route_info/4
	:trailing_slash - a boolean to flag whether or not the helper functions
append a trailing slash. Defaults to false.


  
  examples

  
  Examples


match(:move, "/events/:id", EventController, :move)

match(:*, "/any", SomeController, :any)

  



    

  
    
      
      Link to this macro
    
    options(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a options request to the given path.
options("/events/:id", EventController, :action)
See match/5 for options.

  



    

  
    
      
      Link to this macro
    
    patch(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a patch request to the given path.
patch("/events/:id", EventController, :action)
See match/5 for options.

  



  
    
      
      Link to this macro
    
    pipe_through(pipes)


      
       
       View Source
     


      (macro)

  


  

Defines a list of plugs (and pipelines) to send the connection through.
See pipeline/2 for more information.

  



  
    
      
      Link to this macro
    
    pipeline(plug, list)


      
       
       View Source
     


      (macro)

  


  

Defines a plug pipeline.
Pipelines are defined at the router root and can be used
from any scope.

  
  examples

  
  Examples


pipeline :api do
  plug :token_authentication
  plug :dispatch
end
A scope may then use this pipeline as:
scope "/" do
  pipe_through :api
end
Every time pipe_through/1 is called, the new pipelines
are appended to the ones previously given.

  



    

  
    
      
      Link to this macro
    
    plug(plug, opts \\ [])


      
       
       View Source
     


      (macro)

  


  

Defines a plug inside a pipeline.
See pipeline/2 for more information.

  



    

  
    
      
      Link to this macro
    
    post(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a post request to the given path.
post("/events/:id", EventController, :action)
See match/5 for options.

  



    

  
    
      
      Link to this macro
    
    put(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a put request to the given path.
put("/events/:id", EventController, :action)
See match/5 for options.

  



  
    
      
      Link to this macro
    
    resources(path, controller)


      
       
       View Source
     


      (macro)

  


  

See resources/4.

  



  
    
      
      Link to this macro
    
    resources(path, controller, opts)


      
       
       View Source
     


      (macro)

  


  

See resources/4.

  



  
    
      
      Link to this macro
    
    resources(path, controller, opts, list)


      
       
       View Source
     


      (macro)

  


  

Defines "RESTful" routes for a resource.
The given definition:
resources "/users", UserController
will include routes to the following actions:
	GET /users => :index
	GET /users/new => :new
	POST /users => :create
	GET /users/:id => :show
	GET /users/:id/edit => :edit
	PATCH /users/:id => :update
	PUT /users/:id => :update
	DELETE /users/:id => :delete


  
  options

  
  Options


This macro accepts a set of options:
	:only - a list of actions to generate routes for, for example: [:show, :edit]
	:except - a list of actions to exclude generated routes from, for example: [:delete]
	:param - the name of the parameter for this resource, defaults to "id"
	:name - the prefix for this resource. This is used for the named helper
and as the prefix for the parameter in nested resources. The default value
is automatically derived from the controller name, i.e. UserController will
have name "user"
	:as - configures the named helper exclusively
	:singleton - defines routes for a singleton resource that is looked up by
the client without referencing an ID. Read below for more information


  
  singleton-resources

  
  Singleton resources


When a resource needs to be looked up without referencing an ID, because
it contains only a single entry in the given context, the :singleton
option can be used to generate a set of routes that are specific to
such single resource:
	GET /user => :show
	GET /user/new => :new
	POST /user => :create
	GET /user/edit => :edit
	PATCH /user => :update
	PUT /user => :update
	DELETE /user => :delete

Usage example:
resources "/account", AccountController, only: [:show], singleton: true

  
  nested-resources

  
  Nested Resources


This macro also supports passing a nested block of route definitions.
This is helpful for nesting children resources within their parents to
generate nested routes.
The given definition:
resources "/users", UserController do
  resources "/posts", PostController
end
will include the following routes:
user_post_path  GET     /users/:user_id/posts           PostController :index
user_post_path  GET     /users/:user_id/posts/:id/edit  PostController :edit
user_post_path  GET     /users/:user_id/posts/new       PostController :new
user_post_path  GET     /users/:user_id/posts/:id       PostController :show
user_post_path  POST    /users/:user_id/posts           PostController :create
user_post_path  PATCH   /users/:user_id/posts/:id       PostController :update
                PUT     /users/:user_id/posts/:id       PostController :update
user_post_path  DELETE  /users/:user_id/posts/:id       PostController :delete

  



  
    
      
      Link to this function
    
    route_info(router, method, path, host)


      
       
       View Source
     


  


  

Returns the compile-time route info and runtime path params for a request.
The path can be either a string or the path_info segments.
A map of metadata is returned with the following keys:
	:log - the configured log level. For example :debug
	:path_params - the map of runtime path params
	:pipe_through - the list of pipelines for the route's scope, for example [:browser]
	:plug - the plug to dispatch the route to, for example AppWeb.PostController
	:plug_opts - the options to pass when calling the plug, for example: :index
	:route - the string route pattern, such as "/posts/:id"


  
  examples

  
  Examples


iex> Phoenix.Router.route_info(AppWeb.Router, "GET", "/posts/123", "myhost")
%{
  log: :debug,
  path_params: %{"id" => "123"},
  pipe_through: [:browser],
  plug: AppWeb.PostController,
  plug_opts: :show,
  route: "/posts/:id",
}

iex> Phoenix.Router.route_info(MyRouter, "GET", "/not-exists", "myhost")
:error

  



  
    
      
      Link to this function
    
    routes(router)


      
       
       View Source
     


  


  

Returns all routes information from the given router.

  



  
    
      
      Link to this macro
    
    scope(options, list)


      
       
       View Source
     


      (macro)

  


  

Defines a scope in which routes can be nested.

  
  examples

  
  Examples


scope path: "/api/v1", as: :api_v1, alias: API.V1 do
  get "/pages/:id", PageController, :show
end
The generated route above will match on the path "/api/v1/pages/:id"
and will dispatch to :show action in API.V1.PageController. A named
helper api_v1_page_path will also be generated.

  
  options

  
  Options


The supported options are:
	:path - a string containing the path scope.
	:as - a string or atom containing the named helper scope. When set to
false, it resets the nested helper scopes.
	:alias - an alias (atom) containing the controller scope. When set to
false, it resets all nested aliases.
	:host - a string containing the host scope, or prefix host scope,
ie "foo.bar.com", "foo."
	:private - a map of private data to merge into the connection when a route matches
	:assigns - a map of data to merge into the connection when a route matches
	:log - the level to log the route dispatching under,
may be set to false. Defaults to :debug
	:trailing_slash - whether or not the helper functions append a trailing
slash. Defaults to false.


  



  
    
      
      Link to this macro
    
    scope(path, options, list)


      
       
       View Source
     


      (macro)

  


  

Define a scope with the given path.
This function is a shortcut for:
scope path: path do
  ...
end

  
  examples

  
  Examples


scope "/api/v1", as: :api_v1 do
  get "/pages/:id", PageController, :show
end

  



  
    
      
      Link to this macro
    
    scope(path, alias, options, list)


      
       
       View Source
     


      (macro)

  


  

Defines a scope with the given path and alias.
This function is a shortcut for:
scope path: path, alias: alias do
  ...
end

  
  examples

  
  Examples


scope "/api/v1", API.V1, as: :api_v1 do
  get "/pages/:id", PageController, :show
end

  



  
    
      
      Link to this function
    
    scoped_alias(router_module, alias)


      
       
       View Source
     


  


  

Returns the full alias with the current scope's aliased prefix.
Useful for applying the same short-hand alias handling to
other values besides the second argument in route definitions.

  
  examples

  
  Examples


scope "/", MyPrefix do
  get "/", ProxyPlug, controller: scoped_alias(__MODULE__, MyController)
end

  



    

  
    
      
      Link to this macro
    
    trace(path, plug, plug_opts, options \\ [])


      
       
       View Source
     


      (macro)

  


  

Generates a route to handle a trace request to the given path.
trace("/events/:id", EventController, :action)
See match/5 for options.

Tokens provide a way to generate and verify bearer
tokens for use in Channels or API authentication.
The data stored in the token is signed to prevent tampering
but not encrypted. This means it is safe to store identification
information (such as user IDs) but should not be used to store
confidential information (such as credit card numbers).
Example
When generating a unique token for use in an API or Channel
it is advised to use a unique identifier for the user, typically
the id from a database. For example:
iex> user_id = 1
iex> token = Phoenix.Token.sign(MyAppWeb.Endpoint, "user auth", user_id)
iex> Phoenix.Token.verify(MyAppWeb.Endpoint, "user auth", token, max_age: 86400)
{:ok, 1}
In that example we have a user's id, we generate a token and
verify it using the secret key base configured in the given
endpoint. We guarantee the token will only be valid for one day
by setting a max age (recommended).
The first argument to both sign/4 and verify/4 can be one of:
	the module name of a Phoenix endpoint (shown above) - where
the secret key base is extracted from the endpoint
	Plug.Conn - where the secret key base is extracted from the
endpoint stored in the connection
	Phoenix.Socket - where the secret key base is extracted from
the endpoint stored in the socket
	a string, representing the secret key base itself. A key base
with at least 20 randomly generated characters should be used
to provide adequate entropy

The second argument is a cryptographic salt
which must be the same in both calls to sign/4 and verify/4.
For instance, it may be called "user auth" and treated as namespace
when generating a token that will be used to authenticate users on
channels or on your APIs.
The third argument can be any term (string, int, list, etc.)
that you wish to codify into the token. Upon valid verification,
this same term will be extracted from the token.
Usage
Once a token is signed, we can send it to the client in multiple ways.
One is via the meta tag:
<%= tag :meta, name: "channel_token",
               content: Phoenix.Token.sign(@conn, "user auth", @current_user.id) %>
Or an endpoint that returns it:
def create(conn, params) do
  user = User.create(params)
  render(conn, "user.json",
         %{token: Phoenix.Token.sign(conn, "user auth", user.id), user: user})
end
Once the token is sent, the client may now send it back to the server
as an authentication mechanism. For example, we can use it to authenticate
a user on a Phoenix channel:
defmodule MyApp.UserSocket do
  use Phoenix.Socket

  def connect(%{"token" => token}, socket, _connect_info) do
    case Phoenix.Token.verify(socket, "user auth", token, max_age: 86400) do
      {:ok, user_id} ->
        socket = assign(socket, :user, Repo.get!(User, user_id))
        {:ok, socket}
      {:error, _} ->
        :error
    end
  end

  def connect(_params, _socket, _connect_info), do: :error
end
In this example, the phoenix.js client will send the token in the
connect command which is then validated by the server.
Phoenix.Token can also be used for validating APIs, handling
password resets, e-mail confirmation and more.

      Functions
    


      
        
          decrypt(context, secret, token, opts \\ [])

        


          Decrypts the original data from the token and verifies its integrity.



      


      
        
          encrypt(context, secret, data, opts \\ [])

        


          Encodes, encrypts, and signs data into a token you can send to clients.



      


      
        
          sign(context, salt, data, opts \\ [])

        


          Encodes  and signs data into a token you can send to clients.



      


      
        
          verify(context, salt, token, opts \\ [])

        


          Decodes the original data from the token and verifies its integrity.

      Link to this function
    
    decrypt(context, secret, token, opts \\ [])


      
       
       View Source
     


  


  

Decrypts the original data from the token and verifies its integrity.

  
  options

  
  Options


	:max_age - verifies the token only if it has been generated
"max age" ago in seconds. Defaults to the max age signed in the
token (86400)
	:key_iterations - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 1000
	:key_length - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 32
	:key_digest - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to :sha256


  



    

  
    
      
      Link to this function
    
    encrypt(context, secret, data, opts \\ [])


      
       
       View Source
     


  


  

Encodes, encrypts, and signs data into a token you can send to clients.

  
  options

  
  Options


	:key_iterations - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 1000
	:key_length - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 32
	:key_digest - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to :sha256
	:signed_at - set the timestamp of the token in seconds.
Defaults to System.system_time(:second)


  



    

  
    
      
      Link to this function
    
    sign(context, salt, data, opts \\ [])


      
       
       View Source
     


  


  

Encodes  and signs data into a token you can send to clients.

  
  options

  
  Options


	:key_iterations - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 1000
	:key_length - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 32
	:key_digest - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to :sha256
	:signed_at - set the timestamp of the token in seconds.
Defaults to System.system_time(:second)
	:max_age - the default maximum age of the token. Defaults to
86400 seconds (1 day) and it may be overridden on verify/4.


  



    

  
    
      
      Link to this function
    
    verify(context, salt, token, opts \\ [])


      
       
       View Source
     


  


  

Decodes the original data from the token and verifies its integrity.

  
  examples

  
  Examples


In this scenario we will create a token, sign it, then provide it to a client
application. The client will then use this token to authenticate requests for
resources from the server. See Phoenix.Token summary for more info about
creating tokens.
iex> user_id    = 99
iex> secret     = "kjoy3o1zeidquwy1398juxzldjlksahdk3"
iex> namespace  = "user auth"
iex> token      = Phoenix.Token.sign(secret, namespace, user_id)
The mechanism for passing the token to the client is typically through a
cookie, a JSON response body, or HTTP header. For now, assume the client has
received a token it can use to validate requests for protected resources.
When the server receives a request, it can use verify/4 to determine if it
should provide the requested resources to the client:
iex> Phoenix.Token.verify(secret, namespace, token, max_age: 86400)
{:ok, 99}
In this example, we know the client sent a valid token because verify/4
returned a tuple of type {:ok, user_id}. The server can now proceed with
the request.
However, if the client had sent an expired token, an invalid token, or nil,
verify/4 would have returned an error instead:
iex> Phoenix.Token.verify(secret, namespace, expired, max_age: 86400)
{:error, :expired}

iex> Phoenix.Token.verify(secret, namespace, invalid, max_age: 86400)
{:error, :invalid}

iex> Phoenix.Token.verify(secret, namespace, nil, max_age: 86400)
{:error, :missing}

  
  options

  
  Options


	:max_age - verifies the token only if it has been generated
"max age" ago in seconds. A reasonable value is 1 day (86400
seconds)
	:key_iterations - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 1000
	:key_length - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to 32
	:key_digest - option passed to Plug.Crypto.KeyGenerator
when generating the encryption and signing keys. Defaults to :sha256

Conveniences for testing Phoenix channels.
In channel tests, we interact with channels via process
communication, sending and receiving messages. It is also
common to subscribe to the same topic the channel subscribes
to, allowing us to assert if a given message was broadcast
or not.
Channel testing
To get started, define the module attribute @endpoint
in your test case pointing to your application endpoint.
Then you can directly create a socket and
subscribe_and_join/4 topics and channels:
{:ok, _, socket} =
  socket(UserSocket, "user:id", %{some_assigns: 1})
  |> subscribe_and_join(RoomChannel, "room:lobby", %{"id" => 3})
You usually want to set the same ID and assigns your
UserSocket.connect/3 callback would set. Alternatively,
you can use the connect/3 helper to call your UserSocket.connect/3
callback and initialize the socket with the socket id:
{:ok, socket} = connect(UserSocket, %{"some" => "params"}, %{})
{:ok, _, socket} = subscribe_and_join(socket, "room:lobby", %{"id" => 3})
Once called, subscribe_and_join/4 will subscribe the
current test process to the "room:lobby" topic and start a
channel in another process. It returns {:ok, reply, socket}
or {:error, reply}.
Now, in the same way the channel has a socket representing
communication it will push to the client. Our test has a
socket representing communication to be pushed to the server.
For example, we can use the push/3 function in the test
to push messages to the channel (it will invoke handle_in/3):
push(socket, "my_event", %{"some" => "data"})
Similarly, we can broadcast messages from the test itself
on the topic that both test and channel are subscribed to,
triggering handle_out/3 on the channel:
broadcast_from(socket, "my_event", %{"some" => "data"})
Note only broadcast_from/3 and broadcast_from!/3 are
available in tests to avoid broadcast messages to be resent
to the test process.

While the functions above are pushing data to the channel
(server) we can use assert_push/3 to verify the channel
pushed a message to the client:
assert_push "my_event", %{"some" => "data"}
Or even assert something was broadcast into pubsub:
assert_broadcast "my_event", %{"some" => "data"}
Finally, every time a message is pushed to the channel,
a reference is returned. We can use this reference to
assert a particular reply was sent from the server:
ref = push(socket, "counter", %{})
assert_reply ref, :ok, %{"counter" => 1}
Checking side-effects
Often one may want to do side-effects inside channels,
like writing to the database, and verify those side-effects
during their tests.
Imagine the following handle_in/3 inside a channel:
def handle_in("publish", %{"id" => id}, socket) do
  Repo.get!(Post, id) |> Post.publish() |> Repo.update!()
  {:noreply, socket}
end
Because the whole communication is asynchronous, the
following test would be very brittle:
push(socket, "publish", %{"id" => 3})
assert Repo.get_by(Post, id: 3, published: true)
The issue is that we have no guarantees the channel has
done processing our message after calling push/3. The
best solution is to assert the channel sent us a reply
before doing any other assertion. First change the
channel to send replies:
def handle_in("publish", %{"id" => id}, socket) do
  Repo.get!(Post, id) |> Post.publish() |> Repo.update!()
  {:reply, :ok, socket}
end
Then expect them in the test:
ref = push(socket, "publish", %{"id" => 3})
assert_reply ref, :ok
assert Repo.get_by(Post, id: 3, published: true)
Leave and close
This module also provides functions to simulate leaving
and closing a channel. Once you leave or close a channel,
because the channel is linked to the test process on join,
it will crash the test process:
leave(socket)
** (EXIT from #PID<...>) {:shutdown, :leave}
You can avoid this by unlinking the channel process in
the test:
Process.unlink(socket.channel_pid)
Notice leave/1 is async, so it will also return a
reference which you can use to check for a reply:
ref = leave(socket)
assert_reply ref, :ok
On the other hand, close is always sync and it will
return only after the channel process is guaranteed to
have been terminated:
:ok = close(socket)
This mimics the behaviour existing in clients.
To assert that your channel closes or errors asynchronously,
you can monitor the channel process with the tools provided
by Elixir, and wait for the :DOWN message.
Imagine an implementation of the handle_info/2 function
that closes the channel when it receives :some_message:
def handle_info(:some_message, socket) do
  {:stop, :normal, socket}
end
In your test, you can assert that the close happened by:
Process.monitor(socket.channel_pid)
send(socket.channel_pid, :some_message)
assert_receive {:DOWN, _, _, _, :normal}

          assert_broadcast(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

        


          Asserts the channel has broadcast a message within timeout.



      


      
        
          assert_push(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

        


          Asserts the channel has pushed a message back to the client
with the given event and payload within timeout.



      


      
        
          assert_reply(ref, status, payload \\ Macro.escape(%{}), timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))

        


          Asserts the channel has replied to the given message within
timeout.



      


      
        
          broadcast_from!(socket, event, message)

        


          Same as broadcast_from/3, but raises if broadcast fails.



      


      
        
          broadcast_from(socket, event, message)

        


          Broadcast event from pid to all subscribers of the socket topic.



      


      
        
          close(socket, timeout \\ 5000)

        


          Emulates the client closing the socket.



      


      
        
          connect(handler, params, connect_info \\ quote do
  %{}
end)

        


          Initiates a transport connection for the socket handler.



      


      
        
          join(socket, topic)

        


          See join/4.



      


      
        
          join(socket, topic, payload)

        


          See join/4.



      


      
        
          join(socket, channel, topic, payload \\ %{})

        


          Joins the channel under the given topic and payload.



      


      
        
          leave(socket)

        


          Emulates the client leaving the channel.



      


      
        
          push(socket, event, payload \\ %{})

        


          Pushes a message into the channel.



      


      
        
          refute_broadcast(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))

        


          Asserts the channel has not broadcast a message within timeout.



      


      
        
          refute_push(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))

        


          Asserts the channel has not pushed a message to the client
matching the given event and payload within timeout.



      


      
        
          refute_reply(ref, status, payload \\ Macro.escape(%{}), timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))

        


          Asserts the channel has not replied with a matching payload within
timeout.



      


      
        
          socket(socket_module)

        


          Builds a socket for the given socket_module.



      


      
        
          socket(socket_module, socket_id, socket_assigns)

        


          Builds a socket for the given socket_module with given id and assigns.



      


      
        
          subscribe_and_join!(socket, topic)

        


          See subscribe_and_join!/4.



      


      
        
          subscribe_and_join!(socket, topic, payload)

        


          See subscribe_and_join!/4.



      


      
        
          subscribe_and_join!(socket, channel, topic, payload \\ %{})

        


          Same as subscribe_and_join/4, but returns either the socket
or throws an error.



      


      
        
          subscribe_and_join(socket, topic)

        


          See subscribe_and_join/4.



      


      
        
          subscribe_and_join(socket, topic, payload)

        


          See subscribe_and_join/4.



      


      
        
          subscribe_and_join(socket, channel, topic, payload \\ %{})

        


          Subscribes to the given topic and joins the channel
under the given topic and payload.

      Link to this macro
    
    assert_broadcast(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))


      
       
       View Source
     


      (macro)

  


  

Asserts the channel has broadcast a message within timeout.
Before asserting anything was broadcast, we must first
subscribe to the topic of the channel in the test process:
@endpoint.subscribe("foo:ok")
Now we can match on event and payload as patterns:
assert_broadcast "some_event", %{"data" => _}
In the assertion above, we don't particularly care about
the data being sent, as long as something was sent.
The timeout is in milliseconds and defaults to the :assert_receive_timeout
set on the :ex_unit application (which defaults to 100ms).

  



    

  
    
      
      Link to this macro
    
    assert_push(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))


      
       
       View Source
     


      (macro)

  


  

Asserts the channel has pushed a message back to the client
with the given event and payload within timeout.
Notice event and payload are patterns. This means one can write:
assert_push "some_event", %{"data" => _}
In the assertion above, we don't particularly care about
the data being sent, as long as something was sent.
The timeout is in milliseconds and defaults to the :assert_receive_timeout
set on the :ex_unit application (which defaults to 100ms).
NOTE: Because event and payload are patterns, they will be matched.  This
means that if you wish to assert that the received payload is equivalent to
an existing variable, you need to pin the variable in the assertion
expression.
Good:
expected_payload = %{foo: "bar"}
assert_push "some_event", ^expected_payload
Bad:
expected_payload = %{foo: "bar"}
assert_push "some_event", expected_payload
# The code above does not assert the payload matches the described map.

  



    

    

  
    
      
      Link to this macro
    
    assert_reply(ref, status, payload \\ Macro.escape(%{}), timeout \\ Application.fetch_env!(:ex_unit, :assert_receive_timeout))


      
       
       View Source
     


      (macro)

  


  

Asserts the channel has replied to the given message within
timeout.
Notice status and payload are patterns. This means one can write:
ref = push(channel, "some_event")
assert_reply ref, :ok, %{"data" => _}
In the assertion above, we don't particularly care about
the data being sent, as long as something was replied.
The timeout is in milliseconds and defaults to the :assert_receive_timeout
set on the :ex_unit application (which defaults to 100ms).

  



  
    
      
      Link to this function
    
    broadcast_from!(socket, event, message)


      
       
       View Source
     


  


  

Same as broadcast_from/3, but raises if broadcast fails.

  



  
    
      
      Link to this function
    
    broadcast_from(socket, event, message)


      
       
       View Source
     


  


  

Broadcast event from pid to all subscribers of the socket topic.
The test process will not receive the published message. This triggers
the handle_out/3 callback in the channel.

  
  examples

  
  Examples


iex> broadcast_from(socket, "new_message", %{id: 1, content: "hello"})
:ok

  



    

  
    
      
      Link to this function
    
    close(socket, timeout \\ 5000)


      
       
       View Source
     


  


  

Emulates the client closing the socket.
Closing socket is synchronous and has a default timeout
of 5000 milliseconds.

  



    

  
    
      
      Link to this macro
    
    connect(handler, params, connect_info \\ quote do
  %{}
end)


      
       
       View Source
     


      (macro)

  


  

Initiates a transport connection for the socket handler.
Useful for testing UserSocket authentication. Returns
the result of the handler's connect/3 callback.

  



  
    
      
      Link to this function
    
    join(socket, topic)


      
       
       View Source
     


  


  

See join/4.

  



  
    
      
      Link to this function
    
    join(socket, topic, payload)


      
       
       View Source
     


  


  

See join/4.

  



  
    
      
      Link to this function
    
    join(socket, channel, topic, payload \\ %{})


      
       
       View Source
     


  


  

Joins the channel under the given topic and payload.
The given channel is joined in a separate process
which is linked to the test process.
It returns {:ok, reply, socket} or {:error, reply}.

  



  
    
      
      Link to this function
    
    leave(socket)


      
       
       View Source
     


  


  

      

          @spec leave(Phoenix.Socket.t()) :: reference()


      


Emulates the client leaving the channel.

  



    

  
    
      
      Link to this function
    
    push(socket, event, payload \\ %{})

          @spec push(Phoenix.Socket.t(), String.t(), map()) :: reference()


      


Pushes a message into the channel.
The triggers the handle_in/3 callback in the channel.

  
  examples

  
  Examples


iex> push(socket, "new_message", %{id: 1, content: "hello"})
reference

  



    

  
    
      
      Link to this macro
    
    refute_broadcast(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))


      
       
       View Source
     


      (macro)

  


  

Asserts the channel has not broadcast a message within timeout.
Like assert_broadcast, the event and payload are patterns.
The timeout is in milliseconds and defaults to the :refute_receive_timeout
set on the :ex_unit application (which defaults to 100ms).
Keep in mind this macro will block the test by the
timeout value, so use it only when necessary as overuse
will certainly slow down your test suite.

  



    

  
    
      
      Link to this macro
    
    refute_push(event, payload, timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))


      
       
       View Source
     


      (macro)

  


  

Asserts the channel has not pushed a message to the client
matching the given event and payload within timeout.
Like assert_push, the event and payload are patterns.
The timeout is in milliseconds and defaults to the :refute_receive_timeout
set on the :ex_unit application (which defaults to 100ms).
Keep in mind this macro will block the test by the
timeout value, so use it only when necessary as overuse
will certainly slow down your test suite.

  



    

    

  
    
      
      Link to this macro
    
    refute_reply(ref, status, payload \\ Macro.escape(%{}), timeout \\ Application.fetch_env!(:ex_unit, :refute_receive_timeout))


      
       
       View Source
     


      (macro)

  


  

Asserts the channel has not replied with a matching payload within
timeout.
Like assert_reply, the event and payload are patterns.
The timeout is in milliseconds and defaults to the :refute_receive_timeout
set on the :ex_unit application (which defaults to 100ms).
Keep in mind this macro will block the test by the
timeout value, so use it only when necessary as overuse
will certainly slow down your test suite.

  



  
    
      
      Link to this macro
    
    socket(socket_module)


      
       
       View Source
     


      (macro)

  


  

Builds a socket for the given socket_module.
The socket is then used to subscribe and join channels.
Use this function when you want to create a blank socket
to pass to functions like UserSocket.connect/3.
Otherwise, use socket/3 if you want to build a socket with
existing id and assigns.

  
  examples

  
  Examples


socket(MyApp.UserSocket)

  



  
    
      
      Link to this macro
    
    socket(socket_module, socket_id, socket_assigns)


      
       
       View Source
     


      (macro)

  


  

Builds a socket for the given socket_module with given id and assigns.

  
  examples

  
  Examples


socket(MyApp.UserSocket, "user_id", %{some: :assign})

  



  
    
      
      Link to this function
    
    subscribe_and_join!(socket, topic)


      
       
       View Source
     


  


  

See subscribe_and_join!/4.

  



  
    
      
      Link to this function
    
    subscribe_and_join!(socket, topic, payload)


      
       
       View Source
     


  


  

See subscribe_and_join!/4.

  



  
    
      
      Link to this function
    
    subscribe_and_join!(socket, channel, topic, payload \\ %{})


      
       
       View Source
     


  


  

Same as subscribe_and_join/4, but returns either the socket
or throws an error.
This is helpful when you are not testing joining the channel
and just need the socket.

  



  
    
      
      Link to this function
    
    subscribe_and_join(socket, topic)


      
       
       View Source
     


  


  

See subscribe_and_join/4.

  



  
    
      
      Link to this function
    
    subscribe_and_join(socket, topic, payload)


      
       
       View Source
     


  


  

See subscribe_and_join/4.

  



  
    
      
      Link to this function
    
    subscribe_and_join(socket, channel, topic, payload \\ %{})


      
       
       View Source
     


  


  

Subscribes to the given topic and joins the channel
under the given topic and payload.
By subscribing to the topic, we can use assert_broadcast/3
to verify a message has been sent through the pubsub layer.
By joining the channel, we can interact with it directly.
The given channel is joined in a separate process which is
linked to the test process.
If no channel module is provided, the socket's handler is used to
lookup the matching channel for the given topic.
It returns {:ok, reply, socket} or {:error, reply}.

Conveniences for testing Phoenix endpoints and connection related helpers.
You likely want to use this module or make it part of your ExUnit.CaseTemplate.
Once used, this module automatically imports all functions defined here as
well as the functions in Plug.Conn.
Endpoint testing
Phoenix.ConnTest typically works against endpoints. That's the preferred way
to test anything that your router dispatches to:
@endpoint MyAppWeb.Endpoint

test "says welcome on the home page" do
  conn = get(build_conn(), "/")
  assert conn.resp_body =~ "Welcome!"
end

test "logs in" do
  conn = post(build_conn(), "/login", [username: "john", password: "doe"])
  assert conn.resp_body =~ "Logged in!"
end
The @endpoint module attribute contains the endpoint under testing,
most commonly your application endpoint itself. If you are using the
MyApp.ConnCase generated by Phoenix, it is automatically set for you.
As in your router and controllers, the connection is the main abstraction
in testing. build_conn() returns a new connection and functions in this
module can be used to manipulate the connection before dispatching
to the endpoint.
For example, one could set the accepts header for json requests as
follows:
build_conn()
|> put_req_header("accept", "application/json")
|> get("/")
You can also create your own helpers, such as json_conn() that uses
build_conn/0 and put_req_header/3, so you avoid repeating the connection
setup throughout your tests.
Controller testing
The functions in this module can also be used for controller testing.
While endpoint testing is preferred over controller testing, especially
since the controller in Phoenix plays an integration role between your
domain and your views, unit testing controllers may be helpful in some
situations.
For such cases, you need to set the @endpoint attribute to your controller
and pass an atom representing the action to dispatch:
@endpoint MyAppWeb.HomeController

test "says welcome on the home page" do
  conn = get(build_conn(), :index)
  assert conn.resp_body =~ "Welcome!"
end
Keep in mind that, once the @endpoint variable is set, all tests after
setting it will be affected.
Views testing
Under other circumstances, you may be testing a view or another layer that
requires a connection for processing. For such cases, a connection can be
created using the build_conn/3 helper:
MyApp.UserView.render("hello.html", conn: build_conn(:get, "/"))
While build_conn/0 returns a connection with no request information to it,
build_conn/3 returns a connection with the given request information already
filled in.
Recycling
Browsers implement a storage by using cookies. When a cookie is set in the
response, the browser stores it and sends it in the next request.
To emulate this behaviour, this module provides the idea of recycling.
The recycle/1 function receives a connection and returns a new connection,
similar to the one returned by build_conn/0 with all the response cookies
from the previous connection defined as request headers. This is useful when
testing multiple routes that require cookies or session to work.
Keep in mind Phoenix will automatically recycle the connection between
dispatches. This usually works out well most times, but it may discard
information if you are modifying the connection before the next dispatch:
# No recycling as the connection is fresh
conn = get(build_conn(), "/")

# The connection is recycled, creating a new one behind the scenes
conn = post(conn, "/login")

# We can also recycle manually in case we want custom headers
conn =
  conn
  |> recycle()
  |> put_req_header("x-special", "nice")

# No recycling as we did it explicitly
conn = delete(conn, "/logout")
Recycling also recycles the "accept" and "authorization" headers,
as well as peer data information.

      Functions
    


      
        
          assert_error_sent(status_int_or_atom, func)

        


          Asserts an error was wrapped and sent with the given status.



      


      
        
          build_conn()

        


          Creates a connection to be used in upcoming requests.



      


      
        
          build_conn(method, path, params_or_body \\ nil)

        


          Creates a connection to be used in upcoming requests
with a preset method, path and body.



      


      
        
          bypass_through(conn)

        


          Calls the Endpoint and Router pipelines.



      


      
        
          bypass_through(conn, router)

        


          Calls the Endpoint and Router pipelines for the current route.



      


      
        
          bypass_through(conn, router, pipelines)

        


          Calls the Endpoint and the given Router pipelines.



      


      
        
          clear_flash(conn)

        


          Clears up the flash storage.



      


      
        
          connect(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          delete(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          delete_req_cookie(conn, key)

        


          Deletes a request cookie.



      


      
        
          dispatch(conn, endpoint, method, path_or_action, params_or_body \\ nil)

        


          Dispatches the connection to the given endpoint.



      


      
        
          ensure_recycled(conn)

        


          Ensures the connection is recycled if it wasn't already.



      


      
        
          fetch_flash(conn)

        


          Fetches the flash storage.



      


      
        
          get(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          get_flash(conn)

        


          Gets the whole flash storage.



      


      
        
          get_flash(conn, key)

        


          Gets the given key from the flash storage.



      


      
        
          head(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          html_response(conn, status)

        


          Asserts the given status code, that we have an html response and
returns the response body if one was set or sent.



      


      
        
          init_test_session(conn, session)

        


          Inits a session used exclusively for testing.



      


      
        
          json_response(conn, status)

        


          Asserts the given status code, that we have a json response and
returns the decoded JSON response if one was set or sent.



      


      
        
          options(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          patch(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          path_params(conn, to)

        


          Returns the matched params of the URL for the %Plug.Conn{}'s router.



      


      
        
          post(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          put(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.



      


      
        
          put_flash(conn, key, value)

        


          Puts the given value under key in the flash storage.



      


      
        
          put_req_cookie(conn, key, value)

        


          Puts a request cookie.



      


      
        
          recycle(conn, headers \\ ~w(accept accept-language authorization))

        


          Recycles the connection.



      


      
        
          redirected_params(conn)

        


          Returns the matched params from the URL the connection was redirected to.



      


      
        
          redirected_to(conn, status \\ 302)

        


          Returns the location header from the given redirect response.



      


      
        
          response(conn, given)

        


          Asserts the given status code and returns the response body
if one was set or sent.



      


      
        
          response_content_type(conn, format)

        


          Returns the content type as long as it matches the given format.



      


      
        
          text_response(conn, status)

        


          Asserts the given status code, that we have a text response and
returns the response body if one was set or sent.



      


      
        
          trace(conn, path_or_action, params_or_body \\ nil)

        


          Dispatches to the current endpoint.

      Link to this function
    
    assert_error_sent(status_int_or_atom, func)


      
       
       View Source
     


  


  

      

          @spec assert_error_sent(integer() | atom(), function()) :: {integer(), list(), term()}


      


Asserts an error was wrapped and sent with the given status.
Useful for testing actions that you expect raise an error and have
the response wrapped in an HTTP status, with content usually rendered
by your MyApp.ErrorView.
The function accepts a status either as an integer HTTP status or
atom, such as 404 or :not_found. The list of allowed atoms is available
in Plug.Conn.Status. If an error is raised, a 3-tuple of the wrapped
response is returned matching the status, headers, and body of the response:
{404, [{"content-type", "text/html"} | _], "Page not found"}

  
  examples

  
  Examples


assert_error_sent :not_found, fn ->
  get(build_conn(), "/users/not-found")
end

response = assert_error_sent 404, fn ->
  get(build_conn(), "/users/not-found")
end
assert {404, [_h | _t], "Page not found"} = response

  



  
    
      
      Link to this function
    
    build_conn()


      
       
       View Source
     


  


  

      

          @spec build_conn() :: Plug.Conn.t()


      


Creates a connection to be used in upcoming requests.

  



    

  
    
      
      Link to this function
    
    build_conn(method, path, params_or_body \\ nil)


      
       
       View Source
     


  


  

      

          @spec build_conn(atom() | binary(), binary(), binary() | list() | map() | nil) ::
  Plug.Conn.t()


      


Creates a connection to be used in upcoming requests
with a preset method, path and body.
This is useful when a specific connection is required
for testing a plug or a particular function.

  



  
    
      
      Link to this function
    
    bypass_through(conn)


      
       
       View Source
     


  


  

      

          @spec bypass_through(Plug.Conn.t()) :: Plug.Conn.t()


      


Calls the Endpoint and Router pipelines.
Useful for unit testing Plugs where Endpoint and/or router pipeline
plugs are required for proper setup.
Note the use of get("/") following bypass_through in the examples below.
To execute the plug pipelines, you must issue a request against the router.
Most often, you can simply send a GET request against the root path, but you
may also specify a different method or path which your pipelines may operate
against.

  
  examples

  
  Examples


For example, imagine you are testing an authentication plug in
isolation, but you need to invoke the Endpoint plugs and router
pipelines to set up session and flash related dependencies.
One option is to invoke an existing route that uses the proper
pipelines. You can do so by passing the connection and the
router name to bypass_through:
conn =
  conn
  |> bypass_through(MyAppWeb.Router)
  |> get("/some_url")
  |> MyApp.RequireAuthentication.call([])
assert conn.halted
You can also specify which pipelines you want to run:
conn =
  conn
  |> bypass_through(MyAppWeb.Router, [:browser])
  |> get("/")
  |> MyApp.RequireAuthentication.call([])
assert conn.halted
Alternatively, you could only invoke the Endpoint's plugs:
conn =
  conn
  |> bypass_through()
  |> get("/")
  |> MyApp.RequireAuthentication.call([])

assert conn.halted

  



  
    
      
      Link to this function
    
    bypass_through(conn, router)


      
       
       View Source
     


  


  

      

          @spec bypass_through(Plug.Conn.t(), module()) :: Plug.Conn.t()


      


Calls the Endpoint and Router pipelines for the current route.
See bypass_through/1.

  



  
    
      
      Link to this function
    
    bypass_through(conn, router, pipelines)


      
       
       View Source
     


  


  

      

          @spec bypass_through(Plug.Conn.t(), module(), atom() | list()) :: Plug.Conn.t()


      


Calls the Endpoint and the given Router pipelines.
See bypass_through/1.

  



  
    
      
      Link to this function
    
    clear_flash(conn)


      
       
       View Source
     


  


  

      

          @spec clear_flash(Plug.Conn.t()) :: Plug.Conn.t()


      


Clears up the flash storage.

  



    

  
    
      
      Link to this macro
    
    connect(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



    

  
    
      
      Link to this macro
    
    delete(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



  
    
      
      Link to this function
    
    delete_req_cookie(conn, key)


      
       
       View Source
     


  


  

      

          @spec delete_req_cookie(Plug.Conn.t(), binary()) :: Plug.Conn.t()


      


Deletes a request cookie.

  



    

  
    
      
      Link to this function
    
    dispatch(conn, endpoint, method, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


  


  

Dispatches the connection to the given endpoint.
When invoked via get/3, post/3 and friends, the endpoint
is automatically retrieved from the @endpoint module
attribute, otherwise it must be given as an argument.
The connection will be configured with the given method,
path_or_action and params_or_body.
If path_or_action is a string, it is considered to be the
request path and stored as so in the connection. If an atom,
it is assumed to be an action and the connection is dispatched
to the given action.

  
  parameters-and-body

  
  Parameters and body


This function, as well as get/3, post/3 and friends, accepts the
request body or parameters as last argument:
  get(build_conn(), "/", some: "param")
  get(build_conn(), "/", "some=param&url=encoded")
The allowed values are:
	nil - meaning there is no body

	a binary - containing a request body. For such cases, :headers
must be given as option with a content-type

	a map or list - containing the parameters which will automatically
set the content-type to multipart. The map or list may contain
other lists or maps and all entries will be normalized to string
keys

	a struct - unlike other maps, a struct will be passed through as-is
without normalizing its entries



  



  
    
      
      Link to this function
    
    ensure_recycled(conn)


      
       
       View Source
     


  


  

      

          @spec ensure_recycled(Plug.Conn.t()) :: Plug.Conn.t()


      


Ensures the connection is recycled if it wasn't already.
See recycle/1 for more information.

  



  
    
      
      Link to this function
    
    fetch_flash(conn)


      
       
       View Source
     


  


  

      

          @spec fetch_flash(Plug.Conn.t()) :: Plug.Conn.t()


      


Fetches the flash storage.

  



    

  
    
      
      Link to this macro
    
    get(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



  
    
      
      Link to this function
    
    get_flash(conn)


      
       
       View Source
     


  


  

      

          @spec get_flash(Plug.Conn.t()) :: map()


      


Gets the whole flash storage.

  



  
    
      
      Link to this function
    
    get_flash(conn, key)


      
       
       View Source
     


  


  

      

          @spec get_flash(Plug.Conn.t(), term()) :: term()


      


Gets the given key from the flash storage.

  



    

  
    
      
      Link to this macro
    
    head(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



  
    
      
      Link to this function
    
    html_response(conn, status)


      
       
       View Source
     


  


  

      

          @spec html_response(Plug.Conn.t(), status :: integer() | atom()) ::
  String.t() | no_return()


      


Asserts the given status code, that we have an html response and
returns the response body if one was set or sent.

  
  examples

  
  Examples


assert html_response(conn, 200) =~ "<html>"

  



  
    
      
      Link to this function
    
    init_test_session(conn, session)


      
       
       View Source
     


  


  

      

          @spec init_test_session(Plug.Conn.t(), map() | keyword()) :: Plug.Conn.t()


      


Inits a session used exclusively for testing.

  



  
    
      
      Link to this function
    
    json_response(conn, status)


      
       
       View Source
     


  


  

      

          @spec json_response(Plug.Conn.t(), status :: integer() | atom()) ::
  map() | no_return()


      


Asserts the given status code, that we have a json response and
returns the decoded JSON response if one was set or sent.

  
  examples

  
  Examples


body = json_response(conn, 200)
assert "can't be blank" in body["errors"]

  



    

  
    
      
      Link to this macro
    
    options(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



    

  
    
      
      Link to this macro
    
    patch(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



  
    
      
      Link to this function
    
    path_params(conn, to)


      
       
       View Source
     


  


  

      

          @spec path_params(Plug.Conn.t(), String.t()) :: map()


      


Returns the matched params of the URL for the %Plug.Conn{}'s router.
Useful for extracting path params out of returned URLs, such as those
returned by Phoenix.LiveViewTest's redirected results.

  
  examples

  
  Examples


assert {:error, {:redirect, %{to: "/posts/123" = to}}} = live(conn, "/path")
assert %{id: "123"} = path_params(conn, to)

  



    

  
    
      
      Link to this macro
    
    post(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



    

  
    
      
      Link to this macro
    
    put(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

  



  
    
      
      Link to this function
    
    put_flash(conn, key, value)


      
       
       View Source
     


  


  

      

          @spec put_flash(Plug.Conn.t(), term(), term()) :: Plug.Conn.t()


      


Puts the given value under key in the flash storage.

  



  
    
      
      Link to this function
    
    put_req_cookie(conn, key, value)


      
       
       View Source
     


  


  

      

          @spec put_req_cookie(Plug.Conn.t(), binary(), binary()) :: Plug.Conn.t()


      


Puts a request cookie.

  



    

  
    
      
      Link to this function
    
    recycle(conn, headers \\ ~w(accept accept-language authorization))


      
       
       View Source
     


  


  

      

          @spec recycle(Plug.Conn.t(), [String.t()]) :: Plug.Conn.t()


      


Recycles the connection.
Recycling receives a connection and returns a new connection,
containing cookies and relevant information from the given one.
This emulates behaviour performed by browsers where cookies
returned in the response are available in following requests.
By default, only the headers "accept", "accept-language", and
"authorization" are recycled. However, a custom set of headers
can be specified by passing a list of strings representing its
names as the second argument of the function.
Note recycle/1 is automatically invoked when dispatching
to the endpoint, unless the connection has already been
recycled.

  



  
    
      
      Link to this function
    
    redirected_params(conn)


      
       
       View Source
     


  


  

      

          @spec redirected_params(Plug.Conn.t()) :: map()


      


Returns the matched params from the URL the connection was redirected to.
Uses the provided %Plug.Conn{}s router matched in the previous request.
Raises if the response's location header is not set.

  
  examples

  
  Examples


assert redirected_to(conn) =~ "/posts/123"
assert %{id: "123"} = redirected_params(conn)

  



    

  
    
      
      Link to this function
    
    redirected_to(conn, status \\ 302)


      
       
       View Source
     


  


  

      

          @spec redirected_to(Plug.Conn.t(), status :: non_neg_integer()) :: String.t()


      


Returns the location header from the given redirect response.
Raises if the response does not match the redirect status code
(defaults to 302).

  
  examples

  
  Examples


assert redirected_to(conn) =~ "/foo/bar"
assert redirected_to(conn, 301) =~ "/foo/bar"
assert redirected_to(conn, :moved_permanently) =~ "/foo/bar"

  



  
    
      
      Link to this function
    
    response(conn, given)


      
       
       View Source
     


  


  

      

          @spec response(Plug.Conn.t(), status :: integer() | atom()) :: binary() | no_return()


      


Asserts the given status code and returns the response body
if one was set or sent.

  
  examples

  
  Examples


conn = get(build_conn(), "/")
assert response(conn, 200) =~ "hello world"

  



  
    
      
      Link to this function
    
    response_content_type(conn, format)


      
       
       View Source
     


  


  

      

          @spec response_content_type(Plug.Conn.t(), atom()) :: String.t() | no_return()


      


Returns the content type as long as it matches the given format.

  
  examples

  
  Examples


# Assert we have an html response with utf-8 charset
assert response_content_type(conn, :html) =~ "charset=utf-8"

  



  
    
      
      Link to this function
    
    text_response(conn, status)


      
       
       View Source
     


  


  

      

          @spec text_response(Plug.Conn.t(), status :: integer() | atom()) ::
  String.t() | no_return()


      


Asserts the given status code, that we have a text response and
returns the response body if one was set or sent.

  
  examples

  
  Examples


assert text_response(conn, 200) =~ "hello"

  



    

  
    
      
      Link to this macro
    
    trace(conn, path_or_action, params_or_body \\ nil)


      
       
       View Source
     


      (macro)

  


  

Dispatches to the current endpoint.
See dispatch/5 for more information.

Phoenix.CodeReloader 
    



      
A plug and module to handle automatic code reloading.
To avoid race conditions, all code reloads are funneled through a
sequential call operation.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    
      Functions
    


      
        
          call(conn, opts)

        


          API used by Plug to invoke the code reloader on every request.



      


      
        
          init(opts)

        


          API used by Plug to start the code reloader.



      


      
        
          reload!(endpoint)

        


          Same as reload/1 but it will raise if Mix is not available.



      


      
        
          reload(endpoint)

        


          Reloads code for the current Mix project by invoking the
:reloadable_compilers on the list of :reloadable_apps.



      


      
        
          sync()

        


          Synchronizes with the code server if it is alive.



      


  


      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this function
    
    call(conn, opts)


      
       
       View Source
     


  


  

API used by Plug to invoke the code reloader on every request.

  



  
    
      
      Link to this function
    
    init(opts)


      
       
       View Source
     


  


  

API used by Plug to start the code reloader.

  



  
    
      
      Link to this function
    
    reload!(endpoint)


      
       
       View Source
     


  


  

      

          @spec reload!(module()) :: :ok | {:error, binary()}


      


Same as reload/1 but it will raise if Mix is not available.

  



  
    
      
      Link to this function
    
    reload(endpoint)


      
       
       View Source
     


  


  

      

          @spec reload(module()) :: :ok | {:error, binary()}


      


Reloads code for the current Mix project by invoking the
:reloadable_compilers on the list of :reloadable_apps.
This is configured in your application environment like:
config :your_app, YourAppWeb.Endpoint,
  reloadable_compilers: [:gettext, :elixir],
  reloadable_apps: [:ui, :backend]
Keep in mind :reloadable_compilers must be a subset of the
:compilers specified in project/0 in your mix.exs.
The :reloadable_apps defaults to nil. In such case
default behaviour is to reload the current project if it
consists of a single app, or all applications within an umbrella
project. You can set :reloadable_apps to a subset of default
applications to reload only some of them, an empty list - to
effectively disable the code reloader, or include external
applications from library dependencies.
This function is a no-op and returns :ok if Mix is not available.

  



  
    
      
      Link to this function
    
    sync()


      
       
       View Source
     


  


  

      

          @spec sync() :: :ok


      


Synchronizes with the code server if it is alive.
It returns :ok. If it is not running, it also returns :ok.

Phoenix.Endpoint.Cowboy2Adapter 
    



      
The Cowboy2 adapter for Phoenix.
Endpoint configuration
This adapter uses the following endpoint configuration:
	:http - the configuration for the HTTP server. It accepts all options
as defined by Plug.Cowboy. Defaults
to false

	:https - the configuration for the HTTPS server. It accepts all options
as defined by Plug.Cowboy. Defaults
to false

	:drainer - a drainer process that triggers when your application is
shutting down to wait for any on-going request to finish. It accepts all
options as defined by Plug.Cowboy.Drainer.
Defaults to [], which will start a drainer process for each configured endpoint,
but can be disabled by setting it to false.


Custom dispatch options
You can provide custom dispatch options in order to use Phoenix's
builtin Cowboy server with custom handlers. For example, to handle
raw WebSockets as shown in Cowboy's docs).
The options are passed to both :http and :https keys in the
endpoint configuration. However, once you pass your custom dispatch
options, you will need to manually wire the Phoenix endpoint by
adding the following rule:
{:_, Phoenix.Endpoint.Cowboy2Handler, {MyAppWeb.Endpoint, []}}
For example:
config :myapp, MyAppWeb.Endpoint,
  http: [dispatch: [
          {:_, [
              {"/foo", MyAppWeb.CustomHandler, []},
              {:_, Phoenix.Endpoint.Cowboy2Handler, {MyAppWeb.Endpoint, []}}
            ]}]]
It is also important to specify your handlers first, otherwise
Phoenix will intercept the requests before they get to your handler.

Phoenix.Socket behaviour
    



      
A socket implementation that multiplexes messages over channels.
Phoenix.Socket is used as a module for establishing and maintaining
the socket state via the Phoenix.Socket struct.
Once connected to a socket, incoming and outgoing events are routed to
channels. The incoming client data is routed to channels via transports.
It is the responsibility of the socket to tie transports and channels
together.
Phoenix supports websocket and longpoll options when invoking
Phoenix.Endpoint.socket/3 in your endpoint. websocket is set by default
and longpoll can also be configured explicitly.
socket "/socket", MyAppWeb.Socket, websocket: true, longpoll: false
The command above means incoming socket connections can be made via
a WebSocket connection. Events are routed by topic to channels:
channel "room:lobby", MyAppWeb.LobbyChannel
See Phoenix.Channel for more information on channels.
Socket Behaviour
Socket handlers are mounted in Endpoints and must define two callbacks:
	connect/3 - receives the socket params, connection info if any, and
authenticates the connection. Must return a Phoenix.Socket struct,
often with custom assigns
	id/1 - receives the socket returned by connect/3 and returns the
id of this connection as a string. The id is used to identify socket
connections, often to a particular user, allowing us to force disconnections.
For sockets requiring no authentication, nil can be returned

Examples
defmodule MyAppWeb.UserSocket do
  use Phoenix.Socket

  channel "room:*", MyAppWeb.RoomChannel

  def connect(params, socket, _connect_info) do
    {:ok, assign(socket, :user_id, params["user_id"])}
  end

  def id(socket), do: "users_socket:#{socket.assigns.user_id}"
end

# Disconnect all user's socket connections and their multiplexed channels
MyAppWeb.Endpoint.broadcast("users_socket:" <> user.id, "disconnect", %{})
Socket fields
	:id - The string id of the socket
	:assigns - The map of socket assigns, default: %{}
	:channel - The current channel module
	:channel_pid - The channel pid
	:endpoint - The endpoint module where this socket originated, for example: MyAppWeb.Endpoint
	:handler - The socket module where this socket originated, for example: MyAppWeb.UserSocket
	:joined - If the socket has effectively joined the channel
	:join_ref - The ref sent by the client when joining
	:ref - The latest ref sent by the client
	:pubsub_server - The registered name of the socket's pubsub server
	:topic - The string topic, for example "room:123"
	:transport - An identifier for the transport, used for logging
	:transport_pid - The pid of the socket's transport process
	:serializer - The serializer for socket messages

Using options
On use Phoenix.Socket, the following options are accepted:
	:log - the default level to log socket actions. Defaults
to :info. May be set to false to disable it

	:partitions - each channel is spawned under a supervisor.
This option controls how many supervisors will be spawned
to handle channels. Defaults to the number of cores.


Garbage collection
It's possible to force garbage collection in the transport process after
processing large messages. For example, to trigger such from your channels,
run:
send(socket.transport_pid, :garbage_collect)
Alternatively, you can configure your endpoint socket to trigger more
fullsweep garbage collections more frequently, by setting the :fullsweep_after
option for websockets. See Phoenix.Endpoint.socket/3 for more info.
Client-server communication
The encoding of server data and the decoding of client data is done
according to a serializer, defined in Phoenix.Socket.Serializer.
By default, JSON encoding is used to broker messages to and from clients.
The serializer decode! function must return a Phoenix.Socket.Message
which is forwarded to channels except:
	"heartbeat" events in the "phoenix" topic - should just emit an OK reply
	"phx_join" on any topic - should join the topic
	"phx_leave" on any topic - should leave the topic

Each message also has a ref field which is used to track responses.
The server may send messages or replies back. For messages, the
ref uniquely identifies the message. For replies, the ref matches
the original message. Both data-types also include a join_ref that
uniquely identifies the currently joined channel.
The Phoenix.Socket implementation may also send special messages
and replies:
	"phx_error" - in case of errors, such as a channel process
crashing, or when attempting to join an already joined channel

	"phx_close" - the channel was gracefully closed


Phoenix ships with a JavaScript implementation of both websocket
and long polling that interacts with Phoenix.Socket and can be
used as reference for those interested in implementing custom clients.
Custom sockets and transports
See the Phoenix.Socket.Transport documentation for more information on
writing your own socket that does not leverage channels or for writing
your own transports that interacts with other sockets.
Custom channels
You can list any module as a channel as long as it implements
a child_spec/1 function. The child_spec/1 function receives
the caller as argument and it must return a child spec that
initializes a process.
Once the process is initialized, it will receive the following
message:
{Phoenix.Channel, auth_payload, from, socket}
A custom channel implementation MUST invoke
GenServer.reply(from, {:ok | :error, reply_payload}) during its
initialization with a custom reply_payload that will be sent as
a reply to the client. Failing to do so will block the socket forever.
A custom channel receives Phoenix.Socket.Message structs as regular
messages from the transport. Replies to those messages and custom
messages can be sent to the socket at any moment by building an
appropriate Phoenix.Socket.Reply and Phoenix.Socket.Message
structs, encoding them with the serializer and dispatching the
serialized result to the transport.
For example, to handle "phx_leave" messages, which is recommended
to be handled by all channel implementations, one may do:
def handle_info(
      %Message{topic: topic, event: "phx_leave"} = message,
      %{topic: topic, serializer: serializer, transport_pid: transport_pid} = socket
    ) do
  send transport_pid, serializer.encode!(build_leave_reply(message))
  {:stop, {:shutdown, :left}, socket}
end
We also recommend all channels to monitor the transport_pid
on init and exit if the transport exits. We also advise to rewrite
:normal exit reasons (usually due to the socket being closed)
to the {:shutdown, :closed} to guarantee links are broken on
the channel exit (as a :normal exit does not break links):
def handle_info({:DOWN, _, _, transport_pid, reason}, %{transport_pid: transport_pid} = socket) do
  reason = if reason == :normal, do: {:shutdown, :closed}, else: reason
  {:stop, reason, socket}
end
Any process exit is treated as an error by the socket layer unless
a {:socket_close, pid, reason} message is sent to the socket before
shutdown.
Custom channel implementations cannot be tested with Phoenix.ChannelTest
and are currently considered experimental. The underlying API may be
changed at any moment.

          @type t() :: %Phoenix.Socket{
  assigns: map(),
  channel: atom(),
  channel_pid: pid(),
  endpoint: atom(),
  handler: atom(),
  id: String.t() | nil,
  join_ref: term(),
  joined: boolean(),
  private: map(),
  pubsub_server: atom(),
  ref: term(),
  serializer: atom(),
  topic: String.t(),
  transport: atom(),
  transport_pid: pid()
}

          @callback connect(params :: map(), t()) :: {:ok, t()} | {:error, term()} | :error


      


Receives the socket params and authenticates the connection.

  
  socket-params-and-assigns

  
  Socket params and assigns


Socket params are passed from the client and can
be used to verify and authenticate a user. After
verification, you can put default assigns into
the socket that will be set for all channels, ie
{:ok, assign(socket, :user_id, verified_user_id)}
To deny connection, return :error.
See Phoenix.Token documentation for examples in
performing token verification on connect.

          @callback connect(params :: map(), t(), connect_info :: map()) ::
  {:ok, t()} | {:error, term()} | :error

          @callback id(t()) :: String.t() | nil


      


Identifies the socket connection.
Socket IDs are topics that allow you to identify all sockets for a given user:
def id(socket), do: "users_socket:#{socket.assigns.user_id}"
Would allow you to broadcast a "disconnect" event and terminate
all active sockets and channels for a given user:
MyAppWeb.Endpoint.broadcast("users_socket:" <> user.id, "disconnect", %{})
Returning nil makes this socket anonymous.

Adds key-value pairs to socket assigns.
A single key-value pair may be passed, a keyword list or map
of assigns may be provided to be merged into existing socket
assigns.

  
  examples

  
  Examples


iex> assign(socket, :name, "Elixir")
iex> assign(socket, name: "Elixir", logo: "💧")

      (macro)

  


  

Defines a channel matching the given topic and transports.
	topic_pattern - The string pattern, for example "room:*", "users:*",
or "system"
	module - The channel module handler, for example MyAppWeb.RoomChannel
	opts - The optional list of options, see below


  
  options

  
  Options


	:assigns - the map of socket assigns to merge into the socket on join


  
  examples

  
  Examples


channel "topic1:*", MyChannel

  
  topic-patterns

  
  Topic Patterns


The channel macro accepts topic patterns in two flavors. A splat (the *
character) argument can be provided as the last character to indicate a
"topic:subtopic" match. If a plain string is provided, only that topic will
match the channel handler. Most use-cases will use the "topic:*" pattern to
allow more versatile topic scoping.
See Phoenix.Channel for more information

Phoenix.Socket.Broadcast 
    



      
Defines a message sent from pubsub to channels and vice-versa.
The message format requires the following keys:
	:topic - The string topic or topic:subtopic pair namespace, for example "messages", "messages:123"
	:event- The string event name, for example "phx_join"
	:payload - The message payload

          @type t() :: %Phoenix.Socket.Broadcast{event: term(), payload: term(), topic: term()}


      



  


        

      



  

  
    
    Phoenix.Socket.Message - Phoenix v1.6.9
    
    

    



  
  

    
Phoenix.Socket.Message 
    



      
Defines a message dispatched over transport to channels and vice-versa.
The message format requires the following keys:
	:topic - The string topic or topic:subtopic pair namespace, for
example "messages", "messages:123"
	:event- The string event name, for example "phx_join"
	:payload - The message payload
	:ref - The unique string ref
	:join_ref - The unique string ref when joining

          Converts a map with string keys into a message struct.

          @type t() :: %Phoenix.Socket.Message{
  event: term(),
  join_ref: term(),
  payload: term(),
  ref: term(),
  topic: term()
}

Converts a map with string keys into a message struct.
Raises Phoenix.Socket.InvalidMessageError if not valid.

Defines a reply sent from channels to transports.
The message format requires the following keys:
	:topic - The string topic or topic:subtopic pair namespace, for example "messages", "messages:123"
	:status - The reply status as an atom
	:payload - The reply payload
	:ref - The unique string ref
	:join_ref - The unique string ref when joining

          @type t() :: %Phoenix.Socket.Reply{
  join_ref: term(),
  payload: term(),
  ref: term(),
  status: term(),
  topic: term()
}

Phoenix.Socket.Serializer behaviour
    



      
A behaviour that serializes incoming and outgoing socket messages.
By default Phoenix provides a serializer that encodes to JSON and
decodes JSON messages.
Custom serializers may be configured in the socket.

          Encodes a Phoenix.Socket.Broadcast struct to fastlane format.

          @callback decode!(iodata(), options :: Keyword.t()) :: Phoenix.Socket.Message.t()


      


Decodes iodata into Phoenix.Socket.Message struct.

          @callback encode!(Phoenix.Socket.Message.t() | Phoenix.Socket.Reply.t()) ::
  {:socket_push, :text, iodata()} | {:socket_push, :binary, iodata()}


      


Encodes Phoenix.Socket.Message and Phoenix.Socket.Reply structs to push format.

          @callback fastlane!(Phoenix.Socket.Broadcast.t()) ::
  {:socket_push, :text, iodata()} | {:socket_push, :binary, iodata()}


      


Encodes a Phoenix.Socket.Broadcast struct to fastlane format.

Outlines the Socket <-> Transport communication.
This module specifies a behaviour that all sockets must implement.
Phoenix.Socket is just one possible implementation of a socket
that multiplexes events over multiple channels. Developers can
implement their own sockets as long as they implement the behaviour
outlined here.
Developers interested in implementing custom transports must invoke
the socket API defined in this module. This module also provides
many conveniences that invokes the underlying socket API to make
it easier to build custom transports.
Booting sockets
Whenever your endpoint starts, it will automatically invoke the
child_spec/1 on each listed socket and start that specification
under the endpoint supervisor.
Since the socket supervision tree is started by the endpoint,
any custom transport must be started after the endpoint in a
supervision tree.
Operating sockets
Sockets are operated by a transport. When a transport is defined,
it usually receives a socket module and the module will be invoked
when certain events happen at the transport level.
Whenever the transport receives a new connection, it should invoke
the connect/1 callback with a map of metadata. Different sockets
may require different metadatas.
If the connection is accepted, the transport can move the connection
to another process, if so desires, or keep using the same process. The
process responsible for managing the socket should then call init/1.
For each message received from the client, the transport must call
handle_in/2 on the socket. For each informational message the
transport receives, it should call handle_info/2 on the socket.
Transports can optionally implement handle_control/2 for handling
control frames such as :ping and :pong.
On termination, terminate/2 must be called. A special atom with
reason :closed can be used to specify that the client terminated
the connection.
Example
Here is a simple echo socket implementation:
defmodule EchoSocket do
  @behaviour Phoenix.Socket.Transport

  def child_spec(opts) do
    # We won't spawn any process, so let's return a dummy task
    %{id: __MODULE__, start: {Task, :start_link, [fn -> :ok end]}, restart: :transient}
  end

  def connect(state) do
    # Callback to retrieve relevant data from the connection.
    # The map contains options, params, transport and endpoint keys.
    {:ok, state}
  end

  def init(state) do
    # Now we are effectively inside the process that maintains the socket.
    {:ok, state}
  end

  def handle_in({text, _opts}, state) do
    {:reply, :ok, {:text, text}, state}
  end

  def handle_info(_, state) do
    {:ok, state}
  end

  def terminate(_reason, _state) do
    :ok
  end
end
It can be mounted in your endpoint like any other socket:
socket "/socket", EchoSocket, websocket: true, longpoll: true
You can now interact with the socket under /socket/websocket
and /socket/longpoll.
Security
This module also provides functions to enable a secure environment
on transports that, at some point, have access to a Plug.Conn.
The functionality provided by this module helps in performing "origin"
header checks and ensuring only SSL connections are allowed.

          handle_control(
  {},
  state
)

        


          Handles incoming control frames.



      


      
        
          handle_in(

          Handles incoming socket messages.



      


      
        
          handle_info(message, state)

        


          Handles info messages.



      


      
        
          init(state)

        


          Initializes the socket state.



      


      
        
          terminate(reason, state)

        


          Invoked on termination.

          check_origin(conn, handler, endpoint, opts, sender \\ &Plug.Conn.send_resp/1)

        


          Checks the origin request header against the list of allowed origins.



      


      
        
          check_subprotocols(conn, subprotocols)

        


          Checks the Websocket subprotocols request header against the allowed subprotocols.



      


      
        
          code_reload(conn, endpoint, opts)

        


          Runs the code reloader if enabled.



      


      
        
          connect_info(conn, endpoint, keys)

        


          Extracts connection information from conn and returns a map.



      


      
        
          force_ssl(conn, socket, endpoint, opts)

        


          Forces SSL in the socket connection.

          Logs the transport request.

          @type state() :: term()

          @callback child_spec(keyword()) :: :supervisor.child_spec()


      


Returns a child specification for socket management.
This is invoked only once per socket regardless of
the number of transports and should be responsible
for setting up any process structure used exclusively
by the socket regardless of transports.
Each socket connection is started by the transport
and the process that controls the socket likely
belongs to the transport. However, some sockets spawn
new processes, such as Phoenix.Socket which spawns
channels, and this gives the ability to start a
supervision tree associated to the socket.
It receives the socket options from the endpoint,
for example:
socket "/my_app", MyApp.Socket, shutdown: 5000
means child_spec([shutdown: 5000]) will be invoked.

          @callback connect(transport_info :: map()) :: {:ok, state()} | :error


      


Connects to the socket.
The transport passes a map of metadata and the socket
returns {:ok, state} or :error. The state must be
stored by the transport and returned in all future
operations.
This function is used for authorization purposes and it
may be invoked outside of the process that effectively
runs the socket.
In the default Phoenix.Socket implementation, the
metadata expects the following keys:
	:endpoint - the application endpoint
	:transport - the transport name
	:params - the connection parameters
	:options - a keyword list of transport options, often
given by developers when configuring the transport.
It must include a :serializer field with the list of
serializers and their requirements

      (optional)

  


  

      

          @callback handle_control(
  {message :: term(), opts :: keyword()},
  state()
) ::
  {:ok, state()}
  | {:reply, :ok | :error, {opcode :: atom(), message :: term()}, state()}
  | {:stop, reason :: term(), state()}


      


Handles incoming control frames.
The message is represented as {payload, options}. It must
return one of:
	{:ok, state} - continues the socket with no reply
	{:reply, status, reply, state} - continues the socket with reply
	{:stop, reason, state} - stops the socket

Control frames only supported when using websockets.
The options contains an opcode key, this will be either :ping or
:pong.
If a control frame doesn't have a payload, then the payload value
will be nil.

          @callback handle_in(
  {message :: term(), opts :: keyword()},
  state()
) ::
  {:ok, state()}
  | {:reply, :ok | :error, {opcode :: atom(), message :: term()}, state()}
  | {:stop, reason :: term(), state()}


      


Handles incoming socket messages.
The message is represented as {payload, options}. It must
return one of:
	{:ok, state} - continues the socket with no reply
	{:reply, status, reply, state} - continues the socket with reply
	{:stop, reason, state} - stops the socket

The reply is a tuple contain an opcode atom and a message that can
be any term. The built-in websocket transport supports both :text and
:binary opcode and the message must be always iodata. Long polling only
supports text opcode.

          @callback handle_info(message :: term(), state()) ::
  {:ok, state()}
  | {:push, {opcode :: atom(), message :: term()}, state()}
  | {:stop, reason :: term(), state()}


      


Handles info messages.
The message is a term. It must return one of:
	{:ok, state} - continues the socket with no reply
	{:push, reply, state} - continues the socket with reply
	{:stop, reason, state} - stops the socket

The reply is a tuple contain an opcode atom and a message that can
be any term. The built-in websocket transport supports both :text and
:binary opcode and the message must be always iodata. Long polling only
supports text opcode.

          @callback init(state()) :: {:ok, state()}


      


Initializes the socket state.
This must be executed from the process that will effectively
operate the socket.

  



  
    
      
      Link to this callback

          @callback terminate(reason :: term(), state()) :: :ok


      


Invoked on termination.
If reason is :closed, it means the client closed the socket. This is
considered a :normal exit signal, so linked process will not automatically
exit. See Process.exit/2 for more details on exit signals.

      Link to this function
    
    check_origin(conn, handler, endpoint, opts, sender \\ &Plug.Conn.send_resp/1)

Checks the origin request header against the list of allowed origins.
Should be called by transports before connecting when appropriate.
If the origin header matches the allowed origins, no origin header was
sent or no origin was configured, it will return the given connection.
Otherwise a 403 Forbidden response will be sent and the connection halted.
It is a noop if the connection has been halted.

    check_subprotocols(conn, subprotocols)

Checks the Websocket subprotocols request header against the allowed subprotocols.
Should be called by transports before connecting when appropriate.
If the sec-websocket-protocol header matches the allowed subprotocols,
it will put sec-websocket-protocol response header and return the given connection.
If no sec-websocket-protocol header was sent it will return the given connection.
Otherwise a 403 Forbidden response will be sent and the connection halted.
It is a noop if the connection has been halted.

    code_reload(conn, endpoint, opts)

Runs the code reloader if enabled.

    connect_info(conn, endpoint, keys)

Extracts connection information from conn and returns a map.
Keys are retrieved from the optional transport option :connect_info.
This functionality is transport specific. Please refer to your transports'
documentation for more information.
The supported keys are:
	:peer_data - the result of Plug.Conn.get_peer_data/1

	:trace_context_headers - a list of all trace context headers

	:x_headers - a list of all request headers that have an "x-" prefix

	:uri - a %URI{} derived from the conn

	:user_agent - the value of the "user-agent" request header

Forces SSL in the socket connection.
Uses the endpoint configuration to decide so. It is a
noop if the connection has been halted.

    transport_log(conn, level)

Logs the transport request.
Available for transports that generate a connection.

          Callback implementation for Exception.blame/2.

          Callback implementation for Exception.message/1.

Callback implementation for Exception.blame/2.

Callback implementation for Exception.message/1.

Phoenix.MissingParamError exception
    



      
Raised when a key is expected to be present in the request parameters,
but is not.
This exception is raised by Phoenix.Controller.scrub_params/2 which:
	Checks to see if the required_key is present (can be empty)
	Changes all empty parameters to nils ("" -> nil)

If you are seeing this error, you should handle the error and surface it
to the end user. It means that there is a parameter missing from the request.

Phoenix.NotAcceptableError exception
    



      
Raised when one of the accept* headers is not accepted by the server.
This exception is commonly raised by Phoenix.Controller.accepts/2
which negotiates the media types the server is able to serve with
the contents the client is able to render.
If you are seeing this error, you should check if you are listing
the desired formats in your :accepts plug or if you are setting
the proper accept header in the client. The exception contains the
acceptable mime types in the accepts field.

Exception raised when the URI is malformed on matching.

Phoenix.Router.NoRouteError exception
    



      
Exception raised when no route is found.

Phoenix.Socket.InvalidMessageError exception
    



      
Raised when the socket message is invalid.

Compiles Phoenix source files that support code reloading.
If you are using Elixir v1.11+ or later, there is no longer
a need to use this module as this functionality is now provided
by Elixir. Just remember to update __phoenix_recompile__? to
__mix_recompile__? in any module that may define it.

Updates the Phoenix project generator locally.
$ mix local.phx

Accepts the same command line options as archive.install hex phx_new.

mix phx 
    



      
Prints Phoenix tasks and their information.
$ mix phx

To print the Phoenix version, pass -v or --version, for example:
$ mix phx --version

Digests and compresses static files.
$ mix phx.digest
$ mix phx.digest priv/static -o /www/public

The first argument is the path where the static files are located. The
-o option indicates the path that will be used to save the digested and
compressed files.
If no path is given, it will use priv/static as the input and output path.
The output folder will contain:
	the original file
	the file compressed with gzip
	a file containing the original file name and its digest
	a compressed file containing the file name and its digest
	a cache manifest file

Example of generated files:
	app.js
	app.js.gz
	app-eb0a5b9302e8d32828d8a73f137cc8f0.js
	app-eb0a5b9302e8d32828d8a73f137cc8f0.js.gz
	cache_manifest.json

You can use mix phx.digest.clean to prune stale versions of the assets.
If you want to remove all produced files, run mix phx.digest.clean --all.
vsn
It is possible to digest the stylesheet asset references without the query
string "?vsn=d" with the option --no-vsn.

Removes old versions of compiled assets.
By default, it will keep the latest version and
2 previous versions as well as any digest created
in the last hour.
$ mix phx.digest.clean
$ mix phx.digest.clean -o /www/public
$ mix phx.digest.clean --age 600 --keep 3
$ mix phx.digest.clean --all

Options
	-o, --output - indicates the path to your compiled
assets directory. Defaults to priv/static

	--age - specifies a maximum age (in seconds) for assets.
Files older than age that are not in the last --keep versions
will be removed. Defaults to 3600 (1 hour)

	--keep - specifies how many previous versions of assets to keep.
Defaults to 2 previous versions

	--all - specifies that all compiled assets (including the manifest)
will be removed. Note this overrides the age and keep switches.

Lists all available Phoenix generators.
CRUD related generators
The table below shows a summary of the contents created by the CRUD generators:
	Task	Schema	Migration	Context	Controller	View	LiveView
	phx.gen.embedded	✓					
	phx.gen.schema	✓	✓				
	phx.gen.context	✓	✓	✓			
	phx.gen.live	✓	✓	✓			✓
	phx.gen.json	✓	✓	✓	✓	✓	
	phx.gen.html	✓	✓	✓	✓	✓	

Generates authentication logic for a resource.
$ mix phx.gen.auth Accounts User users

The first argument is the context module followed by the schema module
and its plural name (used as the schema table name).
Additional information and security considerations are detailed in the
mix phx.gen.auth guide.
Password hashing
The password hashing mechanism defaults to bcrypt for
Unix systems and pbkdf2 for Windows systems. Both
systems use the Comeonin interface.
The password hashing mechanism can be overridden with the
--hashing-lib option. The following values are supported:
	bcrypt - bcrypt_elixir
	pbkdf2 - pbkdf2_elixir
	argon2 - argon2_elixir

We recommend developers to consider using argon2, which
is the most robust of all 3. The downside is that argon2
is quite CPU and memory intensive, and you will need more
powerful instances to run your applications on.
For more information about choosing these libraries, see the
Comeonin project.
Web namespace
By default, the controllers and view will be namespaced by the schema name.
You can customize the web module namespace by passing the --web flag with a
module name, for example:
$ mix phx.gen.auth Accounts User users --web Warehouse

Which would generate the controllers, views, templates and associated tests nested in the MyAppWeb.Warehouse namespace:
	lib/my_app_web/controllers/warehouse/user_auth.ex
	lib/my_app_web/controllers/warehouse/user_confirmation_controller.ex
	lib/my_app_web/views/warehouse/user_confirmation_view.ex
	lib/my_app_web/templates/warehouse/user_confirmation/new.html.heex
	test/my_app_web/controllers/warehouse/user_auth_test.exs
	test/my_app_web/controllers/warehouse/user_confirmation_controller_test.exs
	and so on...

Binary ids
The --binary-id option causes the generated migration to use
binary_id for its primary key and foreign keys.
Default options
This generator uses default options provided in the :generators
configuration of your application. These are the defaults:
config :your_app, :generators,
  binary_id: false,
  sample_binary_id: "11111111-1111-1111-1111-111111111111"
You can override those options per invocation by providing corresponding
switches, e.g. --no-binary-id to use normal ids despite the default
configuration.
Custom table names
By default, the table name for the migration and schema will be
the plural name provided for the resource. To customize this value,
a --table option may be provided. For example:
$ mix phx.gen.auth Accounts User users --table accounts_users

This will cause the generated tables to be named "accounts_users" and "accounts_users_tokens".

Generates a self-signed certificate for HTTPS testing.
$ mix phx.gen.cert
$ mix phx.gen.cert my-app my-app.local my-app.internal.example.com

Creates a private key and a self-signed certificate in PEM format. These
files can be referenced in the certfile and keyfile parameters of an
HTTPS Endpoint.
WARNING: only use the generated certificate for testing in a closed network
environment, such as running a development server on localhost.
For production, staging, or testing servers on the public internet, obtain a
proper certificate, for example from Let's Encrypt.
NOTE: when using Google Chrome, open chrome://flags/#allow-insecure-localhost
to enable the use of self-signed certificates on localhost.
Arguments
The list of hostnames, if none are specified, defaults to:
	localhost

Other (optional) arguments:
	--output (-o): the path and base filename for the certificate and
key (default: priv/cert/selfsigned)
	--name (-n): the Common Name value in certificate's subject
(default: "Self-signed test certificate")

Requires OTP 21.3 or later.

Generates a Phoenix channel.
$ mix phx.gen.channel Room

Accepts the module name for the channel
The generated files will contain:
For a regular application:
	a channel in lib/my_app_web/channels
	a channel test in test/my_app_web/channels

For an umbrella application:
	a channel in apps/my_app_web/lib/app_name_web/channels
	a channel test in apps/my_app_web/test/my_app_web/channels

Generates a context with functions around an Ecto schema.
$ mix phx.gen.context Accounts User users name:string age:integer

The first argument is the context module followed by the schema module
and its plural name (used as the schema table name).
The context is an Elixir module that serves as an API boundary for
the given resource. A context often holds many related resources.
Therefore, if the context already exists, it will be augmented with
functions for the given resource.
Note: A resource may also be split
over distinct contexts (such as Accounts.User and Payments.User).

The schema is responsible for mapping the database fields into an
Elixir struct.
Overall, this generator will add the following files to lib/your_app:
	a context module in accounts.ex, serving as the API boundary
	a schema in accounts/user.ex, with a users table

A migration file for the repository and test files for the context
will also be generated.
Generating without a schema
In some cases, you may wish to bootstrap the context module and
tests, but leave internal implementation of the context and schema
to yourself. Use the --no-schema flags to accomplish this.
table
By default, the table name for the migration and schema will be
the plural name provided for the resource. To customize this value,
a --table option may be provided. For example:
$ mix phx.gen.context Accounts User users --table cms_users

binary_id
Generated migration can use binary_id for schema's primary key
and its references with option --binary-id.
Default options
This generator uses default options provided in the :generators
configuration of your application. These are the defaults:
config :your_app, :generators,
  migration: true,
  binary_id: false,
  sample_binary_id: "11111111-1111-1111-1111-111111111111"
You can override those options per invocation by providing corresponding
switches, e.g. --no-binary-id to use normal ids despite the default
configuration or --migration to force generation of the migration.
Read the documentation for phx.gen.schema for more information on
attributes.
Skipping prompts
This generator will prompt you if there is an existing context with the same
name, in order to provide more instructions on how to correctly use phoenix contexts.
You can skip this prompt and automatically merge the new schema access functions and tests into the
existing context using --merge-with-existing-context. To prevent changes to
the existing context and exit the generator, use --no-merge-with-existing-context.

Generates an embedded Ecto schema for casting/validating data outside the DB.
mix phx.gen.embedded Blog.Post title:string views:integer
The first argument is the schema module followed by the schema attributes.
The generated schema above will contain:
	an embedded schema file in lib/my_app/blog/post.ex

Attributes
The resource fields are given using name:type syntax
where type are the types supported by Ecto. Omitting
the type makes it default to :string:
mix phx.gen.embedded Blog.Post title views:integer
The following types are supported:
	:integer

	:float

	:decimal

	:boolean

	:map

	:string

	:array

	:references

	:text

	:date

	:time

	:time_usec

	:naive_datetime

	:naive_datetime_usec

	:utc_datetime

	:utc_datetime_usec

	:uuid

	:binary

	:enum

	:datetime - An alias for :naive_datetime

Generates controller, views, and context for an HTML resource.
mix phx.gen.html Accounts User users name:string age:integer
The first argument is the context module followed by the schema module
and its plural name (used as the schema table name).
The context is an Elixir module that serves as an API boundary for
the given resource. A context often holds many related resources.
Therefore, if the context already exists, it will be augmented with
functions for the given resource.
Note: A resource may also be split
over distinct contexts (such as Accounts.User and Payments.User).

The schema is responsible for mapping the database fields into an
Elixir struct. It is followed by an optional list of attributes,
with their respective names and types. See mix phx.gen.schema
for more information on attributes.
Overall, this generator will add the following files to lib/:
	a context module in lib/app/accounts.ex for the accounts API
	a schema in lib/app/accounts/user.ex, with an users table
	a view in lib/app_web/views/user_view.ex
	a controller in lib/app_web/controllers/user_controller.ex
	default CRUD templates in lib/app_web/templates/user

The context app
A migration file for the repository and test files for the context and
controller features will also be generated.
The location of the web files (controllers, views, templates, etc) in an
umbrella application will vary based on the :context_app config located
in your applications :generators configuration. When set, the Phoenix
generators will generate web files directly in your lib and test folders
since the application is assumed to be isolated to web specific functionality.
If :context_app is not set, the generators will place web related lib
and test files in a web/ directory since the application is assumed
to be handling both web and domain specific functionality.
Example configuration:
config :my_app_web, :generators, context_app: :my_app
Alternatively, the --context-app option may be supplied to the generator:
mix phx.gen.html Sales User users --context-app warehouse
Web namespace
By default, the controller and view will be namespaced by the schema name.
You can customize the web module namespace by passing the --web flag with a
module name, for example:
mix phx.gen.html Sales User users --web Sales
Which would generate a lib/app_web/controllers/sales/user_controller.ex and
lib/app_web/views/sales/user_view.ex.
Customizing the context, schema, tables and migrations
In some cases, you may wish to bootstrap HTML templates, controllers,
and controller tests, but leave internal implementation of the context
or schema to yourself. You can use the --no-context and --no-schema
flags for file generation control.
You can also change the table name or configure the migrations to
use binary ids for primary keys, see mix phx.gen.schema for more
information.

Generates controller, views, and context for a JSON resource.
mix phx.gen.json Accounts User users name:string age:integer
The first argument is the context module followed by the schema module
and its plural name (used as the schema table name).
The context is an Elixir module that serves as an API boundary for
the given resource. A context often holds many related resources.
Therefore, if the context already exists, it will be augmented with
functions for the given resource.
Note: A resource may also be split
over distinct contexts (such as Accounts.User and Payments.User).

The schema is responsible for mapping the database fields into an
Elixir struct. It is followed by an optional list of attributes,
with their respective names and types. See mix phx.gen.schema
for more information on attributes.
Overall, this generator will add the following files to lib/:
	a context module in lib/app/accounts.ex for the accounts API
	a schema in lib/app/accounts/user.ex, with an users table
	a view in lib/app_web/views/user_view.ex
	a controller in lib/app_web/controllers/user_controller.ex

A migration file for the repository and test files for the context and
controller features will also be generated.
The context app
The location of the web files (controllers, views, templates, etc) in an
umbrella application will vary based on the :context_app config located
in your applications :generators configuration. When set, the Phoenix
generators will generate web files directly in your lib and test folders
since the application is assumed to be isolated to web specific functionality.
If :context_app is not set, the generators will place web related lib
and test files in a web/ directory since the application is assumed
to be handling both web and domain specific functionality.
Example configuration:
config :my_app_web, :generators, context_app: :my_app
Alternatively, the --context-app option may be supplied to the generator:
mix phx.gen.json Sales User users --context-app warehouse
Web namespace
By default, the controller and view will be namespaced by the schema name.
You can customize the web module namespace by passing the --web flag with a
module name, for example:
mix phx.gen.json Sales User users --web Sales
Which would generate a lib/app_web/controllers/sales/user_controller.ex and
lib/app_web/views/sales/user_view.ex.
Customizing the context, schema, tables and migrations
In some cases, you may wish to bootstrap JSON views, controllers,
and controller tests, but leave internal implementation of the context
or schema to yourself. You can use the --no-context and --no-schema
flags for file generation control.
You can also change the table name or configure the migrations to
use binary ids for primary keys, see mix phx.gen.schema for more
information.

Generates LiveView, templates, and context for a resource.
mix phx.gen.live Accounts User users name:string age:integer
The first argument is the context module.  The context is an Elixir module
that serves as an API boundary for the given resource. A context often holds
many related resources.  Therefore, if the context already exists, it will be
augmented with functions for the given resource.
The second argument is the schema module.  The schema is responsible for
mapping the database fields into an Elixir struct.
The remaining arguments are the schema module plural name (used as the schema
table name), and an optional list of attributes as their respective names and
types.  See mix help phx.gen.schema for more information on attributes.
When this command is run for the first time, a ModalComponent and
LiveHelpers module will be created, along with the resource level
LiveViews and components, including UserLive.Index, UserLive.Show,
and UserLive.FormComponent modules for the new resource.
Note: A resource may also be split
over distinct contexts (such as Accounts.User and Payments.User).

Overall, this generator will add the following files:
	a context module in lib/app/accounts.ex for the accounts API
	a schema in lib/app/accounts/user.ex, with a users table
	a LiveView in lib/app_web/live/user_live/show.ex
	a LiveView in lib/app_web/live/user_live/index.ex
	a LiveComponent in lib/app_web/live/user_live/form_component.ex
	a helpers module in lib/app_web/live/live_helpers.ex with a modal

After file generation is complete, there will be output regarding required
updates to the lib/app_web/router.ex file.
Add the live routes to your browser scope in lib/app_web/router.ex:

  live "/users", UserLive.Index, :index
  live "/users/new", UserLive.Index, :new
  live "/users/:id/edit", UserLive.Index, :edit

  live "/users/:id", UserLive.Show, :show
  live "/users/:id/show/edit", UserLive.Show, :edit
The context app
A migration file for the repository and test files for the context and
controller features will also be generated.
The location of the web files (LiveView's, views, templates, etc.) in an
umbrella application will vary based on the :context_app config located
in your applications :generators configuration. When set, the Phoenix
generators will generate web files directly in your lib and test folders
since the application is assumed to be isolated to web specific functionality.
If :context_app is not set, the generators will place web related lib
and test files in a web/ directory since the application is assumed
to be handling both web and domain specific functionality.
Example configuration:
config :my_app_web, :generators, context_app: :my_app
Alternatively, the --context-app option may be supplied to the generator:
mix phx.gen.live Accounts User users --context-app warehouse
Web namespace
By default, the LiveView modules will be namespaced by the web module.
You can customize the web module namespace by passing the --web flag with a
module name, for example:
mix phx.gen.live Accounts User users --web Sales
Which would generate the LiveViews in lib/app_web/live/sales/user_live/,
namespaced AppWeb.Sales.UserLive instead of AppWeb.UserLive.
Customizing the context, schema, tables and migrations
In some cases, you may wish to bootstrap HTML templates, LiveViews,
and tests, but leave internal implementation of the context or schema
to yourself. You can use the --no-context and --no-schema flags
for file generation control.
You can also change the table name or configure the migrations to
use binary ids for primary keys, see mix help phx.gen.schema for more
information.

Generates a notifier that delivers emails by default.
$ mix phx.gen.notifier Accounts User welcome_user reset_password confirmation_instructions

This task expects a context module name, followed by a
notifier name and one or more message names. Messages
are the functions that will be created prefixed by "deliver",
so the message name should be "snake_case" without punctuation.
Additionally a context app can be specified with the flag
--context-app, which is useful if the notifier is being
generated in a different app under an umbrella.
$ mix phx.gen.notifier Accounts User welcome_user --context-app marketing

The app "marketing" must exist before the command is executed.

Generates a Presence tracker.
$ mix phx.gen.presence
$ mix phx.gen.presence MyPresence

The argument, which defaults to Presence, defines the module name of the
Presence tracker.
Generates a new file, lib/my_app_web/channels/my_presence.ex, where
my_presence is the snake-cased version of the provided module name.

      





  

  
    
    mix phx.gen.release - Phoenix v1.6.9
    
    

    



  
  

    
mix phx.gen.release 
    



      
Generates release files and optional Dockerfile for release-based deployments.
The following release files are created:
	lib/app_name/release.ex - A release module containing tasks for running
migrations inside a release

	rel/overlays/bin/migrate - A migrate script for conveniently invoking
the release system migrations

	rel/overlays/bin/server - A server script for conveniently invoking
the release system with environment variables to start the phoenix web server


Note, the rel/overlays directory is copied into the release build by default when
running mix release.
To skip generating the migration-related files, use the --no-ecto flag. To
force these migration-related files to be generated, the use --ecto flag.
Docker
When the --docker flag is passed, the following docker files are generated:
	Dockerfile - The Dockerfile for use in any standard docker deployment

	.dockerignore - A docker ignore file with standard elixir defaults


For extended release configuration, the mix release.inittask can be used
in addition to this task. See the Mix.Release docs for more details.

Generates an Ecto schema and migration.
$ mix phx.gen.schema Blog.Post blog_posts title:string views:integer

The first argument is the schema module followed by its plural
name (used as the table name).
The generated schema above will contain:
	a schema file in lib/my_app/blog/post.ex, with a blog_posts table
	a migration file for the repository

The generated migration can be skipped with --no-migration.
Contexts
Your schemas can be generated and added to a separate OTP app.
Make sure your configuration is properly setup or manually
specify the context app with the --context-app option with
the CLI.
Via config:
config :marketing_web, :generators, context_app: :marketing
Via CLI:
$ mix phx.gen.schema Blog.Post blog_posts title:string views:integer --context-app marketing

Attributes
The resource fields are given using name:type syntax
where type are the types supported by Ecto. Omitting
the type makes it default to :string:
$ mix phx.gen.schema Blog.Post blog_posts title views:integer

The following types are supported:
	:integer

	:float

	:decimal

	:boolean

	:map

	:string

	:array

	:references

	:text

	:date

	:time

	:time_usec

	:naive_datetime

	:naive_datetime_usec

	:utc_datetime

	:utc_datetime_usec

	:uuid

	:binary

	:enum

	:datetime - An alias for :naive_datetime


The generator also supports references, which we will properly
associate the given column to the primary key column of the
referenced table:
$ mix phx.gen.schema Blog.Post blog_posts title user_id:references:users

This will result in a migration with an :integer column
of :user_id and create an index.
Furthermore an array type can also be given if it is
supported by your database, although it requires the
type of the underlying array element to be given too:
$ mix phx.gen.schema Blog.Post blog_posts tags:array:string

Unique columns can be automatically generated by using:
$ mix phx.gen.schema Blog.Post blog_posts title:unique unique_int:integer:unique

Redact columns can be automatically generated by using:
$ mix phx.gen.schema Accounts.Superhero superheroes secret_identity:redact password:string:redact

Ecto.Enum fields can be generated by using:
$ mix phx.gen.schema Blog.Post blog_posts title status:enum:unpublished:published:deleted

If no data type is given, it defaults to a string.
table
By default, the table name for the migration and schema will be
the plural name provided for the resource. To customize this value,
a --table option may be provided. For example:
$ mix phx.gen.schema Blog.Post posts --table cms_posts

binary_id
Generated migration can use binary_id for schema's primary key
and its references with option --binary-id.
Default options
This generator uses default options provided in the :generators
configuration of your application. These are the defaults:
config :your_app, :generators,
  migration: true,
  binary_id: false,
  sample_binary_id: "11111111-1111-1111-1111-111111111111"
You can override those options per invocation by providing corresponding
switches, e.g. --no-binary-id to use normal ids despite the default
configuration or --migration to force generation of the migration.

Generates a secret and prints it to the terminal.
$ mix phx.gen.secret [length]

By default, mix phx.gen.secret generates a key 64 characters long.
The minimum value for length is 32.

Generates a Phoenix socket handler.
$ mix phx.gen.socket User

Accepts the module name for the socket
The generated files will contain:
For a regular application:
	a client in assets/js
	a socket in lib/my_app_web/channels

For an umbrella application:
	a client in apps/my_app_web/assets/js
	a socket in apps/my_app_web/lib/app_name_web/channels

You can then generated channels with mix phx.gen.channel.

Creates a new Phoenix project.
It expects the path of the project as an argument.
$ mix phx.new PATH [--module MODULE] [--app APP]

A project at the given PATH will be created. The
application name and module name will be retrieved
from the path, unless --module or --app is given.
Options
	--umbrella - generate an umbrella project,
with one application for your domain, and
a second application for the web interface.

	--app - the name of the OTP application

	--module - the name of the base module in
the generated skeleton

	--database - specify the database adapter for Ecto. One of:
	postgres - via https://github.com/elixir-ecto/postgrex
	mysql - via https://github.com/elixir-ecto/myxql
	mssql - via https://github.com/livehelpnow/tds
	sqlite3 - via https://github.com/elixir-sqlite/ecto_sqlite3

Please check the driver docs for more information
and requirements. Defaults to "postgres".

	--no-assets - do not generate the assets folder.
When choosing this option, you will need to manually
handle JavaScript/CSS if building HTML apps

	--no-ecto - do not generate Ecto files

	--no-html - do not generate HTML views

	--no-gettext - do not generate gettext files

	--no-dashboard - do not include Phoenix.LiveDashboard

	--no-live - comment out LiveView socket setup in assets/js/app.js
and also on the endpoint (the latter also requires --no-dashboard)

	--no-mailer - do not generate Swoosh mailer files

	--binary-id - use binary_id as primary key type in Ecto schemas

	--verbose - use verbose output

	-v, --version - prints the Phoenix installer version


When passing the --no-ecto flag, Phoenix generators such as
phx.gen.html, phx.gen.json, phx.gen.live, and phx.gen.context
may no longer work as expected as they generate context files that rely
on Ecto for the database access. In those cases, you can pass the
--no-context flag to generate most of the HTML and JSON files
but skip the context, allowing you to fill in the blanks as desired.
Similarly, if --no-html is given, the files generated by
phx.gen.html will no longer work, as important HTML components
will be missing.
Installation
mix phx.new by default prompts you to fetch and install your
dependencies. You can enable this behaviour by passing the
--install flag or disable it with the --no-install flag.
Examples
$ mix phx.new hello_world

Is equivalent to:
$ mix phx.new hello_world --module HelloWorld

Or without the HTML and JS bits (useful for APIs):
$ mix phx.new ~/Workspace/hello_world --no-html --no-assets

As an umbrella:
$ mix phx.new hello --umbrella

Would generate the following directory structure and modules:
hello_umbrella/   Hello.Umbrella
  apps/
    hello/        Hello
    hello_web/    HelloWeb
You can read more about umbrella projects using the
official Elixir guide

Creates a new Ecto project within an umbrella project.
This task is intended to create a bare Ecto project without
web integration, which serves as a core application of your
domain for web applications and your greater umbrella
platform to integrate with.
It expects the name of the project as an argument.
$ cd my_umbrella/apps
$ mix phx.new.ecto APP [--module MODULE] [--app APP]

A project at the given APP directory will be created. The
application name and module name will be retrieved
from the application name, unless --module or --app is given.
Options
	--app - the name of the OTP application

	--module - the name of the base module in
the generated skeleton

	--database - specify the database adapter for Ecto. One of:
	postgres - via https://github.com/elixir-ecto/postgrex
	mysql - via https://github.com/elixir-ecto/myxql
	mssql - via https://github.com/livehelpnow/tds
	sqlite3 - via https://github.com/elixir-sqlite/ecto_sqlite3

Please check the driver docs for more information
and requirements. Defaults to "postgres".

	--binary-id - use binary_id as primary key type
in Ecto schemas


Examples
$ mix phx.new.ecto hello_ecto

Is equivalent to:
$ mix phx.new.ecto hello_ecto --module HelloEcto

Creates a new Phoenix web project within an umbrella project.
It expects the name of the OTP app as the first argument and
for the command to be run inside your umbrella application's
apps directory:
$ cd my_umbrella/apps
$ mix phx.new.web APP [--module MODULE] [--app APP]

This task is intended to create a bare Phoenix project without
database integration, which interfaces with your greater
umbrella application(s).
Examples
$ mix phx.new.web hello_web

Is equivalent to:
$ mix phx.new.web hello_web --module HelloWeb

Supports the same options as the phx.new task.
See Mix.Tasks.Phx.New for details.

Prints all routes for the default or a given router.
$ mix phx.routes
$ mix phx.routes MyApp.AnotherRouter

The default router is inflected from the application
name unless a configuration named :namespace
is set inside your application configuration. For example,
the configuration:
config :my_app,
  namespace: My.App
will exhibit the routes for My.App.Router when this
task is invoked without arguments.
Umbrella projects do not have a default router and
therefore always expect a router to be given. An
alias can be added to mix.exs to automate this:
defp aliases do
  [
    "phx.routes": "phx.routes MyAppWeb.Router",
    # aliases...
  ]

Starts the application by configuring all endpoints servers to run.
Note: to start the endpoint without using this mix task you must set
server: true in your Phoenix.Endpoint configuration.
Command line options
	--open - open browser window for each started endpoint

Furthermore, this task accepts the same command-line options as
mix run.
For example, to run phx.server without recompiling:
$ mix phx.server --no-compile

The --no-halt flag is automatically added.
Note that the --no-deps-check flag cannot be used this way,
because Mix needs to check dependencies to find phx.server.
To run phx.server without checking dependencies, you can run:
$ mix do deps.loadpaths --no-deps-check, phx.server

